A CHRONOLOGY OF HP-41C PROGRAMS FOR USE AND EXAMPLE

Thomas W. Beers



A CHRONOLOGY OF HP-41C PROGRAMS FOR USE AND EXAMPLE

Thomas W. Beers

Professor of Forestry

Purdue University

PREFACE

My dictionary defines a "chronology" as "any tabulated arrangement of events of historical or scientific import, in the order of the time of their occurrence." And that is just what is included in this publication. It has been assembled as a personal source of reference (to replace one file drawer of manila folders) with the hope that other HP-41C users and programmers can profit from my learning experiences.

The calculator programmer who anticipates that others will use his program normally tries to make it "user-friendly", but he frequently overlooks making the how-to-use instructions (i.e., the "write-up") user-friendly. The search for a simple, understandable, and useful general format for program documentation and instruction is the underlying thread that bonds this chronology; a project which began very innocently in late 1979 when the APPRAZ program was conceived and written.

The author recognizes and acknowledges: Richard J. Nelson (founder and driving force behind the PPC organization) and the contributors to the PPC Calculator Journal* for publishing various inspirational programming techniques; Linda S. Hampton for typing many of the program descriptions; my wife, Carolyn, for continuity typing and program listing and assembly; and my son, Ted W. Beers, for constant and reliable programming advice and for providing dialogue which enabled the fine-tuning of ideas and concepts.

*Published by Personal Programming Center, PPC, 2545 West Camden Place, Santa Ana, CA 92704.

TABLE OF CONTENTS


```
          +
          44
         4444
        444444
       44444444
      4444444444
     444444444444
    44444444444444
   4444444444444444
  444444444444444444
 44444444444444444444
4444444444444444444444
         1111
         1111
        CCCCCC
```

```
          +
          RR
         RRRR
        RRRRRR
       RRRRRRRR
      RRRRRRRRRR
     RRRRRRRRRRRR
    RRRRRRRRRRRRRR
   RRRRRRRRRRRRRRRR
  RRRRRRRRRRRRRRRRRR
 RRRRRRRRRRRRRRRRRRRR
RRRRRRRRRRRRRRRRRRRRRR
         PPPP
         PPPP
        NNNNNN
```

INTRODUCTION

This chronology is meant to be used as a learning device for novice HP-41C programmers and for direct use by certain 41C owners who can apply specific programs. In either case, the reader should be aware that the programs differ in styles, and, to a certain extent, in sophistication as the chronology develops. Therefore, since the author was learning with time, certain earlier programs would not be written the same, had they been undertaken at a later date (experience does help!). In order to preserve the sense of chronological development, yet provide an opportunity for modest refinement, the program as originally written is presented, followed (in most cases) by a section titled "Retrospective Comments Regarding...." This section was written in the summer of 1982 during which time the author tested and evaluated each program. As a result of the foregoing comments, it is recommended that the "Retrospective Comments" section (if present) be read by the user prior to the serious use of any program.

Although every attempt was made to eliminate technical, programming, and typographical errors, surely a certain number still exist. If these surface and cause major problems or concern, the author would appreciate being made aware, so that other users can be informed.

For those who desire the programs in mini-cassette tape form for use with the HP-82161A cassette drive, a complete assembly, called KRON-1, can be purchased. Presently, no barcode, magnetic card or disc copies of individual programs are available; however, this decision may change as technology advances. Current prices and availability can be obtained from the author, Department of Forestry and Natural Resources, Purdue University, West Lafayette, IN 47907, phone (317) 494-3598 (office) or (317) 463-0807 (home).

ANNOTATED DIRECTORY OF PROGRAMS

Program Number*	Program Name	Page	Purpose and Special Notes
1	APPRAZ	13	To calculate and summarize timber volumes by species and log grade; additionally, numbers of trees by species and DBH class are summarized as are number of logs by species. Grand total volume, number of logs, and number of trees are also provided. NOTES: 1. The use of a compact data input format to save keystrokes is worth noting.
2	CRUZ	25	To process inventory data from fixed-size or horizontal point samples, where the tallied data are: species, tree DBH (diameter breast high), number of logs, and, optionally, soundness. Per-acre and total tract summaries are provided for volume, basal area, and number of trees, by species and overall. Per-acre and total estimates by DBH classes are provided for volume and number of trees. A statistical routine estimates the mean, standard error, percent standard error and .95 confidence intervals for volume and basal area. NOTES: 1. Compact data input as in APPRAZ. 2. The processing part of the program demonstrates that, with very minor differences, one can handle fixed-size plots, horizontal point sampling, and 100% inventories.
5	MIL3	40	To provide on-the-spot solution of the following expression of sawmill "profit margin" using the formula: $\text{margin} = \sum_{i=1}^{6} (\text{lumber tally})_i (V_i) - \text{log value} - \text{sawing cost},$ where log value = (log grade value)(log scale) sawing cost = (cost per minute)(sawing time) V_i = value for various lumber grades; and to accumulate log scale and lumber tally by various species and grade classes, as well as total margin and sawing times for an indefinite number of logs.

*For brevity, only the last digits of the full number (41F001, 41F002, etc.) are given.

Program Number	Name	Page	Purpose and Special Notes
			NOTES: 1. A multiple register clear routine (label MRC) is used to clear a block of registers. 2. A special routine is used to facilitate the loading of the necessary numeric constants and species names (label LOAD). 3. Facility is provided, at the user's discretion, to automatically process keyed-in data without pressing R/S (label F22). 4. Program listing in TRACE printer mode, to facilitate documentation, is included.
7	SINDEX	60	To replace tables and graphs in U.S. Forest Service NC-54 publication. Provides forward or inverse solution of $H = b_1 S(1-e^{b_2 A})^{b_3}$ for nine species-specific sets of b_i's (stored in memory). **NOTES:** 1. Although present in previous programs also, the practice of "clear flag 29 in the presence of FIX 0" to eliminate decimal points after integers, is clearly evident. 2. Using a <u>negative</u> species code input as an indicator that a "reverse solution" is desired is employed; in this way, the height equation is solved backwards to obtain site index.
8	LOGVOL	68	To calculate log volumes by Doyle, Scribner, Int. 1/4-inch, Int. 1/8-inch rules and cubic content by Huber, Smalian, and Newton formulas — United States or Metric units. **NOTES:** 1. Metric or U.S. units input and output are readily handled as is output conversion from one to the other. 2. Order of output can be reversed by setting flag 00.

Program Number	Program Name	Program Page	Purpose and Special Notes
			3. An algorithm was developed specifically for this program to calculate board-foot volume by International 1/8-inch rule. 4. A "stop-pause" subroutine was used (label SP) to enable pause only between answers when flag 01 is set.
9	RANCHEK	84	To evaluate a random number generator (RAN) by simulating the rolling of one die a user-determined number of times. The numbers of 1's, 2's, . . ., 6's are tallied and displayed, as are the calculated chi-square value, sample mean, and the bias in % (deviation of mean from the true: 3.5). NOTES: 1. The random number generator used, based on an algorithm developed by Don Malm, is quite adaptable to 3P (probability proportional to to prediction) sampling, since it generates integers uniformly in the range 1 to an arbitrary N_{max}. 2. An alternative to watching the "goose" traverse the display during a long running program is used: the alpha string "*GENERATING*" fills the display while the number generation is in progress. 3. The use of ISG, DSE, and indirect addressing in the same loop (label 03) constitutes a good demonstration of the use of these concepts.
10	INMULT	95	To calculate, display and print values of $(1 + i)^n$ for arbitrary values of interest rate ($i = \frac{rate}{100}$) and years (n). Two options are provided: a. "Single" solution for specific rate, r, and range of years from 1 to n_{max}. b. Semi-automatic or automatic solution for specified ranges of r and n, using specified intervals. NOTES: 1. The use of the HP-41C to generate segments of "math tables" is exemplified.

Program Number	Program Name	Program Page	Purpose and Special Notes
			2. User-defined "control numbers" are input to enable automatic solution, display and printing for a range of interest rates and years. 3. A function such as SIN is used after AVIEW to achieve a rapid pause and display of an "identifier". 4. Program listings in all three printer modes are included.

Programs 11 through 16 were written to complement and expand the outline published by W. D. Shepperd in U.S. Forest Service Gen. Tech. Report RM-76.

Number	Name	Page	Purpose and Special Notes
11	SLOPE	105	To calculate the horizontal distance and, optionally, the vertical rise from a given slope distance and slope angle in percent, degrees (flag 01 set), or topographic units (flag 02 set).
12	BA	109	To calculate tree basal area in square feet from a keyed-in diameter in inches or vice versa (with flag 01 set) and to summarize a series of such calculations, providing the arithmetic mean diameter, the arithmetic mean basal area, the quadratic mean diameter, the number of entries and the sums and standard deviations of both diameter and basal area. Metric output (basal area in m^2, diameters in cm.) from similar input, can be obtained with flag 02 set. NOTES: 1. Use of BA demonstrates that one program can be used as a conversion routine and, if the user desires, a summary processor to provide various average statistics. 2. The use of negative input for a normally positive variable to set a flag and cause inverse calculations (conversion of basal area to diameter in this case) is worth noting.
13	TH	116	To calculate tree height when horizontal distance is <u>not</u> measured and the clinometer will not read directly. Therefore, tree height is calculated from keyed-in angles to tree top and base, and slope

Program Number	Program Name	Program Page	Purpose and Special Notes
			distance. Provision is made for angles in percent, degrees (flag 01 set), or topographic units (flag 02 set). NOTES: 1. The TH program provides reminders regarding flag settings, making use of the AVIEW, SIN sequence of instructions to provide a quick pause.
14	NTEST	123	For a keyed-in preliminary sample of X_i, this program will calculate and display the preliminary sample size, n_p, sum, mean, standard deviation, standard error, variance, and coefficient of variation. After an arbitrary allowable error is keyed-in, the sample size, n, required to achieve this error is calculated and displayed. An infinite population is assumed but the finite case can be handled by setting flag 00. NOTES: 1. NTEST provides a way of calculating the basic statistics for a sample assumed to be from an infinite population and the processing speed is very fast; thus it can be used as an efficient alternative to the $\Sigma+$ key and MEAN, SDEV functions. Furthermore, the sample size required for given requirements is simply obtained. 2. A simple adjustment formula is used to obtain the required sample size if the population is assumed to be finite.
15	GCAL	131	Angle gauges used in horizontal point sampling are "calibrated," or sometimes designed or assembled making use of the functional relation between basal area factor of the gauge, F, width of a sighting bar, W, and distance, D, from eye to "target." This program solves the relationship for any one of the three variables, using the other two as input; the determination of the basal area factor of wedge prisms is also possible. The use of metric units is enabled with flag 00 set. NOTES: 1. Optional conversion of the basal area factor answer to or from metric units by pressing R/S is provided.

Program			
Number	**Name**	**Page**	**Purpose and Special Notes**
16	PSFIELD	136	To solve certain problems relating to field application of horizontal point sampling. For a given basal area factor F, and if needed, tree DBH, D, and a measure of slope, one can obtain: (a) horizontal distance multiplier; (b) limiting horizontal distance; (c) limiting slope distance; (d) calibrated tape mark to hold; (e) tree factor; (f) associated plot area; (g) a borderline tree check, if actual distance to the tree is provided; and (h) a boundary overlap correction "weight," given the distance to boundary. Slope can be expressed in percent, degrees (flag 01 set), or topographic units (flag 02 set). The technique for slope correction is assumed to be the variable gauge angle approach (see Beers, Jour. For. 67:188-192); however, with flag 00 set, the constant gauge angle technique is assumed. The entire program can be "made metric" by setting flag 03. **NOTES**: 1. PSFIELD is an example of a program which is too diverse to be of much practical value except for educational purposes.
17	SSRS	148	To summarize data from either a simple or stratified sample, obtaining within stratum and overall estimates of the mean and standard error and, optionally, user-specified confidence intervals and other sample statistics. Provision is made to use or not (flag 00 set) a finite population correction. **NOTES**: 1. SSRS was written to remove the tedium from processing data from a stratified random sample. Since each stratum sample is considered a simple random sample, data from this design can also be analyzed. 2. Initial reminders are skipped by setting flag 03 once familiarity with the program is gained.
18	LR	161	To provide simple linear regression calculations which simulate the L.R., $\hat{y}$, and r keys found on the HP-34C, HP-11C and HP-15C. Specifically, one may readily calculate the following, for X and Y data summarized by the $\Sigma+$ key:

Program Number	Program Name	Page	Purpose and Special Notes
			a, the Y-intercept b, the slope of the least-squares fitted line b', the slope assuming the line is forced through the origin r, the simple correlation coefficient r^2, the coefficient of determination t_r, the calculated t to test the hypothesis of zero correlation $\hat{Y}$, a predicted value of Y for any keyed-in X Additional statistics such as confidence interval estimates and standard errors can be calculated using the program SLR (Program No. 41F019) but for brevity (96 program steps, 176 bytes, one magnetic card) only the listed items are included in "LR".
19	SLR	168	To extend the simple linear regression calculations achieved by "LR" (Program No. 41F018) to provide, for ungrouped X and Y data summarized by the $\Sigma+$ key: (1) a, b, b', r, r^2, t_r, and $\hat{Y}$ (see LR program); (2) standard errors: s_{yx}, s_a, s_b, $s_{b'}$; (3) confidence interval estimates assuming mean Y and assuming individual Y for given X_o; and (4) Student's t to test the following hypotheses: $H_o: \rho = o$, $H_o: \alpha = o$, $H_o: \beta = o$, $H_o: \beta' = 1$. NOTES: 1. SLR was written much later than its position in the chronology indicates, therefore the format of the write-up is practically "the latest." 2. Care was taken to make the program "printer compatible" and an automatic print-out mode (flag 00 set) is provided.
20	MSLR	184	MSLR is a steering program to be used in conjunction with SLR (Program No. 41F019) to accommodate ungrouped, grouped, or weighted data and to extend the prompting, correction, and/or deletion capabilities of that program. With the insertion of appropriate subroutines it is anticipated that MSLR can also be used to transform the input data and achieve linear approximations to certain non-linear models.

Program Number	Program Name	Page	Purpose and Special Notes

NOTES:

1. As with SLR, MSLR was written later than its position in the chronology indicates.

2. Detailed examples of ungrouped and grouped cases are provided with printer output.

3. The deletion of data sets, purposefully or to correct for a keyboard error, is provided for and described in detail.

21 VOL17 202

To demonstrate the capacity of the HP-41C to solve a non-linear volume estimation formula using 17 different sets of regression coefficients. The basic form of the equation is:

$$V = b_o + b_1 D^{b_2} + b_3 D^{b_4} H^{b_5}$$

where, V = board foot volume, Int. 1/4" rule
D = tree DBH
H = tree merchantable length
b_i = species unique coefficients developed by the U.S. Forest Survey (refer to U.S.F.S. Research Note NE-271, or parts F and G)

Provision is made in the program to obtain volume totals by species if desired and printed output is provided if the HP-82143A printer is attached.

NOTES:

1. This program makes use of 102 regression coefficients and 17 species groups names, which can be routinely loaded into the proper registers from magnetic cards. When used with the printer, the user is appropriately prompted to load the data (from cards) using the RDTAX command.

22 MG78 213

To provide an alternative to the usual "volume table look up" procedure for individual tree volume determination in board feet by International 1/4 inch, Scribner, and Doyle log rules. The tables obviated by this program are the Form Class 78 tables described by Mesavage & Girard (Tables for Estimating Board-Foot Volume of Timber, U.S.F.S.) using the estimation equations developed by Wiant and Castaneda (Resource Inventory Notes, BLM, March, 1977). Wiant claims agreement with the tabular values within ± 3 percent, for 99, 94 and 97 percent of the cell values for the three log rules.

Program Number	Name	Page	Purpose and Special Notes
			NOTES: 1. MG78 makes use of an audio signal to identify answers; for example, if all three log rules are solved (label A), the standard beep implies Scribner rule, successive tones 9,7,9,7 implies Int. 1/4" rule, and successive tones 7,9,7,9 implies Doyle rule. 2. An "adjustment routine" is used to calculate volumes for form classes other than 78, but this is simply 3% for each class unit removed from 78. This 3% rule is an approximation and should be used with discretion.
23	MYERS	222	To provide a convenient calculation procedure to be used instead of the set of 14 tables of upland hardwood weights and volumes summarized by Myers, et al. (U.S.F.S. Gen. Tech. Rep. NC-60, 1980). Provision is made for all six species cited: black, red, and white oak, hickory, white ash, and yellow poplar. In addition to the obvious advantage of the calculator over the table look-up procedure, "interpolation" between classes given in the tables is routinely handled by the programmed procedure.
24	SCALE	233	The program SCALE provides an accurate alternative to the scale-stick approach to log scaling. Gross scale, net scale and amount of defect can be displayed for each log after scaling diameter, length and defect type and dimensions are input. Provision is made for summaries of gross and net scale and numbers of logs by species (9 groups possible), and by log grade (4 possible) within species. Various flags are used to enable the selection of log rule from Doyle, Scribner, Scribner Decimal C, International 1/4" and International 1/8". The defect calculation formulas are those proposed by Grosenbaugh (1952) and described in Husch, Miller and Beers (FOREST MENSURATION, 1982. John Wiley & Sons). A printed copy of the summary by species is also made available by the program, if the HP-82143A printer is attached. NOTES: 1. SCALE is a long and complicated program necessitated by the flexibility provided.

Program			
Number	Name	Page	Purpose and Special Notes
			It can be used without summary (flag 00 set) as a log scale calculator (like LOGVOL, Program No. 41F008) or as a defect and net scale calculator. If a summary is desired (flag 00 clear) the calculator alone can be used in the field and then a printer attached and used for print-out, or, if one is processing data previously recorded on paper, the printer can be attached (NORM mode) to provide documentation for each log as processed.
25	D2BVVMC	248	To solve the formula relating tree diameter in inches or centimeters, D, and crossectional area ("basal area") in square feet or square meters, B, for either D or B, given the other as input. NOTES: 1. The reminders at the beginning of the program are skipped, not by setting a flag, but by using a local alpha label (LBL A, step 13) placed just after the last reminder prompt. 2. The direction of calculation (D to B or B to D) is reversed at the user's option by pressing R/S in the absence of a numeric input. Flag 22, which is automatically set by a numeric key depression, is tested in the program to accomplish this reversal. 3. Intermittant change of the calculation direction as opposed to a constant "one-way" direction is dictated by flag 01 clear or set, respectively. 4. Conversion of any answer from U.S. units to metric or vice versa is accomplished by a local alpha label (label a).
26	F2CVV	252	To convert either Fahrenheit or Centigrade degrees to the other, with optional conversion to Kelvin units. NOTES: 1. This program is structured very much like D2BVVMC and consequently is executed

Program Number	Name	Page	Purpose and Special Notes
			similarly. The conversion algorithm used is not widely known, but makes for simple programming: add 40 to the number, multiply by either 5/9 or 9/5 depending on the direction of conversion, and then subtract 40. 2. Conversion to Kelvin units is accomplished by use of a local alpha label (label b).
27	CONVRT	256	A general program for commonly occurring conversions. Up to 15 local labels can be used to program user-specified conversions. Those presently programmed are inches-centimeters, feet-meters, square feet per acre-square meters per hectare, cubic feet per acre-cubic meters per hectare, ounces-milliliters (including "fifths" and "quarts"). Either direction of conversion is easily selected. <u>NOTES</u>: 1. CONVRT was written to provide a general program for conversions involving a single multiplier (or divisor) in contrast to D2BVVMC and F2CVV which require formula solution. CONVRT is "open-ended" in that the user can insert his own specific conversion routines and prepare a keyboard overlay to identify the local label.
28	CONBLD	265	A unique program to compose programs similar to CONVRT. Requires use of the PPC ROM. <u>NOTES</u>: 1. Because the local label routines in the CONVRT program are so alike, CONBLD was conceived and written to "automate" the assembly of these subroutines. And, although CONBLD takes more time to assemble the 14 program lines (approx. 3 minutes) than required to key them manually, it is good for the expansive-thinking programmer to be aware of this alternative procedure.
29	BPROB	271	To calculate individual and cumulative probabilities of obtaining k successes in n trials assuming a specified probability of success in one trial, p. Based on the binomial distribution.

Program Number	Program Name	Page	Purpose and Special Notes
			NOTES: 1. BPROB is a simple program which exemplifies the type of applications by which one is able to obviate tables and graphs of distributions such as Poisson, Normal, Student's t, and Chi-square.
30	SCHNUR	276	To calculate height or site index, given the other, after age has been specified. Based on Wiant's (Jour. Forestry, 1975, Vol. 73, page 429) prediction equation for Schnur's classic site index curves for upland oaks. NOTES: 1. SCHNUR makes use of the prompt-reversal technique (R/S depression in the absence of numeric key input), and metric conversion by local label key depression as used in D2BVVMC (Program No. 41F025).
31	TVOLSA	283	To calculate tree volume in cubic feet and lateral surface area in square feet, assuming the following: a. Input is [DBH, upper diameter, merchantable length] for the main stem, and [lower diameter, upper diameter, length] for additional sections. All diameters are consistently outside (or inside) bark in inches and merchantable length is the distance in feet from stump height (here assumed to be .5 foot) to the upper diameter. b. The section between breast height and stump height is a 4-foot cylinder. c. The section above breast height is a frustum of a cone and volume is determined by Newton's formula.
32	QFACTOR	290	To provide a means of fitting data of the form Y = number of trees and X = diameter class to the negative exponential model, $Y = ke^{-aX}$, using the linear least squares approximation $\log Y = b_0 + b_1X$. Estimates of a, k, b_0, and b_1 are displayed, as is the commonly used index, q, which reflects the ratio of trees in successive diameter classes. Provision is made to obtain predicted Y (i.e., $\hat{Y}$) for any X and to obtain q for class widths other than that assumed initially.

Program No. 41F001

Thomas W. Beers
Dept. Forestry, Purdue Univ.
December 1979

APPRAZ

General Description

1. Program title: APPRAZ

2. Calculator: HP-41C with 3 memory modules or HP-41CV

3. Purpose: to calculate timber volumes and summarize same by species and log grade; additionally, numbers of trees by species and DBH class are summarized as are number of logs by species. Grand total volume, number of logs and number of trees are also provided.

4. Application: to be used in timber harvest inventories, on site, or from tally sheets where data are recorded by logs in the form

   dd - LL - G

   where dd = scaling diameter of log in inches
   LL = log length in feet
   G = log grade (5 grades possible)

5. Justification: the primary advantage of calculator processing is that the very time consuming operation of sorting by grade within species is eliminated; secondarily all calculations and sums are automated, thus avoiding table look-up or calculation errors and minimizing transcribing errors.

Program Procedure

The following instructions are rather detailed, assuming that the user is "feeling his way." It is recommended that the procedure be read in its entirety at least once before actual implementation. If a printer is attached to the calculator, displayed output will be printed.

Step

1 XEQ SIZE 141 and load program APPRAZ.

2 To initialize (set flags and clear stack and registers; assign species names) XEQ APPRAZ. Assigning APPRAZ to a key (say x≶y) is only advisable for demonstration purposes, and, in general, USER mode should not be used until all data have been processed (step 6).

3 In response to "SPECIES?"

a. Key in species code for first tree; example: WALNUT = 1, WH.OAK = 2, etc. (9 species possible)

b. If tree count is not desired by DBH class, depress CHS (change sign). The presence or absence of negative on this first tree only dictates absence or presence of the DBH count summary for entire cruise.

4. Depress R/S

a. In the absence of a negative species code, the prompt will be "DBH CLASS?"; go to step 5.

b. If first species code was negative, the prompt will be: "dd.LLG?"

(1) whereupon, for the first log, one "keys-in" the integer scaling diameter (dd), decimal point, the integer length (LL) and log grade (zero through 4; i.e., five grades)

(2) depress R/S and the log will be processed; an audible tone will sound and the prompt will be "NEW TREE? Ø"

(a) for a new log in the same tree, press R/S and prompt will be "dd.LLG?" as for first log (as in 4b)

(b) for a new tree, press zero, then R/S and prompt will be "SPECIES?" as in 3

5 In response to "DBH CLASS? " key-in an integer diameter class (10 to 40) and press R/S.

a. Tree count by DBH class will be processed, and the prompt for the first log will be "dd.LLG?"

b. Proceed as in 4b, eventually depressing 0, R/S for a new tree.

c. For each new tree, the "DBH CLASS?" prompt will automatically appear, unless species code is negative, in which case the DBH prompt and summary will be aborted. To re-activate this option, flag 1 must be set manually, i.e., f SF 01* (or start over by XEQ APPRAZ).

6 When all trees have been processed, summaries are available as follows:

a. In USER mode, depress keys A through E and fA through fD for species 1 through 5 and 6 through 9, respectively. Example:

depress C (i.e., species 3) and output will be:

```
  SPECIES 3       (pause)
  RE.OAK          (pause)
  GR.0 = x x x x  (pause)
  GR.1 = x x x x  (pause)
  GR.2 = x x x x  (pause)
  GR.3 = x x x x  (pause)
  GR.4 = x x x x  (pause)
 Σ VOL. = x x x x (pause)
Σ TREES = x x     (pause)
 Σ LOGS = x x
```

b. Similarly, in USER mode, depress fE to read grand totals of volume by grade (i.e., across species); output will be:

```
  GRAND TOTALS    (pause)
  GR.0 = x x x x  (pause)
  GR.1 = x x x x  (pause)
  GR.2 = x x x x  (pause)
  GR.3 = x x x x  (pause)
  GR.4 = x x x x  (pause)
 Σ VOL. = x x x x (pause)
Σ TREES = x x     (pause)
 Σ LOGS = x x
```

c. Total number of trees by DBH class (DD) can be displayed using any of three options: semiautomatic (DBH); automatic display, one inch classes (AD1); or automatic display, two inch classes (AD2).

---

* Throughout this program, the letter "f" is used to indicate depression of the shift (gold) key.

(1) XEQ DBH (semiautomatic display)

(a) at the prompt "DBH CLASS?", key-in desired DD (10 to 40), depress R/S; "NO. TREES = xx." will be displayed for approx. 2 seconds then the "DBH CLASS?" prompt will reappear.

(b) repeat (a) for another DBH class.

(2) XEQ AD1 (automatic display, one-inch classes; assuming data were recorded by one-inch classes in the cruise)

(a) at the prompt "1ST CLASS:1:?", key-in the class where you want the consecutive automatic display to start, say DD (10 to 40), depress R/S.

(b) starting with this initial class, and then by increments of one-inch, the class will be identified rapidly by "DBH CLASS:DD". Then "NO. TREES = xx". will remain for two seconds.

(c) display will stop after 40-inch class has been displayed (to stop manually, depress R/S)

(3) XEQ AD2 (automatic display, two-inch classes; assuming data were recorded by two-inch classes in the cruise)

Procedure is the same as for AD1, except that the prompt will be "1ST CLASS:2:?", whereupon an initial <u>even</u> DD (10 to 40) class should be keyed-in. Display will then proceed by two-inch increments through the 40-inch class.

7 <u>SPECIAL NOTE</u>: The contents of any two-digit storage register (00 through 99) can be displayed at any time by depressing f VIEW nn. Calculations are not affected since the stack is not disturbed. A depression of ← will restore the x register contents to display. For example, the accumulated volume in species 1 grade 2 can be viewed by f VIEW 12; volume in species 6, grade 0 by f VIEW 60; grand total volume by f VIEW 05; etc. (Refer to storage assignments list for other data locations)

8 ERROR CORRECTIONS. If one observes faulty species code, DBH, or log data in the display <u>prior</u> to processing that tree or log, the correction key (←) can be used to rectify the information.

If one senses that faulty information has <u>just been</u> processed, certain "erasure" routines are possible:

a. To erase the effect of an erroneous species code or DBH class at the time of the first dd.LLG prompt in a tree (i.e., before the audible tone)

(1) XEQ TGOOF (for tree goof)

(2) prompt will then become "SPECIES?"

(3) proceed with correct species code and DBH as in step 3.

b. To erase the effect of erroneous log data at the time of the "NEW TREE? Ø" prompt (i.e., just after the audible tone)

(1) XEQ LGOOF (for log goof)

(2) erroneous volumes and counts will be deducted and prompt will again be "NEW TREE? Ø" after tone sounds.

(3) depress R/S (as for any new log); key-in correct log data, and depress R/S to process the correct data.

9 Program as written assumes bd. ft. volumes, Doyle scale:

$$V_i = (d_i - 4)^2 \frac{L_i}{16}$$

where V_i = log volume in board feet

d_i = log diameter in inches, inside bark at the small end

L_i = log length in feet

If other log rules or cubic foot volume formulas are desired, one can substitute a subroutine with the label LBL 13; for location convenience, it is located last in the APPRAZ program.

Examples:

a. Scribner bd. ft. volume:

$$V_i = (.79d_i^2 - 2d_i - 4)\frac{L_i}{16}$$

Subroutine:

LBL 13, RCL08, x^2, .79, x, RCL08, 2, x, -, 4, -, RCL09, x, 16, ÷, FIX0, XEQ RND, STO98, FIX3, RTN

10 The species codes and names used in the program are as follows:

1 = WALNUT
2 = WH.OAK
3 = RE.OAK
4 = ASH
5 = TULIP
6 = HICKRY (yes, it's misspelled, but for convenient storage one is limited to six characters!)
7 = H.MAPL
8 = S.MAPL
9 = MISC.

To alter this list or use entirely different names or assignments, one can make appropriate changes in subroutine 15, near the end of the program, as follows:

a. To get to this part of the program, make sure that the calculator is <u>not</u> in PRGM mode; depress f GTO 15.

b. Put the calculator in PRGM mode, and use the delete key (←) to delete the unwanted names; then key-in the desired names from top (1) to bottom (9); they will be separated by XEQ 14 instructions.

c. Take calculator out of PRGM mode.

11 If calculator is turned off before a summary is finished, it is most logical (though not necessary) to do this after a given <u>tree</u> has been processed and Ø, R/S have been depressed. To see the prompt for the next tree when the calculator is turned on, depress ALPHA key. Depress ALPHA again (i.e., out of ALPHA mode) before keying-in the next log data.

Typical Data and Example

The hypothetical data given here are meant to be used as a test after APPRAZ has been appropriately loaded in the HP-41C. A four-tree sample (2 black walnut, 1 white oak, and 1 red oak) is shown. Typical summary tables are then presented. In practice these tables should be prepared in skeleton form, then filled-in from observing the calculator display or the paper tape (see example in part C) if a printer has been used. Note that volumes shown are board feet, Doyle Scale. Other volume units can be obtained by program alteration as described in Step 9 of the "Procedure."

A. Data:

			LOG #1		LOG #2		LOG #3	
Tree No.	Spec.	DBH	dd.LLG	vol.	dd.LLG	vol.	dd.LLG	vol.
1	BLW(1)	22	21.060	108	18.082	98		
2	WHO(2)	18	16.100	90	15.101	76		
3	BLW(1)	18	15.121	91	14.081	50	12.083	32
4	REO(3)	24	21.122	217				

B. Results:

	SPECIES									
	1	2	3	4	5	6	7	8	9	GRAND
ITEM	WALNUT	WH.OAK	RE.OAK							TOTALS
Grade 0	108	90	0							198
Grade 1	141	76	0							217
Grade 2	98	0	217							315
Grade 3	32	0	0							32
Grade 4	0	0	0							0
Total volume	379	166	217							762
Tree count	2	1	1							4
Log count	5	2	1							8

DBH CLASS	NUMBER OF TREES
.	.
.	.
18	2
20	0
22	1
24	1
.	.
.	.
40	-

C. Sample printout obtained by pressing A, B, C, and f E and by executing (XEQ) AD2.

```
                    XEQ A
SPECIES: 1
WALNUT
GR. 0=108
GR. 1=141
GR. 2=98
GR. 3=32
GR. 4=0
Σ VOL.=379
Σ TREES=2
Σ LOGS=5
                    XEQ B
SPECIES: 2
WH.OAK
GR. 0=90
GR. 1=76
GR. 2=0
GR. 3=0
GR. 4=0
Σ VOL.=166
Σ TREES=1
Σ LOGS=2
                    XEQ C
SPECIES: 3
RE.OAK
GR. 0=0
GR. 1=0
GR. 2=217
GR. 3=0
GR. 4=0
Σ VOL.=217
Σ TREES=1
Σ LOGS=1
                    XEQ ε
GRAND TOTALS
GR. 0=198
GR. 1=217
GR. 2=315
GR. 3=32
GR. 4=0
Σ VOL.=762
Σ TREES=4
Σ LOGS=8
```

```
                XEQ "AD2"
1ST CLASS:2: ?
                16     RUN
DBH CLASS:16
NO. TREES=0
DBH CLASS:18
NO. TREES=2
DBH CLASS:20
NO. TREES=0
DBH CLASS:22
NO. TREES=1
DBH CLASS:24
NO. TREES=1
DBH CLASS:26
NO. TREES=0
DBH CLASS:28
NO. TREES=0
                     STOP
```

Storage Register Assignments

The location of various intermediate and final calculations can be found in the following table. One can determine the contents of register nn by using f VIEW nn or RCL nn, but this is not recommended until the entire data set has been completed. Otherwise, program processing might be adversely disturbed.

Register	Contents
00	Volume in Grade 0 logs
01	Volume in Grade 1 logs
02	Volume in Grade 2 logs
03	Volume in Grade 3 logs
04	Volume in Grade 4 logs
05	Grand total volume
06	Grand total no. trees
07	Grand total no. logs
08	Current value of dd
09	Current value of LL
10	Volume in species 1, Grade 0
11	Volume in species 1, Grade 1
12	Volume in species 1, Grade 2
13	Volume in species 1, Grade 3
14	Volume in species 1, Grade 4
15	Total volume for species 1
16	Total no. trees for species 1
17	Total no. logs for species 1
18	Current value of G
20-27	Same as 10-17 for species 2
28	Current species code or alteration
30-37	Same as 10-17 for species 3
.	.
.	.
.	.
80-87	Same as 10-17 for species 8
88	DBH class start for automatic display
89	Control number for automatic DBH display
90-97	Same as 10-17 for species 9
98	Current log volume
99	Pointer for DBH summary register
101-109	Species names
110	No. trees in 10" class
111	No. trees in 11" class
.	.
.	.
.	.
140	No. trees in 40" class

UNUSED:

19
29
38,39
48,49
58,59
68,69
78,79
100
111, 113, ..., 137, 139 — If 2-inch DBH classes are used.

Program Listing

APPRAZ

```
4:29PM 03/23
01♦LBL "APPRAZ"
02 CLST
03 CLRG
04 SF 01
05 XEQ 15
06♦LBL 01
07 "SPECIES ?"
08 PROMPT
09 X<0?
10 CF 01
11 ABS
12 1 E01
13 *
14 STO 28
15 FS? 01
16 XEQ 10
17 1
18 ST+ 06
19 6
20 ST+ 28
21 1
22 ST+ IND 28
23 6
24 ST- 28
25♦LBL 11
26 "  dd.LLG ?"
27 PROMPT
28 INT
29 STO 08
30 LASTX
31 FRC
32 1 E02
33 *
34 INT
35 STO 09
36 LASTX
37 FRC
38 1 E01
39 *
40 STO 18
41 XEQ 13
42♦LBL 04
43 ST+ 05
44 5
45 ST+ 28
46 RCL 98
47 ST+ IND 28
48 2
49 ST+ 28
50 1
51 ST+ 07
52 ST+ IND 28
53 7
54 ST- 28
55 RCL 18
56 ST+ 28
57 RCL 98
58 ST+ IND 28
59 ST+ IND 18
60 RCL 18
61 ST- 28
62 RCL 98
63 TONE 9
64 "NEW TREE? 0"
65 PROMPT
66 X=0?
67 GTO 01
68 GTO 11
69♦LBL 10
70 "DBH CLASS?"
71 PROMPT
72 ENTER↑
73 100
74 +
75 STO 99
76 1
77 ST+ IND 99
78 RTN
79♦LBL A
80 1
81 STO 28
82 XEQ 12
83 STOP
84 GTO 01
85♦LBL B
86 2
87 STO 28
88 XEQ 12
89 STOP
90 GTO 01
91♦LBL C
92 3
93 STO 28
94 XEQ 12
95 STOP
96 GTO 01
97♦LBL D
98 4
99 STO 28
100 XEQ 12
101 STOP
102 GTO 01
103♦LBL E
104 5
105 STO 28
106 XEQ 12
107 STOP
108♦LBL a
109 6
110 STO 28
111 XEQ 12
112 STOP
113♦LBL b
114 7
115 STO 28
116 XEQ 12
117 STOP
118♦LBL c
119 8
120 STO 28
121 XEQ 12
122 STOP
123♦LBL d
124 9
125 STO 28
126 XEQ 12
127 STOP
128♦LBL e
129 "GRAND TOTALS"
130 AVIEW
131 PSE
132 0
133 XEQ 03
134 STOP
135 GTO 01
136♦LBL 12
137 FIX 0
138 RCL 28
139 "SPECIES: "
140 ARCL X
141 AVIEW
142 PSE
143 CLA
144 100
145 +
146 ARCL IND X
147 AVIEW
148 PSE
149 RCL 28
150 1 E01
```

```
151 *
152♦LBL 03
153 FIX 0
154 STO 28
155 RCL IND 28
156 "GR. 0="
157 XEQ 02
158 "GR. 1="
159 XEQ 02
160 "GR. 2="
161 XEQ 02
162 "GR. 3="
163 XEQ 02
164 "GR. 4="
165 XEQ 02
166 "Σ VOL.="
167 XEQ 02
168 "Σ TREES="
169 XEQ 02
170 "Σ LOGS="
171♦LBL 02
172 ARCL X
173 AVIEW
174 PSE
175 1
176 ST+ 28
177 RCL IND 28
178 RTN
179♦LBL "DBH"
180 "DBH CLASS?"
181 PROMPT
182 FIX 0
183 100
184 +
185 STO 99
186 RCL IND 99
187 "NO. TREES="
188 ARCL X
189 AVIEW
190 PSE
191 PSE
192 GTO "DBH"
193♦LBL "AD1"
194 CF 02
195 "1ST CLASS:1: ?"
196 PROMPT
197 STO 88
198 1 E02
199 +
200 .140
201♦LBL 17
202 FIX 0
203 +
204 STO 89
205♦LBL 16
206 "DBH CLASS:"
207 ARCL 88
208 AVIEW
209 CLA
210 "NO. TREES="
211 ARCL IND 89
212 AVIEW
213 PSE
214 ISG 89
215 GTO 18
216 STOP
217 FS?C 02
218 GTO "AD2"
219 GTO "AD1"
220♦LBL 18
221 FS? 02
222 2
223 FC? 02
224 1
225 ST+ 88
226 GTO 16
227♦LBL "AD2"
228 "1ST CLASS:2: ?"
229 PROMPT
230 STO 88
231 1 E02
232 +
233 .14002
234 SF 02
235 GTO 17
236♦LBL "LGOOF"
237 7
238 ST+ 28
239 2
240 ST- IND 28
241 ST- 07
242 7
243 ST- 28
244 -1
245 ST* 98
246 RCL 98
247 GTO 04
248♦LBL "TGOOF"
249 6
250 ST+ 28
251 1
252 ST- 06
253 ST- IND 28
254 ST- IND 99
255 GTO 01
256♦LBL 15
257 101.109
258 STO 00
259 "WALNUT"
260 XEQ 14
261 "WH.OAK"
262 XEQ 14
263 "RE.OAK"
264 XEQ 14
265 "ASH"
266 XEQ 14
267 "TULIP"
268 XEQ 14
269 "HICKRY"
270 XEQ 14
271 "H.MAPL"
272 XEQ 14
273 "S.MAPL"
274 XEQ 14
275 "MISC."
276♦LBL 14
277 ASTO IND 00
278 ISG 00
279 RTN
280 0
281 STO 00
282 GTO 01
283♦LBL 13
284 RCL 08
285 4
286 -
287 X↑2
288 RCL 09
289 *
290 16
291 /
292 FIX 0
293 RND
294 STO 98
295 FIX 3
296 RTN
297 .END.
```

APPRAZ:791 BYTES

Retrospective Comments Regarding APPRAZ

1. The data input operation can be made less tedious by eliminating the "NEW TREE? Ø" prompt and adopting the convention: if there is another log in the tree, key the data and press R/S, otherwise simply press R/S; the absence of a number keyed-in thus indicates that the prompt for a new tree is needed. This can be accomplished by changing the program as follows:

   a. delete steps 64 through 67
   b. insert FC? 22
      GTO 01 after step 27 (PROMPT)
   c. insert CF 22 after step 25 (LBL 11)

2. In reviewing the number of trees by diameter class (AD2 and AD1) the class identifier is viewed a little fast; to correct this, insert SIN after step 208 (AVIEW). This will take the sin of whatever is in the X register and the time involved will effectively produce a short pause.

3. The program was written before a printer was available; therefore, if a printer is attached, turned on, and in MAN mode, species, grade, and DBH summaries will be printed but not spaced for legibility. To partially correct this, insert ADV statements after

   step 128 (LBL e)
   step 136 (LBL 12)
   and step 205 (LBL 16)

4. The modifications described in comments 1, 2, and 3 have been incorporated into APPRAZ and the new program, called APPRZ1, is available on the KRON-1 tape.

Program No. 41F002

Thomas W. Beers
Dept. Forestry, Purdue Univ.
January 1980

CRUZ

General Description

1. Program title: CRUZ

2. Calculator: HP-41C with 4 memory modules or HP-41CV

3. Purpose: to process inventory data from fixed-size or horizontal point samples, where the tallied data are: species, tree DBH, number of logs, and, optionally, soundness. Per-acre and total tract summaries are provided for volume, basal area, and number of trees, by species and overall. Per-acre and total estimates by DBH classes are provided for volume and number of trees. A statistical routine estimates the mean, standard error, percent standard error and .95 confidence intervals for volume and basal area.

4. Application: to be used on site or from tally sheets where data are taken in the form

   S - DD. H - SS

   where S = species code (9 possible)
   DD = tree DBH
   H = height in logs
   SS = tree soundness factor (this option available if flag 02 is set)

   Horizontal point sampling is assumed; if fixed-size plots are used, flag 01 must be set.

5. Justification: there are two primary advantages provided by the calculator processing. The time consuming operation of sorting into species and DBH classes is eliminated and the tedious calculations and/or table look-up for tree volume and statistical computations are avoided.

   Although the calculator can be used as an alternative to paper recording of the data it is probably safer to do both or plan on using the calculator after the inventory. In this case data can be recorded, line by line, by plot number and tree number and subsequently processed in an orderly manner. An obvious advantage of this recording format over a dot-dash summary tally is that the calculation of standard error and interval estimates is made possible.

Program Procedure

The following instructions are rather detailed, assuming that the user is "feeling his way." It is recommended that the procedure be read in its entirety at least once before actual implementation. Note that a printer cannot be attached since four modules are required. Use of double or quad-density modules should overcome this limitation, but the spacing may be erratic since the program was written assuming no printer.

Step

1 XEQ SIZE 142

2 Load programs CRUZ, STP, TREEVOL, and SPECIES.*

3 Horizontal point sampling and no tree soundness is assumed:

a. for fixed-size plots, set flag 01 (i.e., f SF 01)**

b. for inventories recognizing tree soundness, set flag 02

4 XEQ CRUZ -- this clears storage registers, stack, and certain internal flags

5 In response to "F = ?" (or "PLOT AREA = ?") key-in basal area factor (or fixed-size plot area in acres) and depress R/S

6 In response to the prompt, "SDD.H" (or "SDD.HSS") key-in the data in the indicated format:

$$\left.\begin{array}{ll} \text{species code} & = \text{S} \\ \text{DBH} & = \text{DD} \\ \text{height in logs} & = \text{H} \\ \text{decimal soundness} & = \text{SS} \end{array}\right\} \quad \text{SDD.H (or SDD.HSS)}$$

and depress R/S

7 An audible tone indicates that the tree has been processed; in response to the prompt

"LAST TREE? Ø"

a. If that tree was the final one at the point (or on the plot) depress zero, then R/S

---

* If these programs are read from magnetic cards, one must first XEQ SIZE 064, and an "f GTO.." operation must occur between each program loaded. Before continuing then, turn calculator OFF, remove card reader, insert fourth memory module, turn calculator ON, and XEQ SIZE 142.

** Throughout this program, the letter "f" is used to indicate depression of the shift (gold) key.

b. If there are more trees at that location, simply depress R/S.

c. Repeat steps 6 and 7 until the "last-tree" condition is met (i.e., 7a is executed)

(1) For the first tree on a new point (or plot), repeat step 6 and 7

(2) When the last tree on the last point (or plot) has been processed, go to step 8. (Note: display will be "SDD.H" or "SDD.HSS", awaiting the first tree at the next location.)

8 XEQ PA (can be assigned to R↓ and USER mode employed) -- the calculator will take approximately 25 seconds to make the necessary calculations and storage manipulations; then it will display "PER ACRE CALC. NOW DONE DEPRESS A THRU e, 1 BY 1."

a. For species 1 through 9, in USER mode, depress keys A through E and fA through fD. Example:

depress A (i.e., species 1) and output will be:

```
SPECIES:  1         (pause)
WALNUT              (pause)
VOL. = xxx.         (pause)
B.A. = xx.xx        (pause)
NO. TREES = xx.x
```

b. Similarly, depress fE to display the overall averages per acre:

```
VOL. = xxx.         (pause)
B.A. = xx.xx        (pause)
NO. TREES = xx.x
```

9 XEQ STP (can be assigned to SIN)

a. In response to the prompt "T = ?, or R/S", the user can key-in Student's t appropriate for .95 confidence intervals (default value assumed by the program is 2.0)

b. Depress R/S -- calculations will be made; then the prompt "A = VOL., B = BA" will appear

In USER mode, depressing A for volume or B for basal area will lead to estimates displayed in the following format: example: depress A

```
VOLUME:            (pause)
MEAN = xxxx.       (pause)
S x BAR = xxxx.    (pause) -- standard error of mean
IN % = xx.         (pause) -- standard error expressed as
                              a % of the mean
.95 CI = ± xxx.            — .95 confidence interval
                              estimate of the mean
```

c. If the mean and interval estimates are to be converted to a tract basis, after the above display depress R/S, and in response to the prompt "FOREST AREA = ?" key-in the area and depress R/S again. Display will be:

TOTAL = xxxxx (pause)
.95 C.I. = ± xxx — on a total basis

(1) Note that if these operations are done as follows: A, R/S, B; or A, B, these both can be reliably repeated; but attempts to repeat after the following (and perhaps others): A, R/S, B, R/S; or A, B, R/S will lead to interchanging the volume and basal area answers. To clear this condition and to enable repeats, simply set flag 05 manually, and depress A.

10 XEQ DBH (can be assigned to COS)

a. In response to the prompt DBH CLASS?, key-in the lowest two-inch class that summaries are desired for.

b. Depress R/S and display will be:

VOLUME = xxxx (pause)
NO. TREES = xx.x

c. Depress R/S again for the next higher class; the class identifier will be displayed rapidly, so be alert!

d. Caution: each time PA is executed after the first, execution of DBH will lead to answers 1/n th the correct values.

11 XEQ TOT (can be assigned to TAN)

a. In response to "FOREST AREA = ?" key-in the area, and depress R/S. (TOT takes about 25 seconds)

b. Display will be TRCT TOT CALC. NOW DONE. DEPRESS A THRU e, 1 BY 1

c. Summaries on a total tract basis will be displayed as described for per acre answers in 8a and 8b; Step 9 (STP program) is no longer appropriate.

d. Caution: if STP has been executed prior to TOT, the answers obtained from pressing fE may have the volume and basal area answers interchanged.

12 The program, as written, assumes tree volumes are board feet, Doyle scale, calculated using the following regression equations:

$V_i = .018D_i^2H_i$ for DBH = D = 10, 12, 14 inches

$V_i = .022D_i^2H_i$ for DBH = D = 16 thru 24 inches

$V_i = .024D_i^2H_i$ for DBH = D > 24 inches

To use another volume estimation formula, a new program called TREEVOL should be prepared, assuming the following register assignments:

R_{20} = source of DBH

R_{30} = source of height in 12-foot logs (the 12 multiplier is located in LBL 03 of CRUZ program; in case 16-foot logs are to be used, this can be changed to 16)

R_{36} = used for intermediate calculations; here to store $D_i^2H_i$

The calculated tree volume should be left in the X register and then should terminate with the

```
statements: FS? 02   (for soundness option)
            RCL 40   (tree soundness)
            FS? 02
            X
            END
```

to return sound volume to the main program if flag 02 is set.

13 The species codes and names used in the program are as follows:

1 = WALNUT
2 = WH.OAK
3 = RE.OAK
4 = ASH
5 = TULIP
6 = HICKRY
7 = H.MAPL
8 = S.MAPL
9 = MISC.

To alter this list or to use entirely different names or assignments, one can make appropriate changes in the program "SPECIES". To access the program:

a. in PRGM and/or USER mode, depress

f GTO alpha SPECIES alpha;

b. with calculator in PRGM mode, single step through the program and make the necessary deletions (←) and insertions;

c. take calculator out of PRGM mode

14 The following general comments should be noted:

a. Although in the program description, various program labels were assigned to keys, the user is cautioned that this might not be a sound practice in actual use since accidental depression of such keys (in USER mode) may obliterate considerable data already entered and processed. As a protection against this problem:

(1) do not assign programs to keys, and/or

(2) keep the calculator out of USER mode until the final summaries are to be output

b. To minimize the chances of clearing the entire program when altering various segments, four stand-alone programs are used. They are

(1)	CRUZ	using labels PA, TOT, DBH, A through e, 01 through 12, 14, 16, 18, and 21
(2)	STP	using labels A, B, 14 (same subroutine as in CRUZ), 19, and 20
(3)	TREEVOL	using labels 13, 17, and 22
(4)	SPECIES	using label 15

c. By reference to the register assignment sheet, the user can recall some values not specifically displayed by the program. For example, after the STP program has been executed (step 9), to display

CV (coefficient of variation) for volume: RCL 44
CV for basal area : RCL 45

standard deviation for volume : RCL 24
standard deviation for basal area: RCL 25

statistical accumulations: RCL 04 through 09

Typical Data and Example

The hypothetical data given here are meant to be used as a test after CRUZ has been appropriately loaded in the HP-41C. Part I assumes a horizontal point sample inventory (flag 01 clear), while part II assumes a fixed size plot (flag 01 set). Either can be used with (flag 02 set) or without soundness as an input variable. Typical summary tables are also shown. In practice, these would be prepared in skeleton form and filled-in with results read from the display. Note that volumes shown are in board feet, Doyle Scale, and that log length is 12 feet (see step 12 of Procedure for alterations.).

I. Horizontal Point Sample (with F = 10.0)

A. Data:

Plot	Tree	Species	DBH	Height in logs	Soundness	Volume* per tree	Volume* per acre	Tree factor	BA factor
1	1	BLW(1)	16	2	.99	135.2(133.8)	968.3(958.3)	7.16	10
	2	WHO(2)	18	3	.75	256.6(192.5)	1452.1(1089.4)	5.66	10
							2420.4(2047.7)	12.82	20
2	1	BLW(1)	16	1	.99	67.6(66.9)	484.2(479.2)	7.16	10
	2	WHO(2)	18	2	.75	171.1(128.3)	968.3(726.1)	5.66	10
	3	REO(3)	20	3	.50	316.8(158.4)	1452.1(726.1)	4.58	10
							2904.6(1931.4)	17.40	30
						Total:	5325.0(3979.1)	30.22	50
						Mean:	2662.5(1989.6)	15.11	25.0

B. Results:

SPECIES	VOLUME*	BA	NO. TREES	
1 WALNUT	726(719)	10.00	7.2	(same as DBH class 16)
2 WH.OAK	1210(908)	10.00	5.7	(same as DBH class 18)
3 RE.OAK	726(363)	5.00	2.3	(same as DBH class 20)
	2662(1990)	25.00	15.2	
MEAN	2662(1989)	25.0	15.1	
S x BAR	242(58)	5.0	-	
IN %	9(3)	20.0	-	
.95 C.I. (t=2)	±484(116)	10.0	-	

*Volumes in parentheses reflect the application of the soundness factor.

II. Fixed Size Plot Sample (one-fifth acre)

A. Data:

Plot	Tree	Species	DBH	Height in logs	Soundness	Volume* per tree	Volume* per acre	Tree factor[a/]	BA factor[b/]
1	1	BLW(1)	16	2	.99	135.2(133.8)	676.0(669.2)	5	6.98
	2	WHO(2)	18	3	.75	256.6(192.5)	1283.0(962.2)	5	8.84
							1959.0(1631.4)	10	15.82
2	1	BLW(1)	16	1	.99	67.6(66.9)	338.0(334.6)	5	6.98
	2	WHO(2)	18	2	.75	171.1(128.3)	855.5(641.6)	5	8.84
	3	REO(3)	20	3	.50	316.8(158.4)	1584.0(792.0)	5	10.91
							2777.5(1768.2)	15	26.73
						Total:	4736.5(3399.7)	25	42.55
						Mean:	2368.2(1699.8)	12.5	21.28

[a/]Tree factor = 1/plot size in acres
[b/]BA factor = (Tree factor)(tree basal area)

B. Results:

SPECIES	VOLUME*	BA	NO. TREES	
1 WALNUT	507(502)	6.98	5.0	(same as DBH class 16)
2 WH.OAK	1069(802)	8.84	5.0	(same as DBH class 18)
3 RE.OAK	792(396)	5.45	2.5	(same as DBH class 20)
	2368(1700)	21.27	12.5	
MEAN	2368(1700)	21.3	12.5	
S x BAR	409(68)	5.5	-	
IN %	17(4)	25.6	-	
.95 C.I. (t=2)	±818(137)	10.9	-	

*Volumes in parentheses reflect the application of the soundness factor.

CRUZ

Storage Register Assignments

Register	Contents		
00	F, or plot size (acres) reciprocal		
01	Vol.	per location per acre sum	then over-all per acre means and totals
02	BA		
03	No. trees		
04	Σx (vol)	sums of per acre for entire sample	
05	Σx^2		
06	Σy (BA)		
07	Σy^2		
08	Σxy		
09	n		
11	Vol.	running totals of per acre sums to date -- species 1	
12	BA		
13	No. trees		
17	Vol.	final per acre spec. 1	then tract total -- species 1
18	BA		
19	No. trees		
21	same -- species 2		
22			
23			
27			
28			
29			
.			
.			
.			
91	same -- species 9		
92			
93			
97			
98			
99			
10	species code		
14	Vol.	current, per tree per acre	
15	BA		
16	No. trees		
20	DD		
30	H in feet		
40	SS		
50	Tract area		
60	DBH pointer		
70	pointer for species Σ per acre		
80	pointer for species per acre		
90	pointer for Σ No. trees (in LBL 06)		

Register	Contents	
24	A_{vol}	acre basis
25	A_{BA}	
26	Student's t	
34	$s_{\bar{v}}$	
35	$s_{\overline{BA}}$	
36	D^2H current	
44	CV_v	
45	CV_{BA}	
54	$ts_{\bar{v}}$	
55	$ts_{\overline{BA}}$	
101-109	Species names	
110-	Volume in 10" class	sample per acre sums, then per acre (PA), then TOTALS (TOT)
111-	No. trees in 10" class	
112-	Vol. in 12" class	
113-	No. trees in 12" class	
.		
.		
.		
140-	Vol. in 40" class	
141-	No. trees in 40" class	

UNUSED:

46, 56
64-66
74-76
84-86
94-96
100

Program Listing

CRUZ

```
5:06PM 03/23
01♦LBL "CRUZ"
02 CLST
03 CLRG
04 ΣREG 04
05 XEQ "SPECIES"
06 CF 03
07 CF 05
08♦LBL 01
09 FS? 01
10 "PLOT AREA= ?"
11 FC? 01
12 "   F= ?"
13 PROMPT
14 FS? 01
15 1/X
16 STO 00
17♦LBL 02
18 FS? 02
19 "  SDD.HSS"
20 FC? 02
21 "  SDD.H"
22 PROMPT
23 1 E02
24 /
25 INT
26 STO 10
27 LASTX
28 1
29 MOD
30 1 E02
31 *
32 INT
33 STO 20
34 LASTX
35 FRC
36 1 E01
37 *
38 INT
39 STO 30
40 FC? 02
41 GTO 03
42 LASTX
43 FRC
44 STO 40
45♦LBL 03
46 12
47 ST* 30
48 RCL 20
49 X↑2
50 PI
51 *
52 576
53 /
54 STO 15
55 FC? 01
56 1/X
57 RCL 00
58 *
59 FS? 01
60 GTO 04
61 STO 16
62 RCL 00
63 STO 15
64 GTO 05
65♦LBL 04
66 STO 15
67 RCL 00
68 STO 16
69♦LBL 05
70 10
71 ST* 10
72 1
73 ST+ 10
74 XEQ "TREEVOL"
75 RCL 16
76 *
77 STO 14
78 ST+ 01
79 ST+ IND 10
80 1
81 ST+ 10
82 RCL 15
83 ST+ 02
84 ST+ IND 10
85 1
86 ST+ 10
87 RCL 16
88 ST+ 03
89 ST+ IND 10
90 100
91 ST+ 20
92 RCL 14
93 ST+ IND 20
94 1
95 ST+ 20
96 RCL 16
97 ST+ IND 20
98 TONE 9
99 "LAST TREE? 0"
100 PROMPT
101 X>0?
102 GTO 02
103 RCL 02
104 RCL 01
105 Σ+
106 0
107 STO 01
108 STO 02
109 STO 03
110 GTO 02
111♦LBL "PA"
112 0
113 STO 03
114 013.09310
115 STO 90
116♦LBL 06
117 RCL IND 90
118 ST+ 03
119 ISG 90
120 GTO 06
121 MEAN
122 STO 01
123 X<>Y
124 STO 02
125 RCL 03
126 RCL 09
127 /
128 STO 03
129 011.013
130 STO 70
131 17
132 STO 80
133♦LBL 07
134 RCL IND 70
135 RCL 09
136 /
137 STO IND 80
138 1
139 ST+ 80
140 ISG 70
141 GTO 07
142 7.010
143 ST+ 70
144 ST+ 80
145 RCL 70
146 100
147 X>Y?
148 GTO 07
149 110.141
150 STO 60
```

```
♦LBL 09
RCL 09
FS? 03
RCL 50
ST/ IND 60
ISG 60
GTO 09
CF 03
"PER ACRE CAL"
♦LBL 11
"├C. NOW DONE"
AVIEW
CLA
"DEPRESS A TH"
"├RU e, 1 BY 1"
AVIEW
CLA
RTN
♦LBL "TOT"
"FOREST AREA?"
PROMPT
SF 03
STO 50
17.019
STO 70
♦LBL 08
RCL 50
ST* IND 70
ISG 70
GTO 08
7.010
ST+ 70
RCL 70
100
X>Y?
GTO 08
110.141
STO 60
♦LBL 10
RCL 50
ST* IND 60
ISG 60
GTO 10
"TRCT TOT CAL"
XEQ 11
STOP
♦LBL A
1
STO 10
XEQ 12
STOP
♦LBL B
2
STO 10
XEQ 12
STOP
♦LBL C
3
STO 10
XEQ 12
STOP
♦LBL D
4
STO 10
XEQ 12
STOP
♦LBL E
5
STO 10
XEQ 12
STOP
♦LBL a
6
STO 10
XEQ 12
STOP
♦LBL b
7
STO 10
XEQ 12
STOP
♦LBL c
8
STO 10
XEQ 12
STOP
♦LBL d
9
STO 10
XEQ 12
STOP
♦LBL e
FS? 03
XEQ 18
1
STO 10
XEQ 16
STOP
♦LBL 12
FIX 0
"SPECIES:"
XEQ 14
100
+
ARCL IND X
AVIEW
PSE
10
ST* 10
7
ST+ 10
♦LBL 16
FIX 0
RCL IND 10
"VOL.="
XEQ 14
1
ST+ 10
RCL IND 10
FIX 2
"B. A. ="
XEQ 14
1
ST+ 10
RCL IND 10
FIX 1
"NO.TREES="
♦LBL 14
ARCL X
AVIEW
PSE
CLA
RTN
♦LBL 18
RCL 50
ST* 01
ST* 02
ST* 03
RTN
♦LBL "DBH"
"DBH CLASS?"
PROMPT
100.140
+
STO 60
♦LBL 21
RCL IND 60
FIX 0
"VOLUME="
XEQ 14
```

```
301 1
302 ST+ 60
303 RCL IND 60
304 FIX 1
305 "NO.TREES="
306 XEQ 14
307 STOP
308 FIX 0
309 ISG 60
310 RCL 60
311 100
312 -
313 INT
314 "DBH CLASS: "
315 ARCL X
316 AVIEW
317 GTO 21
318 RTN
319 .END.
```

CRUZ:773 BYTES

```
01♦LBL "STP"
02 2
03 "T=?, OR R/S"
04 PROMPT
05 STO 26
06 SDEV
07 STO 24
08 X<>Y
09 STO 25
10 RCL 09
11 SQRT
12 /
13 STO 35
14 RCL 24
15 LASTX
16 /
17 STO 34
18 RCL 24
19 RCL 01
20 /
21 1 E02
22 *
23 STO 44
24 RCL 25
25 RCL 02
26 /
27 1 E02
28 *
29 STO 45
30 RCL 26
31 RCL 34
32 *
33 STO 54
34 RCL 26
35 RCL 35
36 *
37 STO 55
38 "A=VOL., B=BA"
39 PROMPT
40♦LBL A
41 "VOLUME:"
42 AVIEW
43 FIX 0
44 FS?C 05
45 XEQ 20
46♦LBL 19
47 PSE
48 RCL 01
49 "MEAN="
50 XEQ 14
51 RCL 34
52 "S XBAR="
53 XEQ 14
54 RCL 01
55 /
56 100
57 *
58 "IN %="
59 XEQ 14
60 RCL 54
61 ".95 C.I.=≠"
62 XEQ 14
63 STOP
64 "FOREST AREA?"
65 PROMPT
66 STO 50
67 RCL 01
68 RCL 50
69 *
70 "TOTAL="
71 XEQ 14
72 RCL 54
73 RCL 50
74 *
75 ".95 C.I.=≠"
76 XEQ 14
77 RTN
78♦LBL B
79 SF 05
80 "BASAL AREA:"
81 AVIEW
82 FIX 1
83 XEQ 20
84 XEQ 19
85♦LBL 20
86 RCL 01
87 X<> 02
88 STO 01
89 RCL 34
90 X<> 35
91 STO 34
92 RCL 54
93 X<> 55
94 STO 54
95 RTN
96♦LBL 14
97 ARCL X
98 AVIEW
99 PSE
100 CLA
101 RTN
102 END
```

STP:270 BYTES

```
01♦LBL "TREEVOL"
02 RCL 30
03 RCL 20
04 X↑2
05 *
06 STO 36
07 24
08 RCL 20
09 X>Y?
10 GTO 13
11 15
12 RCL 20
13 X<=Y?
14 GTO 17
15 RCL 36
16 .022
17 *
18♦LBL 22
19 FS? 02
20 RCL 40
21 FS? 02
22 *
23 RTN
24♦LBL 13
25 RCL 36
26 .028
27 *
28 XEQ 22
29♦LBL 17
30 RCL 36
31 .018
32 *
33 XEQ 22
34 END
```

TREEVOL:77 BYTES

```
01♦LBL "SPECIES"
02 101.109
03 STO 00
04 "WALNUT"
05 XEQ 15
06 "WH.OAK"
07 XEQ 15
08 "RE.OAK"
09 XEQ 15
10 "ASH"
11 XEQ 15
12 "TULIP"
13 XEQ 15
14 "HICKRY"
15 XEQ 15
16 "H.MAPL"
17 XEQ 15
18 "S.MAPL"
19 XEQ 15
20 "MISC."
21♦LBL 15
22 ASTO IND 00
23 ISG 00
24 RTN
25 .END.
```

SPECIES:112 BYTES

Retrospective Comments Regarding CRUZ

1. Similar to the APPRAZ program, the data input operation in CRUZ can be made more efficient by eliminating the "LAST TREE? Ø" prompt and adopting the convention: if there is another tree on the plot, key the data and press R/S; the absence of a number keyed-in (i.e., pressing R/S without keying a number) indicates the end of the current plot. This can be accomplished by changing the program CRUZ as follows:

   a. insert LBL 00 after step 102 (GTO 02)
   b. delete steps 99, 100, and 101
   c. insert FC? 22
      GTO 00 after step 22 (PROMPT)
   d. insert CF 22 after step 17 (LBL 02)

2. CRUZ was written without concern for printed output, therefore when a printer is attached and turned on, lack of spaces between groups of output will diminish legibility. To partially correct this one may:

   a. In CRUZ

      (1) insert ADV after steps
          290 (LBL DBH)
          and 262 (LBL 16)

      (2) insert ADV
          ADV after step 249 (LBL 12)

      (3) insert CF 12 after step 256 (AVIEW)

      (4) insert SF 12 after step 255 (ARCL IND X)

   b. In STP

      (1) insert ADV after steps
          78 (LBL B)
          and 40 (LBL A)

3. Initially it was felt that having separate programs for STP, TREEVOL, and SPECIES would be desirable; subsequent experience has shown this practice to be questionable. To make these programs integral to the CRUZ program, assuming the programs are in the calculator in order (CRUZ, STP, TREEVOL, and SPECIES) one can:

   a. In CRUZ, delete step 319 (END)

b. In STP

   (1) change
   step 38 (A=VOL., B=BA) to[1/] I=VOL., J=BA
   step 40 (LBL A) to LBL I[1/]
   step 78 (LBL B) to LBL J[1/]

   (2) delete step 102 (END)

c. In TREEVOL, change step 34 (END) to RTN

4. The modifications described in comments 1, 2, and 3 have been incorporated into a new program called CRUZ1, available on the KRON-1 tape.

5. Comments from certain users have indicated that CRUZ is too ponderous and is expected to do too much! The author sympathizes with these feelings but does not apologize, since the primary purpose of the program, indeed as for many of the programs in this chronology, was to demonstrate what <u>can</u> be done with a portable calculator. Ideally, efficient specific programs should be written to match the input data format and summary results desired.

6. CRUZ was originally written and described to process either fixed-size plots or variable-size plots (horizontal point sampling), and no mention was made of the fact that a 100% cruise (complete inventory) can also be accommodated by the program. However, to handle a 100% inventory the user can:

   a. select the fixed-size plot option (flag 01 set)
   b. let plot size = 1 at the appropriate prompt
   c. assume the data came from one big plot, therefore

      (1) if using CRUZ, key 0 then R/S only once, after the last tree has been processed
      (2) if using CRUZ1, hit R/S without data entry, only after the last tree has been processed

   d. In this option, "PA" must be executed to calculate the "total" answers, "STP" and "TOT" have no meaning, but "DBH" when executed will provide proper answers by diameter classes.

[1/] Note that if this is done, DBH and TOT must be assigned to keys other than COS and TAN as described in the CRUZ write-up. Shift COS and shift TAN are logical alternatives.

Program No. 41F005

Thomas W. Beers
Dept. Forestry & Natural Resources, Purdue Univ.
July 1980

1. Program title: MIL3

2. Calculator: HP-41CV or HP-41C with 3 memory modules.

3. Purpose: (a) to provide on-the-spot solution of the following expression of sawmill "profit margin" using the formula:

$$\text{margin} = \sum_{i=1}^{6} (\text{lumber tally})_i (V_i) - \text{log value} - \text{sawing cost},$$

where log value = (log grade value)(log scale)
sawing cost = (cost per minute)(sawing time)
V_i = value for various lumber grades.

and (b) to accumulate log scale and lumber tally by various species and grade classes, as well as total margin and sawing times for an indefinite number of logs.

4. Application: the program was prepared primarily as an example of what can be done with the programmable calculator, however, it could conceivably be adapted for a small study to evaluate the "margin" retrievable from an arbitrary run of logs.

5. Justification: the fact that the programmable calculator is portable makes it quite feasible to make on-site determinations where here-to-fore, results could only be done at some time subsequent to the data gathering phase of the study.

   The capacity of the calculator to rapidly and selectively store and accumulate data enables rapid summarization and virtually eliminates hand-sorting and mathematical blunders.

Register Assignment for MIL3

Registers	Used for
00	current "margin" and miscellaneous indexing
01-06	stored lumber values (grades 1-6)
07	stored sawing cost per minute
08	current scale.grade or tally.grade } also miscellaneous indexing
09	current scale or tally } also miscellaneous indexing
10	species 1: margin total
11	species 1: grade 1 lumber tally
12	species 1: grade 2 lumber tally
13	species 1: grade 3 lumber tally
14	species 1: grade 4 lumber tally
15	species 1: grade 5 lumber tally
16	species 1: grade 6 lumber tally
17	species 1: log grade 1 scale
18	species 1: log grade 2 scale
19	species 1: log grade 3 scale
20 . . . 29	Species 2: same as for species 1
. . .	
80 . . . 89	Species 8: same as for species 1
90	grand total of margins
91-93	stored log grade values (grades 1-3)
94	grand total of log scales
95	grand total of lumber tally
96	grand total sawtimes
97	[10(species code) + log grade + 6] or [10(spec. code) + lumber grade]
98	species code
99	general indexing counter
101 . . . 108	stored species names
100	not used

Arbitrary Data and Hand-Calculated Results for MIL3

A. Data:

Log	Species	Scale	Grade	Lumber Tally	Lumber Grade	Sawing Time
1	WALNUT (1)	555	1	150	1	5 minutes
				40	2	
				80	3	
				60	4	
				100	5	
				150	6	
2	WH. OAK (2)	625	2	300	1	6 minutes
				200	2	
				150	4	

B. Prices per board foot:

Lumber:	grade	1	2	3	4	5	6
	value	.60	.50	.40	.30	.20	.10

Logs:	grade	1	2	3
	value	.3	.2	.1

Sawing cost per minute: \$2.00

C. Results:

1. $\text{margin}_1 = 150(.6) + 40(.5) + 80(.4) + 60(.3) + 100(.2) + 150(.10) - 555(.3) - 5(2.00) = \$18.50$

   $\text{margin}_2 = 300(.6) + 200(.5) + 150(.3) - 625(.2) - 6(2.00) = \$188.00$

2. Species summary:

WALNUT	scale = 555 bd. ft.	
	tally = 580 bd. ft.	
WH. OAK	scale = 625 bd. ft.	
	tally = 650 bd. ft.	

3. Margin total: \$206.50

4. Sawing time total: 11 minutes

5. Scale by log grade:

	Grade 1	Grade 2	Grade 3	Total
WALNUT	555	0	0	
WH. OAK	0	625	0	
Total	555	625	0	1180

6. Tally by lumber grade:

	Grade 1	2	3	4	5	6	Total
WALNUT	150	40	80	60	100	150	580
WH. OAK	300	200	0	150	0	0	650
Total	450	240	80	210	100	150	1230

Program Procedure

Step

1 XEQ SIZE 109, f* GTO...

2 Load program MIL3, f GTO.., load program TOT

3 To store the necessary constants, either a or b:

a. manually store data in the following registers:

Register	item
1-6	value per bd. ft., lumber grades 1-6
7	sawing cost, dollars per minute
91-93	value per bd. ft., log grades 1-3
101-108	species names, 6 character limit each;

then, XEQ MIL3 (suggest assignment to LN key); at the prompt "TO LOAD, SF06", depress R/S and go to step 4.

b. XEQ MIL3 and at "TO LOAD, SF06", set flag 06 (f SF06), and proceed to key in the constants as they are prompted for, then go to step 4.

NOTE:

(1) if flag 01 is not set, R/S must be depressed after each constant,

(2) if flag 01 <u>is set</u>, R/S need <u>not</u> be depressed, but you have limited time (1 sec.) to key in each digit.

(3) if only a few constants are to be changed, the manual procedure (3a) should be used.

4 After "STAND BY" (while registers 10-90 and 94-96 are being cleared),

a. for log 1:

---

*throughout this paper "f" represents the shift (gold) key.

Prompt	input(2)	key(1)	output	Example input	Example key	Example output
SPEC. CODE?	a number, 1 to 8	R/S	species name	1	R/S	WALNUT
LOG SSS.G?	log scale.grade	R/S	none	555.1	R/S	
LUM SSS.G?	tally.grade	R/S	none	150.1	R/S	
.	.			.		
.	.			.		
.	.			.		
LUM SSS.G?	tally.grade	R/S	none	150.6	R/S	
LUM SSS.G?	0	R/S	none	0	R/S	
SAW TIME?	sawing time	R/S	"margin"	5	R/S	$18.50**

(1) Note, if flag 01 is set R/S need not be pressed after data input.

(2) To undo a faulty log scale.grade or species code:

(a) at the prompt LUM SSS.G?, f GTO SIN, R/S

(b) Follow the species prompt with same code as just used, RCL SIN, CHS, R/S

(c) repeat step (a) and proceed using correct data

To undo a faulty lumber tally.grade:

(d) at the prompt LUM SSS.G?, RCL SIN, CHS, R/S

(e) proceed with the correct data

To undo a faulty species code only:

(f) at the prompt LOG SSS.G?, f GTO SIN, R/S and proceed with the correct code.

Note: there is no way in the present program of correcting for a log grade over 3 or for a lumber grade over 6; this will really foul up the works and will go undetected until crazy answers appear.

b. for log 2, depress R/S and repeat 4a; using the example:

**Note, these results are identified as "MARGIN=$xx.xx" and if negative, an audible "beep" is sounded.

prompt	Example input	key	output
SPEC. CODE?	2	R/S	WH. OAK
LOG SSS.G?	625.2	R/S	
LUM SSS.G?	300.1	R/S	
LUM SSS.G?	200.2	R/S	
LUM SSS.G?	150.4	R/S	
LUM SSS.G?	0	R/S	
SAW TIME?	6	R/S	$188.00**

c. repeat 4a for all logs in the "study".

5 For totals, XEQ TOT (suggest assignment to TAN), and proceed as prompted:

PRESS: A FOR SPEC. Σ
B FOR LOG Σ
C FOR LUM. Σ

Using the example previously cited:

	key	output
a.	A(Σ+)	WALNUT (pause) SCALE = 555 (pause) TALLY = 580 WH. OAK (pause) SCALE = 625 (pause) TALLY = 650 . . . for all eight species, then MARGIN Σ = $206.50 (pause) SAW TIME Σ = 11.0

Note: the species output list can be stopped, restarted, or aborted by presssing R/S, in which case the margin and saw time sums can be obtained by depressing f A(Σ+)

	key	output
	f A(Σ+)	MARGIN Σ = $206.50 (pause) SAW TIME Σ = 11.0
b.	B(1/x)	SCALE BY LOG GRADE GRADE 1 = 555 GRADE 2 = 625 GRADE 3 = 0
c.	C($\sqrt{x}$)	TALLY BY LUMBER GRADE (pause) GRADE 1 = 450 (pause)

** Note, these results are identifed as "MARGIN = $xx.xx" and if negative, an audible "beep" is sounded.

GRADE 2 = 240 (pause)
GRADE 3 = 80 (pause)
GRADE 4 = 210 (pause)
GRADE 5 = 100 (pause)
GRADE 6 = 150

d. After depressing either key B or key C the log scale and lumber tally, respectively, can be obtained for grades within species; with the user inputting the desired species code. Using the example, the procedure is as follows:

(1) immediately after depressing B and observing the output (or R/S)

key	prompt	input	key*	output
f B(1/x)	SPECIES?	1	R/S	WALNUT: (pause) GRADE 1 = 555 (pause) GRADE 2 = 0 (pause) GRADE 3 = 0 (pause)
	SPECIES?	2	R/S	WH. OAK: (pause) GRADE 1 = 0 (pause) GRADE 2 = 625 (pause) GRADE 3 = 0 (pause)
	SPECIES?	0	R/S	0

(this procedure is necessary to exit from the f B subroutine in order to continue with the TOT program)

(2) immediately after depressing C and observing the output (or R/S)

key	prompt	input	key*	output
f C($\sqrt{x}$)	SPECIES?	1	R/S	WALNUT: (pause) GRADE 1 = 150 (pause) GRADE 2 = 40 (pause) GRADE 3 = 80 (pause) GRADE 4 = 60 (pause) GRADE 5 = 100 (pause) GRADE 6 = 150 (pause)
	SPECIES?	2	R/S	WH. OAK: (pause) GRADE 1 = 300 (pause) GRADE 2 = 200 (pause) GRADE 3 = 0 (pause) GRADE 4 = 150 (pause) GRADE 5 = 0 (pause) GRADE 6 = 0 (pause)
	SPECIES?	0	R/S	0

(this procedure is necessary to exit from the f C subroutine in order to re-do any of the TOT program)

*if flag 01 is set, R/S need not be pressed.

6 To examine the contents of any storage register (up to 99) one can either RCL nn or f VIEW nn (refer to register assignment). For example, after first going to f FIX 2 mode, to find

a. grand total of margins, RCL 90, read 206.50

b. grand total of log scales, RCL 94, read 1180.00

c. grand total of lumber tally, RCL 95, read 1230.00

d. grand total of sawing times, RCL 96, read 11.00

e. the lumber tally for species 2, lumber grade 2, RCL 22, read 200.00

f. the log scale for species 2, log grade 2, RCL 28, read 625.00

etc.

Note that registers over 99 can be viewed only by "indirect addressing"; thus, to recall R_{101}: 101, STO 00, RCL f 00.

7 For a new set of logs using the same constants go to step 4, if new constants are to be entered go to step 3.

Example--printer output

```
              XEQ "MIL3"
TO LOAD, SF06
                     RUN
 STAND BY
SPEC. CODE?
               1
WALNUT
LOG SSS.G?
            555.1
LUM SSS.G?
            150.1
LUM SSS.G?
             40.2
LUM SSS.G?
             80.3
LUM SSS.G?
             60.4
LUM SSS.G?
            100.5
LUM SSS.G?
            150.6
LUM SSS.G?
              0.0
SAW. TIME?
             5.00
MARGIN=$18.50
                     RUN
SPEC. CODE?
               2
WH.OAK
LOG SSS.G?
            625.2
LUM SSS.G?
            300.1
LUM SSS.G?
            200.2
LUM SSS.G?
            150.4
LUM SSS.G?
              0.0
SAW. TIME?
             6.00
MARGIN=$188.00
```

```
               XEQ "TOT"
PRESS:
A FOR SPEC. Σ
B FOR LOG Σ
C FOR LUM. Σ
                   XEQ A
WALNUT
 SCALE= 555
 TALLY= 580

WH.OAK
 SCALE= 625
 TALLY= 650

RE.OAK
 SCALE= 0
 TALLY= 0

ASH
 SCALE= 0
 TALLY= 0

TULIP
 SCALE= 0
 TALLY= 0

HICKRY
 SCALE= 0
 TALLY= 0

MAPLE
 SCALE= 0
 TALLY= 0

MISC.
 SCALE= 0
 TALLY= 0


MARGINΣ=206.50

SAW TIMEΣ=11.0
                   XEQ a

MARGINΣ=206.50

SAW TIMEΣ=11.0
```

```
                      XEQ B
SCALE BY
LOG GRADE

 GRADE 1= 555
 GRADE 2= 625
 GRADE 3= 0
                      XEQ b
SPECIES ?
                  1
WALNUT:
 GRADE 1= 555
 GRADE 2= 0
 GRADE 3= 0

SPECIES ?
                  2
WH.OAK:
 GRADE 1= 0
 GRADE 2= 625
 GRADE 3= 0

SPECIES ?
                  5
TULIP:
 GRADE 1= 0
 GRADE 2= 0
 GRADE 3= 0

SPECIES ?
                  0
```

printer output - continued


```
                    XEQ C                              MEANINGFULL
TALLY BY                                                REGISTER
LUMBER GRADE                                             CONTENTS

 GRADE 1= 450                                          R01=  0.6
 GRADE 2= 240                                          R02=  0.5
 GRADE 3= 80                                           R03=  0.4
 GRADE 4= 210                                          R04=  0.3
 GRADE 5= 100                                          R05=  0.2
 GRADE 6= 150                                          R06=  0.1
                    XEQ c                              R07=  2.0
SPECIES ?                                              R08=  150.0
                 1                                     R09=  16.1
WALNUT:                                                R10=  18.5
 GRADE 1= 150                                          R11=  150.0
 GRADE 2= 40                                           R12=  40.0
 GRADE 3= 80                                           R13=  80.0
 GRADE 4= 60                                           R14=  60.0
 GRADE 5= 100                                          R15=  100.0
 GRADE 6= 150                                          R16=  150.0
                                                       R17=  555.0
SPECIES ?                                              R18=  0.0
                 2                                     R19=  0.0
WH.OAK:                                                R20=  188.0
 GRADE 1= 300                                          R21=  300.0
 GRADE 2= 200                                          R22=  200.0
 GRADE 3= 0                                            R23=  0.0
 GRADE 4= 150                                          R24=  150.0
 GRADE 5= 0                                            R25=  0.0
 GRADE 6= 0                                            R26=  0.0
                                                       R27=  0.0
SPECIES ?                                              R28=  625.0
                 4                                     R29=  0.0
ASH:                                                         .
 GRADE 1= 0                                                  .
 GRADE 2= 0                                                  .
 GRADE 3= 0                                            R89=  0.0
 GRADE 4= 0                                            R90=  206.5
 GRADE 5= 0                                            R91=  0.3
 GRADE 6= 0                                            R92=  0.2
                                                       R93=  0.1
SPECIES ?                                              R94=  1180.0
                 0                                     R95=  0.0
                                                       R96=  11.0
                                                       R97=  24.0
                                                       R98=  20.0
                                                       R99=  47.0
                                                       R100=  0.0
                                                       R101= "WALNUT"
                                                       R102= "WH.OAK"
                                                       R103= "RE.OAK"
                                                       R104= "ASH"
                                                       R105= "TULIP"
                                                       R106= "HICKRY"
                                                       R107= "MAPLE"
                                                       R108= "MISC."
```

Program Listing--MIL3


```
01♦LBL "MIL3"
02 CF 29
03 "TO LOAD, SF06"
04 PROMPT
05 FC?C 06
06 GTO 06
07 SF 05
08 1.007
09 STO 00
10 XEQ "LOAD"
11 91.093
12 STO 00
13 XEQ "LOAD"
14 101.108
15 STO 00
16 AON
17 XEQ "LOAD"
18♦LBL 06
19 SF 03
20 SF 02
21 " STAND BY"
22 AVIEW
23 10.09
24 XEQ "MRC"
25 0
26 STO 94
27 STO 95
28 STO 96
29♦LBL 05
30 0
31 STO 00
32♦LBL 08
33 FIX 0
34 100
35 STO 99
36 "SPEC. CODE?"
37 FS? 01
38 XEQ "F22"
39 FC? 01
40 PROMPT
41 STO 98
42 ST+ 99
43 CLA
44 ARCL IND 99
45 AVIEW
46 PSE
47 FIX 1
48 "LOG SSS.G?"
49 FS? 01
50 XEQ "F22"
51 FC? 01
52 PROMPT
53 XEQ 03
54 RCL 98
55 1 E1
56 *
57 STO 98
58 RCL 99
59 +
60 6
61 +
62 STO 97
63 90
64 ST+ 99
65 RCL 09
66 ST+ 94
67 ST+ IND 97
68 RCL IND 99
69 *
70 ST- 00
71♦LBL 01
72 "LUM SSS.G?"
73 FS? 01
74 XEQ "F22"
75 FC? 01
76 PROMPT
77 X=0?
78 GTO 02
79 XEQ 03
80 RCL 98
81 +
82 STO 97
83 RCL 09
84 ST+ IND 97
85 RCL IND 99
86 *
87 ST+ 00
88 GTO 01
89♦LBL 02
90 FIX 2
91 "SAW. TIME?"
92 FS? 01
93 XEQ "F22"
94 FC? 01
95 PROMPT
96 ST+ 96
97 RCL 07
98 *
99 ST- 00
100 RCL 00
101 ST+ IND 98
102 ST+ 90
103 X<0?
104 BEEP
105 "MARGIN=$"
106 ARCL X
107 AVIEW
108 STOP
109 GTO 05
110♦LBL 03
111 STO 08
112 INT
113 STO 09
114 LASTX
115 FRC
116 1 E1
117 *
118 ABS
119 STO 99
120 RTN
121♦LBL "MRC"
122 0
123♦LBL 04
124 STO IND Y
125 ISG Y
126 GTO 04
127 RTN
128♦LBL "F22"
129 CF 22
130 AVIEW
131♦LBL 10
132 PSE
133 FC?C 22
134 GTO 10
135 RTN
136♦LBL "LOAD"
137 "START REG?"
138 FC? 05
139 PROMPT
140 .4
141 +
142 FC? 05
143 STO 00
144 FIX 0
145 CF 29
146♦LBL 09
147 "REG "
148 ARCL 00
149 "⊢= ?"
150 PROMPT
```

MIL3 - continued

```
151 X<0?
152 GTO 06
153 FS? 48
154 ASTO IND 00
155 FC? 48
156 STO IND 00
157 ISG 00
158 GTO 09
159 AOFF
160 "DONE"
161 AVIEW
162 END

MIL3:414 BYTES
```

```
01♦LBL "TOT"
02 "PRESS:"
03 AVIEW
04 PSE
05 "A FOR SPEC. Σ"
06 AVIEW
07 SIN
08 SIN
09 "B FOR LOG Σ"
10 AVIEW
11 SIN
12 SIN
13 "C FOR LUM. Σ"
14 AVIEW
15 STOP
16♦LBL A
17 FIX 0
18 101.108
19 STO 00
20♦LBL 03
21 0
22 STO 08
23 STO 09
24 CLA
25 ARCL IND 00
26 AVIEW
27 RCL 00
28 1 E1
29 *
30 1000
31 -
32 INT
33 STO 99
34 5
35 +
36 1 E3
37 /
38 ST+ 99
39♦LBL 01
40 RCL IND 99
41 ST+ 09
42 ISG 99
43 GTO 01
44 RCL 99
45 .003
46 ST+ 99
47♦LBL 02
48 RCL IND 99
49 ST+ 08
50 ISG 99
```

```
51 GTO 02
52 " SCALE= "
53 ARCL 08
54 AVIEW
55 PSE
56 " TALLY= "
57 ARCL 09
58 AVIEW
59 ADV
60 PSE
61 ISG 00
62 GTO 03
63♦LBL a
64 FIX 2
65 ADV
66 "MARGINΣ="
67 ARCL 90
68 AVIEW
69 ADV
70 PSE
71 "SAW TIMEΣ="
72 FIX 1
73 ARCL 96
74 AVIEW
75 RTN
76♦LBL B
77 FIX 0
78 CF 07
79 "SCALE BY"
80 AVIEW
81 SIN
82 "LOG GRADE"
83 AVIEW
84 ADV
85 SIN
86 0
87 STO 08
88 17.08710
89♦LBL 06
90 STO 09
91 STO 00
92 XEQ 04
93 " GRADE 1= "
94 XEQ 08
95 XEQ 04
96 " GRADE 2= "
97 XEQ 08
98 XEQ 04
99 " GRADE 3= "
100 XEQ 08
```

MIL3 - continued (2)


```
101 RTN
102♦LBL 04
103 RCL IND 00
104 ST+ 08
105 ISG 00
106 GTO 04
107 RTN
108♦LBL 08
109 ARCL 08
110 AVIEW
111 0
112 STO 08
113 RCL 09
114 1.001
115 +
116 STO 00
117 STO 09
118 RTN
119♦LBL b
120 "SPECIES ?"
121 FS? 01
122 XEQ "F22"
123 FC? 01
124 PROMPT
125 X=0?
126 STOP
127 STO 00
128 STO 99
129 1 E1
130 *
131 FC? 07
132 7.009
133 FS? 07
134 1.006
135 +
136 RCL 99
137 1 E2
138 /
139 +
140 STO 99
141 100
142 ST+ 00
143 CLA
144 ARCL IND 00
145 "⊦:"
146 AVIEW
147 SIN
148 " GRADE 1= "
149 XEQ 05
150 " GRADE 2= "
151 XEQ 05
152 " GRADE 3= "
153 XEQ 05
154 FS? 07
155 GTO 07
156 ADV
157 GTO b
158♦LBL C
159 SF 07
160 "TALLY BY"
161 AVIEW
162 SIN
163 "LUMBER GRADE"
164 AVIEW
165 ADV
166 SIN
167 0
168 STO 08
169 11.08110
170 XEQ 06
171 XEQ 04
172 " GRADE 4= "
173 XEQ 08
174 XEQ 04
175 " GRADE 5= "
176 XEQ 08
177 XEQ 04
178 " GRADE 6= "
179 ARCL 08
180 AVIEW
181 RTN
182♦LBL c
183 XEQ b
184♦LBL 07
185 " GRADE 4= "
186 XEQ 05
187 " GRADE 5= "
188 XEQ 05
189 " GRADE 6= "
190 XEQ 05
191 ADV
192 GTO b
193♦LBL 05
194 ARCL IND 99
195 AVIEW
196 PSE
197 ISG 99
198 END

TOT:594 BYTES
```

Documentation--MIL3

```
 01♦LBL "MIL3"
CF 29  "TO LOAD, SF06"
PROMPT  FC?C 06  GTO 06
SF 05  1.007  STO 00
XEQ "LOAD"  91.093
STO 00  XEQ "LOAD"
101.108  STO 00  AON
XEQ "LOAD"
```

Automatic loading option

```
 18♦LBL 06
SF 03  SF 02
" STAND BY"  AVIEW
10.09  XEQ "MRC"  0
STO 94  STO 95  STO 96
```

Initialization and register clearing

```
 29♦LBL 05
0  STO 00

 32♦LBL 08
FIX 0  100  STO 99
"SPEC. CODE?"  FS? 01
XEQ "F22"  FC? 01
PROMPT  STO 98  ST+ 99
CLA  ARCL IND 99  AVIEW
PSE  FIX 1  "LOG SSS.G?"
FS? 01  XEQ "F22"
FC? 01  PROMPT  XEQ 03
RCL 98  1 E1  *  STO 98
RCL 99  +  6  +  STO 97
90  ST+ 99  RCL 09
ST+ 94  ST+ IND 97
RCL IND 99  *  ST- 00
```

Label 08 entry for error correction

Species, log scale and grade processor

```
 71♦LBL 01
"LUM SSS.G?"  FS? 01
XEQ "F22"  FC? 01
PROMPT  X=0?  GTO 02
XEQ 03  RCL 98  +
STO 97  RCL 09
ST+ IND 97  RCL IND 99
*  ST+ 00  GTO 01
```

Lumber tally and grade processor

```
 89♦LBL 02
FIX 2  "SAW. TIME?"
FS? 01  XEQ "F22"
FC? 01  PROMPT  ST+ 96
RCL 07  *  ST- 00
RCL 00  ST+ IND 98
ST+ 90  X<0?  BEEP
"MARGIN=$"  ARCL X
AVIEW  STOP  GTO 05
```

Sawing time processor and margin display

Documentation--MIL3 - continued

```
110♦LBL 03
STO 08  INT  STO 09
LASTX  FRC  1 E1  *  ABS
STO 99  RTN
```

Separator of scale.grade or tally.grade

```
121♦LBL "MRC"
0

123♦LBL 04
STO IND Y  ISG Y  GTO 04
RTN
```

Register clear routine

```
128♦LBL "F22"
CF 22  AVIEW

131♦LBL 10
PSE  FC?C 22  GTO 10
RTN
```

Prompt avoidance routine (flag 01 set)

```
136♦LBL "LOAD"
"START REG?"  FC? 05
PROMPT  .4  +  FC? 05
STO 00  FIX 0  CF 29

146♦LBL 09
"REG "  ARCL 00  "⊢= ?"
PROMPT  X<0?  GTO 06
FS? 48  ASTO IND 00
FC? 48  STO IND 00
ISG 00  GTO 09  AOFF
"DONE"  AVIEW  END
```

Loading subroutine for numeric or alphabetic constants

Documentation--MIL3 - continued (2)

```
 01♦LBL "TOT"
"PRESS:" AVIEW PSE
"A FOR SPEC. Σ" AVIEW
SIN SIN "B FOR LOG Σ"
AVIEW SIN SIN
"C FOR LUM. Σ" AVIEW
STOP
```

Prompt for various totals

```
 16♦LBL A
FIX 0 101.108 STO 00
```

Initialization

```
 20♦LBL 03
0 STO 08 STO 09 CLA
ARCL IND 00 AVIEW
RCL 00 1 E1 * 1000 -
INT STO 99 5 + 1 E3
/ ST+ 99
```

Species code manipulation and tally and scale counter location

```
 39♦LBL 01
RCL IND 99 ST+ 09
ISG 99 GTO 01 RCL 99
.003 ST+ 99
```

Accumulation of lumber tally

```
 47♦LBL 02
RCL IND 99 ST+ 08
ISG 99 GTO 02
" SCALE= " ARCL 08
AVIEW PSE " TALLY= "
ARCL 09 AVIEW ADV PSE
ISG 00 GTO 03
```

Accumulation of log scale and display of scale and tally

```
 63♦LBL a
FIX 2 ADV "MARGINΣ="
ARCL 90 AVIEW ADV PSE
"SAW TIMEΣ=" FIX 1
ARCL 96 AVIEW RTN
```

Margin and saw time display

```
 76♦LBL B
FIX 0 CF 07 "SCALE BY"
AVIEW SIN "LOG GRADE"
AVIEW ADV SIN 0
STO 08 17.08710
```

Initialization -- scale by log grade

```
 89♦LBL 06
STO 09 STO 00 XEQ 04
" GRADE 1= " XEQ 08
XEQ 04 " GRADE 2= "
XEQ 08 XEQ 04
" GRADE 3= " XEQ 08
RTN
```

Log grade display

Documentation--MIL3 - continued (3)

```
102♦LBL 04
RCL IND 00  ST+ 08
ISG 00  GTO 04  RTN
```

Log grade scale and lumber tally accumulator

```
108♦LBL 08
ARCL 08  AVIEW  0
STO 08  RCL 09  1.001  +
STO 00  STO 09  RTN
```

Incrementer for lumber grade and log grade pointer

```
119♦LBL b
"SPECIES ?"  FS? 01
XEQ "F22"  FC? 01
PROMPT  X=0?  STOP
STO 00  STO 99  1 E1  *
FC? 07  7.009  FS? 07
1.006  +  RCL 99  1 E2
/  +  STO 99  100
ST+ 00  CLA  ARCL IND 00
"⊢:"  AVIEW  SIN
" GRADE 1= "  XEQ 05
" GRADE 2= "  XEQ 05
" GRADE 3= "  XEQ 05
FS? 07  GTO 07  ADV
GTO b
```

Species prompt and log scale by grade display

```
158♦LBL C
SF 07  "TALLY BY"  AVIEW
SIN  "LUMBER GRADE"
AVIEW  ADV  SIN  0
STO 08  11.08110  XEQ 06
XEQ 04  " GRADE 4= "
XEQ 08  XEQ 04
" GRADE 5= "  XEQ 08
XEQ 04  " GRADE 6= "
ARCL 08  AVIEW  RTN
```

Initialization-- tally by lumber grade

```
182♦LBL c
XEQ b
```

Lumber tally by grade display

```
184♦LBL 07
" GRADE 4= "  XEQ 05
" GRADE 5= "  XEQ 05
" GRADE 6= "  XEQ 05
ADV  GTO b
```

```
193♦LBL 05
ARCL IND 99  AVIEW  PSE
ISG 99  END
```

Accumulator for scale or tally within species

Error Example:

```
                XEQ "MIL3"
TO LOAD, SF06
                       RUN
 STAND BY
SPEC. CODE?
             2         RUN
WH.OAK
LOG SSS.G?
                    GTO 08
                       RUN
SPEC. CODE?
             1         RUN
WALNUT
LOG SSS.G?
         255.1         RUN
LUM SSS.G?
                    GTO 08
                       RUN
SPEC. CODE?
             1         RUN
WALNUT
LOG SSS.G?
                    RCL 08
                       CHS
                       RUN
LUM SSS.G?
                    GTO 08
                       RUN
SPEC. CODE?
             1         RUN
WALNUT
LOG SSS.G?
         555.1         RUN
LUM SSS.G?
         156.1         RUN
LUM SSS.G?
                    RCL 08
                       CHS
                       RUN
LUM SSS.G?
         150.1         RUN
LUM SSS.G?
          40.2         RUN
LUM SSS.G?
          80.3         RUN
LUM SSS.G?
          60.4         RUN
LUM SSS.G?
         100.5         RUN
LUM SSS.G?
         150.6         RUN
LUM SSS.G?
           0.0         RUN
SAW. TIME?
          5.00         RUN
MARGIN=$18.50
```

wrong species

wrong scale

wrong tally

```
                       RUN
SPEC. CODE?
             2         RUN
WH.OAK
LOG SSS.G?
         625.2         RUN
LUM SSS.G?
         500.1         RUN
LUM SSS.G?
                    RCL 08
                       CHS
                       RUN
LUM SSS.G?
         300.1         RUN
LUM SSS.G?
         200.3         RUN
LUM SSS.G?
                    RCL 08
                       CHS
                       RUN
LUM SSS.G?
         200.2         RUN
LUM SSS.G?
         150.4         RUN
LUM SSS.G?
           0.0         RUN
SAW. TIME?
          6.00         RUN
MARGIN=$188.00
```

wrong tally

wrong Grade

```
                 XEQ "TOT"
PRESS:
A FOR SPEC. Σ
B FOR LOG Σ
C FOR LUM. Σ
                     XEQ A
WALNUT
 SCALE= 555
 TALLY= 580

WH.OAK
 SCALE= 625
 TALLY= 650

RE.OAK
 SCALE= 0
 TALLY= 0

                      STOP
                     XEQ a

MARGINΣ=206.50

SAW TIMEΣ=11.0
```

```
                     XEQ B
SCALE BY
LOG GRADE

 GRADE 1= 555
 GRADE 2= 625
 GRADE 3= 0
                     XEQ b
SPECIES ?
              1        RUN
WALNUT:
 GRADE 1= 555
 GRADE 2= 0
 GRADE 3= 0

SPECIES ?
              2        RUN
WH.OAK:
 GRADE 1= 0
 GRADE 2= 625
 GRADE 3= 0

SPECIES ?
              0        RUN
                     XEQ C
TALLY BY
LUMBER GRADE

 GRADE 1= 450
 GRADE 2= 240
 GRADE 3= 80
 GRADE 4= 210
 GRADE 5= 100
 GRADE 6= 150
                     XEQ c
SPECIES ?
              1        RUN
WALNUT:
 GRADE 1= 150
 GRADE 2= 40
 GRADE 3= 80
 GRADE 4= 60
 GRADE 5= 100
 GRADE 6= 150

SPECIES ?
              2        RUN
WH.OAK:
 GRADE 1= 300
 GRADE 2= 200
 GRADE 3= 0
 GRADE 4= 150
 GRADE 5= 0
 GRADE 6= 0

SPECIES ?
              0        RUN
```

Retrospective Comments Regarding MIL3

1. The write-up is somewhat difficult to follow, which justifies why efforts continued to develop a "better" format of program description. Subsequent programs in this chronology incorporate more understandable (hopefully!) directions.

2. When used with a printer attached and ON, reasonably spaced printed output is obtained, especially in NORM printer mode.

3. The fact that the output program (TOT) is a separate program may be inconvenient, and one may desire to incorporate TOT into MIL3 and then deal with one large program. Unfortunately, when these programs were written, the same numbers were used for the numeric local labels (01 through 06, and 08 occur in both programs), therefore to incorporate TOT one must carefully and completely change one set of labels. This effort is probably not justified. However, on the KRON-1 tape the two programs (MIL3 and TOT) are available together under the file name MIL3FL to allow simple down-loading to the calculator by a read-all (READA) operation.

Program No. 41F007

Program Name: SINDEX

Calculator: HP-41C/CV

Author: T.W. Beers, Dept. Forestry. Purdue U.

Date: July 1980

Purpose: To replace tables and graphs in NC-54 publication. Provides forward or inverse solution of $H = b_1 S(1-e^{b_2 A})^{b_3}$ for nine species-specific sets of b_i's (stored in memory).

A. Storage assignment

Register	Use
00	control for coefficient loading, retrieval and indexing
01-09	b_1 coefficients for species 1 through 9
10	age
11-19	b_2 coefficients for species 1 through 9
20	site index or height
21-29	b_3 coefficients for species 1 through 9
30	intermediate calculations: $b_1(1-e^{b_2 A})^{b_3}$
31-39	species names
40	species read out control

B. Labels

Name	Use
SINDEX	XEQ to facilitate loading of coefficients and species names ("constants")
SI	XEQ after all constants are stored
LOAD	Prompting for and storing the constants
01	in loading
02	intermediate calculations
03	site index calculations
04	control number storage
05	height calculations and automatic display
09	internal to LOAD

C. Flags

Number	Use
00	set internally by negative species code to enable reverse solution of formula; i.e., to find S, given H and A.
29	cleared in load routine to eliminate decimal in FIX 0; also affects site index output.
48	set in ALPHA mode; used to load species names, tested in LBL 09.

D. Program procedure

1. XEQ SIZE 041, load program SINDEX

2. To load constants initially, XEQ SINDEX, and proceed as prompted:

Prompt	Input	Key	Output	Example Input	Example Key	Example Output
B1 COEFF:						
REG 1=?	b_1(species 1)	R/S		1.890	R/S	
.	.	.		.	.	
.	.	.		.	.	
.	.	.		.	.	
REG 9=?	b_1(species 9)	R/S	DONE	1.598	R/S	DONE
B2 COEFF:						
REG 11=?	b_2(species 1)	R/S		-.01979	R/S	
.	.	.		.	.	
.	.	.		.	.	
.	.	.		.	.	
REG 19=?	b_2(species 9)	R/S	DONE	-.01938	R/S	DONE
B3 COEFF:						
REG 21=?	b_3(species 1)	R/S		1.3892	R/S	
.	.	.		.	.	
.	.	.		.	.	
.	.	.		.	.	
REG 29=?	b_3(species 9)	R/S	DONE	.9824	R/S	DONE
SPEC. NAMES:	(alpha mode is automatic)					
REG 31=?	species 1	R/S		R.PINE	R/S	
.	.	.		.	.	
.	.	.		.	.	
.	.	.		.	.	
REG 39=?	species 9	R/S		PBIRCH	R/S	DONE

3. XEQ SI (suggest assignment to SIN key)

D. Program procedure *(Continued)*

4. To solve for site index go to step 5; to solve for a series of heights, given age and site index:

Prompt	Input	Key	Output	Input (*Example*)	Key (*Example*)	Output (*Example*)
SPEC. CODE?	1 to 9	R/S	species name	1	R/S	R.PINE
AGE?	age	R/S		70	R/S	
FIRST S.I.?	initial S.I.	R/S	height in meters	10	R/S	HT.= 12.7 M.
			•			HT.= 13.9 M.
			•			HT.= 15.2 M.
			•			HT.= 16.5 M.
						HT.= 17.7 M.
						HT.= 19.0 M.
			↓			HT.= 20.3 M.
			height in meters		R/S	HT.= 21.5 M.

Note: the displayed heights correspond to the initial site index keyed-in and successive site index classes using an increment of 1. The display can be stopped at any point by another R/S depression, and restarted at that point by another R/S depression.

For a new age or initial site index, go to step 3.

5. To solve the equation for site index, species code is keyed in as a negative value. Therefore, XEQ SI and proceed as prompted:

Prompt	Input	Key	Output	Input (*Example*)	Key (*Example*)	Output (*Example*)
SPEC. CODE?	1 to 9, CHS	R/S	species name	1, CHS	R/S	R.Pine
AGE?	age	R/S		70	R/S	
HEIGHT?	height	R/S	site index	17.7	R/S	S.INDEX= 14

For a new age and height, depress R/S and prompt is AGE?

For a different species, XEQ SI and follow prompts using negative species code.

E. Program

```
01◆LBL "SINDEX"
02 1.009
03 STO 00
04 "B1 COEFF:"
05 XEQ 01
06 11.019
07 STO 00
08 "B2 COEFF:"
09 XEQ 01
10 21.029
11 STO 00
12 "B3 COEFF:"
13 XEQ 01
14 31.039
15 STO 00
16 "SPEC. NAMES:"
17 AVIEW
18 SIN
19 AON
20 XEQ "LOAD"
21 RTN
22◆LBL "SI"
23 CF 00
24 "SPEC. CODE?"
25 PROMPT
26 X<0?
27 SF 00
28 ABS
29 STO 00
30 STO 40
31 30
32 ST+ 40
33 CLA
34 ARCL IND 40
35 AVIEW
36 PSE
37 XEQ 04
38 FS? 00
39 GTO 03
40 "AGE ?"
41 PROMPT
42 STO 10
43 "FIRST S.I. ?"
44 PROMPT
45 STO 20
46 XEQ 02
47◆LBL 05
48 RCL 20
49 *
50 FIX 1
51 " HT. = "
52 ARCL X
53 "⊢ M."
54 AVIEW
55 PSE
56 PSE
57 1
58 ST+ 20
59 ISG 00
60 XEQ 02
61 GTO 05
62◆LBL 01
63 AVIEW
64 SIN
65 XEQ "LOAD"
66 RTN
67◆LBL "LOAD"
68 FIX 0
69 CF 29
70◆LBL 09
71 "REG "
72 ARCL 00
73 "⊢= ?"
74 PROMPT
75 FS? 48
76 ASTO IND 00
77 FC? 48
78 STO IND 00
79 ISG 00
80 GTO 09
81 AOFF
82 "DONE"
83 AVIEW
84 RTN
85◆LBL 03
86 "AGE ?"
87 PROMPT
88 STO 10
89 "HEIGHT ?"
90 PROMPT
91 STO 20
92 XEQ 02
93 1/X
94 RCL 20
95 *
96 FIX 0
97 "S. INDEX= "
98 ARCL X
99 AVIEW
100 STOP
101 ISG 00
102 FS? 00
103 GTO 03
104 STOP
105◆LBL 02
106 1
107 RCL 10
108 RCL IND 00
109 *
110 E↑X
111 -
112 ISG 00
113 RCL IND 00
114 Y↑X
115 20
116 ST- 00
117 RDN
118 RCL IND 00
119 *
120 STO 30
121 RTN
122◆LBL 04
123 RCL 00
124 10.99910
125 +
126 STO 00
127 END

SINDEX:351 BYTES
```

F. Sample printout and stored constants

```
              XEQ "SI"
SPEC. CODE?
              1     RUN
R.PINE
AGE ?
              70    RUN
FIRST S.I. ?
              10    RUN
 HT. = 12.7 M.
 HT. = 13.9 M.
 HT. = 15.2 M.
 HT. = 16.5 M.
 HT. = 17.7 M.
 HT. = 19.0 M.
 HT. = 20.3 M.
 HT. = 21.5 M.
                  STOP
                   RUN
 HT. = 22.8 M.
 HT. = 24.1 M.
                  STOP


              XEQ "SI"
SPEC. CODE?
            -1.0    RUN
R.PINE
AGE ?
            70.0    RUN
HEIGHT ?
            17.7    RUN
S. INDEX= 14
                    RUN
AGE ?
              70    RUN
HEIGHT ?
            21.5    RUN
S. INDEX= 17
```

```
STORED
 CONSTANTS:

R01=   1.890
R02=   1.633
R03=   1.966
R04=   1.437
R05=   1.762
R06=   1.547
R07=   1.973
R08=   1.480
R09=   1.598

R11= -0.01979
R12= -0.02233
R13= -0.02399
R14= -0.02266
R15= -0.02011
R16= -0.02246
R17= -0.01535
R18= -0.02140
R19= -0.01938

R21=   1.3892
R22=   1.2419
R23=   1.8942
R24=   0.9381
R25=   1.2307
R26=   1.1129
R27=   1.0895
R28=   0.9377
R29=   0.9824

R31= "R.PINE
"
R32= "J.PINE
"
R33= "W.PINE
"
R34= "  FIR
"
R35= "SPRUCE
"
R36= "LARCH
"
R37= "CEDAR
"
R38= "ASPEN
"
R39= "PBIRCH
"
```

G. Documentation

```
 01♦LBL "SINDEX"
1.009  STO 00
"B1 COEFF:"  XEQ 01
11.019  STO 00
"B2 COEFF:"  XEQ 01
21.029  STO 00
"B3 COEFF:"  XEQ 01
31.039  STO 00
"SPEC. NAMES:"  AVIEW
SIN  AON  XEQ "LOAD"
RTN
```

Store constants

```
 22♦LBL "SI"
CF 00  "SPEC. CODE?"
PROMPT  X<0?  SF 00  ABS
STO 00  STO 40  30
ST+ 40  CLA  ARCL IND 40
AVIEW  PSE  XEQ 04
FS? 00  GTO 03  "AGE ?"
PROMPT  STO 10
"FIRST S.I. ?"  PROMPT
STO 20  XEQ 02

 47♦LBL 05
RCL 20  *  FIX 1
" HT. = "  ARCL X
"├ M."  AVIEW  PSE  PSE
1  ST+ 20  ISG 00
XEQ 02  GTO 05
```

Initialize and display species

Calc. of height.

```
 62♦LBL 01
AVIEW  SIN  XEQ "LOAD"
RTN
```

Constants prompt and store.

```
 67♦LBL "LOAD"
FIX 0  CF 29

 70♦LBL 09
"REG "  ARCL 00  "├= ?"
PROMPT  FS? 48
ASTO IND 00  FC? 48
STO IND 00  ISG 00
GTO 09  AOFF  "DONE"
AVIEW  RTN

 85♦LBL 03
"AGE ?"  PROMPT  STO 10
"HEIGHT ?"  PROMPT
STO 20  XEQ 02  1/X
RCL 20  *  FIX 0
"S. INDEX= "  ARCL X
AVIEW  STOP  ISG 00
FS? 00  GTO 03  STOP
```

calculation of site index

```
105♦LBL 02
1  RCL 10  RCL IND 00  *
E↑X  -  ISG 00
RCL IND 00  Y↑X  20
ST- 00  RDN  RCL IND 00
*  STO 30  RTN
```

intermediate calc. for R_{30}

```
122♦LBL 04
RCL 00  10.99910  +
STO 00  END
```

coefficient recall indexer

USDA FOREST SERVICE
GENERAL TECHNICAL REPORT **NC-54**

Metric site index curves for aspen, birch and conifer in the Lake States

Paul R. Laidly

Table 1.—*Parameters of the equation describing metric site index curves*[1] (derived from Lundgren and Dolid 1970)

Species	Parameters b_1	b_2	b_3	Standard error	Maximum error
				-------meters------	
Red pine	1.890	−0.01979	1.3892	.19	.43
Jack pine	1.633	−0.02233	1.2419	.15	.34
White pine	1.966	−0.02399	1.8942	.20	.52
Balsam fir	1.437	−0.02266	0.9381	.21	.58
Black spruce	1.762	−0.02011	1.2307	.22	.58
Tamarack	1.547	−0.02246	1.1129	.16	.43
Northern white-cedar	1.973	−0.01535	1.0895	.20	.52
Aspen	1.480	−0.02140	0.9377	.12	.34
Paper birch	1.598	−0.01938	0.9824	.10	.18

[1]Height = b_1 (Site Index (meters)) $(1-e^{b_2 \text{(Age)}})^{b_3}$.

Table 2.— *Red pine height as related to site index and age*
(In meters)

Site Index	Age (years) 20	30	40	50	60	70	80	90	100	110	120
10	4.0	6.2	8.2	9.9	11.4	12.7	13.7	14.6	15.4	16.0	16.5
11	4.4	6.8	9.0	10.9	12.5	13.9	15.1	16.1	16.9	17.6	18.2
12	4.8	7.4	9.8	11.9	13.7	15.2	16.5	17.6	18.4	19.2	19.8
13	5.2	8.0	10.6	12.9	14.8	16.5	17.9	19.0	20.0	20.8	21.5
14	5.6	8.7	11.4	13.9	16.0	17.7	19.2	20.5	21.5	22.4	23.1
15	6.0	9.3	12.3	14.9	17.1	19.0	20.6	21.9	23.1	24.0	24.8
16	6.4	9.9	13.1	15.9	18.2	20.3	22.0	23.4	24.6	25.6	26.4
17	6.8	10.5	13.9	16.8	19.4	21.5	23.3	24.9	26.1	27.2	28.1
18	7.2	11.1	14.7	17.8	20.5	22.8	24.7	26.3	27.7	28.8	29.7
19	7.6	11.8	15.5	18.8	21.7	24.1	26.1	27.8	29.2	30.4	31.4
20	8.0	12.4	16.3	19.8	22.8	25.3	27.5	29.3	30.7	32.0	33.0
21	8.4	13.0	17.2	20.8	23.9	26.6	28.8	30.7	32.3	33.6	34.7
22	8.8	13.6	18.0	21.8	25.1	27.9	30.2	32.2	33.8	35.2	36.3
23	9.2	14.2	18.8	22.8	26.2	29.1	31.6	33.6	35.4	36.8	38.0
24	9.6	14.9	19.6	23.8	27.4	30.4	33.0	35.1	36.9	38.4	39.6
25	10.0	15.5	20.4	24.8	28.5	31.7	34.3	36.6	38.4	40.0	41.3

Retrospective Comments Regarding SINDEX

1. The write-up for SINDEX approaches that which is used in most of the remaining programs.

2. The user who chooses to use the program as is, with the coefficients and species names as cited, can avoid the manual loading of these data (if the KRON-1 tape is available) by the following:

   a. insert after step 01 (LBL "SINDEX"):
      FS? 01
      GTO SI

   b. down-load the data from the KRON-1 tape by:

      (1) place SIDATA in alpha register
      (2) CLX, then XEQ SEEKR
      (3) key 1.039, then XEQ READRX

   c. set flag 01, then XEQ SINDEX; or, if flag 01 is not set, simply XEQ SI

3. The program SINDEX which is on the KRON-1 tape incorporates the changes indicated in 2a.

4. Printed output is properly spaced and identified only when printer is in NORM mode.

Program No. 41F008

Program Name: LOGVOL

Calculator: HP-41C with two memory modules or HP-41CV

Author: T. W. Beers, Dept. of Forestry, Purdue University

Date: August, 1980

Purpose: To calculate log volumes by Doyle, Scribner, Int. ¼-inch, Int. 1/8-inch rules and cubic content by Huber, Smalian, and Newton formulas - United States or Metric units.

A. Storage assignment

Register	Use
00	alpha storage of "BF, "CF", or "M"
01	log length
02	d_u for board foot calculations
03	intermediate use in Int. algorithm $(d_u + \frac{n}{2})$, etc.
04	n = integer part of n.f
05	not used
06	d_b
07	d_m in cubic foot and cubic meters calculations
08	d_u
09	L
10	not used
11	Doyle volume in board feet
12	Scribner volume in board feet
13	Int. 1/4 volume in board feet
14	Int. 1/8 volume in board feet
15	Huber volume in cubic feet or cubic meters
16	Smalian volume in cubic feet or cubic meters
17	Newton volume in cubic feet or cubic meters

B. Labels

Name	Use
LOGVOL	XEQ to prompt for BF, CF, or M
BF	prompts for d_u, L input and for A to D, and E depression
A	calculates and displays Doyle volume
B	calculates and displays Scribner volume
C	calculates and displays Int. 1/4 volume
D	calculates and displays Int. 1/8 volume
E	calculates and displays all four board-foot volumes
ABC	displays previously calculated Doyle, Scribner, Int. 1/4 volumes
AB	displays previously calculated Doyle, Scribner volumes
AC	displays previously calculated Doyle, Int. 1/4 volumes
BC	displays previously calculated Scribner, Int. 1/4 volumes
CF	prompts for a, b, c, or e depression
a	prompts for d_m, L and calculates and displays Huber volume

Name	Use
b	prompts for d_b, d_u, L and calculates and displays Smalian volume
c	prompts for d_b, d_m, d_u, L and calculates and displays Newton volume
e	prompts for d_b, d_m, d_u, L and calculates and displays all three volumes
M	sets flag 01 and directs control to CF assuming metric volume calculations
SP	"stop-pause" routine (pause with flag 01 set)
F23	R/S avoidance routine (if flag 01 set); alphabetic input
00	in C to convert Int. 1/8 to Int. 1/4
01	in A for Doyle display
02	in B for Scribner display
03	in C for Int. 1/4 display
04	in D for Int. 1/8 display
05	in ABC for display reversal
06,07	in C for Int. algorithm solution
08	in AB for display reversal
09	in AC for display reversal
10	in CF for cubic feet to cubic meter conversion
11	in F23 for internal control
13	in e for internal looping
15	in BF for input data storage
16	in CF for input data storage
17	in CF for branch link from M
18	in CF for cubic meter to cubic feet conversion
20	in e for internal looping
21	in a for internal looping
22	in b for internal looping
23	in c for internal looping
24	in a for Huber calculation and display
25	in b for Smalian calculation and display
26	in c for Newton calculation and display

C. Flags

Number	Use
00	set externally to reverse display in ABC, AB, AC, and BC
01	set externally to achieve "automatic" display or input; i.e., to avoid pressing R/S.
04	set internally in M to use CF for metric input, calculation and display
06,07	internal use in C and D for Int. volume calculation
22	numeric data input flag; used in CF to indicate input for new log has been keyed in.
23	alphabetic data input flag; used in F23 to indicate alpha data (BF, CF, or M) has been keyed in.

D. Program procedure

1. XEQ SIZE 018, load program LOGVOL

2. Select number of decimals, suggest 1 for board feet, 2 for cubic measure; therefore f* FIX 1 initially.

   * f is used throughout to indicate the shift (gold) key.

3. Set appropriate flags for desired input/output format:
   flag 01 set for "automatic" input and display (where appropriate);
   flag 00 set to reverse display within certain groups of board-foot calculations.

The following directions will assume F 01 is set and initially F 00 is clear; therefore f SF 01, f CF 00.

4. For board-foot volume calculations,

   XEQ LOGVOL (suggest assignment to TAN key), then depress BF(1/x, x≷y) at the initial prompt and proceed as prompted:

	Prompt	Input	Key	Output	Example Input	Example Key	Example Output
a.	BF CF or M?	BF			BF		
b.	KEY DU ↑ L	d_u	ENTER↑	d_u	14	ENTER↑	14.0
		L	R/S		10	R/S	Key A TO D or E
c.	KEY A TO D OR E	none	A	DOYLE= xx.x		A	DOYLE= 62.5
d.			B	SCRIB.= xx.x		B	SCRIB.= 76.8
e.			C	INT.1/4= xx.x		C	INT.1/4= 80.2
f.			D	INT.1/8= xx.x		D	INT.1/8= 88.6
g.			E	DOYLE= xx.x(pause)		E	DOYLE= 62.5(pause)
				SCRIB.= xx.x(pause)			SCRIB.= 76.8 (pause)
				INT.1/4= xx.x(pause)			INT.1/4= 80.2 (pause)
				INT.1/8= xx.x			INT.1/8= 88.6

h. After A, B, C, and D or after E, results can be recalled in groups as follows:

XEQ ABC (assign to x≷y) to display Doyle, Scribner, Int. 1/4
XEQ AB (assign to R↓) to display Doyle, Scribner
XEQ AC (assign to SIN) to display Doyle, Int. 1/4
XEQ BC (assign to COS) to display Scribner, Int. 1/4

By setting flag 00, one can reverse the order of display, except for ABC which wll then display Int. 1/4, Doyle, Scribner

i. Note that to calculate volumes for several logs using same log rule, at the 4c. prompt (KEY A TO D OR E) choose the desired A to D key and proceed as follows; for example, to calculate Doyle volume (key A) for a series of logs: (14", 10'; 10", 4'; 14", 16').

Prompt	Input	Key	Output	Example Input	Example Key	Example Output
		XEQ BF			XEQ BF	
KEY DU↑L	d_u	ENTER↑	d_u	14	ENTER↑	14.0
	L	R/S		10	R/S	KEY A TO D OR E
KEY A TO DO OR E	none	A	DOYLE= xx.x		A	DOYLE= 62.5
	d_u	ENTER↑	d_u	10	ENTER↑	10.0
	L	R/S	DOYLE= xx.x	4	R/S	DOYLE= 9.0
	d_u	ENTER↑	d_u	14	ENTER↑	14.0
	L	R/S	DOYLE= xx.x	16	R/S	DOYLE= 100.0
			etc.			etc.

5. For cubic-foot input and volume calculations, XEQ LOGVOL (TAN key), then depress CF(√x, x≷y) at the initial prompt, and proceed as prompted: (assume f FIX 2)

	Prompt	Input	Key	Output	Example Input	Example Key	Example Output
a.	BF CF OR M?	CF			CF		
b.	KEY a,b,c, OR e	none	a(i.e.fA)	HUBER VOLUME		a	HUBER VOLUME
	KEY DM↑L	d_m	ENTER↑	d_m	10	ENTER↑	10.00
		L	R/S	HUB.= x.xxCF	14	R/S	HUB.= 7.64CF
		(optional)	R/S	= x.xxM↑3	(opt.)	R/S	= 0.22 M↑3
c.			b	SMALIAN VOLUME		b	SMALIAN VOLUME
	KEY DB↑DU↑L	d_u	ENTER↑	d_b	12.4	ENTER↑	12.40

					Example		
	Prompt	Input	Key	Output	Input	Key	Output
		d_u	ENTER↑	d_u	9.1	ENTER↑	9.10
		L	R/S	SMAL.= x.xxCF	14	R/S	SMAL.= 9.03CF
d.			c	NEWTON VOLUME		c	NEWTON VOLUME
	DB↑DM↑DU↑,L	d_b	ENTER↑	d_b	12.4	ENTER↑	12.40
		d_m	ENTER↑	d_m	10	ENTER↑	10.00
		d_u	ENTER↑	d_u	9.1	ENTER↑	9.10
		L	R/S	NEWT.= x.xxCF	14	R/S	NEWT.= 8.10CF
e.			e	none		e	
	DB↑DM↑DU↑L	d_b	ENTER↑	d_b	12.4	ENTER↑	12.40
		d_m	ENTER↑	d_m	10	ENTER↑	10.00
		d_u	ENTER↑	d_y	9.1	ENTER↑	9.10
		L	R/S	HUB.= x.xxCF	14	R/S	HUB.= 7.64CF (pause)
				SMAL.= x.xxCF			SMAL.= 9.03CF (pause)
				NEWT.= x.xxCF			NEWT.= 8.10CF

f. Note the optional feature of metric conversion shown in step 5b. This is available by R/S depression immediately after the cubic-foot display in labels a, b, or c.

g. To calculate volumes for several logs using the same formula, after the answer is displayed for the first log (before or after metric conversion) simply key in the next log dimensions and depress R/S.

6. For metric input and volume calculations, XEQ LOGVOL (TAN key), then depress M (RCL) at the initial prompt and proceed as prompted. Refer to step 5 instructions and keep in mind that:

   a. diameters are assumed to be in centimeters
   b. lengths are assumed to be in meters
   c. answers will be labeled in m^3 (M↑3)
   d. the optional R/S conversion feature will lead to answers in cu. ft. (CF)

E. Sample Printer Output


```
                    SF 01
                    FIX 1
             XEQ "LOGVOL"
BF CF OR M?
BF
KEY DU ↑ L
             14.0 ENTER↑
             10.0    RUN
KEY A TO D OR E
                    XEQ A
DOYLE= 62.5
                    XEQ B
SCRIB.= 76.8
                    XEQ C
INT. 1/4= 80.2
                    XEQ D
INT. 1/8= 88.6
                    XEQ E
DOYLE= 62.5
SCRIB.= 76.8
INT. 1/4= 80.2
INT. 1/8= 88.6
              XEQ "ABC"
DOYLE= 62.5
SCRIB.= 76.8
INT. 1/4= 80.2
               XEQ "AB"
DOYLE= 62.5
SCRIB.= 76.8
               XEQ "AC"
DOYLE= 62.5
INT. 1/4= 80.2
               XEQ "BC"
SCRIB.= 76.8
INT. 1/4= 80.2
                    SF 00
              XEQ "ABC"
INT. 1/4= 80.2
DOYLE= 62.5
SCRIB.= 76.8
               XEQ "AB"
SCRIB.= 76.8
DOYLE= 62.5
               XEQ "AC"
INT. 1/4= 80.2
DOYLE= 62.5
               XEQ "BC"
INT. 1/4= 80.2
SCRIB.= 76.8
```

```
                    CF 00
                    SF 01
                    FIX 2
             XEQ "LOGVOL"
BF CF OR M?
CF
KEY a,b,c, OR e
                    XEQ a
HUBER VOLUME
KEY DM ↑ L
            10.00 ENTER↑
            14.00    RUN
HUB.= 7.64 CF
                      RUN
   = 0.22 M↑3
                    XEQ b
SMALIAN VOLUME
KEY DB↑ DU↑ L
            12.40 ENTER↑
             9.10 ENTER↑
            14.00    RUN
SMAL.= 9.03 CF
                      RUN
   = 0.26 M↑3
                    XEQ c
NEWTON VOLUME
DB↑ DM↑ DU↑,L
            12.40 ENTER↑
            10.00 ENTER↑
             9.10 ENTER↑
            14.00    RUN
NEWT.= 8.10 CF
                      RUN
   = 0.23 M↑3
                    XEQ e
DB↑ DM↑ DU↑,L
            12.40 ENTER↑
            10.00 ENTER↑
             9.10 ENTER↑
            14.00    RUN
HUB.= 7.64 CF
SMAL.= 9.03 CF
NEWT.= 8.10 CF
```

```
                    SF 01
                    FIX 2
             XEQ "LOGVOL"
BF CF OR M?
M
KEY a,b,c, OR e
                    XEQ a
HUBER VOLUME
KEY DM ↑ L
            25.40 ENTER↑
             4.27    RUN
HUB.= 0.22 M↑3
                      RUN
   = 7.64 CF
                    XEQ b
SMALIAN VOLUME
KEY DB↑ DU↑ L
            31.50 ENTER↑
            23.10 ENTER↑
             4.27    RUN
SMAL.= 0.26 M↑3
                      RUN
   = 9.04 CF
                    XEQ c
NEWTON VOLUME
DB↑ DM↑ DU↑,L
            31.50 ENTER↑
            25.40 ENTER↑
            23.10 ENTER↑
             4.27    RUN
NEWT.= 0.23 M↑3
                      RUN
   = 8.11 CF
                    XEQ e
DB↑ DM↑ DU↑,L
            31.50 ENTER↑
            25.40 ENTER↑
            23.10 ENTER↑
             4.27    RUN
HUB.= 0.22 M↑3
SMAL.= 0.26 M↑3
NEWT.= 0.23 M↑3
```

F. Program Listing


```
01♦LBL "LOGVOL"
02 AON
03 "BF CF OR M?"
04 FS? 01
05 XEQ "F23"
06 FC? 01
07 PROMPT
08 AOFF
09 ASTO 00
10 GTO IND 00
11♦LBL "BF"
12 "KEY DU ↑ L"
13 PROMPT
14 STO 01
15 X<>Y
16 STO 02
17 "KEY A TO D OR E"
18 PROMPT
19♦LBL A
20 RCL 02
21 4
22 -
23 X↑2
24 RCL 01
25 16
26 /
27 *
28 STO 11
29♦LBL 01
30 "DOYLE= "
31 ARCL 11
32 XEQ "SP"
33 RTN
34 XEQ 15
35 GTO A
36♦LBL B
37 RCL 02
38 X↑2
39 .79
40 *
41 RCL 02
42 2
43 *
44 -
45 4
46 -
47 RCL 01
48 16
49 /
50 *
51 STO 12
52♦LBL 02
53 "SCRIB.= "
54 ARCL 12
55 XEQ "SP"
56 RTN
57 XEQ 15
58 GTO B
59♦LBL C
60 SF 06
61 RCL 01
62 4
63 /
64 INT
65 STO 04
66 LASTX
67 FRC
68 XEQ 07
69 CF 06
70 *
71 STO 14
72 RCL 04
73 X=0?
74 GTO 00
75♦LBL 06
76 1
77 ST- 04
78♦LBL 07
79 RCL 04
80 2
81 /
82 RCL 02
83 +
84 STO 03
85 X↑2
86 .22
87 *
88 RCL 03
89 .71
90 *
91 -
92 FS? 06
93 RTN
94 ST+ 14
95 RCL 04
96 X>0?
97 GTO 06
98♦LBL 00
99 FS?C 07
100 RTN
101 RCL 14
102 .905
103 *
104 STO 13
105♦LBL 03
106 "INT. 1/4= "
107 ARCL 13
108 XEQ "SP"
109 RTN
110 XEQ 15
111 GTO C
112♦LBL D
113 SF 07
114 XEQ C
115♦LBL 04
116 "INT. 1/8= "
117 ARCL 14
118 XEQ "SP"
119 RTN
120 XEQ 15
121 GTO D
122♦LBL E
123 XEQ A
124 XEQ B
125 XEQ C
126 XEQ D
127 RTN
128♦LBL "ABC"
129 FS? 00
130 XEQ 05
131 XEQ 01
132 PSE
133 XEQ 02
134 FS? 00
135 RTN
136 PSE
137♦LBL 05
138 XEQ 03
139 PSE
140 RTN
141♦LBL "AB"
142 FS? 00
143 XEQ 08
144 XEQ 01
145 FS? 00
146 RTN
147 PSE
148♦LBL 08
149 XEQ 02
150 PSE
```

F. Program Listing (Continued 1)


```
RTN
◆LBL "AC"
FS? 00
XEQ 09
XEQ 01
FS? 00
RTN
PSE
◆LBL 09
XEQ 03
PSE
RTN
◆LBL "BC"
FS? 00
XEQ 09
XEQ 02
FS? 00
RTN
PSE
XEQ 09
RTN
◆LBL "SP"
AVIEW
FS? 01
PSE
FC? 01
STOP
RTN
◆LBL "F23"
CF 23
AVIEW
AON
◆LBL 11
PSE
FC?C 23
GTO 11
AOFF
RTN
◆LBL 15
STO 01
X<>Y
STO 02
RTN
◆LBL "CF"
CF 04
◆LBL 17
"KEY a,b,c, OR e"
PROMPT
◆LBL a
"HUBER VOLUME"
AVIEW
SIN
SIN
"KEY DM ↑ L"
PROMPT
◆LBL 21
STO 09
X<>Y
STO 07
◆LBL 24
RCL 09
RCL 07
X↑2
*
PI
*
FS? 04
40000
FC? 04
576
/
STO 15
"HUB.= "
ARCL 15
FS? 04
"├ M↑3"
FC? 04
"├ CF"
XEQ "SP"
CF 22
RTN
FS? 22
GTO 21
FS? 04
XEQ 18
FC? 04
XEQ 10
RTN
FS? 22
GTO 21
◆LBL b
"SMALIAN VOLUME"
AVIEW
SIN
SIN
"KEY DB↑ DU↑ L"
PROMPT
◆LBL 22
STO 09
X<>Y
STO 08
RDN
RDN
STO 06
◆LBL 25
RCL 09
PI
*
FS? 04
80000
FC? 04
1152
/
RCL 06
X↑2
RCL 08
X↑2
+
*
STO 16
"SMAL.= "
ARCL 16
FS? 04
"├ M↑3"
FC? 04
"├ CF"
XEQ "SP"
CF 22
RTN
FS? 22
GTO 22
FS? 04
XEQ 18
FC? 04
XEQ 10
RTN
FS? 22
GTO 22
◆LBL c
"NEWTON VOLUME"
AVIEW
SIN
SIN
"DB↑ DM↑ DU↑,L"
PROMPT
◆LBL 23
XEQ 16
◆LBL 26
RCL 06
X↑2
```

F. Program Listing (Continued 2)

```
301 RCL 08
302 X↑2
303 +
304 RCL 07
305 X↑2
306 4
307 *
308 +
309 PI
310 *
311 RCL 09
312 *
313 FS? 04
314 240000
315 FC? 04
316 3456
317 /
318 STO 17
319 "NEWT.= "
320 ARCL 17
321 FS? 04
322 "⊢ M↑3"
323 FC? 04
324 "⊢ CF"
325 XEQ "SP"
326 CF 22
327 RTN
328 FS? 22
329 GTO 23
330 FS? 04
331 XEQ 18
332 FC? 04
333 XEQ 10
334 RTN
335 FS? 22
336 GTO 23
337♦LBL 10
338 .028317
339 *
340 "   = "
341 ARCL X
342 "⊢ M↑3"
343 AVIEW
344 RTN
345♦LBL 16
346 STO 09
347 RDN
348 STO 08
349 RDN
350 STO 07
351 RDN
352 STO 06
353 RTN
354♦LBL e
355 "DB↑ DM↑ DU↑,L"
356 PROMPT
357♦LBL 20
358 XEQ 16
359♦LBL 13
360 XEQ 24
361 XEQ 25
362 XEQ 26
363 CF 22
364 RTN
365 FS? 22
366 GTO 20
367 GTO 13
368♦LBL "M"
369 SF 04
370 GTO 17
371 RTN
372♦LBL 18
373 35.3144
374 *
375 "   = "
376 ARCL X
377 "⊢ CF"
378 AVIEW
379 RTN
380 END
```

```
LBL"LOGVOL
LBL"BF
LBL"ABC
LBL"AB
LBL"AC
LBL"BC
LBL"SP
LBL"F23
LBL"CF
LBL"M
END          927 BYTES
```

G. Documentation

Initial prompt

```
 01♦LBL "LOGVOL"
AON  "BF CF OR M?"
FS? 01  XEQ "F23"
FC? 01  PROMPT  AOFF
ASTO 00  GTO IND 00
```

Data prompt and type of log rule

```
 11♦LBL "BF"
"KEY DU ↑ L"  PROMPT
STO 01  X<>Y  STO 02
"KEY A TO D OR E"
PROMPT
```

Doyle log rule calculation and display

```
 19♦LBL A
RCL 02  4  -  X↑2
RCL 01  16  /  *  STO 11

 29♦LBL 01
"DOYLE= "  ARCL 11
XEQ "SP"  RTN  XEQ 15
GTO A
```

Scribner log rule calculation and display

```
 36♦LBL B
RCL 02  X↑2  .79  *
RCL 02  2  *  -  4  -
RCL 01  16  /  *  STO 12

 52♦LBL 02
"SCRIB.= "  ARCL 12
XEQ "SP"  RTN  XEQ 15
GTO B
```

Int. log rule algorithm solution

```
 59♦LBL C
SF 06  RCL 01  4  /  INT
STO 04  LASTX  FRC
XEQ 07  CF 06  *  STO 14
RCL 04  X=0?  GTO 00

 75♦LBL 06
1  ST- 04

 78♦LBL 07
RCL 04  2  /  RCL 02  +
STO 03  X↑2  .22  *
RCL 03  .71  *  -
FS? 06  RTN  ST+ 14
RCL 04  X>0?  GTO 06
```

Int. 1/4 log rule calculation

```
 98♦LBL 00
FS?C 07  RTN  RCL 14
.905  *  STO 13
```

Int. 1/4 display

```
105♦LBL 03
"INT. 1/4= "  ARCL 13
XEQ "SP"  RTN  XEQ 15
GTO C
```

Int. 1/8 log rule calculation and display

```
112♦LBL D
SF 07  XEQ C

115♦LBL 04
"INT. 1/8= "  ARCL 14
XEQ "SP"  RTN  XEQ 15
GTO D
```

All four log rules calculation and display

```
122♦LBL E
XEQ A  XEQ B  XEQ C
XEQ D  RTN
```

Doyle, Scribner, Int. 1/4 display

```
128♦LBL "ABC"
FS? 00  XEQ 05  XEQ 01
PSE  XEQ 02  FS? 00  RTN
PSE

137♦LBL 05
XEQ 03  PSE  RTN
```

Doyle, Scribner display

```
141♦LBL "AB"
FS? 00  XEQ 08  XEQ 01
FS? 00  RTN  PSE

148♦LBL 08
XEQ 02  PSE  RTN
```

Doyle, Int. 1/4 display

```
152♦LBL "AC"
FS? 00  XEQ 09  XEQ 01
FS? 00  RTN  PSE

159♦LBL 09
XEQ 03  PSE  RTN
```

Scribner, Int. 1/4 display

```
163♦LBL "BC"
FS? 00  XEQ 09  XEQ 02
FS? 00  RTN  PSE  XEQ 09
RTN
```

Stop-pause subroutine

```
172♦LBL "SP"
AVIEW  FS? 01  PSE
FC? 01  STOP  RTN
```

R/S avoidance subroutine

```
179♦LBL "F23"
CF 23  AVIEW  AON

183♦LBL 11
PSE  FC?C 23  GTO 11
AOFF  RTN
```

G. Documentation (Continued)


```
189♦LBL 15
STO 01  X<>Y  STO 02
RTN
```

Input data storage subroutine (BF)

```
194♦LBL "CF"
CF 04

196♦LBL 17
"KEY a,b,c, OR e"
PROMPT
```

Cubic measure prompt for type of formula

```
199♦LBL a
"HUBER VOLUME"  AVIEW
SIN  SIN  "KEY DM ↑ L"
PROMPT

206♦LBL 21
STO 09  X<>Y  STO 07

210♦LBL 24
RCL 09  RCL 07  X↑2  *
PI  *  FS? 04  40000
FC? 04  576  /  STO 15
"HUB.= "  ARCL 15
FS? 04  "⊢ M↑3"  FC? 04
"⊢ CF"  XEQ "SP"  CF 22
RTN  FS? 22  GTO 21
FS? 04  XEQ 18  FC? 04
XEQ 10  RTN  FS? 22
GTO 21
```

Huber formula calculation and display

```
241♦LBL b
"SMALIAN VOLUME"  AVIEW
SIN  SIN
"KEY DB↑ DU↑ L"  PROMPT

248♦LBL 22
STO 09  X<>Y  STO 08
RDN  RDN  STO 06

255♦LBL 25
RCL 09  PI  *  FS? 04
80000  FC? 04  1152  /
RCL 06  X↑2  RCL 08  X↑2
+  *  STO 16  "SMAL.= "
ARCL 16  FS? 04  "⊢ M↑3"
FC? 04  "⊢ CF"  XEQ "SP"
CF 22  RTN  FS? 22
GTO 22  FS? 04  XEQ 18
FC? 04  XEQ 10  RTN
FS? 22  GTO 22
```

Smalian formula calculation and display

```
289♦LBL c
"NEWTON VOLUME"  AVIEW
SIN  SIN
"DB↑ DM↑ DU↑,L"  PROMPT

296♦LBL 23
XEQ 16

298♦LBL 26
RCL 06  X↑2  RCL 08  X↑2
+  RCL 07  X↑2  4  *  +
PI  *  RCL 09  *  FS? 04
240000  FC? 04  3456  /
STO 17  "NEWT.= "
ARCL 17  FS? 04  "⊢ M↑3"
FC? 04  "⊢ CF"  XEQ "SP"
CF 22  RTN  FS? 22
GTO 23  FS? 04  XEQ 18
FC? 04  XEQ 10  RTN
FS? 22  GTO 23
```

Newton formula calculation and display

```
337♦LBL 10
.028317  *  "  = "
ARCL X  "⊢ M↑3"  AVIEW
RTN
```

cu.ft. to m³ conversion subroutine

```
345♦LBL 16
STO 09  RDN  STO 08  RDN
STO 07  RDN  STO 06  RTN
```

CF input data storage subroutine

```
354♦LBL e
"DB↑ DM↑ DU↑,L"  PROMPT

357♦LBL 20
XEQ 16

359♦LBL 13
XEQ 24  XEQ 25  XEQ 26
CF 22  RTN  FS? 22
GTO 20  GTO 13
```

Cubic measure all formula calculation and display

```
368♦LBL "M"
SF 04  GTO 17  RTN
```

Metric measure flag set and branch

```
372♦LBL 18
35.3144  *  "  = "
ARCL X  "⊢ CF"  AVIEW
RTN  END
```

m³ to cu.ft. conversion subroutine

H. Formulas used

Reference to any reputable mensuration text (e.g., Husch, Miller, and Beers, 1972 and 1982 Forest Mensuration, John Wiley and Sons.) will make clear that the Doyle and International log rules were based originally on formulas and therefore should present no problem being adapted for calculator solution. As will be discussed later, this is not necessarily the case.

Unlike the computer with its virtually unlimited storage space, the portable calculator can rarely afford the luxury of a "table-look-up" procedure, therefore to determine log volumes for the Scribner rule (originally a "diagram" rule) one must employ a regression fit such as that developed by Bruce and described in Husch, Miller and Beers (1982).

As in the Scribner case, the calculation of cubic volumes presents no real problem either, since basic mensurational formulas are traditionally used.

An irregularity which surfaces in the calculation of "short log" board-foot volumes is rarely faced except by the forester charged with developing a volume table from felled-tree or dendrometer data. That is, the calculation of volumes for irregular-length or very short sections. The general rule for Doyle and Scribner rules seems to be: calculate the volume of a 16-foot log and multiply by the factor (log length/16). For the International rule, an additional complication arises because the original rule assumes a taper allowance (from the top end) of one-half inch for every four feet of length.

Recognizing the above conventions and constraints, the following formulas were used in the program previously described:

1. Board-foot Calculations:

assume d_u = scaling diameter in inches i.e., diameter inside bark at the small end of the log.

L = log length in feet.

V = log volume in board feet.

a. Doyle rule:

$$V = (d_u - 4)^2 L/16 \quad , L \leq 16$$

b. Scribner rule:

$$V = (.79d_u^2 - 2d_u - 4)L/16, \quad L \leq 16$$

H. Formulas used *(continued)*

c. International 1/4 inch rule:

$V = (.905)$ (volume for Int. 1/8 inch rule)

d. International 1/8 inch rule:

Here we have a complication! The rule specifies that starting at the top end of the log, calculate the volume in 4-foot sections, using the scaling diameter, from the formula:

$$V = .22d_u^2 - .71\ d_u\ ,$$

then add one-half inch to d_u and apply the formula using $(d_u + .5)$, etc., until the length $(L \leq 20)$ is exhausted. For the last section (large end) a partial 4-foot section might be encountered, but its volume is calculated as if 4 feet long, then adjusted by the multiplicative factor, $\frac{\text{section length}}{4}$. In the past, using the above taper assumptions, formulas have been derived for common even-foot lengths (e.g., 14, 12, 10, and 8) less than 16 feet (Husch, Miller, and Beers, 1982).

Using this multitude of formulas is perhaps satisfactory for the old rotary calculator (and lots of labor time available) or for the modern computer, but storing or generating the various required coefficients exhausts storage capacities or program steps rapidly and/or can become tedious to place in calculator memory. Therefore, the following algorithm was developed and used:

(1) Calculate $L/4$ to obtain n.f, where

$1 \leq L \leq 20$

n = the integer component, $0 \leq n \leq 5$

f = the fractional component: .00, .25, .50, or .75.

(2) When $4 \leq L \leq 20$, calculate log volume, V, from the two-part formula:

$$V = \sum_{j=1}^{n} \{.22(d_u + \tfrac{n-j}{2})^2 - .71(d_u + \tfrac{n-j}{2})\} + f\{.22(d_u + \tfrac{n}{2})^2 - .71(d_u + \tfrac{n}{2})\}$$

H. Formulas used *(continued)*

(3) If $L < 4$, $n = 0$, implying the summation part of the formula is meaningless and the second part reduces to:

$$V = f(.22d_u^2 - .71d_u)$$

(4) If $L > 20$ the log should be scaled as two logs.

2. Cubic calculations

Assume

D_b, D_m, D_u = diameters in inches (or centimeters) at the log base, middle and upper (small) end.

L = section length in feet (or meters)

V = section volume in cubic feet (or cubic meters)

a. Huber's formula:

$$V = L \frac{\Pi D_m^2}{576} \text{ in cubic feet}$$

$$= L \frac{\Pi D_m^2}{40{,}000} \text{ in cubic meters}$$

b. Smalian's formula:

$$V = \frac{L}{2} \frac{\Pi}{576} (D_b^2 + D_u^2) \text{ in cubic feet}$$

$$= \frac{L}{2} \frac{\Pi}{40{,}000} (D_b^2 + D_u^2) \text{ in cubic meters}$$

c. Newton's formula:

$$V = \frac{L}{6} \frac{\Pi}{576} (D_b^2 + 4 D_m^2 + D_u^2) \text{ in cubic feet}$$

$$= \frac{L}{6} \frac{\Pi}{40{,}000} (D_b^2 + 4 D_m^2 + D_u^2) \text{ in cubic meters}$$

Figure 1. Representation of Options, Input, and Display of the HP-41C Program: LOGVOL.

XEQ LOGVOL (assigned to one key)

prompt for keys to press

B F — C F — M

B F

prompt for input: d_u, L
prompt for key to press:

A : DOYLE=xx.x[*1]
B : SCRIB.=xx.x[*1]
C : INT. 1/4=xx.x[*1]
D : INT. 1/8=xx.x[*1]

E[*5]: DOYLE=xx.x
SCRIB.=xx.x
INT. 1/4=xx.x
INT. 1/8=xx.x

C F

prompt for key to press:

a : input D_m, L, R/S : HUB.=xx.xxCF[*1,3]
b : input D_b, D_u, L, R/S : SMAL.=xx.xxCF[*1,3]
c : input D_b, D_m, D_u, L R/S : NEWT.=xx.xxCF[*1,3]
e[*5]: input D_b, D_m, D_u, L, R/S : HUB.=xx.xxCF
SMAL.=xx.xxCF
NEWT.=xx.xxCF

M

same procedure as for CF except input units are cm. and m., and display units are m^3.

a : HUB.=xx.xxM↑3[*1,4]
b : SMAL.=xx.xxM↑3[*1,4]
c : NEWT.=xx.xxM↑3[*1,4]
e[*5]: HUB.=xx.xxM↑3
SMAL.=xx.xxM↑3
NEWT.=xx.xxM↑3

Then: XEQ ABC, AB, AC, BC
(each assigned to a key)

ABC : Doyle, Scribner, Int. 1/4[*2]
AB : Doyle, Scribner[*2]
AC : Doyle, Int. 1/4[*2]
BC : Scribner, Int. 1/4[*2]

---

*1 For the same rule or formula, but new log, input new dimensions, then R/S

*2 If flag 00 is set, last answer within group is displayed first.

*3 Depress R/S to convert answer to cubic meters.

*4 Depress R/S to convert answer to cubic feet.

*5 If flag 01 is set, display is automatic (with 1 second pause), otherwise R/S must be pressed between answers.

Retrospective Comments Regarding LOGVOL

1. The means by which the board feet, cubic feet, or metric options are chosen (keying BF, CF or M at the initial prompt) may not be pleasing to all users, but it represents an alternative to using flags or alphabetic local labels. Similarly, the use of global labels ABC, AB, AC, and BC to display groups of board-foot output could be replaced by logical alphabetic local labels.

2. An alternative to using the Int. 1/8-inch algorithm (label C) is available if one chooses to use the detailed equations developed by Grosenbaugh (1952, Shortcuts for Cruisers and Scalers); answers will be equivalent and the programming logic will be somewhat easier to follow.

3. Printed output is adequately spaced and identified only when printer is in NORM mode.

Program No. 41F009

Progam Name: RANCHEK

Calculator: HP-41C/CV

Author: T.W. Beers, Dept. Forestry, Purdue U.

Date: August 1980

Purpose: To evaluate a random number generator (RAN) by simulating the rolling of one die a user-determined number of times. The occurrences of 1's, 2's, . . ., 6's are tallied and displayed, as is the calculated chi-square value, sample mean, and the bias in % (deviation of mean from the true: 3.5).

A. Storage assignment

Register	Use
00	N_{max} for RNG; here, $N_{max} = 6$
01-06	No. of 1's (n_1), n_2, ..., n_6
07	seed for RNG
08	roll count, calculated internally, = n
09	decrementing counter, from input n' to zero and in label 04
10	general indexing counter
11-16	statistical accumulators

B. Labels

Name	Use
RANCHEK	XEQ to initialize and start program
F22	prompt avoidance routine
MRC	selective register clearing
RAN	random number generator (RNG); one used here in the "9821X" based on Malm's algorithm.
01	in MRC
02	in RAN
03	main program; generate, count, and accumulate
04	calculate ΣO_i^2 for chi-square
05	count display subroutine

B. Labels *(Continued)*

Name	
06	in F22
07	display calling for counts
08	pause-stop option
A	to repeat read-out of counts
B	to repeat chi-square display
C	to repeat mean display
D	to repeat bias % display

C. Flags

Number	Use
01	set externally to select input option; when set, R/S depression is minimized, and "prompt stops" are avoided.
05	to avoid prompting in RAN, set internally
22	numeric data input flag, used in F22

D. Program procedure

1. XEQ SIZE 017, f[1] GTO . ., load program RANCHEK

2. For "automatic" input and count display, f SF 01.

   If flag 01 is clear, R/S must be pressed after desired roll count is keyed and between count displays.

3. Store seed ($0 \leq$ seed < 1) in R_{07} (Example: 0 STO 07)

4. XEQ RANCHEK (suggest assignment to LN) and follow prompts, assuming flag 01 has been set:

Prompt	Input	Key	Output	Example Input	Example Key	Example Output
NO. ROLLS?	n' arbitrary*	none	*GENERATING* then audible beep.	6		*GENERATING* then audible beep.
DONE: HIT R/S	none	R/S	ROLL CT. = n N1 = n_1 N2 = n_2 N3 = n_3		R/S	ROLL CT. = 6 N1 = 1 N2 = 2 N3 = 1

[1] f is used to indicate the shift (gold) key.

D. Program procedure *(Continued)*

4. *(Continued)*

Prompt	Input	Key	Output	Input (Example)	Key (Example)	Output (Example)
			N4 = n_4			N4 = 2
			N5 = n_5			N5 = 0
			N6 = n_6			N6 = 0
none	none	R/S	CHI SQ. = X.XX		R/S	CHI SQ. = 4.00
none	none	R/S	MEAN = X.XX		R/S	MEAN = 2.67
none	none	R/S	BIAS = X.XX%		R/S	BIAS = -23.81%

*Note: approximately 15 seconds are needed when n' = 10

5. For repeat of the output, in USER mode:

   depress A for count display
   " B for chi square
   " C for mean
   " D for bias percent

   Note that the order of depression is immaterial <u>except</u> that C must be pressed before D.

6. For a new set of rolls, go to step 4; seed for the RNG will be the "unknown" contents of R_{07}.

E. Formulas and results

1. Formulas

   a. in RAN, Malm's algorithm for pseudo-random numbers, modified to provide integers from 1 to arbitrary N_{max}:

$$N_i = \text{INT}\left[N_{max}(u_i)\right] + 1$$

$$\text{where} \quad u_i = \text{FRAC}\left[9821(u_{i-1}) + .211327\right]$$

   b. chi square: based on $\chi^2 = \sum\frac{(O_i-E_i)^2}{E_i}$ one can show

$$\chi^2 = \sum\frac{O_i^2}{E_i} - n, \quad \text{where} \quad n = \sum O_i = \sum E_i$$

$$\text{but when} \quad E_1 = E_2 = \ldots = E_6 = \frac{\Sigma E_i}{6} = \frac{n}{6}$$

$$\chi^2 = \frac{6\Sigma O_i^2}{n} - n \quad , \quad n = \text{roll count}$$

E. Formulas and results *(Continued)*

c. true mean = $\mu = 3.5$ by definition

d. sample mean $= \overline{X} = \sum_{i=1}^{n} X_i / n$, X_i = 1,2,3,4,5, or 6

e. bias % = $\left(\frac{\overline{X} - \mu}{\mu}\right) 100$

2. Some results (compare with 34C results)

Trial	seed	time	n	Number of 1	2	3	4	5	6	χ^2	$\overline{X}$	Bias %	$\chi^2.05$
1a.	0	12 sec.	6	1	2	1	2	0	0	4.00	2.67	-23.81	11.07
2a.	0	15 "	10	2	2	2	4	0	0	6.80	2.80	-20.00	↓
3a.	0	30 "	20	3	6	4	4	0	3	5.80	3.05	-12.86	↓
4a.	0	2 min. 30 sec.	100	18	14	23	22	11	12	7.88	3.30	-5.71	↓
b.	?	"	100	13	22	18	25	13	9	11.12*	3.30	-5.71	↓
c.	?	"	100	17	17	21	18	13	14	2.48	3.35	-4.29	↓
d.	?	"	100	20	16	14	16	17	17	1.16	3.45	-1.43	↓
e.	?	"	100	18	21	18	20	12	11	5.24	3.20	-8.57	↓
f.	?	"	100	19	12	15	23	15	16	4.40	3.51	0.29	↓

F. Program Listing

```
01♦LBL "RANCHEK"
02 CF 29
03 FIX 0
04 "NO. ROLLS?"
05 FS? 01
06 XEQ "F22"
07 FC? 01
08 PROMPT
09 STO 09
10 "*GENERATING*"
11 AVIEW
12 ADV
13 SF 05
14 CLΣ
15 1.006
16 XEQ "MRC"
17 0
18 STO 08
19 STO 10
20 6
21 STO 00
22♦LBL 03
23 XEQ "RAN"
24 3.5
25 X<>Y
26 Σ+
27 LASTX
28 ISG IND X
29 BEEP
30 ISG 08
31 BEEP
32 DSE 09
33 GTO 03
34 BEEP
35 "DONE: HIT R/S"
36 PROMPT
37 0
38 STO 09
39 1.006
40 STO 10
41♦LBL 04
42 RCL IND 10
43 X↑2
44 ST+ 09
45 ISG 10
46 GTO 04
47♦LBL A
48 1.006
49 STO 10
50 FIX 0
51 "ROLL CT.= "
52 ARCL 08
53 FS? 01
54 XEQ 08
55 FC? 01
56 PROMPT
57♦LBL 07
58 XEQ 05
59 ISG 10
60 GTO 07
61 RTN
62 6
63 ST* 09
64 RCL 08
65 ST/ 09
66 ST- 09
67♦LBL B
68 FIX 2
69 "CHI SQ.= "
70 ARCL 09
71 AVIEW
72 RTN
73♦LBL C
74 MEAN
75 "MEAN= "
76 ARCL X
77 AVIEW
78 RTN
79♦LBL D
80 %CH
81 "BIAS= "
82 ARCL X
83 "⊦%"
84 AVIEW
85 RTN
86 GTO "RANCHEK"
87♦LBL 05
88 RCL 10
89 INT
90 " N"
91 ARCL X
92 "⊦= "
93 ARCL IND 10
94 FS? 01
95 XEQ 08
96 FC? 01
97 PROMPT
98 RTN
99♦LBL "F22"
100 CF 22
101 AVIEW
102♦LBL 06
103 PSE
104 FC?C 22
105 GTO 06
106 RTN
107♦LBL "MRC"
108 0
109♦LBL 01
110 STO IND Y
111 ISG Y
112 GTO 01
113 RTN
114♦LBL 08
115 AVIEW
116 PSE
117 RTN
118♦LBL "RAN"
119 FS? 05
120 GTO 02
121 "MAX VALUE?"
122 PROMPT
123 STO 00
124♦LBL 02
125 RCL 07
126 9821
127 *
128 .211327
129 +
130 FRC
131 STO 07
132 RCL 00
133 *
134 1
135 +
136 INT
137 END

RANCHEK:341 BYTES
```

G. Program Documentation

```
 01♦LBL "RANCHEK"
CF 29  FIX 0
"NO. ROLLS?"  FS? 01
XEQ "F22"  FC? 01
PROMPT  STO 09
"*GENERATING*"  AVIEW
ADV  SF 05  CLΣ  1.006
XEQ "MRC"  0  STO 08
STO 10  6  STO 00
```

Initialize and prompt for roll count

```
 22♦LBL 03
XEQ "RAN"  3.5  X<>Y  Σ+
LASTX  ISG IND X  BEEP
ISG 08  BEEP  DSE 09
GTO 03  BEEP
"DONE: HIT R/S"  PROMPT
0  STO 09  1.006  STO 10
```

Call for RN, increment n_i, decrement n′

```
 41♦LBL 04
RCL IND 10  X↑2  ST+ 09
ISG 10  GTO 04
```

Calculate ΣO_i^2

```
 47♦LBL A
1.006  STO 10  FIX 0
"ROLL CT.= "  ARCL 08
FS? 01  XEQ 08  FC? 01
PROMPT
```

Display accumulated roll count

```
 57♦LBL 07
XEQ 05  ISG 10  GTO 07
RTN  6  ST* 09  RCL 08
ST/ 09  ST- 09
```

Display individual counts and calculate chi-sq.

```
 67♦LBL B
FIX 2  "CHI SQ.= "
ARCL 09  AVIEW  RTN
```

Display chi-square

```
 73♦LBL C
MEAN  "MEAN= "  ARCL X
AVIEW  RTN
```

Calculate and display $\overline{X}$

```
 79♦LBL D
%CH  "BIAS= "  ARCL X
"⊦%"  AVIEW  RTN
GTO "RANCHEK"
```

Calculate and display Bias %

```
 87♦LBL 05
RCL 10  INT  " N"
ARCL X  "⊦= "
ARCL IND 10  FS? 01
XEQ 08  FC? 01  PROMPT
RTN
```

Individual count display, called in A

```
 99♦LBL "F22"
CF 22  AVIEW

102♦LBL 06
PSE  FC?C 22  GTO 06
RTN
```

prompt avoidance

```
107♦LBL "MRC"
0

109♦LBL 01
STO IND Y  ISG Y  GTO 01
RTN
```

register clearing

```
114♦LBL 08
AVIEW  PSE  RTN
```

pause-stop option

```
118♦LBL "RAN"
FS? 05  GTO 02
"MAX VALUE?"  PROMPT
STO 00

124♦LBL 02
RCL 07  9821  *  .211327
+  FRC  STO 07  RCL 00
*  1  +  INT  END
```

Random number generator

H. Printer Output -- with and without flag 01 set.

```
               0 STO 07
                  SF 01
         XEQ "RANCHEK"
NO. ROLLS?
                6
*GENERATING*

DONE: HIT R/S
                    RUN
ROLL CT.= 6
 N1= 1
 N2= 2
 N3= 1
 N4= 2
 N5= 0
 N6= 0
                    RUN
CHI SQ.= 4.00
                    RUN
MEAN= 2.67
                    RUN
BIAS= -23.81%
```

```
TIME NEEDED:

15 SEC. PER 10 NOS.
```

```
            0.00 STO 07
                  CF 01
         XEQ "RANCHEK"
NO. ROLLS?
               6    RUN
*GENERATING*

DONE: HIT R/S
                    RUN
ROLL CT.= 6
                    RUN
 N1= 1
                    RUN
 N2= 2
                    RUN
 N3= 1
                    RUN
 N4= 2
                    RUN
 N5= 0
                    RUN
 N6= 0
                    RUN
                    RUN
CHI SQ.= 4.00
                    RUN
MEAN= 2.67
                    RUN
BIAS= -23.81%
                    RUN
NO. ROLLS?
```

I. Some Results using the HP-34C

Trial	Seed	Time	n	Number of 1	2	3	4	5	6	χ^2	$\overline{X}$	B%	χ^2 .05, 5df
1 a.	0	26 sec.	6	1	2	1	2	0	0	4.00	2.67	-23.81	11.07
b.	?	"	6	2	0	2	2	0	0	6.00	2.67	-23.81	
c.	?	"	6	0	3	1	0	0	2	8.00	3.50	0.00	
d.	?	"	6	1	1	1	0	1	2	2.00	3.83	9.52	
e.	?	"	6	1	1	0	1	1	2	2.00	4.00	14.29	
2 a.	0	43 sec.	10	2	2	2	4	0	0	6.80	2.80	-20.00	
b.	?	"	10	1	4	2	0	0	3	8.00	3.30	-5.71	
c.	?	"	10	2	1	1	1	2	3	2.00	3.90	11.43	
d.	?	"	10	1	0	3	4	1	1	6.80	3.70	5.71	
e.	?	"	10	0	1	2	4	2	1	5.60	4.00	14.29	
3 a.	0	85 sec.	20	3	6	4	4	0	3	5.80	3.05	-12.86	
b.	?	"	20	3	1	4	5	3	4	2.80	3.80	8.57	
c.	?	"	20	2	1	5	6	4	2	5.80	3.75	7.14	
d.	?	"	20	3	4	6	4	3	0	5.80	3.00	-14.29	
e.	?	"	20	7	2	4	3	1	3	6.40	2.90	-17.14	
4 a.	0	7 min, 2 sec.	100	18	14	23	22	11	12	7.88	3.30	-5.71	
b.	?	"	100	13	22	18	25	13	9	11.12*	3.30	-5.71	
c.	?	"	100	17	17	21	18	13	14	2.48	3.35	-4.29	
d.	?	"	100	20	16	14	16	17	17	1.16	3.45	-1.43	
e.	?	"	100	18	21	18	20	12	11	5.24	3.20	-8.57	
5 a.	0	42 min, 9 sec.	600	105	102	109	124	81	79	14.88*	3.35	-4.24	
b.	?	"	600	104	97	100	90	97	112	2.78	3.53	0.71	
c.	?	"	600	101	93	93	119	92	102	5.28	3.52	0.67	
d.	?	"	600	96	102	89	95	109	109	3.28	3.58	2.19	
e.	?	"	600	69	120	117	97	83	114	21.44*	3.58	2.24	
6 a.	0	approx. 8.5 hrs.	6000	961	1007	998	1053	979	1002	4.83	3.51	0.42	
b.	?	"	6000	986	1026	981	979	1033	995	2.79	3.51	0.15	
c.	?	"	6000	978	1020	995	961	1029	1017	3.56	3.52	0.45	
d.	?	"	6000	951	1024	984	1002	1029	1010	4.18	3.53	0.78	
e.	?	"	6000	1040	1021	999	993	947	1000	4.90	3.46	-1.02	

Retrospective Comments Regarding RANCHEK

1. RANCHEK as originally written is extremely inefficient in places; for example, the use of global labels such as MRC and RAN, and statements such as GTO RANCHEK take many bytes to represent and force the calculator to search its entire resident and peripheral memory for these labels. It is usually better to employ local numeric labels even though some program clarity is sacrificed.

2. Since the statistical registers are assumed to be 11 through 16, the statement ΣREG 11 should be inserted just after step 01.

3. The program was written to simulate the roll of a six-sided die, generating the integers 1 through 6 from a uniform distribution, and to demonstrate the calculation of certain statistics to evaluate the performance of an arbitrary "random" number-generating algorithm. Realistically, one might want the integers 0 through 9, rather than 1 through 6, in the evaluation. A very short program to do this is described by Dearing (1982, Calculator Tips and Routines), or, one can use a refined version of RANCHEK, called RCHEK1, found on the KRON-1 tape.

4. In RCHEK1, some of the inefficiencies of RANCHEK were removed, the program was shortened to 129 steps, and it was made to simulate a ten-sided die assuming the integers 0 through 9. Other comments regarding RCHEK1:

   a. With flag 04 set, other random number generators, such as RAN1 shown here

```
01♦LBL "RAN1"
02 RCL 27
03 R-D
04 FRC
05 STO 27
06 10
07 *
08 INT
09 END
```

   can be evaluated by this method:

   (1) key global name of generator, here RAN1
   (2) ASTO 20
   (3) SF 04
   (4) store seed in R_{27} (optional, but in RAN1 zero or multiples of π won't work!)
   (5) XEQ RCHEK1

The evaluation of RAN1 and the "resident" Malm generator (label 10 in RCHEK1) are shown below:

```
RAN1            ASTO 20
                  SF 04
        .123456 STO 27
          XEQ "RCHEK1"
NO. ROLLS?
         50.000
SEED=0.123456
*GENERATING*

ROLL CT.= 50
 N0= 6
 N1= 9
 N2= 4
 N3= 3
 N4= 3
 N5= 8
 N6= 4
 N7= 5
 N8= 5
 N9= 3

CHI SQ.= 8.00
MEAN= 4.08
BIAS= -9.33%
```

```
                  CF 04
           0.00 STO 27
          XEQ "RCHEK1"
NO. ROLLS?
          50.00
SEED=0.000000
*GENERATING*

ROLL CT.= 50
 N0= 4
 N1= 4
 N2= 5
 N3= 2
 N4= 9
 N5= 9
 N6= 4
 N7= 4
 N8= 4
 N9= 5

CHI SQ.= 9.20
MEAN= 4.60
BIAS= 2.22%
```

b. An additional characteristic of the generators, time required, can be evaluated efficiently if one employs an HP Time Module (HP 82182 A) as follows:

(1) In the generator routine itself, insert RUNSW just after the label (LBL 10 or LBL RAN1 in the example above), and insert STOPSW just before the END (or RTN) statement.

(2) From the keyboard:
XEQ STOPSW
key 0
XEQ SETSW

(3) Run the RCHEK1 program as usual.

(4) At the "DONE, HIT R/S" prompt (or later):
depress FIX 6
clear alpha register (CLA)
XEQ RCLSW
XEQ ATIME24
go to alpha mode and observe elapsed time for generator to generate the number of rolls specified.

(5) Note: using the RAN1 example for ROLL CT= 50, the result after step (4) will be 00:00:18.20 (approximately) indicating that RAN1 takes 18.20 seconds to generate the 50 numbers, even though the time elapsed, during which *GENERATING* fills the display, is a little over a minute.

Program No. 41F010

Calculator: HP-41C/CV

Program Name: INMULT (interest multiplier)

Author: Thomas W. Beers

Date: October, 1980

Purpose: To calculate, display and print values of $(1 + i)^n$ for arbitrary values of interest rate ($i = \frac{rate}{100}$) and years (n). Two options are provided:

1. "Single" solution for specific rate, r, and range of years from 1 to n_{max}.
2. Semi-automatic or automatic solution for specified ranges of r and n, using certain specified intervals.

A. Storage assignments

Register	Use
00	n_{max}, or active control number for years (BB.EEEII)
01	interest rate, r, as a percent
02	$\frac{r}{100}$, then $(1 + \frac{r}{100})$, i.e., $(1 + i)$
03	control number for interest (BB.EEEII)
04	stored control number for years

B. Labels

Name	Use
INMULT	program name; should be assigned.
F22	used to achieve automatic input and output if flag 01 is set.
00	automatic solution routine, reached if flag 00 is set
01	primary routine for "single" solution option
02	rate display subroutine
03	years incrementing loop
04	interest rate incrementing loop

B. (continued)

Name	Use
05	subroutine to achieve .5 increment in rate; active if flag 02 is set.
06	internal loop in F22
07	internal access for .5 increment option

C. Flags

Number	Use
00	if set, option 2 (range of r and n) is selected.
01	if set, input and output are automatic in option 1 (i.e. R/S need not be pressed between data); in option 2 this flag is only operative on output.
02	for use only in option 2; if set, interest rate is incremented in steps of 0.5% above those specified in the control number.
06	internal use in conjunction with flag 02.
12	double wide printing flag; used to print rate in bold type.
22	numeric input flag; used to detect input in F22 subroutine.
29	digit grouping flag; used to suppress printing and display of decimal point for years.

D. Program procedure and example

Step

1. XEQ SIZE 005, f* GTO..
2. Load program INMULT
3. To select option 1 (single solution) insure that flag 00 is clear, then

*Throughout this paper "f" represents the shift (gold) key.

D. (continued)

3. a. XEQ INMULT (suggest assignment to COS key) and follow the prompts:

Prompt	input	key1/	output	Example input	Example key1/	Example output
RATE=?	rate=r	R/S	-	18	R/S	-
FINAL YR=?	n_{max}	R/S	rate % YR.1: X.XXX	5	R/S	RATE=18.00% YR. 1: 1.180
		R/S	YR.2: X.XXX		R/S	YR. 2: 1.392
		⋮	⋮ ⋮		R/S	YR. 3: 1.643
		R/S	YR.n: X.XXX		R/S	YR. 4: 1.939
		R/S	beep then n.000		R/S	YR. 5: 2.288
		(if flag 01 is set, final output is YR.n:XXX)			R/S or	beep 5.000 beep YR.5:2.288

1/note, if flag 01 is set, R/S need not be pressed after inputs or between outputs.

b. For new interest rate or final number of years (n_{max}) repeat step 3a.

4. To select option 2 (automatic solution), set flag 00, then

a. Develop control numbers for interest rate and for years, using the format BB.EEEII,

where BB = beginning rate or year
EEE = ending rate or year
II = desired increment (if not specified II is assumed = 01)

Example:

if rate: 16.02202 and years: 5.02505, then $(1 + i)^n$ would be calculated and displayed for n = 5, 10, 15, 20, and 25 for interest rates 16, 18, 20, and 22.

b. XEQ INMULT (flag 00 set) and follow the prompts:

4. b. (continued)

Prompt	input	key[1]	output	Example input	Example key[1]	Example output
			INPUT RATE:			INPUT RATE:
BB.EEEII	BB.EEEII	R/S		6.008	R/S	
			INPUT YEARS:			INPUT YEARS:
BB.EEEII	BB.EEEII	R/S	RATE=X.XX% YR.--(1+I)$\uparrow$N Y_1 X.XXX	1.00502	R/S	RATE=6.00% YR--(1+I)$\uparrow$N 1 1.060
		R/S	Y_2 X.XXX		R/S	3 1.191
		⋮	⋮			
		R/S	Y_n X.XXX		R/S	5 1.338
		R/S	RATE=X.XX% YR.--(1+I)$\uparrow$N Y_1 X.XXX		R/S	RATE=7.00% YR--(1+I)$\uparrow$N 1 1.070
		R/S	Y_2 X.XXX		R/S	3 1.225
			etc. till finally		R/S	5 1.403
		R/S	beep, then BB.EEE for year (if flag 01 is set, final output is Y_n and X.XXX)		R/S	RATE=8.00% YR.--(1+I)$\uparrow$N 1 1.080
					R/S	3 1.260
					R/S	5 1.469
					R/S	beep, then 1.005
					or	beep, then 5 1.469

[1] note, if flag 01 is set, R/S need not be pressed between output answers, i.e., display of $(1+i)^n$ is automatic.

c. An additional feature in option 2 only is the incrementing of interest rates by one-half percents. To employ this refinement:

(1) set flag 02 (as well as flag 00)

(2) proceed as in 4a and 4b

(3) if the same example is used, the interest rates now considered will be 6.00, 6.50, 7.00, 7.50, 8.00, and 8.50, years will still be 1, 3, and 5.

E1. Program listing--with printer control in MAN mode. Note that statements are "left-justified and not set off into label groups.

```
01♦LBL "INMULT"
02 FS? 00
03 GTO 00
04 "RATE = ?"
05 FS? 01
06 XEQ "F22"
07 FC? 01
08 PROMPT
09 STO 01
10 1 E2
11 /
12 STO 02
13 "FINAL YR = ?"
14 FS? 01
15 XEQ "F22"
16 FC? 01
17 PROMPT
18 1 E3
19 /
20 1
21 ST+ 02
22 +
23 STO 00
24 XEQ 02
25♦LBL 01
26 RCL 02
27 RCL 00
28 INT
29 Y↑X
30 LASTX
31 FIX 0
32 "YR. "
33 10
34 X>Y?
35 "⊦ "
36 RDN
37 CF 29
38 ARCL X
39 "⊦:  "
40 FIX 3
41 ARCL Y
42 AVIEW
43 FS? 01
44 PSE
45 FC? 01
46 STOP
47 SF 29
48 ISG 00
49 GTO 01
50 BEEP
51 ADV
52 ADV
53 ADV
54 ADV
55 RTN
56♦LBL "F22"
57 CF 22
58 AVIEW
59♦LBL 06
60 PSE
61 FC?C 22
62 GTO 06
63 RTN
64♦LBL 00
65 "INPUT RATE:"
66 AVIEW
67 SIN
68 " BB.EEEII"
69 PROMPT
70 STO 03
71 "INPUT YEARS:"
72 AVIEW
73 SIN
74 " BB.EEEII"
75 PROMPT
76 STO 00
77 STO 04
78♦LBL 04
79 RCL 03
80 INT
81 STO 01
82 1 E2
83 /
84 1
85 +
86 STO 02
87♦LBL 07
88 XEQ 02
89 "YR.--(1+I)↑N"
90 AVIEW
91 SIN
92♦LBL 03
93 RCL 02
94 RCL 00
95 INT
96 Y↑X
97 CF 29
98 FIX 0
99 CLA
100 ARCL L
101 "⊦      "
102 FIX 3
103 ARCL X
104 AVIEW
105 FS? 01
106 PSE
107 FC? 01
108 STOP
109 SF 29
110 ISG 00
111 GTO 03
112 RCL 04
113 STO 00
114 ADV
115 ADV
116 ADV
117 FS? 06
118 RTN
119 FS? 02
120 XEQ 05
121 ISG 03
122 GTO 04
123 BEEP
124 RTN
125♦LBL 02
126 FIX 2
127 SF 12
128 ADV
129 "RATE="
130 ARCL 01
131 "⊦ %"
132 AVIEW
133 CF 12
134 ADV
135 PSE
136 RTN
137♦LBL 05
138 .5
139 ST+ 01
140 1 E2
141 /
142 ST+ 02
143 SF 06
144 XEQ 07
145 CF 06
146 RTN
147 END

INMULT:329 BYTES
```

E2. Program listing--with printer control in TRACE mode. Note that statements are left-justified, and in "paragraph" form grouped by label designation.


```
 01♦LBL "INMULT"
FS? 00  GTO 00
"RATE = ?"  FS? 01
XEQ "F22"  FC? 01
PROMPT  STO 01  1 E2  /
STO 02  "FINAL YR = ?"
FS? 01  XEQ "F22"
FC? 01  PROMPT  1 E3  /
1  ST+ 02  +  STO 00
XEQ 02

 25♦LBL 01
RCL 02  RCL 00  INT  Y↑X
LASTX  FIX 0  "YR. "  10
X>Y?  "⊦ "  RDN  CF 29
ARCL X  "⊦:  "  FIX 3
ARCL Y  AVIEW  FS? 01
PSE  FC? 01  STOP  SF 29
ISG 00  GTO 01  BEEP
ADV  ADV  ADV  ADV  RTN

 56♦LBL "F22"
CF 22  AVIEW

 59♦LBL 06
PSE  FC?C 22  GTO 06
RTN

 64♦LBL 00
"INPUT RATE:"  AVIEW
SIN  " BB.EEEII"  PROMPT
STO 03  "INPUT YEARS:"
AVIEW  SIN  " BB.EEEII"
PROMPT  STO 00  STO 04

 78♦LBL 04
RCL 03  INT  STO 01
1 E2  /  1  +  STO 02

 87♦LBL 07
XEQ 02  "YR.--(1+I)↑N"
AVIEW  SIN

 92♦LBL 03
RCL 02  RCL 00  INT  Y↑X
CF 29  FIX 0  CLA
ARCL L  "⊦      "  FIX 3
ARCL X  AVIEW  FS? 01
PSE  FC? 01  STOP  SF 29
ISG 00  GTO 03  RCL 04
STO 00  ADV  ADV  ADV
FS? 06  RTN  FS? 02
XEQ 05  ISG 03  GTO 04
BEEP  RTN

125♦LBL 02
FIX 2  SF 12  ADV
"RATE="  ARCL 01  "⊦ %"
AVIEW  CF 12  ADV  PSE
RTN

137♦LBL 05
.5  ST+ 01  1 E2  /
ST+ 02  SF 06  XEQ 07
CF 06  RTN  END
```

E3. Program listing--with printer control in NORM mode. Note that statements are essentially right justified with a space before all labels. The fastest listing is obtained in this mode.

```
01♦LBL "INMULT"
02 FS? 00
03 GTO 00
04 "RATE = ?"
05 FS? 01
06 XEQ "F22"
07 FC? 01
08 PROMPT
09 STO 01
10 1 E2
11 /
12 STO 02
13 "FINAL YR = ?"
14 FS? 01
15 XEQ "F22"
16 FC? 01
17 PROMPT
18 1 E3
19 /
20 1
21 ST+ 02
22 +
23 STO 00
24 XEQ 02

25♦LBL 01
26 RCL 02
27 RCL 00
28 INT
29 Y↑X
30 LASTX
31 FIX 0
32 "YR. "
33 10
34 X>Y?
35 "├ "
36 RDN
37 CF 29
38 ARCL X
39 "├:  "
40 FIX 3
41 ARCL Y
42 AVIEW
43 FS? 01
44 PSE
45 FC? 01
46 STOP
47 SF 29
48 ISG 00
49 GTO 01
50 BEEP
51 ADV
52 ADV
53 ADV
54 ADV
55 RTN

56♦LBL "F22"
57 CF 22
58 AVIEW

59♦LBL 06
60 PSE
61 FC?C 22
62 GTO 06
63 RTN

64♦LBL 00
65 "INPUT RATE:"
66 AVIEW
67 SIN
68 " BB.EEEII"
69 PROMPT
70 STO 03
71 "INPUT YEARS:"
72 AVIEW
73 SIN
74 " BB.EEEII"
75 PROMPT
76 STO 00
77 STO 04

78♦LBL 04
79 RCL 03
80 INT
81 STO 01
82 1 E2
83 /
84 1
85 +
86 STO 02

87♦LBL 07
88 XEQ 02
89 "YR.--(1+I)↑N"
90 AVIEW
91 SIN

92♦LBL 03
93 RCL 02
94 RCL 00
95 INT
96 Y↑X
97 CF 29
98 FIX 0
99 CLA
100 ARCL L
101 "├      "
102 FIX 3
103 ARCL X
104 AVIEW
105 FS? 01
106 PSE
107 FC? 01
108 STOP
109 SF 29
110 ISG 00
111 GTO 03
112 RCL 04
113 STO 00
114 ADV
115 ADV
116 ADV
117 FS? 06
118 RTN
119 FS? 02
120 XEQ 05
121 ISG 03
122 GTO 04
123 BEEP
124 RTN

125♦LBL 02
126 FIX 2
127 SF 12
128 ADV
129 "RATE="
130 ARCL 01
131 "├ %"
132 AVIEW
133 CF 12
134 ADV
135 PSE
136 RTN

137♦LBL 05
138 .5
139 ST+ 01
140 1 E2
141 /
142 ST+ 02
143 SF 06
144 XEQ 07
145 CF 06
146 RTN
147 END
```

F1. Sample printer output--option 1 (flag 00 clear), with flag 01 clear, then set (1A).


```
EXAMPLE: OPTION 1

                  CF 00
                  CF 01
            XEQ "INMULT"
RATE = ?
          18.000    RUN
FINAL YR = ?
           5.000    RUN

RATE=18.00 %

YR.  1:  1.180
                    RUN
YR.  2:  1.392
                    RUN
YR.  3:  1.643
                    RUN
YR.  4:  1.939
                    RUN
YR.  5:  2.288
                    RUN


            XEQ "INMULT"
RATE = ?
          19.000    RUN
FINAL YR = ?
           3.000    RUN

RATE=19.00 %

YR.  1:  1.190
                    RUN
YR.  2:  1.416
                    RUN
YR.  3:  1.685
                    RUN
```

```
EXAMPLE: OPTION 1A

                  CF 00
                  SF 01
            XEQ "INMULT"
RATE = ?
          18.000
FINAL YR = ?
           5.000

RATE=18.00 %

YR.  1:  1.180
YR.  2:  1.392
YR.  3:  1.643
YR.  4:  1.939
YR.  5:  2.288


            XEQ "INMULT"
RATE = ?
          19.000
FINAL YR = ?
           3.000

RATE=19.00 %

YR.  1:  1.190
YR.  2:  1.416
YR.  3:  1.685
```

F2. Sample printer output--option 2 (flag 00 set), with flag 01 clear, then flag 01 set (2A), and then also with flag 02 set (2B).


```
EXAMPLE: OPTION 2

                       SF 00
                       CF 01
              XEQ "INMULT"
INPUT RATE:
 BB.EEEII
              6.008    RUN
INPUT YEARS:
 BB.EEEII
           1.00502    RUN

RATE=6.00 %

YR.--(1+I)↑N
1       1.060
                      RUN
3       1.191
                      RUN
5       1.338
                      RUN

RATE=7.00 %

YR.--(1+I)↑N
1       1.070
                      RUN
3       1.225
                      RUN
5       1.403
                      RUN

RATE=8.00 %

YR.--(1+I)↑N
1       1.080
                      RUN
3       1.260
                      RUN
5       1.469
                      RUN
```

```
EXAMPLE: OPTION 2A

                       SF 00
                       SF 01
              XEQ "INMULT"
INPUT RATE:
 BB.EEEII
              6.008    RUN
INPUT YEARS:
 BB.EEEII
           1.00502    RUN

RATE=6.00 %

YR.--(1+I)↑N
1       1.060
3       1.191
5       1.338

RATE=7.00 %

YR.--(1+I)↑N
1       1.070
3       1.225
5       1.403

RATE=8.00 %

YR.--(1+I)↑N
1       1.080
3       1.260
5       1.469
```

```
EXAMPLE: OPTION 2B
                       SF 00
                       SF 01
                       SF 02
              XEQ "INMULT"
INPUT RATE:
 BB.EEEII
              6.007    RUN
INPUT YEARS:
 BB.EEEII
           1.00502    RUN

RATE=6.00 %

YR.--(1+I)↑N
1       1.060
3       1.191
5       1.338

RATE=6.50 %

YR.--(1+I)↑N
1       1.065
3       1.208
5       1.370

RATE=7.00 %

YR.--(1+I)↑N
1       1.070
3       1.225
5       1.403

RATE=7.50 %

YR.--(1+I)↑N
1       1.075
3       1.242
5       1.436
```

Retrospective Comments Regarding INMULT

1. INMULT has certain inefficiencies similar to those discussed for the RANCHEK program, but in general it is fairly efficient. The following minor changes might be made:

   a. Since the "F22" routine is part of the INMULT program, the global label designation is not needed, therefore it could be changed to "LBL 22"; that is

      (1) change steps 06 and 15 to XEQ 22
      (2) change step 56 to LBL 22

   b. Delete step 146 since both RTN and END are not needed to terminate label 05.

Program No. 41F011

Calculator: HP-41C/CV

Program Name: SLOPE

Author: T.W. Beers

Date: January 1981

Purpose: To calculate the horizontal distance and, optionally, the vertical rise from a given slope distance and slope angle in percent, degrees (flag 01 set) or topographic units (flag 02 set). (Refer to "Slope to Horizontal Distance" program in Rocky Mtn. Gen. Tech. Rep. RM-76 by Shepperd.)

A. Storage assignments

Register	Use
01	keyed-in slope distance
02	keyed-in angle, then angle in degrees
03	calculated horizontal distance

B. Labels

Global	Use
SLOPE	start of program, conversion of percent to degree measure
Local	
a	calculation of vertical rise (after calculation of horizontal distance)
01	calculation and display of horizontal distance
02	conversion of topo to degree measure

C. Flags

Number	Use
none set	assumes angle is in percent
01	if set, assumes angle is in degrees

C. Flags *(Continued)*

Number	Use
02	if set, assumes angle is in topographic units

D. Program procedure and example

I. In PRGM mode, load program "SLOPE"
(SIZE needed is 004)

II. In RUN mode:

1. Check flags for the three options:

a. none set implies angle in percent
b. 01 set implies angle in degrees
c. 02 set implies angle in topo

2. XEQ SLOPE (suggest assignment to 1/X) and follow the prompts.

	Prompt	Input	Key	Output	Example Input	Example Key	Example Output
a.	DIST.↑∢?	slope distance	↑	slope distance	100	↑	100.00
		slope percent	R/S	H.DIST.=XX.XX	10	R/S	H.DIST.= 99.50
		-	fA*	V.RISE=XX.XX	-	fA*	V.RISE= 9.95

*the "f" is to be interpreted as the shift (gold) key, and the underline implies USER mode.

b. If flag 01 is set the input angle will be in degrees and the example:

Input	Key	Output
100	↑	100.00
10	R/S	H.DIST.=98.48
-	fA	V.RISE= 17.36

c. If flag 02 is set the input angle will be in topo and the example:

Input	Key	Output
100	↑	100.00
10	R/S	H.DIST.= 98.87
-	fA	V.RISE= 14.98

3. General comments:

a. After calculation of either horizontal distance or vertical rise, a depression of R/S will recover the distance-angle prompt for the next set of data.

E. Program Listing and Printer Output

```
01♦LBL "SLOPE"
02 "DIST. ↑ ∡?"
03 PROMPT
04 STO 02
05 X<>Y
06 STO 01
07 FS? 01
08 GTO 01
09 FS? 02
10 GTO 02
11 RCL 02
12 100
13 /
14 ATAN
15 STO 02
16♦LBL 01
17 RCL 02
18 COS
19 RCL 01
20 *
21 FIX 2
22 STO 03
23 "H. DIST.= "
24 ARCL X
25 AVIEW
26 STOP
27 GTO "SLOPE"
28♦LBL a
29 RCL 02
30 TAN
31 RCL 03
32 *
33 "V. RISE= "
34 ARCL X
35 AVIEW
36 STOP
37 GTO "SLOPE"
38♦LBL 02
39 RCL 02
40 66
41 /
42 ATAN
43 STO 02
44 GTO 01
45 RTN
46 END

SLOPE:109 BYTES
```

```
                 XEQ "SLOPE"
DIST. ↑ ∡?
              100.00 ENTER↑
               10.00    RUN
H. DIST.= 99.50
                      XEQ a
V. RISE= 9.95


                      SF 01
                XEQ "SLOPE"
DIST. ↑ ∡?
              100.00 ENTER↑
               10.00    RUN
H. DIST.= 98.48
                      XEQ a
V. RISE= 17.36


                      CF 01
                      SF 02
                XEQ "SLOPE"
DIST. ↑ ∡?
              100.00 ENTER↑
               10.00    RUN
H. DIST.= 98.87
                      XEQ a
V. RISE= 14.98
```

F. Formulas used.

1. Angle conversion (assuming DEG mode):

angles input in percent or topographic units are converted to degrees by

$$\text{degrees} = \arctan\left(\frac{\%}{100}\right)$$

or

$$\text{degrees} = \arctan\left(\frac{\text{topo}}{66}\right)$$

2. Horizontal distance (HD) and vertical rise (VR):

the HD and VR calculations make use of basic trigonometric functions;

$$HD = (\text{slope distance})\ (\cos S),$$

and

$$VR = (HD)\ (\tan S),$$

where S = slope angle in degrees.

Program No. 41F012

Calculator: HP-41C/CV

Program Name: BA (for basal area)

Author: T.W. Beers

Date: January 1981

Purpose: To calculate tree basal area in square feet from a keyed-in diameter in inches or vice versa (with flag 01 set) and to summarize a series of such calculations, providing the arithmetic mean diameter, the arithmetic mean basal area, the quadratic mean diameter, the number of entries and the sums and standard deviations of both diameter and basal area. Metric output (basal area in m^2, diameters in cm.) from similar input, can be obtained with flag 02 set. (Refer to "Basal Area Computation" program in Rocky Mtn. Gen. Tech. Rep. RM-76 by Shepperd).

A. Storage assignments

Register	Use
00	internal storage of the constant .005454 or .00007854 (with flag 02 set)
01	individual basal area
02	individual diameter
03	arithmetic mean basal area
04	arithmetic mean diameter
05	quadratic mean diameter
06	standard deviation of basal area
07	standard deviation of diameter
11-16	statistical summaries; ΣB, ΣB^2, ΣD, ΣD^2, ΣBD, n

B. Labels

Global	Use
BA	start of program, initialization steps and initial prompt
Local	
a	calculation and display of $\overline{B}$, $\overline{D}$, QMD, and number of observations
b	recall and display of sums of diameters and basal areas
c	calculation and display of standard deviation of diameter and basal area
01	when flag 01 is set, calculation and display of diameter from keyed-in basal area
02	calculation and display of basal area from keyed-in diameter

C. Flags

Number	Use
none set	assumes U.S. units and diameter as input
01	if set, assumes basal area as input. Set internally if initial input is negative (indicating B is to be converted to D).
02	if set, assumes metric units (input and output)

D. Program procedure and example

I. In PRGM mode, load program "BA"
(SIZE needed is 017)

II. In RUN mode:

1. Check flags for the various options:

   a. none set implies input of diameter in inches and output of basal area in sq. ft.

   b. 01 set implies input of basal area, output of diameter

   c. 02 set implies metric units-- centimeters for diameter, sq. meters for basal area

2. XEQ BA (suggest assignment to $\sqrt{X}$) and follow the prompts. (Assume no flags are set).

	Prompt	Input	Key	Output	Example Input	Example Key	Example Output
a.	KEY D OR-BA	diameter	R/S	1 (pause)	10	R/S	1 then B.A.= 0.545
				B.A.=X.XXX			
	NEXT DIAM?	diameter	R/S	2 (pause)	12	R/S	2 then B.A.=0.785
		.		B.A.=X.XXX			
		.					
		for all observations			14	R/S	3 then B.A.=1.069
b.			fA*	AVE.D.=X.XX		fA*	AVE.D.=12.00
				AVE.B.A.=X.XXX			AVE.B.A.=0.800
				Q.M.D.=X.XX			Q.M.D.=12.11
				NUMBER=X			NUMBER=3
c.			fB	ΣD=X.XX		fB	ΣD=36.00
				ΣBA=X.XXX			ΣBA=2.400
d.			fC	S.D.(D)=X.XX		fC	S.D.(D)=2.00
				S.D.(B)=X.XXX			S.D.(B)=0.262

*the "f" is to be interpreted as the shift (gold) key, and the underline implies USER mode.

3. General comments:

   a. Examples of "BA" executed with flags 01 or 02 set are shown as printer output in section E.

   b. For repeat viewing of the various answers shown in 2b through 2d the user can do one or more of the following:

      (1) repeat depression of fA, fB, or fC;

      (2) recall the appropriate storage register manually;

      (3) replace the PSE statements in the program with R/S statements, which will necessitate pressing the R/S key between answers.

E. Program Listing and Printer Output


```
01♦LBL "BA"
02 FIX 2
03 CLΣ
04 .005454
05 FS? 02
06 .00007854
07 STO 00
08 "KEY D OR -BA"
09 PROMPT
10 X<0?
11 SF 01
12 FS? 01
13 GTO 01
14♦LBL 02
15 STO 02
16 X↑2
17 RCL 00
18 *
19 STO 01
20 RCL 02
21 X<>Y
22 Σ+
23 FIX 0
24 PSE
25 FIX 3
26 "B.A.= "
27 ARCL 01
28 AVIEW
29 PSE
30 PSE
31 FIX 2
32 "NEXT DIAM.?"
33 PROMPT
34 GTO 02
35♦LBL a
36 MEAN
37 STO 03
38 X<>Y
39 STO 04
40 RCL 03
41 RCL 00
42 /
43 SQRT
44 STO 05
45 FIX 2
46 "AVE. D.= "
47 ARCL 04
48 AVIEW
49 PSE
50 FIX 3
51 "AVE. BA= "
52 ARCL 03
53 AVIEW
54 PSE
55 FIX 2
56 "Q.M.D.= "
57 ARCL 05
58 AVIEW
59 PSE
60 FIX 0
61 "NUMBER= "
62 ARCL 16
63 AVIEW
64 RTN
65♦LBL 01
66 ABS
67 STO 01
68 RCL 00
69 /
70 SQRT
71 STO 02
72 RCL 01
73 Σ+
74 FIX 0
75 PSE
76 FIX 2
77 "DIAM.= "
78 ARCL 02
79 AVIEW
80 PSE
81 PSE
82 "NEXT B.A.?"
83 PROMPT
84 GTO 01
85 RTN
86♦LBL b
87 FIX 2
88 "ΣD= "
89 ARCL 13
90 AVIEW
91 PSE
92 FIX 3
93 "ΣBA= "
94 ARCL 11
95 AVIEW
96 RTN
97♦LBL c
98 FIX 2
99 SDEV
100 STO 06
101 X<>Y
102 STO 07
103 "S.D. <D>= "
104 ARCL X
105 AVIEW
106 PSE
107 FIX 3
108 "S.D. <B>= "
109 ARCL Y
110 AVIEW
111 END

BA:274 BYTES
```

E. Printer Output (Continued)


```
                XEQ "BA"
KEY D OR -BA
            10.00    RUN
B.A.= 0.545
NEXT DIAM.?
            12.00    RUN
B.A.= 0.785
NEXT DIAM.?
            14.00    RUN
B.A.= 1.069
NEXT DIAM.?

                   XEQ a
AVE. D.= 12.00
AVE. BA= 0.800
Q.M.D.= 12.11
NUMBER= 3
                   XEQ b
ΣD= 36.00
ΣBA= 2.400
                   XEQ c
S.D. <D>= 2.00
S.D. <B>= 0.262
```

```
                   CF 01
                   SF 02
                XEQ "BA"
KEY D OR -BA
            30.00    RUN
B.A.= 0.071
NEXT DIAM.?
            40.00    RUN
B.A.= 0.126
NEXT DIAM.?
            50.00    RUN
B.A.= 0.196
NEXT DIAM.?

                   XEQ a
AVE. D.= 40.00
AVE. BA= 0.131
Q.M.D.= 40.82
NUMBER= 3
                   XEQ b
ΣD= 120.00
ΣBA= 0.393
                   XEQ c
S.D. <D>= 10.00
S.D. <B>= 0.063
```

```
                XEQ "BA"
KEY D OR -BA
            -.545    RUN
DIAM.= 10.00
NEXT B.A.?
             .785    RUN
DIAM.= 12.00
NEXT B.A.?
            1.069    RUN
DIAM.= 14.00
NEXT B.A.?

                   XEQ a
AVE. D.= 12.00
AVE. BA= 0.800
Q.M.D.= 12.11
NUMBER= 3
                   XEQ b
ΣD= 35.99
ΣBA= 2.399
                   XEQ c
S.D. <D>= 2.00
S.D. <B>= 0.262
```

```
                XEQ "BA"
KEY D OR -BA
            -.071    RUN
DIAM.= 30.07
NEXT B.A.?
             .126    RUN
DIAM.= 40.05
NEXT B.A.?
             .196    RUN
DIAM.= 49.96
NEXT B.A.?

                   XEQ a
AVE. D.= 40.03
AVE. BA= 0.131
Q.M.D.= 40.84
NUMBER= 3
                   XEQ b
ΣD= 120.08
ΣBA= 0.393
                   XEQ c
S.D. <D>= 9.94
S.D. <B>= 0.063
```

F. Formulas used

1. Basal area, B, and diameter, D, are calculated for U.S. units, by

$$B = .005454D^2 \quad \text{or the inverse, } D = \sqrt{\frac{B}{.005454}}$$

For metric units, the constant is replaced by .00007854

2. The quadratic mean diameter, Q.M.D., (i.e., the diameter corresponding to the arithmetic mean basal area) is calculated using the inverse form of the above equation with the arithmetic mean basal area, $\overline{B}$ substituted for B:

$$\text{Q.M.D.} = \sqrt{\frac{\overline{B}}{.005454}} \qquad \text{(U.S. units)}$$

or

$$\text{Q.M.D.} = \sqrt{\frac{\overline{B}}{.00007854}} \qquad \text{(metric units)}$$

3. Arithmetic means, $\overline{D}$, and $\overline{B}$, and the standard deviations, s_D and s_B are "calculated" using the HP defined functions "MEAN" and "SDEV", Respectively. The formulas used therein are:

$$\overline{X} = \frac{\Sigma X}{n}$$

$$s = \sqrt{\frac{\Sigma X^2 - (\Sigma X)^2/n}{n - 1}}$$

Retrospective Comments Regarding BA

1. To ensure that the statistical registers are 11 through 16 as assumed in the program, the command ΣREG 11 should be inserted immediately after step 01 (LBL "BA"). Also, to avoid the decimal point appearing in integer output (FIX 0), the instruction CF 29 should be placed early in the program.

Program No. 41F013

Calculator: HP-41C/CV

Program Name: TH (for tree height)

Author: T.W. Beers

Date: January 1981

Purpose: To calculate tree height when horizontal distance is not measured and the clinometer will not read directly. Therefore, tree height is calculated from keyed-in angles to tree top and base, and slope distance. Provision is made for angles in percent, degrees (flag 01 set) or topographic units (flag 02 set). (Refer to "Tree Heights" program in Rocky Mtn. Gen. Tech. Rep. RM-76 by Shepperd.)

A Storage assignments

Register	Use
01	keyed-in angle to top, then angle in degrees (B_t)
02	keyed-in angle to base, then angle in degrees (B_b)
03	angle sum, $B = B_t+B_b$ or $B = B_t-B_b$ (flag 00 set)
04	slope distance

B. Labels

Global	Use
TH	start of program
Local	
01	prompt for slope distance; calculation and display of tree height
15	prompt for top and base angles; conversion of angles to degrees if needed

C. Flags

Number	Use
none set	assumes angles are physically different in sign and are expressed in percent
00	if set, implies angles are physically the same sign
01	if set, assumes angles in degrees
02	if set, assumes angles in topo scale

D. Program procedure and example

I. In PRGM mode, load program "TH"
(SIZE needed is 005)

II. In RUN mode:

1. Check flags for the various conditions

   a. none set assumes angles in percent
   b. 00 set assumes angles are same sign
   c. 01 set assumes angles are in degrees
   d. 02 set assumes angles are in topo scale

2. XEQ TH (suggest assignment to LOG) and follow the prompts

	Prompt	Input	Key	Output	Example Input	Example Key	Example Output
a.	TOP∢↑BASE∢	∢↑∢	R/S	-	40↑20	R/S	-
	SLOPE DISTANCE?	distance	R/S	HEIGHT=X.XX	56	R/S	HEIGHT=32.9

b. If both angles are positive or both are negative, flag 00 can be set externally <u>or</u> slope distance can be keyed as a negative value.

Input	Key	Output
40↑20	R/S	-
56	CHS	-56
-	R/S	HEIGHT=11.0

c. Assume angles are in degrees, then SF 01 and proceed.

Input	Key	Output
21.8↑11.3	R/S	-
56	R/S	HEIGHT=32.9

d. Assume angles are in topo units, then SF 02 and proceed.

Input	Key	Output
26.4↑13.2	R/S	-
56	R/S	HEIGHT=32.9

D. Program procedure and example *(Continued)*

III. General comments:

1. In the usual application, "TH" will be executed, then the user will consistently be using one of procedures 2a, c, or d. Therefore once one of these is chosen, the proper flag set (or cleared), and the first tree height calculated, pressing R/S will recover the prompt for keying the next set of angles.

2. Use of procedure 2b will occur rarely, therefore after use, pressing R/S will automatically clear flag 00.

3. Special note: observe that for simplicity angles are always keyed-in as positive numbers, although it is recognized that the angles actually measured are usually different in sign.

E. Program Listing and Printer Output

```
01♦LBL "TH"
02 "FLAGS SET?"
03 AVIEW
04 PSE
05 " %: NONE"
06 AVIEW
07 SIN
08 "DEGREES: 01"
09 AVIEW
10 SIN
11 "TOPO: 02"
12 AVIEW
13 SIN
14♦LBL 15
15 "TOP∡ ↑ BASE∡"
16 PROMPT
17 STO 02
18 X<>Y
19 STO 01
20 FS? 01
21 GTO 01
22 FS? 02
23 66
24 FC? 02
25 100
26 /
27 ATAN
28 STO 01
29 X<>Y
30 FS? 02
31 66
32 FC? 02
33 100
34 /
35 ATAN
36 STO 02
37♦LBL 01
38 "SLOPE DIST.?"
39 PROMPT
40 X<0?
41 SF 00
42 ABS
43 STO 04
44 RCL 01
45 RCL 02
46 FS? 00
47 CHS
48 +
49 STO 03
50 SIN
51 RCL 04
52 *
53 90
54 RCL 01
55 -
56 SIN
57 /
58 FIX 1
59 "HEIGHT= "
60 ARCL X
61 AVIEW
62 RTN
63 CF 00
64 GTO 15
65 END

TH:162 BYTES
```

```
                    XEQ "TH"
FLAGS SET?
 %: NONE
DEGREES: 01
TOPO: 02
TOP∡ ↑ BASE∡
             40.0 ENTER↑
             20.0     RUN
SLOPE DIST.?
             56.0     RUN
HEIGHT= 32.9
```

```
                         RUN
TOP∡ ↑ BASE∡
             40.0 ENTER↑
             20.0     RUN
SLOPE DIST.?
            -56.0     RUN
HEIGHT= 11.0
```

```
                    XEQ "TH"
FLAGS SET?
 %: NONE
DEGREES: 01
TOPO: 02
TOP∡ ↑ BASE∡
                      SF 01
             21.8 ENTER↑
             11.3     RUN
SLOPE DIST.?
             56.0     RUN
HEIGHT= 11.0
```

```
                         RUN
TOP∡ ↑ BASE∡
             21.8 ENTER↑
             11.3     RUN
SLOPE DIST.?
            -56.0     RUN
HEIGHT= 11.0
```

```
                    XEQ "TH"
FLAGS SET?
 %: NONE
DEGREES: 01
TOPO: 02
TOP∡ ↑ BASE∡
                      SF 02
             26.4 ENTER↑
             13.2     RUN
SLOPE DIST.?
             56.0     RUN
HEIGHT= 32.9
```

```
                         RUN
TOP∡ ↑ BASE∡
             26.4 ENTER↑
             13.2     RUN
SLOPE DIST.?
            -56.0     RUN
HEIGHT= 11.0
```

F. Formulas used

1. Angle conversion (assuming DEG mode)

angles input in percent or topographic scale are converted to degrees by

$$\text{degrees} = \arctan\left(\frac{\%}{100}\right)$$

or

$$\text{degrees} = \arctan\left(\frac{\text{topo}}{66}\right)$$

2. Tree height calculation

a. Usual case; where slope signs are opposite:



$$\text{tree height, } b = \frac{a \sin(B_t+B_b)}{\sin(90-B_t)}$$

b. Abnormal case; where slope signs are alike:



$$\text{tree height, } b = \frac{a \sin(B_t-B_b)}{\sin(90-B_t)}$$

F. Formulas used *(Continued)*

c. The formulas cited above were used in the program "TH" to conform with those in U.S.F.S. Gen. Tech. Rep. RM-76. A more "traditional" approach would be to use the following:



since $\tan B_t = \frac{t}{H}$ and $\tan B_b = \frac{b}{H}$

then $h = t+b = H(\tan B_t + \tan B_b)$

or, since $\cos B_b = \frac{H}{S}$

$$h = S \cos B_b (\tan B_t + \tan B_b)$$

This approach makes clear the practical simplicity of using the percent scale when assessing tree height, for, by definition $\tan X = \frac{X(\%)}{100}$, therefore:

if $H = 100$

$$h = 100 \left[\frac{B_t(\%)}{100} + \frac{B_b(\%)}{100}\right]$$

$$= B_t(\%) + B_b(\%)$$

or if $H \neq 100$

$$h = \frac{H}{100} \left[B_t(\%) + B_b(\%)\right]$$

Retrospective Comments Regarding TH

1. Although the flag setting reminders are helpful, one eventually tires of seeing them each time TH is executed. Thus, the option to bypass their display can be provided with a flag setting (say flag 03) by inserting the following instructions just after step 01 (LBL "TH"):

   ```
   FS? 03
   GTO 15
   ```

2. In similar fashion, the basic prompt for angle input can be obtained automatically after each height "answer" by inserting the following three statements after step 61 (AVIEW):

   ```
   FS? 03
   PSE
   FC? 03
   ```

3. The program called TH1, found on the KRON-1 tape, incorporates both of the above refinements.

Program No. 41F014

Calculator: HP-41C/CV

Program Name: NTEST (sample size determination)

Author: T.W. Beers

Date: January 1981

Purpose: For a keyed-in preliminary sample of X_i, this program will calculate and display the preliminary sample size, n_p, sum, mean, standard deviation, standard error, variance, and coefficient of variation. After an arbitrary allowable error is keyed-in, the sample size, n, required to achieve this error is calculated and displayed. An infinite population is assumed but the finite case can be handled by setting flag 00. (Refer to "Adequacy of Sample Test" program in Rocky Mtn. Gen. Tech. Rep. RM-76 by Shepperd.)

A. Storage assignments

Register	Use	
01	n_p, sample size	determined from preliminary sample
02	ΣX, sum of X	
03	$\overline{X}$, arithmetic mean	
04	s, standard deviation	
05	$s_{\overline{X}}$, standard error	
06	s^2, variance	
07	CV, coefficient of variation = $\frac{s}{\overline{X}}$ (100)	
08	AE, allowable error in %	
09	n, calculated sample size	
10	not used	
11-16	statistical registers	

B. Labels

Global	Use
NTEST	start of program, prompt for first X
Local	
a	calculates and displays n_p, ΣX, $\overline{X}$, and s
b	prompts for allowable error and Student's t, then calculates required sample size n.
00	adjustment for finite population
01	summarization loop
03	calculates and displays $s_{\overline{X}}$, s^2, and CV
04	prompt for new value of t
05	calculating loop in label b

C. Flags

Number	Use
00	if set, implies sampling from finite population
22	data-entry flag, used to detect keyboard entry of t in label b
29	digit-grouping flag, cleared to suppress decimal point and comma in FIX 0 status

D. Program procedure and example

I. In PRGM mode, load program "NTEST"
(Size needed is 017)

II. In RUN mode:

Basic assumption is for sampling with replacement (i.e., infinite population); for sample size determination assuming sampling without replacement (i.e., finite population) flag 00 must be set prior to step lc <u>or</u> the allowable error can be keyed-in as a negative value.

1. XEQ NTEST (suggest assignment to LN) and follow the prompts.

	Prompt	Input	Key	Output	Example Input	Example Key	Example Output
a.	EACH X,R/S	X_1	R/S	1	10	R/S	1
		X_2	R/S	2	20	R/S	2
			⋮		15	R/S	3
		X_{n_p}	R/S	n_p	18	R/S	4
					12	R/S	5
b.	-	-	a	NO.OBS.=X	-	a	NO.OBS.=5
				ΣX=X.X	-		ΣX=75.0
				X BAR=X.XX	-		X BAR=15.00
				S. DEV.=X.XX	-		S.DEV.=4.12
	R/S FOR MORE				-		
			R/S	S X BAR=X.XX	-	R/S	S X BAR=1.84
				S↑2=X.XX	-		S↑2=17.00
				C.V.=X.XX%	-		C.V.=27.5%
c.	PRESS b	-	b		-	b	
	A.E. IN %?	AE(%)[1/]	R/S	-	20[1/]	R/S	-
	KEY T OR R/S	-[2/]	R/S	N=XX	-[2/]	R/S	N=8
			R/S		-	R/S	
	NEW T?	new t	R/S	N=XX	2.365	R/S	N=11
						R/S	
					2.228	R/S	N=9
						R/S	
					2.306	R/S	N=10
						R/S	
					2.262	R/S	N=10

[1/] if AE(%) is negative, this automatically sets Flag 00, implying finite population

[2/] if no t is keyed-in, default value of t=2 is used, implying .05 level

D. Program procedure and example *(Continued)*

2. General comments

   a. As shown in the above example one can use a refined t value and repeat the solution until N settles down. Note that the program assumes initially that t = 2, unless otherwise keyed in step 1c.

   b. The preliminary sample is always assumed to be from an infinite population, therefore the standard error is uncorrected.

E. Program Listing and Printer Output


```
01◆LBL "NTEST"
02 CLΣ
03 FIX 0
04 CF 29
05 "EACH X, R/S"
06 PROMPT
07◆LBL 01
08 Σ+
09 STOP
10 GTO 01
11◆LBL a
12 CF 29
13 "NO. OBS.= "
14 ARCL 16
15 AVIEW
16 PSE
17 STO 01
18 FIX 1
19 "ΣX= "
20 ARCL 11
21 AVIEW
22 PSE
23 STO 02
24 FIX 2
25 MEAN
26 STO 03
27 "X BAR= "
28 ARCL X
29 AVIEW
30 PSE
31 SDEV
32 STO 04
33 "S.DEV.= "
34 ARCL X
35 AVIEW
36 PSE
37 "R/S FOR MORE"
38 AVIEW
39 STOP
40◆LBL 03
41 FIX 2
42 RCL 04
43 RCL 16
44 SQRT
45 /
46 STO 05
47 "S XBAR= "
48 ARCL X
49 AVIEW
50 PSE
51 RCL 04
52 X↑2
53 STO 06
54 "S↑2= "
55 ARCL X
56 AVIEW
57 PSE
58 FIX 1
59 RCL 04
60 RCL 03
61 /
62 100
63 *
64 STO 07
65 "C.V.= "
66 ARCL X
67 "├ %"
68 AVIEW
69 PSE
70 "PRESS b"
71 PROMPT
72◆LBL b
73 FIX 0
74 "A.E. IN %?"
75 PROMPT
76 X<0?
77 SF 00
78 ABS
79 STO 08
80 CF 22
81 "KEY T OR R/S"
82 PROMPT
83 FS? 22
84 X↑2
85 FC? 22
86 4
87◆LBL 05
88 RCL 07
89 X↑2
90 *
91 RCL 08
92 X↑2
93 /
94 FS? 00
95 XEQ 00
96 "N = "
97 ARCL X
98 AVIEW
99 BEEP
100 STOP
101 GTO 04
102◆LBL 00
103 STO 09
104 "POP. SIZE ?"
105 PROMPT
106 /
107 1
108 +
109 RCL 09
110 X<>Y
111 /
112 RTN
113◆LBL 04
114 "NEW T ?"
115 PROMPT
116 X↑2
117 GTO 05
118 END

NTEST:284 BYTES
```

E. Printer Output (Continued)


```
            XEQ "NTEST"
EACH X, R/S
               10    RUN
               20    RUN
               15    RUN
               18    RUN
               12    RUN
                   XEQ a
NO. OBS.= 5
ΣX= 75.0
X BAR= 15.00
S.DEV.= 4.12
R/S FOR MORE
                     RUN
S XBAR= 1.84
S↑2= 17.00
C.V.= 27.5 %
PRESS b
                   XEQ b
A.E. IN %?
               20    RUN
KEY T OR R/S
                     RUN
N = 8
                     RUN
NEW T ?
            2.365    RUN
N = 11
                     RUN
NEW T ?
            2.228    RUN
N = 9
                     RUN
NEW T ?
            2.306    RUN
N = 10
                     RUN
NEW T ?
            2.262    RUN
N = 10
```

Example using finite population

```
                   XEQ b
A.E. IN %?
             -7.5    RUN
KEY T OR R/S
                     RUN
POP. SIZE ?
               50    RUN
N = 26
                     RUN
NEW T ?
            2.060    RUN
POP. SIZE ?
               50    RUN
N = 27
                     RUN
NEW T ?
            2.056    RUN
POP. SIZE ?
               50    RUN
N = 27
```

F. Formulas used

1. For the preliminary sample.

$$\text{X BAR} = \frac{\Sigma X}{n}$$

$$\text{S.DEV.} = \sqrt{\frac{n\Sigma X^2 - (\Sigma X)^2}{n(n-1)}} = s$$

$$\text{S X BAR} = \frac{s}{\sqrt{n}}$$

$$s\uparrow 2 = s^2 = \text{variance}$$

$$\text{C.V.} = \frac{s}{\bar{X}}(100)$$

2. For sample size determination.

a. Infinite population:

$$n = \frac{t^2(\text{C.V.})^2}{(\text{A.E.})^2}$$

where both C.V. and A.E. are in percent

and t = Student's t

b. Finite population:

$$n_{adj} = \frac{n}{1+\frac{n}{N}}$$

where

n = calculated as in 2a

and

N = population size

Retrospective Comments Regarding NTEST

1. The program can be made more smooth in the iteration phase (i.e., where a refined Student's t is to be input) by deleting step 100 (STOP) and inserting one or two pauses (PSE), depending upon how long one needs to view the current value of n.

2. Since registers 11 through 16 are assumed to be the statistical registers, ΣREG 11 should be inserted immediately after step 01 (LBL "NTEST").

3. In the finite population case, label 00, as written, requires the re-entry of population size at each iteration; in order to avoid this, one can use flag 02, label 02, and storage 10 and make the following changes:

   a. Insert CF 02 after step 72 (LBL b)

   b. Insert FS? 02
      GTO 02 after step 103 (STO 09)

   c. Insert STO 10
      X≷Y
      LBL 02
      RCL 10 after step 105 (PROMPT)

   d. Insert SF 02 after step 113 (LBL 04)

4. All of the above changes are incorporated into the program NTEST1, found on the KRON-1 tape.

Program No. 41F015

Calculator: HP-41C/CV

Program Name: GCAL (Gauge calibration)

Author: T.W. Beers

Date: January 1981

Purpose: Angle gauges used in horizontal point sampling are "calibrated," or sometimes designed or assembled making use of the functional relation between basal area factor of the gauge, F, width of a sighting bar, W, and distance, D, from eye to "target." This program solves the relationship for any one of the three variables, using the other two as input; the determination of the basal area factor of wedge prisms is also possible. The use of metric units is enabled with flag 00 set. (Refer to "BAF" Gauge Calibration program in Rocky Mtn. Gen. Tech. Rep. RM-76 by shepperd.)

A. Storage assignments

Register	Use
00	internally stored constant, 43560 or 10000 (flag 00 set)

B. Labels

Global	Use
GCAL	start of program, prompts for the 3 options, and stores constants

Local	
a	calculates F, from prompted W and D
b	calculates W, from prompted F and D
c	calculates D, from prompted F and W

C. Flags

Number	Use
00	if set, metric units are assumed and are implied in output; also in label a, conversion of F to sq. meters is enabled.

D. Program procedure and example

I. In PRGM mode, load program "GCAL"
(SIZE needed is 001)

II. In RUN mode:

1. XEQ GCAL (suggest assignment to X≶Y) and follow the prompts (choosing a, b, or c)

	Prompt	Input	Key	Output	*Example* Input	Key	Output
a.	KEY a,b, OR c	-	a	-	-	a	-
	BAR WIDTH?	W	R/S	-	1	R/S	-
	DISTANCE?	D	R/S	BAF=X.XX	33	R/S	BAF=10.00
	-	-	R/S	=X.XX SQ.M.	-	R/S	=2.30 SQ.M./A
b.	KEY a,b,OR c	-	b	-	-	b	-
	BAF?	F	R/S	-	10	R/S	-
	DISTANCE?	D	R/S	WIDTH=X.XX	33	R/S	WIDTH=1.00
c.	KEY a,b,OR c	-	c	-	-	c	-
	BAF?	F	R/S	-	10	R/S	-
	WIDTH?	W	R/S	DIST.=X.XX	1	R/S	DIST.=33.00

2. If input units are metric, SF 00, XEQ GCAL (to load proper constant) and proceed as above; an example for 1a would be:

Prompt	Input	Key	Output
KEY a,b, OR c	-	a	-
BAR WIDTH	2	R/S	-
DISTANCE?	70.7	R/S	BAF=2.00
		R/S	=8.71 SQ.FT./A

III. General comments:

1. After label a,b, or c has been selected, pressed, and the first pair of data has been processed, depression of R/S will keep the user within that routine, and the appropriate prompts will be repeated.

2. For input in U.S. units, F is assumed to be square feet per acre; W and D can be either feet or inches but both must be in the same units. For metric input, F is assumed to be square meters per hectare, while W and D can be either cm. or meters (both the same).

E. Program Listing and Printer Output

```
01♦LBL "GCAL"
02 FIX 2
03 "BAF?  KEY a"
04 AVIEW
05 SIN
06 SIN
07 "WIDTH? KEY b"
08 AVIEW
09 SIN
10 SIN
11 "DIST.? KEY c"
12 AVIEW
13 SIN
14 SIN
15 43560
16 FS? 00
17 10000
18 STO 00
19 "KEY a, b, OR c"
20 PROMPT
21♦LBL a
22 "BAR WIDTH?"
23 PROMPT
24 "DISTANCE?"
25 PROMPT
26 /
27 2
28 /
29 ATAN
30 SIN
31 X↑2
32 RCL 00
33 *
34 "BAF= "
35 ARCL X
36 AVIEW
37 STOP
38 .2295684
39 FS? 00
40 1/X
41 *
42 "= "
43 ARCL X
44 FS? 00
45 "├ SQ.FT./A"
46 FC? 00
47 "├ SQ.M./H"
48 AVIEW
49 STOP
50 GTO a
```

```
51♦LBL b
52 "BAF?"
53 PROMPT
54 "DISTANCE?"
55 PROMPT
56 X<>Y
57 RCL 00
58 /
59 SQRT
60 ASIN
61 TAN
62 *
63 2
64 *
65 "WIDTH= "
66 ARCL X
67 AVIEW
68 STOP
69 GTO b
70♦LBL c
71 "BAF?"
72 PROMPT
73 "WIDTH?"
74 PROMPT
75 2
76 /
77 X<>Y
78 RCL 00
79 /
80 SQRT
81 ASIN
82 TAN
83 /
84 "DIST.= "
85 ARCL X
86 AVIEW
87 STOP
88 GTO c
89 END

GCAL:262 BYTES
```

```
                 XEQ "GCAL"
BAF?  KEY a
WIDTH? KEY b
DIST.? KEY c
KEY a, b, OR c
                      XEQ a
BAR WIDTH?
                1.00    RUN
DISTANCE?
               33.00    RUN
BAF= 10.00
                        RUN
= 2.30 SQ.M./H

                      XEQ b
BAF?
               10.00    RUN
DISTANCE?
               33.00    RUN
WIDTH= 1.00

                      XEQ c
BAF?
               10.00    RUN
WIDTH?
                1.00    RUN
DIST.= 33.00

                      SF 00
                 XEQ "GCAL"
BAF?  KEY a
WIDTH? KEY b
DIST.? KEY c
KEY a, b, OR c
                      XEQ a
BAR WIDTH?
                2.00    RUN
DISTANCE?
               70.70    RUN
BAF= 2.00
                        RUN
= 8.71 SQ.FT./A
```

F. Formulas used

1. Assuming U.S. units as input

a. The basic formula, solved in label a, is

$$F = 43560 \sin^2 A \qquad (1)$$

$$\text{where} \quad A = \arctan \frac{W/2}{D} \qquad (2)$$

note: F = basal area factor
W = bar width or "target" width
D = distance to bar or "target"
$A = \frac{1}{2}$ the angle generated by the gauge



b. When solving for bar width, as in label b, one first finds angle A from

$$A = \arcsin \sqrt{\frac{F}{43560}} \qquad (3)$$

then

$$W = 2\,D(\tan A) \qquad (4)$$

c. When solving for distance, as in label c, angle A is found from formula (3),

then

$$D = \frac{W/2}{\tan A} \qquad (5)$$

2. Assuming metric units as input, the above formulas apply if 43560 is replaced by 10000

3. For conversion of F from U.S. to metric or vice versa the following are appropriate:

$$F_{\text{sq. m. per hectare}} = .2295684\ F_{\text{sq. ft. per acre}}$$

and

$$F_{\text{sq. ft. per acre}} = \frac{1}{.2295684}\ F_{\text{sq. m. per hectare}}$$

Retrospective Comments Regarding GCAL

1. If the initial reminder messages display too rapidly, insertion of an additional SIN command after steps 04, 08, and 12 should produce sufficient slow-down.

Calculator: HP-41C/CV

Program Name: PSFIELD (Point sampling, field)

Author: T.W. Beers

Date: January 1981

Purpose: To solve certain problems relating to field application of horizontal point sampling. For a given basal area factor F, and if needed, tree DBH, D, and a measure of slope, one can obtain:

a. horizontal distance multiplier

b. limiting horizontal distance

c. limiting slope distance

d. calibrated tape mark to hold

e. tree factor

f. associated plot area

g. a borderline tree check, if actual distance to the tree is provided

h. a boundary overlap correction "weight", given the distance to boundary

Slope can be expressed in percent, degrees (flag 01 set), or topographic units (flag 02 set). The technique for slope correction is assumed to be the variable gauge angle approach (see Beers, Jour. For. 67: 188-192); however, with flag 00 set, the constant gauge angle technique is assumed. The entire program can be "made metric" by setting flag 03. (Refer to "Limiting Distance" program in Rocky Mountain Gen. Tech. Rep. RM-76 by Shepperd.)

a. Storage Assignments

Register	Use
00	F, basal area factor

A. Storage Assignments *(continued)*

Register	Use
01	HDM, horizontal distance multiplier
02	D, tree diameter at breast height
03	R, horizontal associated plot radius
04	S, slope angle in degrees
05	R_s, plot radius on the slope
06	D_t, calibrated tape mark to hold
07	F_t, tree factor
08	a, plot area
09	M, slope multiplier (using "constant slope correction")
10	B, horizontally measured distance from point to boundary
11	W, tree "weight" if boundary overlap is corrected by direct weighting method

B. Labels

Global	Use
PSFIELD	start of program, flag reminders, prompt for basal area factor, and calculation of HDM
Local	
a	prompts for actual distance, compares this with plot radius (R or R_s) and displays whether tree is IN or OUT.
b	calculates and displays the "D to hold" mark for use with a calibrated point sampling tape
c	calculates and displays the tree factor and optionally, the associated plot area
d	for use when flag 00 is set; calculates and displays the "adjusted F" and the adjusted tree factor
e	prompts for horizontal distance to the boundary, B, then calculates and displays the tree "weight", adjusted F and adjusted tree factor, F_t.
00	when flag 00 is set, labels and stores horizontal radius as slope radius

Local *(continued)*

01	angle in degrees prompt and storage
02	angle in topo prompt, conversion to degrees and storage
15	DBH prompt, calculates and displays horizontal radius
16	calculates and displays slope radius
17	angle in percent prompt, conversion to degrees and storage
18	when flag 00 set, calculates and displays slope multiplier
19	when flag 03 set, calculates HDM for metric use

C. Flags

Number	Use
None	assumes U.S. input and output; F in sq. ft. per acre, D in inches, R in feet and area in acres. Also, slope angle is assumed to be expressed in percent, and variable gauge angle slope adjustment applies
00	if set, assumes the constant gauge angle slope adjustment technique is used
01	if set, assumes slope angle is in degrees
02	if set, slope angle is in topo units
03	if set, metric input and output; F in sq. m. per hectare, D in cm., R in meters, and area in hectares

D. Program procedure and example

I. In PRGM mode, load program "PSFIELD: (size needed is 010)

II. In RUN mode:

1. Check flags for the appropriate conditions:
    a. none set assumes U.S. units, angle in percent, variable slope
    b. 00 set assumes constant slope adjustment
    c. 01 set assumes angle expressed in degrees

D. Program procedure and example *(Continued)*

d. 02 set assumes angle expressed in topo units

e. 03 set assumes metric units

2. XEQ PSFIELD (suggest assignment to R↓) and follow the prompts, noting first the flag reminders.

a. No flags set

Prompt	Input	Key	Output	Example Input	Example Key	Example Output
F?	F	R/S	HDM=X.XXX	10	R/S	HDM=2.750
DBH?	D	R/S	HOR.RAD=XX.XX	10	R/S	HOR.RAD=27.50
		R/S			R/S	
∢ in %?	α	R/S	SLP.RAD=XX.XX	30	R/S	SLP.RAD=28.71
-	-	a		-	a	
ACTUAL DIST?		R/S	TREE IS IN or TREE IS OUT	28	R/S	TREE IS IN
-	-	b	D TO HOLD=XX.XX	-	b	D TO HOLD=10.44
-	-	c	TREFAC=XX.XX	-	c	TREFAC=18.34
		R/S	AREA=X.XX		R/S	AREA=0.05
-	-	e		-	e	
HORIZ. B?	B	R/S	W=X.XX	20	R/S	W=1.09
		R/S	WTD. F=XX.XX		R/S	WTD. F=10.89
			WTD. F(T)=XX.XX			WTD. F(T)=19.97

b. If flag 01 or flag 02 is set, the procedure is similar to a., except that degrees or topo units are used. See part E for an example of printer output for these cases.

c. If flag 00 is set, the normal procedure is as follows:

Prompt	Input	Key	Output	Example Input	Example Key	Example Output
F?	F	R/S	HDM=X.XX	10	R/S	HDM=2.750
DBH?	D	R/S	SLP.RAD=XX.XX	10	R/S	SLP.RAD=27.50
-	-	R/S		-	R/S	
∢ in %?	α	R/S	MULT.=X.XXX	30	R/S	MULT.=1.044

D. Program procedure and example *(Continued)*

Prompt	Input	Key	Output	Example Input	Example Key	Example Output
-	-	a		-	a	
ACTUAL DIST?		R/S	TREE IS IN or TREE IS OUT	28	R/S	TREE IS OUT
-	-	c	TREFAC=XX.XX	-	c	TREFAC=18.34
-	-	R/S	AREA=X.XX	-	R/S	AREA=0.05
-	-	d	ADJ.F=XX.XX	-	d	ADJ.F=10.44
-	-	R/S	ADJ.F(T)=XX.XX	-	R/S	ADJ.F(T)=19.14

Note: In this constant gauge angle approach, Label b (depression of key b) is not necessary since the tape mark to hold is the same as tree diameter; also, depression of key c provides the unadjusted tree factor and plot area on the slope.

d. If flag 03 is set, the procedures outlined above in a or c are still appropriate; the only difference being the input and output are in metric units. See part E for an example of printer output in this case.

III. General Comments:

1. In general, PSFIELD need only be executed once, enabling either U.S. or metric units mode, and the storage of the basal area factor. After that, the basic prompt DBH? can be obtained by XEQ 15 or by successive depression of R/S since labels a, b, c, d, and e all end in GTO 15 statements.

2. The calculation of an overlap correction weight in label e is recommended only if the mirage method (see Beers, South. Jour. App. For. 1:16-18) is not feasible. Also, use of label e with the constant gauge angle method (flag 00 set) is questionable since then two adjustment multipliers are necessary.

E. Program Listing and Printer Output

```
01♦LBL "PSFIELD"
02 "FLAGS SET?"
03 AVIEW
04 PSE
05 "CNST∠: SF 00"
06 AVIEW
07 SIN
08 SIN
09 "VAR.∠: NONE"
10 AVIEW
11 PSE
12 "  %: NONE"
13 AVIEW
14 SIN
15 SIN
16 "DEG.: SF 01"
17 AVIEW
18 SIN
19 SIN
20 "TOPO: SF 02"
21 AVIEW
22 PSE
23 "METRIC: SF 03"
24 AVIEW
25 SIN
26 SIN
27 "  F ?"
28 PROMPT
29 STO 00
30 10
31 X<>Y
32 /
33 SQRT
34 2.75
35 *
36 FS? 03
37 XEQ 19
38 STO 01
39 FIX 3
40 "HDM= "
41 ARCL X
42 AVIEW
43 PSE
44♦LBL 15
45 "DBH?"
46 PROMPT
47 STO 02
48 RCL 01
49 *
50 STO 03
51 STO 05
52 FIX 2
53 FS? 00
54 GTO 00
55 FC? 00
56 "HOR.RAD= "
57 ARCL X
58 PROMPT
59♦LBL 17
60 FS? 01
61 GTO 01
62 FS? 02
63 GTO 02
64 "∠ IN %?"
65 PROMPT
66 100
67 /
68 ATAN
69 STO 04
70 FS? 00
71 RTN
72♦LBL 16
73 COS
74 1/X
75 RCL 03
76 *
77 "SLP.RAD= "
78 ARCL X
79 AVIEW
80 STO 05
81 STOP
82 GTO 15
83♦LBL a
84 RCL 05
85 "ACTUAL DIST?"
86 PROMPT
87 X<=Y?
88 "TREE IS IN"
89 X>Y?
90 "TREE IS OUT"
91 AVIEW
92 STOP
93 GTO 15
94♦LBL b
95 FIX 2
96 RCL 04
97 COS
98 1/X
99 RCL 02
100 *
101 STO 06
102 "D TO HOLD="
103 ARCL X
104 AVIEW
105 STOP
106 GTO 15
107♦LBL c
108 FIX 2
109 RCL 00
110 RCL 02
111 X↑2
112 FC? 03
113 .005454
114 FS? 03
115 .00007854
116 *
117 /
118 STO 07
119 "TREFAC= "
120 ARCL X
121 AVIEW
122 STOP
123 1/X
124 STO 08
125 "AREA= "
126 ARCL X
127 AVIEW
128 STOP
129 GTO 15
130♦LBL 01
131 "∠IN DEGREES?"
132 PROMPT
133 STO 04
134 FS? 00
135 GTO 18
136 GTO 16
137♦LBL 02
138 "∠ IN TOPO?"
139 PROMPT
140 66
141 /
142 ATAN
143 STO 04
144 FS? 00
145 GTO 18
146 GTO 16
147♦LBL 00
148 "SLP.RAD= "
149 ARCL X
150 STO 05
```

E. Program Listing and Printer Output (Continued)


```
151 PROMPT
152 XEQ 17
153◆LBL 18
154 RCL 04
155 COS
156 1/X
157 STO 09
158 FIX 3
159 "MULT.= "
160 ARCL X
161 AVIEW
162 RTN
163◆LBL d
164 FIX 2
165 RCL 09
166 RCL 00
167 *
168 "ADJ. F= "
169 ARCL X
170 AVIEW
171 STOP
172 RCL 02
173 X↑2
174 FC? 03
175 .005454
176 FS? 03
177 .00007854
178 *
179 /
180 "ADJ. F(T)= "
181 ARCL X
182 AVIEW
183 STOP
184 GTO 15
185◆LBL 19
186 RCL 00
187 SQRT
188 2
189 *
190 1/X
191 RTN
192◆LBL e
193 RAD
194 "HORIZ. B ?"
195 PROMPT
196 STO 10
197 RCL 03
198 /
199 ACOS
200 CHS
```

```
201 PI
202 +
203 RCL 03
204 X↑2
205 *
206 LASTX
207 RCL 10
208 X↑2
209 -
210 SQRT
211 RCL 10
212 *
213 +
214 RCL 00
215 *
216 RCL 02
217 X↑2
218 PI
219 *
220 FS? 03
221 .25
222 FC? 03
223 75.625
224 *
225 X<>Y
226 /
227 STO 11
228 "W = "
229 ARCL X
230 AVIEW
231 DEG
232 STOP
233 RCL 00
234 *
235 "WTD. F= "
236 ARCL X
237 AVIEW
238 PSE
239 RCL 07
240 RCL 11
241 *
242 "WTD. F(T)= "
243 ARCL X
244 AVIEW
245 STOP
246 GTO 15
247 END

PSFIELD:634 BYTES
```

```
                  XEQ "PSFIELD"
FLAGS SET?
CNST∡: SF 00
VAR.∡: NONE
  %: NONE
DEG.: SF 01
TOPO: SF 02
METRIC: SF 03
  F ?
                   10     RUN
HDM= 2.750
DBH?
               10.000     RUN
HOR.RAD= 27.50
                          RUN
∡ IN %?
                30.00     RUN
SLP.RAD= 28.71
                        XEQ a
ACTUAL DIST?
                28.00     RUN
TREE IS IN
                        XEQ b
D TO HOLD=10.44
                        XEQ c
TREFAC= 18.34
                          RUN
AREA= 0.05
                        XEQ e
HORIZ. B ?
                20.00     RUN
W = 1.09
                          RUN
WTD. F= 10.89
WTD. F(T)= 19.97
```

E. Printer Output (Continued)

```
                    SF 01
                   XEQ 15
DBH?
           10.00    RUN
HOR.RAD= 27.50
                    RUN
∡IN DEGREES?
           16.70    RUN
SLP.RAD= 28.71
                  XEQ a
ACTUAL DIST?
           29.00    RUN
TREE IS OUT
                  XEQ b
D TO HOLD=10.44
                  XEQ c
TREFAC= 18.34
                    RUN
AREA= 0.05
```

```
                    CF 01
                    SF 02
                   XEQ 15
DBH?
           10.00    RUN
HOR.RAD= 27.50
                    RUN
∡ IN TOPO?
           19.80    RUN
SLP.RAD= 28.71
                  XEQ a
ACTUAL DIST?
           28.50    RUN
TREE IS IN
                  XEQ b
D TO HOLD=10.44
                  XEQ c
TREFAC= 18.34
                    RUN
AREA= 0.05
```

```
                    CF 01
                    CF 02
                    SF 00
                   XEQ 15
DBH?
           10.00    RUN
SLP.RAD= 27.50
                    RUN
∡ IN %?
           30.00    RUN
MULT.= 1.044
                    RUN
ADJ. F= 10.44
                    RUN
ADJ. F(T)= 19.14
                  XEQ a
ACTUAL DIST?
           28.00    RUN
TREE IS OUT
                  XEQ c
TREFAC= 18.34
                    RUN
AREA= 0.05
                  XEQ d
ADJ. F= 10.44
                    RUN
ADJ. F(T)= 19.14
```

```
                    CF 00
                    SF 03
          XEQ "PSFIELD"
FLAGS SET?
CNST∡: SF 00
VAR.∡: NONE
  %: NONE
DEG.: SF 01
TOPO: SF 02
METRIC: SF 03
  F ?
            2.30    RUN
HDM= 0.330
DBH?
          25.400    RUN
HOR.RAD= 8.37
                    RUN
∡ IN %?
           30.00    RUN
SLP.RAD= 8.74
                  XEQ a
ACTUAL DIST?
            8.60    RUN
TREE IS IN
                  XEQ b
D TO HOLD=26.52
                  XEQ c
TREFAC= 45.39
                    RUN
AREA= 0.02
                  XEQ e
HORIZ. B ?
           6.096    RUN
W = 1.09
                    RUN
WTD. F= 2.50
WTD. F(T)= 49.42
```

F. Formulas used: (see part A for definition of symbols)

1. Basic horizontal point sampling

a. Horizontal distance multiplier

$$HDM = \frac{33\sqrt{10}}{12\sqrt{F}} = \frac{2.75\sqrt{10}}{\sqrt{F}} = 2.75\sqrt{\frac{10}{F}}\text{, U.S. units}$$

$$\text{or } HDM = \frac{1}{2\sqrt{F}}\text{, metric units}$$

b. associated plot radius, horizontal

$$R = HDM\ (D)$$

c. tree factor

$$F_t = \frac{F}{\text{tree basal area}}$$

$$= \frac{F}{.005454\ D^2}\text{, U.S. units}$$

$$\text{or } F_t = \frac{F}{.00007854\ D^2}\text{, metric units}$$

d. plot area

$$a = \frac{1}{F_t}$$

2. Slope conversion

a. $\text{angle in degrees} = \arctan\left(\frac{\text{angle in \%}}{100}\right)$

b. $\text{angle in degrees} = \arctan\left(\frac{\text{topo angle}}{66}\right)$

3. Related to slope correction techniques

a. using variable gauge angle approach

(1) associated plot radius <u>on the slope</u>

$$R_s = R \sec S = \frac{D(HDM)}{\cos S}$$

note: S is the slope angle (in degrees) from sample point to the tree

(2) calibrated tape diameter mark to hold; tape used <u>on the slope</u>

(2) *continued*

$$D_t = D \sec S = \frac{D}{\cos S}$$

(3) tree qualification test, given actual distance, R_a

IN if $R_a \leq R_s$, OUT otherwise

b. using the constant gauge angle approach

(1) associated plot radius <u>on the slope</u>

$R_s = R$ i.e., horizontal associated plot radius is established on the slope as if the terrain were level

(2) slope multiplier, assuming one "prevailing" slope measurement at the point

$$M = \sec S = \frac{1}{\cos S}$$

(3) adjusted basal area factor

adj. $F = M\ (F)$

(4) adjusted tree factor

adj. $F_t = M\ (F_t)$

4. Boundary overlap correction

a. mirage method is recommended and appropriate for either variable or constant gauge angle <u>on the slope</u>.

b. in case mirage method is impractical, the direct weighting procedure can be used (label e); the weight calculated from

$$W = \frac{\pi D^2}{576} \cdot \frac{43560}{F(I)} = \frac{75.625\ \pi D^2}{F(I)}\text{, U.S. units}$$

$$\text{or, } W = \frac{\pi D^2}{40000} \cdot \frac{10000}{F(I)} = \frac{.25\pi D^2}{F(I)}\text{, metric units}$$

where, in either system of units, the plot area inside the boundary, I, is found by

$$I = R^2(\pi - \arccos \frac{B}{R}) + B\sqrt{R^2-B^2}\text{, radian mode}$$

Note: in field application, B, the distance from sample point to boundary is:

(1) measured horizontally in the variable gauge angle technique.

(2) measured on the slope in the constant gauge angle technique.

c. for the direct weighting procedure,

Weighted $F = (W)(F)$ and

Weighted $F_t = (W)(F_t)$ are calculated and can be used as an alternative to tallying trees as the weight, W.

Retrospective Comments Regarding PSFIELD

1. PSFIELD is a ponderous program which probably has no practical use except to demonstrate the potential field application of the HP-41C. For this reason it is included in this compilation.

2. Foresters wishing to apply such a program would be well advised to extract the parts of PSFIELD appropriate for their purpose and prepare a specific program. The formulas and references cited in the write-up should prove useful.

Program No. 41F017

Calculator: HP-41C/CV

Program Name: SSRS (Simple and Stratified Random Sampling)

Author: T.W. Beers

Date: February 1981

Purpose: To summarize data from either a simple or stratified sample, obtaining within stratum and overall estimates of the mean and standard error and, optionally, user specified confidence intervals and other sample statistics. Provision is made to use or not (Flag 00 set) a finite population correction.

A. Storage assignments

Register	Use
00-05	statistical: ΣX, ΣX^2, ΣY, ΣY^2, ΣXY, n
06	N, population size or N_h, stratum size
07	$\overline{X}$, sample mean, or $\overline{X}_h$, stratum sample mean, then $\overline{X}_{st}$, overall stratified sample mean
08	$\Sigma N_h \overline{X}_h$
09	ΣN_h
10	s, sample standard deviation, or s_h
11	$s_{\overline{X}}$, sample standard error, or $s_{\overline{X}_h}$, the $s_{\overline{X}_{st}}$, overall standard error
12	$\Sigma N_h{}^2 s^2_{\overline{X}_h}$ used in overall standard error calculation
13	f.p.c., finite population correction
14	Student's t used for confidence interval calculations (user input, or default to t = 2)

B. Labels

Global	Use
SSRS	start of program, flag reminder, clears all registers and prompts for simple (A) or stratified sampling (a)
FPC	prompts for population size in simple sampling and calculates finite population correction for simple $s_{\bar{X}}$ or for $s_{\bar{X}_h}$
NSTRAT	provides a NEXT SIZE? prompt for the second and subsequent strata in a stratified sample
STRAT	prompts for stratum size, processes this, sets flag 01, and directs execution to SUMRY
SUMRY	primary routine to summarize the basic data, clears statistical registers and prompts for each X
Local	
A	clears flag 01 and directs execution to SUMRY label
a	directs execution to STRAT label
B	basic routine to calculate mean and standard error for simple sample or for within stratum estimates
b	for the stratified sample, calculates overall mean, standard error and standard error in percent, and clears accumulators for next problem
C	extended part of label B, used to conveniently display within stratum standard error in percent, no. observations, and coefficient of variation
c	prompts for Student's t and calculates lower and upper bounds of confidence interval estimate for $\bar{X}$, $\bar{X}_h$ or $\bar{X}_{st}$
01	internal loop in SUMRY to summarize the data
02	used in SSRS to skip the initial prompts (if flag 03 is set)
03	used in label c to enable a repeat calculation and display of confidence interval bounds (with flag 01 clear)
04	used in FPC to skip the prompt in stratified sampling (flag 01 set)
05	used in STRAT to skip the STR. SIZE? prompt after the initial stratum

C. Flags

Number	Use
NONE	assumes simple random sampling, finite population correction to be applied, and initial prompts are to be displayed
00	if set (externally), the f.p.c. is not applied and the prompt for population size in simple sampling is avoided
01	set automatically in STRAT to signify stratified sampling mode
03	if set (externally) several initial prompts are avoided

D. Program procedure and example

I. In PRGM mode, load program "SSRS" (SIZE needed is 015)

II. In RUN mode:

1. Check flags for proper condition

   a. none set, assumes finite population correction (f.p.c.) is to be applied
   b. 00 set, assumes no f.p.c. to be applied
   c. 03 set, will skip preliminary prompts

2. XEQ SSRS (suggest assignment to LN) and follow the prompts, noting that key A will select simple random sampling, while key a (i.e., shift, A) will select the stratified sample mode. Four cases will be described:

   a. stratified sampling with correction applied
   b. stratified sampling without correction applied
   c. simple random sampling with correction applied
   d. simple random sampling without correction applied

The stratified sample used in the example, as shown below, is from a finite population. The standard errors assuming no correction are shown in parentheses, and of course no correction is made for the means. Stratum 2 is used for the simple sampling example.

	Stratum 1	Stratum 2	Stratum 3	Total
Data:	3,0,2	12,8,15,13	18,22	
N_h	30	50	20	100
$\bar{X}_h$	1.7	12.0	20.0	$10.5 = \bar{X}_{st}$
s_h	1.53	2.94	2.83	
$s_{\bar{X}_h}$	0.84 (0.88)	1.41 (1.47)	1.90 (2.00)	$0.84\ (0.88) = s_{\bar{X}_{st}}$

a. No flags set-- stratified with correction (key a)

Prompt	Input	Key	Output	Example Input	Example Key	Example Output
A OR a (STRT)	-	a	-	-	a	-
STR. SIZE?	N_1	R/S	-	30	R/S	-
EACH X, R/S	X_1	R/S	1	3	R/S	1
(repeat for all X in stratum 1)				0	R/S	2
				2	R/S	3
-	-	B	MEAN = X.X S.E. = X.XX	-	B	MEAN = 1.7 S.E. = 0.84
NEXT SIZE?	N_2	R/S	-	50	R/S	-
EACH X, R/S	X_1	R/S	1	12	R/S	1
(repeat for all X in stratum 2)				8	R/S	2
				15	R/S	3
				13	R/S	4
-	-	B	MEAN = X.X S.E. = X.XX	-	B	MEAN = 12.0 S.E. = 1.41
NEXT SIZE?	N_3	R/S	-	20	R/S	-
EACH X, R/S	X_1	R/S	1	18	R/S	1
(repeat for all X in stratum 3)				22	R/S	2
-	-	B	MEAN = X.X S.E. = X.XX	-	B	MEAN = 20.0 S.E. = 1.90
NEXT SIZE?	-	b	MEAN (ST) = X.X S.E. (ST) = X.XX	-	b	MEAN (ST) = 10.5 S.E. (ST) = 0.84
-	-	R/S	S.E. IN % = X.X	-	R/S	S.E. IN % = 8.0
-	-	R/S	-	-	R/S	-
T? OR R/S	-	R/S	C.I. LOW. = X.X C.I. UP. = X.X	2 (by default)	R/S	C.I. LOW. = 8.8 C.I. UP. = 12.2

b. With flag 00 set-- stratified with no finite population correction.
Instructions are the same as in 2a with the following results:

	Key	Output
stratum 1:	B	MEAN = 1.7
		S.E. = 0.88
stratum 2:	B	MEAN = 12.0
		S.E. = 1.47
stratum 3:	B	MEAN = 20.0
		S.E. = 2.00
overall results:	b	MEAN(ST)= 10.5
		S.E. (ST)= 0.88
	R/S	S.E. IN % = 8.4
	R/S	-
	R/S	C.I. LOW. = 8.7
		C.I. UP. = 12.3

c. No flags set-- simple random sampling with correction (key A)

Prompt	Input	Key	Output	Example Input	Example Key	Example Output
A or a (STRT.)	-	A	-	-	A	-
EACH X, R/S	X_1	R/S	1	12	R/S	1
(repeat for all X)				8	R/S	2
				15	R/S	3
				13	R/S	4

c. *Continued*

Prompt	Input	Key	Output	Example Input	Example Key	Example Output
-	-	B	-	-	B	-
POP. SIZE?	N	R/S	MEAN = X.X	50	R/S	MEAN= 12.0
			S.E. = X.XX			S.E.= 1.41
-	-	R/S	S.E. IN % = X.XX	-	R/S	S.E. IN %= 11.77
			NO. OBS. = X			NO. OBS.= 4
			C.V. IN % = X.X			C.V. IN %= 24.5
-	-	c	-	-	c	-
T? OR R/S	-	R/S	C.I. LOW. = X.X	2 (by default)	R/S	C.I. LOW.= 9.2
			C.I. UP. = X.X			C.I. UP. = 14.8

d. With flag 00 set-- simple random sampling with no correction (key A)

Prompt	Input	Key	Output	Example Input	Example Key	Example Output
A OR a (STRT.)	-	A	-	-	A	-
EACH X, R/S	X_1	R/S	1	12	R/S	1
(repeat for all X)				8	R/S	2
				15	R/S	3
				13	R/S	4
-	-	B	MEAN= X.X	-	B	MEAN= 12.0
			S.E.= X.XX			S.E.= 1.47
-	-	R/S	S.E. IN % =	-	R/S	S.E. IN %= 12.27
-	-		NO. OBS. =			NO. OBS.= 4
			C.V. IN % =			C.V. IN %= 24.5

d. *Continued*

Prompt	Input	Key	Output	Example Input	Example Key	Example Output
-	-	c	-	-	c	-
T? OR R/S	-	R/S	C.I. LOW. = X.X	2 (by default)	R/S	C.I. LOW.= 9.1
			C.I. UP. = X.X		R/S	C.I. UP.= 14.9

3. General comments:

   a. For simple random sampling use, SSRS need be executed only once for a given calculation session; thereafter key A can be used to initialize the registers; for several stratified sample data sets, SSRS must be executed for each new problem.

   b. The initial reminder prompts can be avoided by setting flag 03.

   c. Label C, which calculates and displays S.E. IN %, NO. OBS, and C.V. IN %, can be activated and used for within stratum answers (by key C) after the MEAN and S.E. results have been obtained by key B. After key C, then, for the NEXT SIZE? prompt, press R/S and input data for the next stratum. Pressing key C is not appropriate for the final stratified answers (i.e.. after key b has been pressed).

E. Program Listing and Printer Output

```
01♦LBL "SSRS"
02 ΣREG 00
03 CLRG
04 CF 29
05 FS? 03
06 GTO 02
07 " SF 00 FOR"
08 AVIEW
09 PSE
10 " NO F.P.C."
11 AVIEW
12 PSE
13 "THEN PRESS"
14 AVIEW
15 PSE
16♦LBL 02
17 "A OR a(STRT.)"
18 PROMPT
19♦LBL A
20 CF 01
21 GTO "SUMRY"
22♦LBL a
23 GTO "STRAT"
24♦LBL "SUMRY"
25 FIX 0
26 CLΣ
27 "EACH X, R/S"
28 PROMPT
29♦LBL 01
30 Σ+
31 STOP
32 GTO 01
33♦LBL B
34 FC? 00
35 XEQ "FPC"
36 FIX 1
37 MEAN
38 STO 07
39 "MEAN= "
40 ARCL X
41 AVIEW
42 PSE
43 RCL 06
44 *
45 ST+ 08
46 SDEV
47 STO 10
48 X↑2
49 RCL 05
50 /
51 FC? 00
52 RCL 13
53 FS? 00
54 1
55 *
56 SQRT
57 STO 11
58 X↑2
59 RCL 06
60 X↑2
61 *
62 ST+ 12
63 FIX 2
64 "S.E. = "
65 ARCL 11
66 AVIEW
67 PSE
68 FS? 01
69 GTO "NSTRAT"
70 STOP
71♦LBL C
72 RCL 11
73 RCL 07
74 /
75 1 E2
76 *
77 "S.E. IN %="
78 ARCL X
79 AVIEW
80 PSE
81 FIX 0
82 "NO. OBS.= "
83 ARCL 05
84 AVIEW
85 PSE
86 RCL 10
87 RCL 07
88 /
89 1 E2
90 *
91 FIX 1
92 "C.V. IN %="
93 ARCL X
94 AVIEW
95 STOP
96 FC? 01
97 GTO "SUMRY"
98♦LBL "NSTRAT"
99 "NEXT SIZE?"
100 PROMPT
101 GTO 05
102♦LBL "STRAT"
103 SF 01
104 "STR. SIZE?"
105 PROMPT
106♦LBL 05
107 STO 06
108 ST+ 09
109 GTO "SUMRY"
110♦LBL b
111 RCL 08
112 RCL 09
113 /
114 FIX 1
115 STO 07
116 "MEAN(ST)="
117 ARCL X
118 AVIEW
119 PSE
120 RCL 12
121 RCL 09
122 X↑2
123 /
124 SQRT
125 FIX 2
126 STO 11
127 "S.E.(ST)="
128 ARCL X
129 AVIEW
130 0
131 STO 08
132 STO 09
133 STO 12
134 STOP
135 RCL 11
136 RCL 07
137 /
138 1 E2
139 *
140 FIX 1
141 "S.E. IN %="
142 ARCL X
143 AVIEW
144 RTN
145♦LBL c
146 CF 22
147 "T? OR R/S"
148 PROMPT
149 FC? 22
150 2
```

E. Program Listing (continued) and Printer Output


```
151 STO 14
152♦LBL 03
153 RCL 07
154 RCL 14
155 RCL 11
156 *
157 -
158 "C.I. LOW.= "
159 ARCL X
160 AVIEW
161 PSE
162 LASTX
163 RCL 07
164 +
165 "C.I. UP.= "
166 ARCL X
167 AVIEW
168 STOP
169 FS? 01
170 GTO "NSTRAT"
171 GTO 03
172♦LBL "FPC"
173 FS? 01
174 GTO 04
175 "POP. SIZE?"
176 PROMPT
177 STO 06
178♦LBL 04
179 RCL 06
180 RCL 05
181 -
182 RCL 06
183 /
184 STO 13
185 RTN
186 END

SSRS:505 BYTES
```

Stratified-- with f.p.c. correction

```
                    XEQ "SSRS"
 SF 00 FOR
 NO F.P.C.
THEN PRESS
A OR a<STRT.>
                        XEQ a
STR. SIZE?
                30      RUN
EACH X, R/S
                 3      RUN
                 0      RUN
                 2      RUN
                        XEQ B
MEAN= 1.7
S.E. = 0.84
NEXT SIZE?
             50.00      RUN
EACH X, R/S
                12      RUN
                 8      RUN
                15      RUN
                13      RUN
                        XEQ B
MEAN= 12.0
S.E. = 1.41
NEXT SIZE?
             20.00      RUN
EACH X, R/S
                18      RUN
                22      RUN
                        XEQ B
MEAN= 20.0
S.E. = 1.90
NEXT SIZE?
                        XEQ b
MEAN<ST>=10.5
S.E.<ST>=0.84
                          RUN
S.E. IN %=8.0
                          RUN
T? OR R/S
                          RUN
C.I. LOW.= 8.8
C.I. UP.= 12.2
```

Stratified-- without f.p.c. correction

```
              XEQ "SSRS"
 SF 00 FOR
 NO F.P.C.
THEN PRESS
A OR a<STRT.>
                   SF 00
                   XEQ a
STR. SIZE?
           30.0      RUN
EACH X, R/S
              3      RUN
              0      RUN
              2      RUN
                   XEQ B
MEAN= 1.7
S.E. = 0.88
NEXT SIZE?
          50.00      RUN
EACH X, R/S
             12      RUN
              8      RUN
             15      RUN
             13      RUN
                   XEQ B
MEAN= 12.0
S.E. = 1.47
NEXT SIZE?
          20.00      RUN
EACH X, R/S
             18      RUN
             22      RUN
                   XEQ B
MEAN= 20.0
S.E. = 2.00
NEXT SIZE?
                   XEQ b
MEAN<ST>=10.5
S.E.<ST>=0.88
                     RUN
S.E. IN %=8.4
                     RUN
T? OR R/S
                     RUN
C.I. LOW.= 8.7
C.I. UP.= 12.3
```

Simple-- with f.p.c.

```
                   CF 00
                   XEQ A
EACH X, R/S
             12      RUN
              8      RUN
             15      RUN
             13      RUN
                   XEQ B
POP. SIZE?
             50      RUN
MEAN= 12.0
S.E. = 1.41
                     RUN
S.E. IN %=11.77
NO. OBS.= 4
C.V. IN %=24.5
                   XEQ c
T? OR R/S
                     RUN
C.I. LOW.= 9.2
C.I. UP.= 14.8
```

Simple-- without f.p.c.

```
                   SF 00
                   XEQ A
EACH X, R/S
             12      RUN
              8      RUN
             15      RUN
             13      RUN
                   XEQ B
MEAN= 12.0
S.E. = 1.47
                     RUN
S.E. IN %=12.27
NO. OBS.= 4
C.V. IN %=24.5
                   XEQ c
T? OR R/S
                     RUN
C.I. LOW.= 9.1
C.I. UP.= 14.9
```

F. Formulas used:

I. Simple random sampling

a. mean $= \overline{X} = \frac{\Sigma X}{n}$

b. standard error

$$s_{\overline{X}} = \sqrt{\frac{s^2}{n}}$$

or

$$s_{\overline{X}} = \sqrt{\frac{s^2}{n}\left(\frac{N-n}{N}\right)}$$ if finite population is assumed

c. coefficient of variation

$$C.V. = \frac{s}{\overline{X}}(100)$$

d. standard error in %

$$s_{\overline{X}}(\%) = \frac{s_{\overline{X}}}{\overline{X}}(100)$$

e. confidence interval

$\overline{X} \pm t\, s_{\overline{X}}$, value of t determines degree of "confidence"

II. Stratified sampling

a. through e., same as above for within-stratum estimates

f. overall mean, $\overline{X}_{st} = \frac{\Sigma N_h \overline{X}_h}{N}$

where N_h = population size for stratum h

$\overline{X}_h$ = sample mean for stratum h

and N = total population size $= \Sigma N_h$

g. overall standard error, $s_{\overline{X}_{st}} = \sqrt{\frac{1}{N^2}(\Sigma N_h^2 s^2_{\overline{X}_h})}$

where

$s^2_{\overline{X}_h}$ = squared standard error for stratum h, corrected, if appropriate, for finite population

h. confidence intervals commonly used

t = 2 for .05 probability level

and t = 2.6 for .01 probability level.

i. $s_{\overline{X}_{st}} (\%) = \frac{s_{\overline{X}_{st}}}{\overline{X}_{st}} (100)$

Retrospective Comments Regarding SSRS

1. While the use of descriptive global labels such as FPC, STRAT, and SUMRY make programming logic easier to follow, they require more memory space than numeric labels, and they slow down a "CAT1" function execution to a considerable degree. Therefore, such labels should be used with discretion.

Program No. 41F018

Calculator: HP-41C/CV

Program Name: LR (for Linear Regression)

Author: T.W. Beers

Date: March 1981

Purpose: To provide simple linear regression calculations which simulate the L.R., $\hat{y}$, and r keys found on the HP-34C. Specifically, one may readily calculate the following, for X and Y data summarized by the $\Sigma+$ key:

a, the Y-intercept
b, the slope of the least-squares fitted line
b', the slope assuming the line is forced through the origin
r, the simple correlation coefficient
r^2, the coefficient of determination
t_r, the calculated t to test the hypothesis of zero correlation
$\hat{Y}$, a predicted value of Y for any keyed-in X

Additional statistics such as confidence interval estimates and standard errors can be calculated using the program SLR (program number 41F019) but for brevity (96 program steps, 176 bytes, one magnetic card) only the listed items are included in "LR".

A. Storage assignments

Register	Use
00	B.F.O., the slope, forced through the origin
01	$\overline{X}$, arithmetic mean X
02	Σx^2, corrected sum of squares for X
03	$\overline{Y}$, arithmetic mean Y
04	Σy^2, corrected sum of squares for Y
05	Σxy, corrected sum of products

A. Storage assignments *(continued)*

Register	Use
06	Y-INT., the Y intercept (= a)
07	SLOPE, the change in Y per unit change in X(= b)
08	r, the simple correlation coefficient
09	not used
10	not used
11	ΣX, sum of X
12	ΣX^2, sum of squared X
13	ΣY, sum of Y
14	ΣY^2, sum of squared Y
15	ΣXY, sum of products of X and Y
16	n, number of pairs of data

B. Labels

Global	Use
LR	calculates the mean X and Y, corrected sums of squares and products, and calculates and displays the Y-intercept (a), slope (b), and optionally (by R/S), the slope of the line forced through the origin (B.F.O.)

Local	
H	prompts for X to be input, then calculates and displays the predicted Y (Y-HAT) obtained by solving the equation, $\hat{Y} = a + bX$, for $\hat{Y}$
I	calculates and displays the simple correlation coefficient (r) and optionally (by R/S), its square, the coefficient of determination (r^2)
J	calculates and displays the calculated t value, t_r, used to test the hypothesis of zero correlation between X and Y
01	used in label H to skip the "KEY X, R/S" prompt, expediting additional $\hat{Y}$ solutions

C. Flags -- none used

D. Program procedure and example

I. In PRGM mode, load the program "LR" (SIZE needed is 017)

II. In RUN mode: (make sure the current statistical registers are 11-16, i.e., XEQ ΣREG 11)

1. Obtain summations for the two-variable data by the normal use of the Σ+ key. For example, if the sample data are:

Y	X
7	5
5	3
9	8

The steps would be

a. f*CLΣ (to clear the summation registers)

b. 7↑5, Σ+ (read 1.00)

5↑3, Σ+ (read 2.00)

9↑8 Σ+ (read 3.00, i.e., n)

*f means the shift (gold) key

2. XEQ LR (suggest assignment to R↓ key). The above example is used, and USER mode is assumed. Example assumes f FIX 2.

	Prompt	Input	Key	Output
a. least squares line	-	-	R↓	Y-INT.= 2.79 SLOPE= 0.79
			R/S	B.F.O.= 1.24
b. predicted Y	-	-	H(SIN key)	-
	KEY X, R/S	6	R/S	Y-HAT= 7.53
		8	R/S	Y-HAT= 9.11
		10	R/S	Y-HAT= 10.68
		etc., for more X_i		
c. correlation	-	-	I(COS key)	R= 0.99
			R/S	R↑2= 0.99
d. t statistic	-	-	J(TAN key)	T(R)= 8.66

D. Program procedure and example *(Continued)*

e. If one has previously assigned the MEAN function to a key (say LOG) and the SDEV function to a key (say f LOG) then the folowing can be obtained:

Key	Output	
LOG	5.33	($\overline{X}$)
X$\lessgtr$Y	7.00	($\overline{Y}$)
f LOG	2.52	(s_X)
X$\lessgtr$Y	2.00	(s_Y)

f. In addition, the stored items (see Section A) can be observed by RCL nn:

Key	Output	
RCL 00	1.24	(b')
RCL 01	5.33	($\overline{X}$)
RCL 02	12.67	(Σx^2)
RCL 03	7.00	($\overline{Y}$)
RCL 04	8.00	(Σy^2)
RCL 05	10.00	(Σxy)
RCL 06	2.79	(a)
RCL 07	0.79	(b)
RCL 08	0.99	(r)
RCL 11	16.00	(ΣX)
RCL 12	98.00	(ΣX^2)
RCL 13	21.00	(ΣY)
RCL 14	155.00	(ΣY^2)
RCL 15	122.00	(ΣXY)
RCL 16	3.00	(n)

III. Notes and cautions

1. LR must be executed before the other labels.
2. Label I (COS key) must be pressed before label J(TAN key).
3. Erroneous data pairs in the summation process can be corrected by the usual use of the $\Sigma-$ key.

E. Program listing and printer output

```
01♦LBL "LR"
02 MEAN
03 STO 01
04 X↑2
05 STO 02
06 X<>Y
07 STO 03
08 X↑2
09 STO 04
10 RCL 01
11 RCL 03
12 *
13 STO 05
14 RCL 16
15 CHS
16 ST* 02
17 ST* 04
18 ST* 05
19 RCL 12
20 ST+ 02
21 RCL 14
22 ST+ 04
23 RCL 15
24 ST+ 05
25 RCL 05
26 RCL 02
27 /
28 STO 07
29 RCL 01
30 *
31 CHS
32 RCL 03
33 +
34 STO 06
35 "Y-INT.= "
36 ARCL X
37 AVIEW
38 PSE
39 "SLOPE= "
40 ARCL 07
41 AVIEW
42 STOP
43 RCL 15
44 RCL 12
45 /
46 STO 00
47 "B.F.O.= "
48 ARCL X
49 AVIEW
50 RTN
51♦LBL H
52 "KEY X, R/S"
53 PROMPT
54♦LBL 01
55 RCL 07
56 *
57 RCL 06
58 +
59 "Y-HAT= "
60 ARCL X
61 AVIEW
62 STOP
63 GTO 01
64♦LBL I
65 RCL 05
66 RCL 02
67 RCL 04
68 *
69 SQRT
70 /
71 STO 08
72 "R= "
73 ARCL X
74 AVIEW
75 STOP
76 X↑2
77 "R↑2= "
78 ARCL X
79 AVIEW
80 RTN
81♦LBL J
82 RCL 16
83 2
84 -
85 1
86 RCL 08
87 X↑2
88 -
89 /
90 SQRT
91 RCL 08
92 *
93 "T(R)= "
94 ARCL X
95 AVIEW
96 END

LR:176 BYTES
```

E. Printer output *(Continued)*

Printer in NORM mode

```
              CLΣ
       7.00 ENTER↑
       5.00     Σ+
       5.00 ENTER↑
       3.00     Σ+
       9.00 ENTER↑
       8.00     Σ+
          XEQ "LR"
Y-INT.= 2.79
SLOPE= 0.79
               RUN
B.F.O.= 1.24
             XEQ H
KEY X, R/S
       6.00    RUN
Y-HAT= 7.53
       8.00    RUN
Y-HAT= 9.11
      10.00    RUN
Y-HAT= 10.68
             XEQ I
R= 0.99
               RUN
R↑2= 0.99
             XEQ J
T(R)= 8.66


              MEAN
       5.33    ***
              X<>Y
       7.00    ***
              SDEV
       2.52    ***
              X<>Y
       2.00    ***
```

Register contents

Actual

```
      0.016
           PRREGX

R00=  1.24
R01=  5.33
R02=  12.67
R03=  7.00
R04=  8.00
R05=  10.00
R06=  2.79
R07=  0.79
R08=  0.99
R09=  0.00
R10=  2.94
R11=  16.00
R12=  98.00
R13=  21.00
R14=  155.00
R15=  122.00
R16=  3.00
```

Symbolic

```
R00= "B.F.O."
R01= "X-BAR"
R02= "CΣXX"
R03= "Y-BAR"
R04= "CΣYY"
R05= "CΣXY"
R06= "Y-INT"
R07= "SLOPE"
R08= "R"
R09= "REG 9 "
R10= "REG 10"
R11= "ΣX"
R12= "ΣXX"
R13= "ΣY"
R14= "ΣYY"
R15= "ΣXY"
R16= "N"
```

Printer in MAN mode

```
Y-INT.= 2.79
SLOPE= 0.79
B.F.O.= 1.24
Y-HAT= 7.53
Y-HAT= 9.11
Y-HAT= 10.68
R= 0.99
R↑2= 0.99
T(R)= 8.66
```

F. Formulas used

Many variations exist for the calculation of simple linear regression statistics. The following were chosen for programming simplicity.

1. Means -- $\overline{X}$ and $\overline{Y}$ calculated using the standard MEAN function

2. standard deviations -- s_X and s_Y calculated using the standard SDEV function

3. corrected sum of squares and products, i.e., sum of squared (or product) deviations about the mean (designated by lower case leters)

$$\Sigma x^2 = \Sigma X^2 - n\overline{X}^2$$

$$\Sigma y^2 = \Sigma Y^2 - n\overline{Y}^2$$

$$\Sigma xy = \Sigma XY - n\overline{X}\overline{Y}$$

4. least squares slope and Y-intercept

$$b = \frac{\Sigma xy}{\Sigma x^2}$$

$$a = \overline{Y} - b\overline{X}$$

5. slope when line is forced through the origin

$$b' \text{ (or B.F.O.)} = \frac{\Sigma XY}{\Sigma X^2}$$

6. correlation coefficient and coefficient of determination

$$r = \frac{\Sigma xy}{\sqrt{\Sigma x^2 \Sigma y^2}}$$

$$R\uparrow 2 = r^2$$

7. Student's t to test the hypothesis that the true correlation is zero (H_o:ρ=0), which is identical to the t which tests the hypothesis of zero slope (H_o:β=0)

$$t_r = r\sqrt{\frac{n-2}{1-r^2}}$$

Program No. 41F019

Calculator: HP-41C/CV

Program Name: SLR (Simple Linear Regression)

Author: T.W. Beers

Date: 1982

Purpose: To extend the simple linear regression calculations achieved by "LR" (Program No. 41F018) to provide, for ungrouped X and Y data summarized by the Σ+ key:

a, b, b', r, r^2, t_r, and $\hat{Y}$ (see LR program)

standard errors: s_{yx}, s_a, s_b, $s_{b'}$

confidence interval estimates assuming mean Y and assuming individual Y for given X_0

Students t to test the following hypotheses:

H_0: $\rho = 0$

H_0: $\alpha = 0$

H_0: $\beta = 0$

H_0: $\beta' = 1$

A. Storage assignments

Register	Use
00	the beginning and incremented number in a sequence of X's to facilitate automatic printing of $\hat{Y}$ and confidence intervals
01	$\overline{X}$, arithmetic mean X
02	Σx^2, corrected sum of squares for X
03	$\overline{Y}$, arithmetic mean Y
04	Σy^2, corrected sum of squares for Y
05	Σxy, corrected sum of products
06	Y-INT., the Y intercept (= a)
07	SLOPE, the change in Y per unit increase in X (= b)

A. Storage assignments *(continued)*

Register	Use
08	r, the simple correlation coefficient
09	s_a, standard error of a
10	s_b, standard error of b
11	ΣX, sum of X
12	ΣX^2, sum of squared X
13	ΣY, sum of Y
14	ΣY^2, sum of squared Y
15	ΣXY, sum of products of X and Y
16	n, number of pairs of data
17	t, Student's t, keyed-in; default value = 2.00
18	X_0, arbitrary X, keyed-in
19	$\hat{Y}$, Y-HAT or predicted Y; $\hat{Y} = a + bX_0$
20	C, intermediate value in confidence interval calculations
21	$t\ s_{yx}\sqrt{C}$ or $t\ s_{yx}\sqrt{1+C}$
22	B.F.O, slope forced through the origin (= b')
23	s'_{yx}, standard error of estimate about the b' line
24	$s_{b'}$, standard error of b'
25	s_{yx}, standard error of estimate
26	c, the number of "classes" of data (appropriate only if the program MSLR is used for grouped data)
27	to store the alpha string MCI= or ICI= , meaning "mean confidence interval" or "individual confidence interval" respectively; used in the "auto print" option (flag 00 set)
28	to save the current decimal fix setting, when "control number" is printed in label PCI
29	the control number printed in label PCI of the form bbbb.dddII, where bbbb is the beginning X, ddd ($\leq$999) is the difference between the beginning and ending X, and II ($\leq$99) is the increment value for which $\hat{Y}$ and confidence interval estimates are to be calculated.

B. Labels

Global	Use
SLR	calculates the mean X and Y, and corrected sums of squares and products; calculates and displays a, b, and optionally (by R/S) b'
PCI	preserves the current decimal fix setting and provides the calculation and printing of $\hat{Y}$ and confidence intervals; executed only if flag 00 is set

Local	
B	an alternative to "SLR" after SLR has been executed once
C	calculates and displays s_{yx}, s_a, s_b, and optionally (by R/S) $s_{b'}$
D	calculates and displays r and optionally (by R/S) r^2
E	prompts for X_0, then calculates and displays the corresponding $\hat{Y}$ ("Y - HAT")
b	calculates and displays t_a and t_b to test the hypotheses $\alpha = 0$ and $\beta = 0$ and optionally (by R/S) $t_{b'}$ to test the hypothesis that β forced through the origin ("BFO") equals 1
d	calculates and displays t_r to test the hypothesis that the true correlation coefficient $\rho = 0$, i.e., that there is no linear correlation between Y and X
e	calculates and displays (assuming user-provided Student's t and X_0) predicted Y, $\hat{Y}_0$, then lower and upper confidence limits for the mean Y for that X_0; optionally (by R/S) the lower and upper confidence limits for the corresponding individual Y are calculated and displayed
01	used in label E
02	used in label e to enable individual confidence interval estimates
03	used in label e to skip the Student's t prompt for X_0's after the first
07	used in label PCI as the re-entry point in the ISG loop
08	used in label PCI to enable calculation of individual confidence interval estimates

B. Labels *(continued)*

Local	Use
09	used in label PCI to store "ICI= " and adjust the intermediate value C to C+1 for individual confidence interval estimation
10	used in label PCI to print the beginning and ending "bar" in the table of $\hat{Y}$ and confidence intervals

C. Flags

Type and Number	Use
User:	
00	tested throughout the program, and if set, the data are processed and printed automatically and $\hat{Y}$ and confidence intervals are automatically calculated and printed for the range of X specified by the keyed-in numbers: bb↑ee↑II, where bb = beginning X value, ee = ending X value, II = increment value
02	tested in label C to enable grouped regression if flag 02 is set (applicable only if program MSLR is used)
03	set the first time through label e to skip the Student's t prompt the next times through; cleared early in label SLR
04	set, tested, and cleared in label 02 (in label e) to select the proper confidence interval prompt, calculation formula and display
08	used in conjunction with flag 22 in label e to enable (if set) being "locked in" to the mean confidence interval calculations if a numeric value is keyed
12	the double wide flag; set and cleared in label 10
21	the printer enable flag; cleared occasionally to suppress printing of messages meant for display only
22	the numeric data input flag; tested in label e and used to set flag 08 and direct calculations for confidence interval estimates
System:	
36-39	number of digits flags; tested in label PCI to determine and save the "current" decimal fix
55	printer existence flag; tested in label 08 (in PCI) to suppress the display of long scrolling alpha strings so that the automatic printing will proceed more rapidly

D. Size and key assignments

SIZE: $\geq$ 030

Suggested key assignments:

"SLR" on R↓

B, C, D, E, and b, d, e are all assigned internally and are available once SLR has been executed

E. Program procedure and example

I. Load the program into the calculator.

II. Assume a printer is <u>not</u> attached; make sure the calculator is not in PRGM mode, not in USER mode and decimal control is FIX 2.

1. Obtain summations for the two-variable data by the normal use of the Σ+ key. For example if the sample data are:

Y	X
7	5
5	3
9	8

The steps would be

a. shift CLΣ (to clear the summation registers)
b. 7↑5, Σ+ (read 1.00)
   5↑3, Σ+ (read 2.00)
   9↑8, Σ+ (read 3.00, i.e., n)

2. Go into USER mode

3. XEQ SLR (press R↓ key) and follow the procedure in the example which follows, using the above data as input

Step	Prompt	Input	*Example* Key	Output
a. least squares line	-	-	R↓	Y-INT.= 2.79 SLOPE= 0.79
	-	-	R/S	B.F.O.= 1.24
b. standard errors	-	-	C[1/]	S<YX>= 0.32 S<a>= 0.52 S<b>= 0.09
	-	-	R/S	S<BFO>= 0.13

[1/] Rather than press each local label as shown in this example, one can simply press R/S for the next "group" of calculations.

E. Program procedure and example *(continued)*

Step	Prompt	Input	*Example* Key	Output
c. correlation	-	-	D	R= 0.99 (.9934)
	-	-	R/S	R↑2= 0.99 (.9868)
d. predicted Y	-	-	E	-
	KEY X, R/S	4	R/S	X= 4.00
				Y-HAT= 5.95
	KEY X, R/S	6	R/S	X= 6.00
				Y-HAT= 7.53
	etc., for any X_0			
e. t-tests	-	-	d (shift D)	T<R>= 8.66
(1)	-	-	R/S	T<a>= 5.35
				T<b>= 8.66
	-	-	R/S	T<BFO=1>= 1.94
or (2)	-	-	b (shift B)	T<a>= 5.35
				T<b>= 8.66
	-	-	R/S	T<BFO=1>= 1.94
f. confidence intervals	-	-	e (shift E)	-
	KEY T, R/S	-[2]	R/S	T= 2.00
	KEY X, R/S	4	R/S	X= 4.00
				Y-HAT= 5.95
				C.I.= 5.50 TO 6.39
	IND. CI?, R/S	-	R/S[3]	IND. CI= 5.16 TO 6.74
	KEY X, R/S	6	R/S	X= 6.00
				Y-HAT= 7.53
				C.I.= 7.13 TO 7.92
	IND. CI?, R/S	-	R/S	IND. CI= 6.77 TO 8.29

etc., for any X_0

[2] If a specific value is keyed-in at this point it replaces the default value of t=2 obtained (as here) by R/S depression.

[3] If the "individual" confidence interval is not wanted, the next X_0 can be keyed and R/S pressed, providing the mean confidence interval estimate for that X_0.

E. Program procedure and example *(continued)*

III. If a printer is attached and in MAN mode the use of flag 00 provides for manual and automatic mode. Assuming that the data have been properly summarized by the Σ+ key these options are described below, using the preceding example.

1. Manual mode -- flag 00 clear, MAN printer mode

If the steps described in part II-3 are followed the following print-out will be obtained:

```
Y-INT.= 2.79
SLOPE= 0.79
B.F.O.= 1.24
S<YX>= 0.32
S<a>= 0.52
S<b>= 0.09
S<BFO>= 0.13
R= 0.99
R↑2= 0.99
X= 4.00
Y-HAT= 5.95

X= 6.00
Y-HAT= 7.53

T<R>= 8.66
T<a>= 5.35
T<b>= 8.66
T<BFO=1>=1.94
T= 2.00
X= 4.00
Y-HAT= 5.95

C.I.= 5.50
    TO
      6.39

IND. CI= 5.16
    TO
         6.74

X= 6.00
Y-HAT= 7.53

C.I.= 7.13
    TO
      7.92
```

E. Program procedure and example *(continued)*

2. Manual mode -- flag 00 clear, NORM printer mode

Doing the same example but with the printer in NORM mode provides a somewhat better picture as to what keys have been pressed. Starting with step II-1, the example would appear as follows:

```
                    CLΣ
            7.00 ENTER↑
            5.00     Σ+
            5.00 ENTER↑
            3.00     Σ+
            9.00 ENTER↑
            8.00     Σ+
              XEQ "SLR"

Y-INT.= 2.79
SLOPE= 0.79
                    RUN
B.F.O.= 1.24
                  XEQ C
S(YX)= 0.32
S(a)= 0.52
S(b)= 0.09
                    RUN
S(BFO)= 0.13
                  XEQ D
R= 0.99
                    RUN
R↑2= 0.99
                  XEQ E
KEY X, R/S
            4.00    RUN
X= 4.00
Y-HAT= 5.95

KEY X, R/S
            6.00    RUN
X= 6.00
Y-HAT= 7.53

KEY X, R/S
                  XEQ d
T(R)= 8.66
                    RUN
T(a)= 5.35
T(b)= 8.66
                    RUN
T(BFO=1)=1.94

                  XEQ e
KEY T, R/S
                    RUN
T= 2.00
KEY X, R/S
            4.00    RUN
X= 4.00
Y-HAT= 5.95

C.I.= 5.50
    TO
      6.39

IND. CI?, R/S
                    RUN
IND. CI= 5.16
    TO
         6.74

KEY X, R/S
            6.00    RUN
X= 6.00
Y-HAT= 7.53

C.I.= 7.13
    TO
      7.92

IND. CI?, R/S
                    RUN
IND. CI= 6.77
    TO
         8.29

KEY X, R/S
```

E. Program procedure and example *(continued)*

3. Automatic mode -- flag 00 set, MAN printer mode

Starting with part II-3, if flag 00 is set, the following will be obtained automatically upon execution of SLR (R↓ in USER mode):

```
Y-INT.= 2.79
SLOPE= 0.79
B.F.O.= 1.24

S<YX>= 0.32
S<a>= 0.52
S<b>= 0.09
S<BFO>= 0.13

R= 0.99
R↑2= 0.99

T<R>= 8.66
T<a>= 5.35
T<b>= 8.66
T<BFO=1>=1.94
```

The calculator display should now show the prompt bb↑ee↑II, R/S. If one wants $\hat{Y}$ and confidence interval estimates for a beginning $X_0 = bb$, and ending with $X_0 = ee$, incremented by II, simply key these numbers in and press R/S. For example for $X_0 = 4$ to $X_0 = 10$ by increments of 2, and using a Student's t of 2.00:

Prompt	Input	Key	Output
-	4	↑	4.00
-	10	↑	10.00
-	2	R/S	-
KEY T, R/S	-	R/S	(see below)

```
*************
bbb.dddII: 4.00602

STUDENTS T= 2.00

X: 4.00
 Y-HAT= 5.95
 MCI= 5.50 TO 6.39
 ICI= 5.16 TO 6.74

X: 6.00
 Y-HAT= 7.53
 MCI= 7.13 TO 7.92
 ICI= 6.77 TO 8.29

X: 8.00
 Y-HAT= 9.11
 MCI= 8.49 TO 9.72
 ICI= 8.21 TO 10.00

X: 10.00
 Y-HAT= 10.68
 MCI= 9.75 TO 11.61
 ICI= 9.55 TO 11.82

*************
```

E. Program procedure and example *(continued)*

Note that the "bbb.dddII" number at the top of the table indicates bbb, the beginning X_0, ddd, the difference between beginning X_0 and ending X_0, and II, the increment value. The program was written so that any reasonable number of digits for bbb could be handled, but with the constraints that <u>ddd≤999</u> and <u>II≤99</u>. Exceed these constraints and results will be hard to interpret, therefore one should revert to the manual mode (flag 00 clear) for confidence interval estimation.

V. Additional notes and comments

1. Mean and standard deviation.

If one has previously assigned the MEAN and SDEV functions to say the COS and shift COS keys, using the example the following can be obtained:

Key	Output
COS	5.33 ($\bar{X}$)
X≷Y	7.00 ($\bar{Y}$)
shift COS	2.52 (s_x)
X≷Y	2.00 (s_y)

2. Recalling stored items.

Stored calculations and statistics (see Section A) can be displayed by RCL nn or VIEW nn, and with a printer attached and ON a complete listing can be obtained by keying .029 and XEQ PRREGX (print registers according to X). The example with printer in NORM mode follows:

```
        .029
            PRREGX
```

Register	Value		Register	Value	
R00=	0.00		R17=	2.00	t
R01=	5.33	$\bar{x}$	R18=	6.00	
R02=	12.67	Σx^2	R19=	7.53	
R03=	7.00	$\bar{y}$	R20=	0.37	
R04=	8.00	Σy^2	R21=	0.76	
R05=	10.00	Σxy	R22=	1.24	b'
R06=	2.79	a	R23=	1.25	s'_{YX}
R07=	0.79	b	R24=	0.13	$s_{b'}$
R08=	0.99	r	R25=	0.32	s_{YX}
R09=	0.52	s_a	R26=	0.00	
R10=	0.09	s_b	R27=	0.00	
R11=	16.00	ΣX	R28=	0.00	
R12=	98.00	ΣX^2	R29=	0.00	
R13=	21.00	ΣY			
R14=	155.00	ΣY^2			
R15=	122.00	ΣXY			
R16=	3.00	n			

E. Program procedure and example *(continued)*

3. Deletion of data

   Pairs of X and Y data can be deleted at any time (before or after SLR has been executed) by the use of the shift Σ- key as described in the HP-41C manual.

4. Use of the Σ+ key

   Summarizing the basic data using the Σ+ key will proceed much faster if the calculator is not in USER mode.

5. Order of calculations

   After SLR has been executed, label C (calculation of standard errors) must precede the t calculations (labels d or b) and/or the confidence interval calculations (label e); otherwise the order of pressing the various local label keys (after XEQ SLR) is immaterial, and each can be repeated if desired.

6. Use of the printer

   The program was not written to cope with all printer associated stoppages; however, the underlying philosophy used, and recommended action in this regard is summarized in the following table:

Status of Flag 00	Printer Status	Problem	Solution or Comment
clear	unattached	none	-
	attached, OFF	program stops at first AVIEW	turn printer ON or unplug it
	attached, ON	none	use MAN or NORM mode for printout
set	unattached	fast display then stoppage, displaying NONEXISTENT if control number is keyed and R/S pressed	plug in printer and turn ON
	attached, OFF	program stops at first AVIEW	turn printer ON
	attached, ON	none	use MAN or NORM mode for printout

In summary, to avoid problems, if the printer is attached, turn it ON; if the printer is not attached, don't set flag 00.

F. Formulas used

Many variations exist for the calculation of simple linear regression statistics. The following formulas in general were chosen for programming simplicity. Where feasible, the standard MEAN and SDEV function were employed. The Σ+ key provides the gross (raw) sums of squares and products. The following summary is according to the order of calculations itemized in Section E.

Note: lower case letters indicate deviations from the mean, i.e. "corrected", whereas capital letters are used to represent the gross sum of squares and products.

a. least squares line

(1) corrected sums of squares and products

$$\Sigma x^2 = \Sigma X^2 - n\bar{X}^2 \qquad \text{i.e.,} \qquad \Sigma x_i^2 = \Sigma(X_i - \bar{X})^2$$

$$\Sigma y^2 = \Sigma Y^2 - n\bar{Y}^2 \qquad \Sigma y_i^2 = \Sigma(Y_i - \bar{Y})^2$$

$$\Sigma xy = \Sigma XY - n\bar{X}\bar{Y} \qquad \Sigma x_i y_i = \Sigma(X_i - \bar{X})(Y_i - \bar{Y})$$

(2) regression coefficients

$$b = \frac{\Sigma xy}{\Sigma x^2}$$

$$a = \bar{Y} - b\bar{X}$$

$$b' = \frac{\Sigma XY}{\Sigma X^2}$$, b' = the slope of the line forced through the origin, $X = 0$, $Y = 0$

b. standard errors

$$s_{YX} = \sqrt{\frac{\Sigma y^2 - b\Sigma xy}{n-2}}$$

$$s_a = s_{YX}\sqrt{\frac{\Sigma X^2}{n\Sigma x^2}}$$

$$s_b = \frac{s_{YX}}{\sqrt{\Sigma x^2}}$$

$$s_{b'} = \frac{s'_{YX}}{\sqrt{\Sigma X^2}} \quad , \quad \text{where } s'_{YX} = \sqrt{\frac{\Sigma Y^2 - b'\Sigma XY}{n-1}}$$

F. Formulas used *(continued)*

c. correlation coefficient

$$r = \frac{\Sigma xy}{\sqrt{\Sigma x^2 \Sigma y^2}}$$

d. predicted Y

$$\hat{Y}_0 = a + bX_0$$

e. t tests

$$t_r = r\sqrt{\frac{n-2}{1-r^2}}$$

$$t_a = \frac{a}{s_a}$$

$$t_b = \frac{b}{s_b}$$

$$t_{b'=1} = \frac{b'-1}{s_{b'}}$$

f. confidence interval estimates

the general formula used was

C.I. for given $X_0 = \hat{Y}_0 \pm t s_{YX}\sqrt{C}$

where t = appropriate keyed-in Student's t

s_{YX} = standard error of estimate

and

$$C = \frac{1}{n} + \frac{(X_0 - \overline{X})^2}{\Sigma x^2} \quad \text{for } \underline{\text{mean}} \text{ C.I.}$$

or

$$C = 1 + \frac{1}{n} + \frac{(X_0 - \overline{X})^2}{\Sigma x^2} \quad \text{for } \underline{\text{individual}} \text{ C.I.}$$

G. Program listing

```
01♦LBL "SLR"
02♦LBL B
03 ΣREG 11
04 CF 03
05 MEAN
06 STO 01
07 X↑2
08 STO 02
09 X<>Y
10 STO 03
11 X↑2
12 STO 04
13 RCL 01
14 RCL 03
15 *
16 STO 05
17 RCL 16
18 CHS
19 ST* 02
20 ST* 04
21 ST* 05
22 RCL 12
23 ST+ 02
24 RCL 14
25 ST+ 04
26 RCL 15
27 ST+ 05
28 RCL 05
29 RCL 02
30 /
31 STO 07
32 RCL 01
33 *
34 CHS
35 RCL 03
36 +
37 STO 06
38 ADV
39 "Y-INT.= "
40 ARCL X
41 AVIEW
42 PSE
43 "SLOPE= "
44 ARCL 07
45 AVIEW
46 FC? 00
47 STOP
48 RCL 15
49 RCL 12
50 /
51 STO 22
52 "B.F.O.= "
53 ARCL X
54 AVIEW
55 FC? 00
56 STOP
57 ADV
58♦LBL C
59 RCL 04
60 RCL 05
61 RCL 07
62 *
63 -
64 FS? 02
65 RCL 26
66 FC? 02
67 RCL 16
68 2
69 -
70 /
71 SQRT
72 STO 25
73 RCL 02
74 SQRT
75 /
76 STO 10
77 RCL 12
78 RCL 16
79 RCL 02
80 *
81 /
82 SQRT
83 RCL 25
84 *
85 STO 09
86 "S(YX)= "
87 ARCL 25
88 AVIEW
89 FC? 00
90 PSE
91 "S(a)= "
92 ARCL 09
93 AVIEW
94 FC? 00
95 PSE
96 "S(b)= "
97 ARCL 10
98 AVIEW
99 FC? 00
100 STOP
101 RCL 14
102 RCL 22
103 RCL 15
104 *
105 -
106 FS? 02
107 RCL 26
108 FC? 02
109 RCL 16
110 1
111 -
112 /
113 SQRT
114 STO 23
115 RCL 12
116 SQRT
117 /
118 STO 24
119 "S(BFO)= "
120 ARCL X
121 AVIEW
122 FC? 00
123 RTN
124 ADV
125♦LBL D
126 RCL 05
127 RCL 02
128 RCL 04
129 *
130 SQRT
131 /
132 STO 08
133 "R= "
134 ARCL X
135 AVIEW
136 FC? 00
137 STOP
138 X↑2
139 "R↑2= "
140 ARCL X
141 AVIEW
142 FC? 00
143 RTN
144 ADV
145♦LBL d
146 FS? 02
147 RCL 26
148 FC? 02
149 RCL 16
150 2
```

G. Program listing *(continued)*


```
151 -
152 1
153 RCL 08
154 X↑2
155 -
156 /
157 SQRT
158 RCL 08
159 *
160 "T(R)= "
161 ARCL X
162 AVIEW
163 FC? 00
164 STOP
165♦LBL b
166 RCL 06
167 RCL 09
168 /
169 "T(a)= "
170 ARCL X
171 AVIEW
172 FC? 00
173 PSE
174 RCL 07
175 RCL 10
176 /
177 "T(b)= "
178 ARCL X
179 AVIEW
180 FC? 00
181 STOP
182 RCL 22
183 1
184 -
185 RCL 24
186 /
187 "T(BFO=1)="
188 ARCL X
189 AVIEW
190 FC? 00
191 STOP
192 ADV
193 FS? 00
194 GTO "PCI"
195♦LBL E
196 "KEY X, R/S"
197 PROMPT
198 CF 22
199♦LBL 01
200 STO 18
201 "X= "
202 ARCL X
203 AVIEW
204 RCL 07
205 *
206 RCL 06
207 +
208 STO 19
209 "Y-HAT= "
210 ARCL X
211 AVIEW
212 ADV
213 PSE
214 FS? 03
215 RTN
216 GTO E
217♦LBL e
218 FS? 03
219 GTO 03
220 2
221 "KEY T, R/S"
222 PROMPT
223 CF 22
224 STO 17
225 "T= "
226 ARCL 17
227 AVIEW
228 SF 03
229♦LBL 03
230 FC? 08
231 XEQ E
232 FS?C 08
233 XEQ 01
234 PSE
235 RCL 18
236 RCL 01
237 -
238 X↑2
239 RCL 02
240 /
241 RCL 16
242 1/X
243 +
244 STO 20
245♦LBL 02
246 SQRT
247 RCL 17
248 *
249 RCL 25
250 *
251 STO 21
252 CHS
253 RCL 19
254 +
255 FS? 04
256 "IND. CI= "
257 FC? 04
258 "C.I.= "
259 ARCL X
260 AVIEW
261 PSE
262 "    TO"
263 AVIEW
264 PSE
265 RCL 21
266 +
267 RCL 21
268 +
269 FC? 04
270 "      "
271 FS? 04
272 "        "
273 ARCL X
274 AVIEW
275 ADV
276 PSE
277 FS?C 04
278 GTO 03
279 "IND. CI?, R/S"
280 PROMPT
281 FS? 22
282 SF 08
283 FS?C 22
284 GTO e
285 RCL 20
286 1
287 +
288 SF 04
289 GTO 02
290♦LBL "PCI"
291 CLST
292 FS? 36
293 8
294 FS? 37
295 4
296 +
297 FS? 38
298 2
299 +
300 FS? 39
```

G. Program listing *(continued)*


```
1
+
STO 28
CF 21
"X CONTROL:"
AVIEW
PSE
SF 21
"bb↑ee↑II, R/S"
PROMPT
X<> Z
STO 29
-
1 E3
/
X<> Y
1 E5
/
+
STO 00
XEQ 10
FIX 5
"bbb.dddII: "
RCL 29
+
ARCL X
ACA
PRBUF
ADV
2
"KEY T, R/S"
PROMPT
STO 17
FIX IND 28
"STUDENTS T= "
ARCL 17
ACA
PRBUF
ADV
♦LBL 07
RCL 29
RCL 00
INT
+
STO 18
RCL 07
*
RCL 06
+
STO 19
"X: "
ARCL 18
AVIEW
" Y-HAT= "
ARCL 19
AVIEW
" MCI= "
ASTO 27
RCL 18
RCL 01
-
X↑2
RCL 02
/
RCL 16
1/X
+
STO 20
SF 04
♦LBL 08
SQRT
RCL 17
*
RCL 25
*
STO 21
CHS
RCL 19
+
CLA
ARCL 27
ARCL X
RCL 21
+
RCL 21
+
"⊢ TO "
ARCL X
FC? 55
AVIEW
FS? 55
PRA
FC? 04
ADV
FS?C 04
GTO 09
ISG 00
GTO 07
FIX IND 28
XEQ 10
ADV
ADV
ADV
ADV
ADV
ADV
STOP
♦LBL 10
SF 12
"************"
ACA
ADV
CF 12
RTN
♦LBL 09
" ICI= "
ASTO 27
RCL 20
1
+
GTO 08
END
```

```
                    CAT 1
LBL'SLR
LBL'PCI
END          873 BYTES
```

Program No. 41F020

Calculator: HP-41C/CV

Program Name: MSLR (Master Simple Linear Regression)

Author: T.W. Beers

Date: 1982

Purpose: MSLR is a steering program to be used in conjunction with SLR (Program No. 41F019) to accomodate ungrouped, grouped, or weighted data and to extend the prompting, correction, and/or deletion capabilities of that program. With the insertion of appropriate subrountines it is anticipated that MSLR can also be used to transform the input data and achieve linear approximations to certain non-linear models.

A. Storage registers

Register	Use
00	the current frequency, F_i, for grouped data (flag 01 set) or the current weight, W_i for weighted data (flag 02 set)
01	the current value of X (i.e., X_i) for grouped or weighted data
02	the current value of Y (i.e., Y_i) for grouped or weighted data
03	current F_iY_i or W_iY_i
04	current $F_iY_i^2$ or $W_iY_i^2$
05	current F_iX_i or W_iX_i
06	current $F_iX_i^2$ or $W_iX_i^2$
07	current $F_iX_iY_i$ or $W_iX_iY_i$
08-10	used only in SLR program
11	ΣX, ΣFX, or ΣWX

A. Storage registers *(continued)*

Register	Use
12	ΣX^2, ΣFX^2, or ΣWX^2
13	ΣY, ΣFY, or ΣWY
14	ΣY^2, ΣFY^2, or ΣWY^2
15	ΣXY, ΣFXY, or ΣWXY
16	n = number of observations if flags 01 and 02 are clear ΣF = sum of the frequencies if flag 01 is set ΣW = sum of the weights if flag 02 is set
17-25	used only in SLR program
26	c = number of classes of data if either flag 01 or 02 is set
27-29	used only in SLR program

B. Labels

Global	Use
MSLR	program start, assigns statistical registers, clears registers 00-29, and prompts for desired flag setting (if flag 03 is clear)
DLT	for deletion of data after summary (by R/S depression) <u>and</u> after another data point has been summarized (before or after SLR); assumes the to-be-deleted data point has been keyed as either Y↑X, Y↑X↑F, or Y↑X↑W, as appropriate
<u>Local</u>	
c	for deletion of data <u>before</u> SLR has been used and the "faulty" data point has <u>just been</u> summarized (by R/S depression); in this case, the faulty data need not be re-keyed
00	in MSLR to clear registers 00-29
01	used by label c and DLT for correction or deletion of data; ungrouped case
02	used by label c and DLT for correction or deletion of data; grouped and weighted cases
04	used in label 01 and by label DLT to print the deleted data point, ungrouped case
05	used by label DLT to process and print, delete and print (twice) the deleted data point; grouped and weighted cases

B. Labels *(continued)*

Local	Use
10	prints and summarizes ungrouped data point
11	prompts for grouped data point (Y↑X↑F)
12	prompts for weighted data point (Y↑X↑W)
13	prints and summarizes grouped and weighted data points
14	directs control to labels 11 or 12 or prompts for ungrouped data point (Y↑X)

C. Flags

Type and Number	Use
User:	
01	if set (externally) implies the grouped data case; frequencies are prompted for and required
02	if set (externally) implies the weighted data case; weights are prompted for and required
03	if set (externally) the initial flag reminders and prompts are skipped; note that flag 03 is cleared when SLR is executed and therefore may have to be re-set manually if MSLR is repeated
07	set in label 05 to enable executing labels 13 and 02 as subroutines as required in the label DLT option
12	the double-wide flag; set and cleared early in MSLR
System:	
55	the printer existence flag; tested throughout to enable printing of information if printer is attached

D. Size and key assignments

SIZE: $\geq$ 030 (since MSLR assumes SLR will be used, the SIZE must be at least as large as SLR requires, i.e., 030)

D. Size and key assignments *(continued)*

Suggested key assignments:

MSLR: on shift X≷Y (logic: this is "CLΣ" which normally would precede a statistical summary!)

DLT: on shift TAN (this is the "previously summarized" correction routine and the deletion routine)

c: on shift C (this is done internally and is the "just summarized" correction key)

E. Program procedure and example

I. Load the program into the calculator

II. XEQ MSLR (shift X≷Y in USER mode) and observe the following reminders:

```
 *CLEARING*        (registers 00-29 are set
                   to zero)
 FLAG SET ?
00 AUTO PRNT
01 FOR FREQ.
02 FOR WTD.
SF XX, R/S
```

Thus, the options are:

(1) no flags set implies ungrouped data
(2) flag 01 set implies grouped data
(3) flag 02 set implies weighted data
(4) flag 00 set implies automatic printing of the results
(5) note that if flag 03 is set, these reminders are skipped upon execution of MSLR

Let's first assume that the printer is <u>not</u> attached (therefore make sure flag 00 is clear) and that the same example as described in the SLR program for ungrouped data is used:

1. Ungrouped data:

Y	X
7	5
5	3
9	8

E. Program procedure and example *(continued)*

The procedure would be as follows: (assume FIX 2)

	Step	Prompt	Input	Key (*Example*)	Output
a.	start and	SF XX, R/S	-	R/S	
	data entry	KEY Y↑X, R/S	7↑5	R/S	1.00
		-	5↑3	R/S	2.00
		-	9↑8	R/S	3.00
b.	execute SLR	-	-	R↓	Y-INT= 2.79 SLOPE= 0.79

c. continue as described in SLR

d. data correction options:

(1) just summarized data -- assume the second observation was keyed in as 5↑33 and R/S pressed; if we realize this before keying in the next data point, the faulty point can be deleted by shift C (i.e., c); the prompt, KEY Y↑X, R/S, will be displayed, at which time we then key in the correct values, 5↑3, R/S and proceed with 9↑8, R/S, then R↓ to XEQ SLR

(2) previously summarized data -- assume we again had used 5↑33, R/S but did not realize it until after 9↑8, R/S; in this case to undo the error, we must re-key the faulty point then XEQ DLT (shift TAN in USER mode). Therefore, assuming 7↑5 R/S
5↑33 R/S
9↑8 R/S

oops! (5↑33 should be 5↑3),
so 5↑33, shift TAN to delete,
then at the KEY Y↑X, R/S prompt,
5↑3, R/S, then R↓ to XEQ SLR

(Note, this procedure can also be used to delete data points after SLR has been executed).

2. Grouped data: (i.e., flag 01 must be set)

Y	X	F (= frequency)
7	5	3
5	3	4
9	8	2

E. Program procedure and example *(continued)*

Again assume FIX 2 and assume we have XEQ MSLR and the prompt is SF XX, R/S.

	Step	Prompt	Example Input	Example Key	Example Output
a.	set flag	SF XX, R/S	-	shift SF 01	0.00
	and start	-	-	R/S	-
b.	data entry	Y↑X↑F, R/S	7↑5↑3	R/S	1.00
		-	5↑3↑4	R/S	2.00
			9↑8↑2	R/S	3.00
c.	execute SLR	-	-	R↓	Y-INT.= 2.70
					SLOPE= 0.81

d. continue as described in SLR

e. data correction options: same as described for ungrouped data

3. Weighted data: (i.e., flag 02 must be set, and flag 01 clear!)

Y	X	W (= weight)
0	1	1
1	2	.25
5	5	.04
10	10	.01

Again assume FIX 2 and the prompt stands at SF XX, R/S.

	Step	Prompt	Example Input	Example Key	Example Output
a.	clear flag,	SF XX, R/S	-	shift CF 01	0.00
	set flag and		-	shift SF 02	0.00
	start		-	R/S	-
b.	data entry	Y↑X↑W, R/S	0↑1↑1	R/S	1.00
			1↑2↑.25	R/S	2.00
			5↑5↑.04	R/S	3.00
			10↑10↑.01	R/S	4.00
c.	execute SLR	-	-	R↓	Y-HAT=-1.17
					SLOPE= 1.15

d. continued as described in SLR

e. data correction options: same as described for ungrouped data

E. Program procedure and example *(continued)*

III. Printed output with flag 00 clear

Using the same three examples described in section II, and assuming R/S is pressed between the SLR groups of answers, the printed output will look like this (up to the calculation of $\hat{Y}$ values)

1. Ungrouped case:

Printer in MAN mode

```
 *CLEARING*
 FLAG SET ?
00 AUTO PRNT
01 FOR FREQ.
02 FOR WTD.
SF XX, R/S
KEY Y↑X, R/S

7.00↑ 5.00
5.00↑ 3.00
9.00↑ 8.00

Y-INT.= 2.79
SLOPE= 0.79
B.F.O.= 1.24

S<YX>= 0.32
S<a>= 0.52
S<b>= 0.09
S<BFO>= 0.13

R= 0.99
R↑2= 0.99

T<R>= 8.66
T<a>= 5.35
T<b>= 8.66
T<BFO=1>=1.94
```

Printer in NORM mode

```
                  XEQ "MSLR"
   *CLEARING*
 FLAG SET ?
00 AUTO PRNT
01 FOR FREQ.
02 FOR WTD.
SF XX, R/S
                         RUN
KEY Y↑X, R/S

                7.00 ENTER↑
                5.00     RUN
7.00↑ 5.00
                5.00 ENTER↑
                3.00     RUN
5.00↑ 3.00
                9.00 ENTER↑
                8.00     RUN
9.00↑ 8.00
                   XEQ "SLR"

Y-INT.= 2.79
SLOPE= 0.79
                         RUN
B.F.O.= 1.24
                         RUN

S<YX>= 0.32
S<a>= 0.52
S<b>= 0.09
                         RUN
S<BFO>= 0.13
                         RUN

R= 0.99
                         RUN
R↑2= 0.99
                         RUN

T<R>= 8.66
                         RUN
T<a>= 5.35
T<b>= 8.66
                         RUN
T<BFO=1>=1.94
```

III. *(continued)*

2. Grouped case with an error detected just after summarizing the second observation (i.e., use key c to correct error).

Printer in MAN mode

```
  *CLEARING*
 FLAG SET ?
00 AUTO PRNT
01 FOR FREQ.
02 FOR WTD.
SF XX, R/S
 Y↑X↑F, R/S

7.00↑ 5.00↑ 3.00
5.00↑ 3.00↑ 44.00
5.00, 3.00, 44.00 <->
 Y↑X↑F, R/S

5.00↑ 3.00↑ 4.00
9.00↑ 8.00↑ 2.00

Y-INT.= 2.70
SLOPE= 0.81
B.F.O.= 1.29

S<YX>= 0.21
S<a>= 0.19
S<b>= 0.04
S<BFO>= 0.07

R= 0.99
R↑2= 0.99

T<R>= 21.96
T<a>= 14.22
T<b>= 21.96
T<BFO=1>=4.16
```

Printer in NORM mode

```
                  XEQ "MSLR"
  *CLEARING*
 FLAG SET ?
00 AUTO PRNT
01 FOR FREQ.
02 FOR WTD.
SF XX, R/S
                      SF 01
                        RUN
 Y↑X↑F, R/S

                7.00 ENTER↑
                5.00 ENTER↑
                3.00    RUN
7.00↑ 5.00↑ 3.00
                5.00 ENTER↑
                3.00 ENTER↑
               44.00    RUN
5.00↑ 3.00↑ 44.00
                      XEQ c
5.00, 3.00, 44.00 <->
 Y↑X↑F, R/S

                5.00 ENTER↑
                3.00 ENTER↑
                4.00    RUN
5.00↑ 3.00↑ 4.00
                9.00 ENTER↑
                8.00 ENTER↑
                2.00    RUN
9.00↑ 8.00↑ 2.00
```

```
                  XEQ "SLR"
Y-INT.= 2.70
SLOPE= 0.81
                        RUN
B.F.O.= 1.29
                        RUN

S<YX>= 0.21
S<a>= 0.19
S<b>= 0.04
                        RUN
S<BFO>= 0.07
                        RUN

R= 0.99
                        RUN
R↑2= 0.99
                        RUN

T<R>= 21.96
                        RUN
T<a>= 14.22
T<b>= 21.96
                        RUN
T<BFO=1>=4.16
```

III. *(continued)*

3. Weighted case with an error detected in the second observation, after the third and fourth observations have been summarized (i.e., the DLT subroutine, assigned to shift TAN, must be used).

Printer in MAN mode

```
 *CLEARING*
 FLAG SET ?
00 AUTO PRNT
01 FOR FREQ.
02 FOR WTD.
SF XX, R/S
 Y↑X↑W, R/S

0.00↑ 1.00↑ 1.00
1.00↑ 22.00↑ 0.25
5.00↑ 5.00↑ 0.04
10.00↑ 10.00↑ 0.01
1.00↑ 22.00↑ 0.25
1.00, 22.00, 0.25 <->
1.00, 22.00, 0.25 <->
 Y↑X↑W, R/S

1.00↑ 2.00↑ 0.25

Y-INT.= -1.17
SLOPE= 1.15
B.F.O.= 0.63

S(YX)= 0.08
S(a)= 0.11
S(b)= 0.07
S(BFO)= 0.24

R= 1.00
R↑2= 0.99

T(R)= 17.73
T(a)= -10.29
T(b)= 17.73
T(BFO=1)=-1.57
```

Printer in NORM mode

```
                   XEQ "MSLR"
 *CLEARING*
 FLAG SET ?
00 AUTO PRNT
01 FOR FREQ.
02 FOR WTD.
SF XX, R/S
                        CF 01
                        SF 02
                          RUN
 Y↑X↑W, R/S

               0.00 ENTER↑
               1.00 ENTER↑
               1.00    RUN
0.00↑ 1.00↑ 1.00
               1.00 ENTER↑
              22.00 ENTER↑
                .25    RUN
1.00↑ 22.00↑ 0.25
               5.00 ENTER↑
               5.00 ENTER↑
                .04    RUN
5.00↑ 5.00↑ 0.04
              10.00 ENTER↑
              10.00 ENTER↑
                .01    RUN
10.00↑ 10.00↑ 0.01
               1.00 ENTER↑
              22.00 ENTER↑
                .25
                 XEQ "DLT"
1.00↑ 22.00↑ 0.25
1.00, 22.00, 0.25 <->
1.00, 22.00, 0.25 <->
 Y↑X↑W, R/S

               1.00 ENTER↑
               2.00 ENTER↑
                .25    RUN
1.00↑ 2.00↑ 0.25
```

```
                 XEQ "SLR"

Y-INT.= -1.17
SLOPE= 1.15
                       RUN
B.F.O.= 0.63
                       RUN

S(YX)= 0.08
S(a)= 0.11
S(b)= 0.07
                       RUN
S(BFO)= 0.24
                       RUN

R= 1.00
                       RUN
R↑2= 0.99
                       RUN

T(R)= 17.73
                       RUN
T(a)= -10.29
T(b)= 17.73
                       RUN
T(BFO=1)=-1.57
```

E. Program procedure and example *(continued)*

IV. Printed output with flag 00 set ("automatic" mode)

The automatic printout facilitates rapid analysis for data, and when used in conjunction with the delete subroutine, DLT, enables re-analysis with selected points deleted.

1. Ungrouped data

Consider the following data described in a problem posed by Draper and Smith (p. 68 in Applied Regression Analysis, 2nd Ed., 1981, John Wiley & Sons, N.Y.)

Per bottle price of vintage port wine (Y)	Year (X)
$ 50.00	1890
35.00	1900
25.00	1920
11.98	1931
15.00	1934
13.00	1935
6.98	1940
10.00	1941
5.99	1944
8.98	1948
6.98	1950
4.99	1952
5.98	1955
4.98	1960

Say we want the complete analysis, then $\hat{Y}$ and .95 confidence intervals for the years 1900 to 1960 by increments of 20. The pertinent steps, then, are:

1. XEQ MSLR (shift X$\gtrless$Y)
2. at the prompt SF XX R/S, SF 00
   $\left.\begin{array}{l}\text{CF 01}\\ \text{CF 02}\end{array}\right\}$ if needed
3. key in data
4. XEQ SLR (R↓)
5. at the prompt bb↑ee↑II, R/S, 1900↑
   1960↑
   20 R/S
6. at the prompt KEY T, R/S, 2.18 R/S
   (since .95 Student's t for 14 - 2 = 12df is 2.18)

Now, let's assume that the 1941 price is an "outlier" and we want to delete it and run the analysis again. Therefore:

7. Key 10↑ 1941 and XEQ DLT (shift TAN)
8. XEQ SLR (R↓)

E. Program procedure and example *(continued)*

9. use the same bb, ee, and II but now Student's t = 2.20 for 11 degrees of freedom and .95 confidence level

The results are shown on the next page for the two cases described above. The careful observer will note that linear regression (without a transformation) is not appropriate for this problem, which is implicit in the presentation by Draper and Smith.

1. *(continued)*

Ungrouped case showing complete analysis and with one point deleted. Printer in MAN mode.

Complete case

```
*CLEARING*
FLAG SET ?
00 AUTO PRNT
01 FOR FREQ.
02 FOR WTD.
SF XX, R/S
KEY Y↑X, R/S

50.00↑ 1890.00
35.00↑ 1900.00
25.00↑ 1920.00
11.98↑ 1931.00
15.00↑ 1934.00
13.00↑ 1935.00
6.98↑ 1940.00
10.00↑ 1941.00
5.99↑ 1944.00
8.98↑ 1948.00
6.98↑ 1950.00
4.99↑ 1952.00
5.98↑ 1955.00
4.98↑ 1960.00

Y-INT.= 1233.81
SLOPE= -0.63
B.F.O.= 0.01

S<YX>= 3.80
S<a>= 100.62
S<b>= 0.05
S<BFO>= 1.85E-3

R= -0.96
R↑2= 0.92

T<R>= -12.12
T<a>= 12.26
T<b>= -12.12
T<BFO=1>=-535.81
```

```
*************
bbb.dddII: 1900.06020

STUDENTS T= 2.18

X: 1900.00
 Y-HAT= 37.13
 MCI= 32.51 TO 41.74
 ICI= 27.65 TO 46.60

X: 1920.00
 Y-HAT= 24.53
 MCI= 21.69 TO 27.37
 ICI= 15.78 TO 33.28

X: 1940.00
 Y-HAT= 11.93
 MCI= 9.67 TO 14.20
 ICI= 3.35 TO 20.51

X: 1960.00
 Y-HAT= -0.66
 MCI= -4.19 TO 2.87
 ICI= -9.66 TO 8.34

*************
```

Deleted case

```
10.00, 1941.00 <->
KEY Y↑X, R/S

Y-INT.= 1231.20
SLOPE= -0.63
B.F.O.= 0.01

S<YX>= 3.94
S<a>= 104.81
S<b>= 0.05
S<BFO>= 1.99E-3

R= -0.96
R↑2= 0.92

T<R>= -11.60
T<a>= 11.75
T<b>= -11.60
T<BFO=1>=-498.25

*************
bbb.dddII: 1900.06020

STUDENTS T= 2.20

X: 1900.00
 Y-HAT= 37.18
 MCI= 32.33 TO 42.02
 ICI= 27.24 TO 47.12

X: 1920.00
 Y-HAT= 24.61
 MCI= 21.59 TO 27.63
 ICI= 15.42 TO 33.80

X: 1940.00
 Y-HAT= 12.04
 MCI= 9.57 TO 14.51
 ICI= 3.02 TO 21.06

X: 1960.00
 Y-HAT= -0.53
 MCI= -4.33 TO 3.27
 ICI= -10.00 TO 8.95

*************
```

E. Program procedure and example *(continued)*

2. Grouped or weighted data

Since the difference in the analysis is minor if one uses either the grouped or the weighted approach, and since the weighted situation is probably more common, only that case will be described. The differences between the two are described in Section F.

For an example, consider data given by Meyer (H. Arthur Meyer, 1953. Forest Mensuration. Penns Valley Publishers. State College, PA.) concerning average tree diameter growth by diameter classes:

Average diameter growth Y	Average diameter X	Number of observations (weight) W
inches	inches	
0.137	4.2	6
0.203	5.3	26
0.164	6.4	14
0.195	7.5	13
0.166	8.5	24
0.166	9.4	15
0.122	10.6	9
0.161	11.4	6
0.158	12.5	11
0.129	13.2	4
0.167	14.5	3
0.104	15.4	3
0.172	16.3	6
0.068	17.3	1
0.132	20.7	1
0.068	23.6	1
		143

The pertinent steps in the analysis are:

1. XEQ MSLR (shift X$\gtrless$Y)
2. at the prompt SF XX R/S, SF 00
SF 02 (for weighted data)
CF 01 if needed
3. key in data (use FIX 3)
4. XEQ SLR (R↓)
5. at the prompt bb↑ee↑II, R/S, 6↑
24↑
2 R/S
6. at the prompt KEY T, R/S, 2.145 R/S
(since c = 16 = number of classes, and .95 Student's t for 16 - 2 = 14 df is 2.145)

E. Program procedure and example *(continued)*

Noting that the last 3 data points are based on only 1 tree each let's delete them and re-run the analysis. Furthermore, say we want the deleted data printed in FIX 3, the various fitted coefficients in FIX 5, then the predicted Y and confidence interval table in FIX 3 again. Therefore:

7. key .068 ↑ 17.3 ↑ 1 XEQ DLT (shift TAN)
   key .132 ↑ 20.7 ↑ 1 XEQ DLT (shift TAN)
   key .068 ↑ 23.6 ↑ 1 XEQ DLT (shift TAN)
8. shift FIX 5
9. XEQ SLR (R↓)
10. 3 STO 28 (register 28 is used at this point in the program to control the decimal FIX)
11. for $\hat{Y}$ and CI, use 6↑18↑4, but now Student's $t = 2.201$ for 11 degrees of freedom and .95 confidence level

2. *(continued)* Weighted case, showing complete analysis, and analysis with three classes deleted. Printer in MAN mode.

Complete case

```
*CLEARING*
 FLAG SET ?
00 AUTO PRNT
01 FOR FREQ.
02 FOR WTD.
SF XX, R/S
 Y↑X↑W, R/S

0.137↑ 4.200↑ 6.000
0.203↑ 5.300↑ 26.000
0.164↑ 6.400↑ 14.000
0.195↑ 7.500↑ 13.000
0.166↑ 8.500↑ 24.000
0.166↑ 9.400↑ 15.000
0.122↑ 10.600↑ 9.000
0.161↑ 11.400↑ 6.000
0.158↑ 12.500↑ 11.000
0.129↑ 13.200↑ 4.000
0.167↑ 14.500↑ 3.000
0.104↑ 15.400↑ 3.000
0.172↑ 16.300↑ 6.000
0.068↑ 17.300↑ 1.000
0.132↑ 20.700↑ 1.000
0.068↑ 23.600↑ 1.000

Y-INT.= 0.208
SLOPE= -0.005
B.F.O.= 0.015

S(YX)= 0.070
S(a)= 0.016
S(b)= 0.002
S(BFO)= 0.002

R= -0.598
R↑2= 0.358

T(R)= -2.793
T(a)= 13.034
T(b)= -2.793
T(BFO=1)=-467.734
```

```
************
bbb.dddII: 6.01802

STUDENTS T= 2.145

X: 6.000
 Y-HAT= 0.181
 MCI= 0.164 TO 0.197
 ICI= 0.029 TO 0.332

X: 8.000
 Y-HAT= 0.172
 MCI= 0.158 TO 0.185
 ICI= 0.020 TO 0.323

X: 10.000
 Y-HAT= 0.162
 MCI= 0.149 TO 0.175
 ICI= 0.011 TO 0.314

X: 12.000
 Y-HAT= 0.153
 MCI= 0.137 TO 0.170
 ICI= 0.002 TO 0.305

X: 14.000
 Y-HAT= 0.144
 MCI= 0.123 TO 0.166
 ICI= -0.008 TO 0.296

X: 16.000
 Y-HAT= 0.135
 MCI= 0.108 TO 0.162
 ICI= -0.018 TO 0.288

X: 18.000
 Y-HAT= 0.126
 MCI= 0.092 TO 0.160
 ICI= -0.028 TO 0.280

X: 20.000
 Y-HAT= 0.117
 MCI= 0.076 TO 0.157
 ICI= -0.039 TO 0.273

X: 22.000
 Y-HAT= 0.108
 MCI= 0.061 TO 0.155
 ICI= -0.050 TO 0.265

X: 24.000
 Y-HAT= 0.098
 MCI= 0.045 TO 0.152
 ICI= -0.061 TO 0.258

************
```

Deleted case

```
0.068↑ 17.300↑ 1.000
0.068, 17.300, 1.000 <->
0.068, 17.300, 1.000 <->
 Y↑X↑W, R/S

0.132↑ 20.700↑ 1.000
0.132, 20.700, 1.000 <->
0.132, 20.700, 1.000 <->
 Y↑X↑W, R/S

0.068↑ 23.600↑ 1.000
0.068, 23.600, 1.000 <->
0.068, 23.600, 1.000 <->
 Y↑X↑W, R/S


Y-INT.= 0.20388
SLOPE= -0.00402
B.F.O.= 0.01640

S(YX)= 0.07592
S(a)= 0.01892
S(b)= 0.00202
S(BFO)= 0.00222

R= -0.51569
R↑2= 0.26594

T(R)= -1.99626
T(a)= 10.77358
T(b)= -1.99626
T(BFO=1)=-442.34533

*************
bbb.dddII: 6.01204

STUDENTS T= 2.201

X: 6.000
 Y-HAT= 0.180
 MCI= 0.161 TO 0.199
 ICI= 0.012 TO 0.348

X: 10.000
 Y-HAT= 0.164
 MCI= 0.149 TO 0.179
 ICI= -0.004 TO 0.331

X: 14.000
 Y-HAT= 0.148
 MCI= 0.121 TO 0.174
 ICI= -0.022 TO 0.317

X: 18.000
 Y-HAT= 0.131
 MCI= 0.088 TO 0.175
 ICI= -0.041 TO 0.304

*************
```

F. Formulas used

1. Ungrouped data

The same formulas as shown in section F of program SLR were used in the case of ungrouped data after the basic sums, sums of squares and products were accumulated using the $\Sigma+$ function.

2. Grouped and weighted data

Here, the procedure was to accumulate the weighted sums, weighted sums of squares and products, and store them in the statistical registers, i.e., ΣFX or ΣWX in R_{11}, ΣFX^2 or ΣWX^2 in R_{12}, etc. After that, nearly all the formulas used for the ungrouped case apply, with appropriate symbol changes; for example:

$$\Sigma Fx^2 = \Sigma FX^2 - n\overline{X}^2, \quad \text{where} \quad n = \Sigma F$$

$$\text{and} \quad \overline{X} = \frac{\Sigma FX}{\Sigma F}$$

or

$$\Sigma Wy^2 = \Sigma WY^2 - n\overline{Y}^2, \quad \text{where} \quad n = \Sigma W$$

$$\text{and} \quad \overline{Y} = \frac{\Sigma WY}{\Sigma W}$$

The minor exceptions to this procedure are discussed below.

a. Grouped data

It is important to realize that this is the case where <u>identical</u> Y's and X's occur more than once, and repetitive entry of such pairs is not desirable. Thus, in this situation, the number of observations, n, is equal to $\sum_{i=1}^{c} F$, where there are c groups. Therefore, since registers 11 through 16 are "loaded" with ΣFX, ΣFX^2, ΣFY, ΣFY^2, ΣFXY, and ΣF, the ungrouped procedure in SLR is totally appropriate.

b. Weighted data

Weighted data arise in regression generally because (1) one has c classes of data where $(\overline{X}_i, \overline{Y}_i)$ are based on different numbers of observations (W_i) or (2) one wants to weight the c data points to correct for (or lessen) the condition known as heterogeneous variance, or (3) a combination of (1) and (2). In any of these cases, no error will be incurred by proceding as in the grouped case described in 2a, and letting ΣW replace ΣF <u>except</u> for the following; in weighted regression:

F. Formulas used *(continued)*

$$s_{YX} = \sqrt{\frac{\Sigma Wy^2 - b\Sigma Wxy}{c - 2}} \quad ,$$

$$t_r = r\sqrt{\frac{c - 2}{1 - r^2}} \quad ,$$

and $$s'_{YX} = \sqrt{\frac{\Sigma WY^2 - b'\Sigma WXY}{c - 2}}$$

Examination of the SLR program will show that in the calculation of these three estimates, R_{26} (containing c) is recalled and used for the weighted case (flag 02 set) rather than R_{16} (containing n, ΣF or ΣW).

G. Program listing

```
01♦LBL "MSLR"
02 ΣREG 11
03 .029
04 0
05 SF 12
06 " *CLEARING*"
07 AVIEW
08 CF 12
09♦LBL 00
10 STO IND Y
11 ISG Y
12 GTO 00
13 FS? 03
14 GTO 14
15 " FLAG SET ?"
16 AVIEW
17 PSE
18 "00 AUTO PRNT"
19 AVIEW
20 PSE
21 "01 FOR FREQ."
22 AVIEW
23 PSE
24 "02 FOR WTD."
25 AVIEW
26 PSE
27 "SF XX, R/S"
28 AVIEW
29 STOP
30♦LBL 14
31 FS? 01
32 GTO 11
33 FS? 02
34 GTO 12
35 "KEY Y↑X, R/S"
36 AVIEW
37 ADV
38 STOP
39♦LBL 10
40 CLA
41 ARCL Y
42 "⊦↑ "
43 ARCL X
44 FS? 55
45 PRA
46 Σ+
47 STOP
48 GTO 10
49♦LBL 01
50 CLA
51 RDN
52 LASTX
53♦LBL 04
54 ARCL Y
55 "⊦, "
56 ARCL X
57 "⊦ (-)"
58 FS? 55
59 PRA
60 Σ-
61 GTO 14
62♦LBL c
63 FS? 01
64 GTO 02
65 FS? 02
66 GTO 02
67 GTO 01
68♦LBL 02
69 CLA
70 RCL 00
71 ST- 16
72 RCL 03
73 ST- 13
74 RCL 04
75 ST- 14
76 RCL 05
77 ST- 11
78 RCL 06
79 ST- 12
80 RCL 07
81 ST- 15
82 ARCL 02
83 "⊦, "
84 ARCL 01
85 "⊦, "
86 ARCL 00
87 "⊦ (-)"
88 FS? 55
89 PRA
90 1
91 ST- 26
92 FS? 07
93 RTN
94 GTO 14
95♦LBL 05
96 SF 07
97 XEQ 13
98 XEQ 02
99 CF 07
100 GTO 02
101♦LBL 11
102 " Y↑X↑F, R/S"
103 AVIEW
104 ADV
105 STOP
106 GTO 13
107♦LBL 12
108 " Y↑X↑W, R/S"
109 AVIEW
110 ADV
111 STOP
112♦LBL 13
113 CLA
114 ARCL Z
115 "⊦↑ "
116 ARCL Y
117 "⊦↑ "
118 ARCL X
119 FS? 55
120 PRA
121 STO 00
122 ST+ 16
123 RDN
124 STO 01
125 RDN
126 STO 02
127 RCL 00
128 *
129 STO 03
130 ST+ 13
131 RCL 02
132 *
133 STO 04
134 ST+ 14
135 RCL 01
136 RCL 00
137 *
138 STO 05
139 ST+ 11
140 RCL 01
141 *
142 STO 06
143 ST+ 12
144 RCL 00
145 RCL 01
146 RCL 02
147 *
148 *
149 STO 07
150 ST+ 15
151 1
152 ST+ 26
153 RCL 26
154 FS? 07
155 RTN
156 STOP
157 GTO 13
158♦LBL "DLT"
159 CLA
160 FS? 01
161 GTO 05
162 FS? 02
163 GTO 05
164 GTO 04
165 END

LBL'MSLR
LBL'DLT
END          383 BYTES
```

Calculator: HP-41CV or HP-41C with 3 single density memory modules

Program Name: VOL17 (17 species board foot volume calculations)

Author: T.W. Beers

Date: April, 1981

Purpose: To demonstrate the capacity of the HP-41C to solve a non-linear volume estimation formula using 17 different sets of regression coefficients. The basic form of the equation is:

$$V = b_0 + b_1 D^{b_2} + b_3 D^{b_4} H^{b_5}$$

where, V = board foot volume, Int. ¼" rule
D = tree DBH
H = tree merchantable length
b_i = species unique coefficients developed by the U.S. Forest Survey (refer to U.S.F.S. Research Note NE-271, or parts F and G)

Provision in the program is made to obtain volume totals by species if desired and printed output is provided if the HP-82143A printer is attached.

A. Storage assignments

Register	Use
00	control number for recall of b coefficients by species (see part F for formula)
01-17	volume sum for species 1 through 17
18	overall volume total
21	D, tree diameter at breast height in inches
22	H, tree height in feet
23	V, individual tree volume
24	control number for species names used in labels B and a
25	control number to recall volume sums used in labels B and a
29	preservation of control number used in R_{00}
30	current species number code
31-47	b_0 for species 1 through 17
48-64	b_1 for species 1 through 17
65-81	b_2 for species 1 through 17

A. Storage assignments *(Continued)*

Register	Use
82-98	b_3 for species 1 through 17
99-115	b_4 for species 1 through 17
116-132	b_5 for species 1 through 17
141-157	6-letter species abbreviations
158	the word TOTAL
19,20,26-28, 133-140	not used

B. Labels

Global	Use
VOL17	program start; clears registers 1 through 18 and prompts for input of data cards containing the 102 coefficients (with flag 01 clear), then sets flag 01.
BAR	generates and prints table header and end in label B
CRAX	clears registers according to X (R_1 to R_{18}) in label VOL17
VOL	prompts for D and H and calculates individual tree volume

Local	
A	prints the contents of R_1 through R_{18}
a	displays and/or prints species name and volume for that species; species having zero volume are skipped
B	prints in tabular form the species name and volume; zero volume species are skipped
00	incrementing loop in label CRAX
01	used in VOL17 to prompt for and display species code and to construct the species control number
02	used in label a to store control numbers
03	incrementing, display, and printing loop in label a
04	used in label B to test for zero volume and skip spaces in printed table
10	used in label VOL to skip steps when printer is not attached
11	used in label B to skip printing when volume is zero

B. Labels *(Continued)*

Local	Use
12	same as 11 but in label a
13	provides the "NEED PRINTER" message when printer is not attached.

C. Flags

Type and Number	Use
User:	
01	to skip prompts for data cards and for read data statement in label VOL17
02	same as flag 01, but in label a
12	printer double-wide flag used in labels VOL17, B, and BAR
21	printer enable flag, used in label VOL to skip printer functions with no printer
29	digit grouping flag, used in label 01 to suppress decimal point
System:	
55	printer existence flag, used variously to skip printer functions or prompt for printer to be attached

D. Program procedure and example

I. In PRGM mode, key in the program "VOL17"; or in RUN mode, read the program (SIZE needed is 159) from magnetic cards.

II. In RUN mode

1. Be prepared to load the 102 regression coefficients into R_{31} through R_{132}, and the species names in R_{141} through R_{158} by one of the following:

   a. key the data in manually; making use of the program "LOAD" (No. 41U007), then set flags 01 and 02 before proceeding

   b. have the coefficients and species names recorded on magnetic cards and the card reader attached; to obtain the appropriate prompts, flags 01 and 02 should be clear initially.

2. Assuming the data are on cards and the card reader is attached, go to step 3. A four-tree example will be used assuming the coefficients and names as listed in part G.

D. Program procedure and example *(Continued)*

3. XEQ VOL17 (suggest assignment to X≷Y key), and follow the prompts; USER mode is assumed, with flags 01 and 02 clear.

Example

	Prompt	Input	Key	Output
a. Start	-	-	X≷Y	-
	LOAD COEFFS. FROM DATA CARDS			

b. (Insert the first data card and follow subsequent prompts; 7 tracks needed)

	Prompt	Input	Key	Output
c. Volume calculations	SPEC: 0 OR ?	4	R/S	-
	KEY D↑H, R/S	18↑16	R/S	VOLUME = 115.8
	SPEC: 4 OR ?	7	R/S	-
	KEY D↑H, R/S	20↑32	R/S	VOLUME = 276.9
	SPEC: 7 OR ?	-	R/S	-
	KEY D↑H, R/S	16↑24	R/S	VOLUME = 137.6
	SPEC: 7 OR ?	16	R/S	-
	KEY D↑H, R/S	24↑12	R/S	VOLUME = 188.1
	SPEC: 16 OR ?			

d. Volume summary:

	Prompt	Input	Key	Output
(1) with no printer, set flag 02	-	-	SF02	-
	-	-	a	HEMLOK = 115.8 BF SUMAPL = 414.5 BF HICKRY = 188.1 BF TOTAL = 718.5 BF

(If species names have been keyed manually, names will display, otherwise 0.0 will show for all species codes.)

	Prompt	Input	Key	Output
(2) with printer attached and flag 02 clear or set		-	A	the contents of R_1 through R_{18} will be printed without species names

with flag 02 clear:

Prompt	Input	Key	Output
-	-	a	-
LOAD NAMES FROM DATA CARDS			

(Insert the species name data card, and follow subsequent prompts; 2 tracks needed)

with flag 02 set: the "load prompt" will be skipped.

D. Program procedure and example *(Continued)*

Example

Prompt	Input	Key	Output
			display will be the same as in (1) above and results will also be printed
-	-	B	species totals will be printed in tabular form

e. For a new group of trees go to step 3a, flags 01 and 02 will be automatically set and the card-loading prompts will be skipped.

E. Program listing and printer output

```
01♦LBL "VOL17"
02 XEQ "CRAX"
03 31.132
04 FS? 01
05 GTO 01
06 "LOAD COEFFS."
07 AVIEW
08 PSE
09 "   FROM"
10 AVIEW
11 PSE
12 "DATA CARDS"
13 AVIEW
14 XEQ "RDTAX"
15 SF 01
16♦LBL 01
17 CF 29
18 FIX 0
19 RCL 30
20 "SPEC: "
21 ARCL X
22 "⊦ OR ?"
23 PROMPT
24 "SPECIES: "
25 ARCL X
26 SF 12
27 FS? 55
28 PRA
29 CF 12
30 STO 30
31 30
32 +
33 STO 00
34 RCL 30
35 115
36 +
37 1 E3
38 /
39 ST+ 00
40 .00017
41 ST+ 00
42 RCL 00
43 STO 29
44♦LBL "VOL"
45 FIX 1
46 RCL 29
47 STO 00
48 "KEY D↑H, R/S"
49 PROMPT
50 STO 22
51 X<>Y
52 STO 21
53 FC? 21
54 GTO 10
55 "D:"
56 XEQ "ACA"
57 XEQ "ACX"
58 "  H:"
59 XEQ "ACA"
60 RCL 22
61 XEQ "ACX"
62 ADV
63♦LBL 10
64 RCL IND 00
65 STO 23
66 ISG 00
67 RCL IND 00
68 ISG 00
69 RCL 21
70 RCL IND 00
71 Y↑X
72 *
73 ST+ 23
74 ISG 00
75 RCL IND 00
76 RCL 21
77 ISG 00
78 RCL IND 00
79 Y↑X
80 *
81 RCL 22
82 ISG 00
83 RCL IND 00
84 Y↑X
85 *
86 ST+ 23
87 RCL 30
88 RCL 23
89 ST+ IND Y
90 ST+ 18
91 "VOLUME= "
92 ARCL X
93 AVIEW
94 ADV
95 PSE
96 GTO 01
97♦LBL A
98 FIX 1
99 FC? 55
100 GTO 13
101 1.018
102 XEQ "PRREGX"
103 RTN
104♦LBL a
105 FIX 1
106 141.158
107 FS? 02
108 GTO 02
109 "LOAD NAMES"
110 AVIEW
111 PSE
112 "   FROM"
113 AVIEW
114 PSE
115 "DATA CARDS"
116 AVIEW
117 XEQ "RDTAX"
118 SF 02
119♦LBL 02
120 ADV
121 141.158
122 STO 24
123 1.018
124 STO 25
125♦LBL 03
126 RCL IND 25
127 X=0?
128 GTO 12
129 CLA
130 ARCL IND 24
131 "⊦= "
132 ARCL IND 25
133 "⊦ BF"
134 AVIEW
135 PSE
136♦LBL 12
137 ISG 25
138 X<> X
139 ISG 24
140 GTO 03
141 RTN
142♦LBL B
143 FIX 1
144 FC? 55
145 GTO 13
146 ADV
147 XEQ "BAR"
148 141.158
149 STO 24
150 1.018
```

E. Program listing and printer output *(Continued)*


```
151 STO 25
152 "SPECIES"
153 ACA
154 11
155 SKPCHR
156 "VOLUME"
157 ACA
158 ADV
159 SF 12
160 "------------"
161 PRA
162 CF 12
163♦LBL 04
164 RCL IND 25
165 X=0?
166 GTO 11
167 CLA
168 ARCL IND 24
169 ACA
170 RCL IND 25
171 LOG
172 INT
173 CHS
174 14
175 +
176 SKPCHR
177 RCL IND 25
178 ACX
179 ADV
180♦LBL 11
181 ISG 25
182 X<> X
183 ISG 24
184 GTO 04
185♦LBL "BAR"
186 SF 12
187 "************"
188 PRA
189 CF 12
190 RTN
191♦LBL 13
192 "NEED PRINTER"
193 PROMPT
194♦LBL "CRAX"
195 1.018
196 0
197 STO 30
198♦LBL 00
199 STO IND Y
200 ISG Y
201 GTO 00
202 END

VOL17:583 BYTES
```

Individual tree results

```
SPECIES: 4
      D: 18.0  H: 16.0
VOLUME= 115.8

SPECIES: 7
      D: 20.0  H: 32.0
VOLUME= 276.9

SPECIES: 7
      D: 16.0  H: 24.0
VOLUME= 137.6

SPECIES: 16
      D: 24.0  H: 12.0
VOLUME= 188.1
```

Label a output

```
HEMLOK= 115.8 BF
SUMAPL= 414.5 BF
HICKRY= 188.1 BF
TOTAL= 718.5 BF
```

Label A output

```
R01=  0.0
R02=  0.0
R03=  0.0
R04=  115.8
R05=  0.0
R06=  0.0
R07=  414.5
R08=  0.0
R09=  0.0
R10=  0.0
R11=  0.0
R12=  0.0
R13=  0.0
R14=  0.0
R15=  0.0
R16=  188.1
R17=  0.0
R18=  718.5
```

Label B output

```
************
SPECIES     VOLUME
------------
HEMLOK       115.8
SUMAPL       414.5
HICKRY       188.1
 TOTAL       718.5
************
```

F. Formulas used and background material

1. Formula

The individual tree volume formula used is that cited by Scott (1979) in U.S.F.S. Res. Note NE-271:

$$V = b_0 + b_1D^{b_2} + b_3D^{b_4}H^{b_5}$$

where,

V = Int. ¼-inch board-foot volume
D = tree diameter at breast height in inches
H = bole length in feet

2. The coefficients are stored in R_{31} to R_{132} as follows:

Species code

	1	2	. . .	17
b_0	R_{31}	R_{32}	. . .	R_{47}
b_1	R_{48}	R_{49}	. . .	R_{64}
b_2	R_{65}	R_{66}	. . .	R_{81}
b_3	R_{82}	R_{83}	. . .	R_{98}
b_4	R_{99}	R_{100}	. . .	R_{115}
b_5	R_{116}	R_{117}	. . .	R_{132}

For a given species code S, the appropriate set of coefficients is retrieved by indirect addressing, making use of a control number developed (in label 01) from the following algorithm:

$$\text{control number} = (S + 30) + \frac{S + 115}{1000} + .00017$$

The coefficients for species 3, for example, would be retrieved by the control number:

$$\text{C.N.} = 3 + 30 + \frac{3 + 115}{1000} + .00017$$

$$= 33.11817 \text{ (i.e., } R_{33}, R_{50}, \ldots, R_{118})$$

F. Formulas used and background material *(Continued)*

3. Species group names were abbreviated as follows:

Species code	abbreviation	species included
1	W,RPNE	white, red pine
2	R,W,BS	red, white, black spruce
3	BALFIR	balsam fir
4	HEMLOK	hemlock
5	HP,T,N	hard pines, tamarack, Norway spruce
6	CEDARS	cedar species
7	SUMAPL	sugar maple
8	SMP,YP	soft maple, yellow-poplar
9	ASHASP	ash species, aspen species
10	BLCHRY	black cherry
11	BIRCHS	birch species
12	ABEECH	american beech
13	BASSWD	basswood
14	RO,GUM	redoaks, sweetgum, blackgum
15	CHSOAK	chestnut oak
16	HICKRY	hickory
17	OTHERS	other hardwoods

G. Coefficients and species names

R31=	-12.25000
R32=	-13.03000
R33=	-12.29000
R34=	-8.36000
R35=	-6.78000
R36=	-8.89000
R37=	3.73000
R38=	2.84000
R39=	9.20000
R40=	1.58000
R41=	8.23000
R42=	-0.84000
R43=	2.66000
R44=	1.01000
R45=	4.46000
R46=	-1.24000
R47=	0.03000
R48=	-0.02418
R49=	-0.05197
R50=	-0.08212
R51=	-0.01433
R52=	-0.00841
R53=	-0.07324
R54=	-0.00182
R55=	-0.00557
R56=	0.00052
R57=	-0.00151
R58=	0.00039
R59=	-0.01207
R60=	-0.00313
R61=	-0.00192
R62=	-0.00061
R63=	-0.00385
R64=	-0.00196
R65=	2.68650
R66=	2.52480
R67=	2.56410
R68=	2.78780
R69=	2.70010
R70=	2.45560
R71=	3.37660
R72=	3.18080
R73=	3.00000
R74=	3.38780
R75=	3.00000
R76=	3.00430
R77=	3.27800
R78=	3.31880
R79=	3.59720
R80=	3.16480
R81=	3.32360

R82=	0.09610
R83=	0.12000
R84=	0.14160
R85=	0.07710
R86=	0.06450
R87=	0.12160
R88=	0.02620
R89=	0.02960
R90=	0.01930
R91=	0.02870
R92=	0.02060
R93=	0.04190
R94=	0.02820
R95=	0.02460
R96=	0.01820
R97=	0.03120
R98=	0.02630
R99=	2.22810
R100=	2.19990
R101=	2.26570
R102=	2.25930
R103=	2.19380
R104=	2.23820
R105=	2.42910
R106=	2.46060
R107=	2.21650
R108=	2.38750
R109=	2.21160
R110=	2.39510
R111=	2.44160
R112=	2.42680
R113=	2.48040
R114=	2.38880
R115=	2.41620
R116=	0.42220
R117=	0.42270
R118=	0.37440
R119=	0.42020
R120=	0.47130
R121=	0.32490
R122=	0.61390
R123=	0.57710
R124=	0.80430
R125=	0.63560
R126=	0.80190
R127=	0.59120
R128=	0.59400
R129=	0.60000
R130=	0.59220
R131=	0.60670
R132=	0.60120

R141=	"W,RPNE"
R142=	"R,W,BS"
R143=	"BALFIR"
R144=	"HEMLOK"
R145=	"HP,T,N"
R146=	"CEDARS"
R147=	"SUMAPL"
R148=	"SMP,YP"
R149=	"ASHASP"
R150=	"BLCHRY"
R151=	"BIRCHS"
R152=	"ABEECH"
R153=	"BASSWD"
R154=	"RO,GUM"
R155=	"CHSOAK"
R156=	"HICKRY"
R157=	"OTHERS"
R158=	"TOTAL"

Retrospective Comments Regarding VOL17

1. The program was written originally assuming a printer is attached; consequently, if no printer is used the write-up is a little rough in spots. Specifically, one should load the seventeen species names into registers 141 through 157 and "TOTAL" into register 158. If these names are on a card one can simply key 141.158 then XEQ RDTAX and follow the prompt. A logical time to do this is immediately after the coefficients have been loaded into memory.

2. The numeric coefficients and the species names are present on the KRON-1 tape under the name VLDATA, and can be downloaded to the calculator by the following (make sure SIZE $\geq$ 159):

   a. with VLDATA in alpha register, CLX (to put zero in the X register), then XEQ SEEKR

   b. key 31.158, then XEQ READRX

3. A slightly modified version of VOL17 (with SF 01 and SF 02 inserted as steps 02 and 03) and all the numeric coefficients and the species names, are grouped under the file name VL17FL and can be downloaded in one step from the KRON-1 tape by:

   a. VL17FL in the alpha register
   b. XEQ READA

Program No. 41F022

Calculator: HP-41CV or HP-41C with one single density module

Program Name: MG78 (for Mesavage & Girard form class 78)

Author: T.W. Beers

Date: May 1981

Purpose: To provide an alternative to the usual "volume table look up" procedure for individual tree volume determination in board feet by International ¼ inch, Scribner, and Doyle log rules. The tables obviated by this program are the Form Class 78 tables described by Mesavage & Girard (Tables for Estimating Board-Foot Volume of Timber, U.S.F.S.) using the estimation equations developed by Wiant and Castaneda (Resource Inventory Notes, BLM 4, March, 1977).

Wiant claims agreement with the tabular values within ± 3 percent, for 99, 94 and 97 percent of the cell values for the three log rules.

A. Storage assignments

Register	Use
01	D, tree diameter at breast height in inches
02	N, number of 16-foot logs
03	D^2
04	N^2
05	calculated volume
06	externally stored form class (needed when flag 04 is set)

B. Labels

Global

MG78	program start, initialization and reminders
ADJUST	to enable adjustment for form classes other than 78
INPUT	data prompting and storage for D, D^2, N, and N^2

B. Labels *(Continued)*

Global	Use
TONEB	tone routine for Int. ¼ log rule
TONED	tone routine for Doyle log rule
TONEDD	tone routine for Doyle log rule
TONES	tone routine for Scribner log rule
Local	
A	calculates and displays tree volume for all three log rules
B	calculates and displays volume by Int. ¼ rule
C	calculates and displays volume by Scribner rule
D	calculates and displays volume by Doyle rule
01	used to skip identifier and flag clear in label B
02	used to skip identifier and flag clear in label C
03	used to skip identifier and flag clear in label D
04	used to skip the prompt in INPUT when flag 22 is set

C. Flags

User	Use
00	set automatically in label A to signify the calculation of volume by all three log rules
04	set externally to display the form class and calculate the volume according to the form class stored in register 06
22	numeric data input flag, when set automatically by numeric data entry, skips the prompt in INPUT
29	digit grouping flag, cleared in MG78 to eliminate decimal point (in FIX 0)

D. Program procedure and example

I. In PRGM mode, load the program "MG78"
(SIZE needed is 007)

II. In RUN mode:

1. XEQ MG78 (suggest assignment to TAN key) and follow the prompts.

Example

	Prompt	Input	Key	Output
a. start	KEY A, B, C OR D A FOR ALL 3 B FOR INT. ¼ C FOR SCRIB. D FOR DOYLE			
b. all 3		-	A	
	DBH ↑ LOGS, R/S	16↑2	R/S	*INT. ¼* then VOL. = 177 B.FT. *SCRIBNER* then VOL. = 156 B.FT. *DOYLE* then VOL. = 114 B.FT.
	DBH ↑ LOGS, R/S	18↑4	R/S	*INT. ¼* then VOL. = 376 B.FT. etc. 330 for Scribner 248 for Doyle
		etc.		

(1) Note that a distinctive 4-note tone is heard just before each answer (Int. ¼ = 9,7,9,7; Scribner = standard "beep"; Doyle = 7,9,7,9).

c. one table only; for example, Doyle:

	Prompt	Input	Key	Output
		-	D	DOYLE
	DBH ↑ LOGS, R/S	18↑4	R/S	*DOYLE* then VOL. = 248 B.FT.
	DBH ↑ LOGS, R/S	16↑2	R/S	*DOYLE* then VOL. = 114 B.FT.
		etc.		

(1) Note, in this "single action" mode, the audible signal is <u>one</u> tone 9 for Int. ¼, <u>two</u> tone 9's for Scribner and <u>three</u> tone 9's for Doyle.

(2) To change to another log rule (i.e., table) or to the automatic mode, one can either (a) press B,C,D, or A, note the identifier and key in data at the DBH ↑ LOGS, R/S prompt or (b) key in the DBH, ↑, No. of logs, <u>then</u> press B,C,D, or A.

D. Program procedure and example *(Continued)*

III. Notes and cautions

1. The audio part of the display can be disabled by clearing flag 26 manually (shift, CF 26); but note that turning the calculator off, then on, resets this flag. "Permanent" disabling of the audio can be done by inserting a CF 26 statement in the initialization part of the program.

2. A two-second pause is programmed after any answer, before the next prompt; if the answer is missed, it can be retrieved by a single depression of the correction key (←), or by VIEW 05 or RCL 05.

3. The program can be used to calculate volumes for form classes other than 78, by

   a. store the desired form class in register 06
   b. set flag 04
   c. proceed as in part II

4. No error check is made on the reasonableness of the diameter and number of logs-- let the user beware!

E. Program Listing

```
01♦LBL "MG78"
02 FIX 0
03 CF 29
04 CF 22
05 CF 00
06 "KEY A,B,C OR D"
07 AVIEW
08 SIN
09 SIN
10 "A FOR ALL 3"
11 AVIEW
12 PSE
13 "B FOR INT.1/4"
14 AVIEW
15 PSE
16 "C FOR SCRIB."
17 AVIEW
18 PSE
19 "D FOR DOYLE"
20 PROMPT
21♦LBL A
22 "AUTO.--ALL 3"
23 AVIEW
24 SF 00
25 GTO "INPUT"
26♦LBL B
27 "INT.1/4 INCH"
28 AVIEW
29 CF 00
30♦LBL 01
31 FC? 00
32 XEQ "INPUT"
33 "  *INT. 1/4*"
34 AVIEW
35 1.52968
36 RCL 04
37 *
38 9.58615
39 RCL 02
40 *
41 +
42 13.35212
43 -
44 1.7962
45 ENTER↑
46 .27465
47 RCL 04
48 *
49 -
50 2.59995
51 RCL 02
52 *
53 -
54 RCL 01
55 *
56 +
57 .04482
58 ENTER↑
59 .00961
60 RCL 04
61 *
62 -
63 .45997
64 RCL 02
65 *
66 +
67 RCL 03
68 *
69 +
70 STO 05
71 FS? 04
72 XEQ "ADJUST"
73 FC? 00
74 TONE 9
75 FS? 00
76 XEQ "TONEB"
77 "VOL.= "
78 ARCL X
79 "⊦ B.FT."
80 AVIEW
81 PSE
82 PSE
83 FC? 00
84 GTO 01
85 FS? 00
86 GTO 02
87♦LBL C
88 "  SCRIBNER"
89 AVIEW
90 CF 00
91♦LBL 02
92 FC? 00
93 XEQ "INPUT"
94 " *SCRIBNER*"
95 AVIEW
96 17.53508
97 RCL 02
98 *
99 .59242
100 RCL 04
101 *
102 -
103 22.50365
104 -
105 3.02988
106 ENTER↑
107 .02302
108 RCL 04
109 *
110 -
111 4.34381
112 RCL 02
113 *
114 -
115 RCL 01
116 *
117 +
118 .51593
119 RCL 02
120 *
121 .02035
122 RCL 04
123 *
124 -
125 .01969
126 -
127 RCL 03
128 *
129 +
130 STO 05
131 FS? 04
132 XEQ "ADJUST"
133 FS? 00
134 BEEP
135 FC? 00
136 XEQ "TONES"
137 "VOL.= "
138 ARCL X
139 "⊦ B.FT."
140 AVIEW
141 PSE
142 PSE
143 FC? 00
144 GTO 02
145 FS? 00
146 GTO 03
147♦LBL D
148 "   DOYLE"
149 AVIEW
150 CF 00
```

E. Program Listing *(Continued)*

```
151♦LBL 03
152 FC? 00
153 XEQ "INPUT"
154 "  *DOYLE*"
155 AVIEW
156 .55743
157 RCL 04
158 *
159 41.51275
160 RCL 02
161 *
162 +
163 29.37337
164 -
165 2.78043
166 ENTER↑
167 .04516
168 RCL 04
169 *
170 -
171 8.77272
172 RCL 02
173 *
174 -
175 RCL 01
176 *
177 +
178 .04177
179 ENTER↑
180 .01578
181 RCL 04
182 *
183 -
184 .59042
185 RCL 02
186 *
187 +
188 RCL 03
189 *
190 +
191 STO 05
192 FS? 04
193 XEQ "ADJUST"
194 FC? 00
195 XEQ "TONEDD"
196 FS? 00
197 XEQ "TONED"
198 "VOL.= "
199 ARCL X
200 "⊢ B.FT."
201 AVIEW
202 PSE
203 PSE
204 FC? 00
205 GTO 03
206♦LBL "INPUT"
207 FS? 22
208 GTO 04
209 "DBH↑LOGS, R/S"
210 PROMPT
211♦LBL 04
212 STO 02
213 X↑2
214 STO 04
215 X<>Y
216 STO 01
217 X↑2
218 STO 03
219 CF 22
220 FC? 00
221 RTN
222 GTO 01
223♦LBL "ADJUST"
224 RCL 06
225 "F. CLASS: "
226 ARCL X
227 AVIEW
228 78
229 -
230 .03
231 *
232 1
233 +
234 RCL 05
235 *
236 RTN
237♦LBL "TONEB"
238 TONE 9
239 TONE 7
240 TONE 9
241 TONE 7
242 RTN
243♦LBL "TONES"
244 TONE 9
245 TONE 9
246 RTN
247♦LBL "TONED"
248 TONE 7
249 TONE 9
250 TONE 7
251 TONE 9
252 RTN
253♦LBL "TONEDD"
254 TONE 9
255 TONE 9
256 TONE 9
257 END

MG78:795 BYTES
```

F. Formulas used.

1. Assuming Girard Form Class 78, the formulas for the three volume tables are those published by Wiant and Castaneda:

Log rule	Equation
Int. ¼"	Vol. = $(1.52968N^2 + 9.58615N - 13.35212) +$
	$(1.79620 - 0.27465N^2 - 2.59995N)\ D +$
	$(0.04482 - 0.00961N^2 + 0.45997N)\ D^2$
Scribner	Vol. = $(17.53508N - 0.59242N^2 - 22.50365) +$
	$(3.02988 - 0.02302N^2 - 4.34381N)\ D +$
	$(0.51593N - 0.02035N^2 - 0.01969)\ D^2$
Doyle	Vol. = $(0.55743N^2 + 41.51275N - 29.37337) +$
	$(2.78043 - 0.04516N^2 - 8.77272N)\ D +$
	$(0.04177 - 0.01578N^2 + 0.59042N)\ D^2$

Where: N = number of 16-foot logs

D = dbh

2. In this program if form class is not 78, the often cited rule: "volume changes by 3 percent for each class change from 78" is applied to unrounded class 78 volumes calculated using the above formulas. The extent of the error involved with this adjustment is undetermined.

FORM CLASS 78

TABLE 21.—*Gross volume of tree, International ¼-inch rule*

Tree diameter (inches)	VOLUME (board feet) BY NUMBER OF USABLE 16-FOOT LOGS										
	1	1½	2	2½	3	3½	4	4½	5	5½	6
10	36	48	59	66	73						
11	46	61	76	86	96						
12	56	74	92	106	120	128	137				
13	67	90	112	130	147	158	168				
14	78	105	132	153	174	187	200				
15	92	124	156	182	208	225	242				
16	106	143	180	210	241	263	285				
17	121	164	206	242	278	304	330				
18	136	184	233	274	314	344	374				
19	154	209	264	311	358	392	427				
20	171	234	296	348	401	440	480	511	542		
21	191	262	332	391	450	496	542	579	616		
22	211	290	368	434	500	552	603	647	691		
23	231	318	404	478	552	608	663	714	766		
24	251	346	441	523	605	664	723	782	840		
25	275	380	484	574	665	732	800	865	930		
26	299	414	528	626	725	801	877	949	1,021		
27	323	448	572	680	788	870	952	1,032	1,111		
28	347	482	616	733	850	938	1,027	1,114	1,201	1,280	1,358
29	375	521	667	794	920	1,016	1,112	1,210	1,308	1,398	1,488
30	403	560	718	854	991	1,094	1,198	1,306	1,415	1,517	1,619
31	432	602	772	921	1,070	1,184	1,299	1,412	1,526	1,640	1,754
32	462	644	826	988	1,149	1,274	1,400	1,518	1,637	1,762	1,888
33	492	686	880	1,053	1,226	1,360	1,485	1,622	1,750	1,888	2,026
34	521	728	934	1,119	1,304	1,447	1,590	1,727	1,864	2,014	2,165
35	555	776	998	1,196	1,394	1,548	1,702	1,851	2,000	2,156	2,312
36	589	836	1,063	1,274	1,485	1,650	1,814	1,974	2,135	2,298	2,461
37	622	873	1,124	1,351	1,578	1,752	1,926	2,099	2,272	2,444	2,616
38	656	921	1,186	1,428	1,670	1,854	2,038	2,224	2,410	2,590	2,771
39	694	976	1,258	1,514	1,769	1,968	2,166	2,359	2,552	2,741	2,937
40	731	1,029	1,326	1,598	1,868	2,081	2,294	2,494	2,693	2,898	3,103

FORM CLASS 78

TABLE 73.—*Gross volume of tree, Doyle log rule*

Tree diameter (inches)	VOLUME (board feet) BY NUMBER OF USABLE 16-FOOT LOGS										
	1	1½	2	2½	3	3½	4	4½	5	5½	6
10	14	17	20	21	22						
11	22	27	32	35	38						
12	29	36	43	48	53	54	56				
13	38	48	59	66	73	76	80				
14	48	62	75	84	93	98	103				
15	60	78	96	108	121	128	136				
16	72	94	116	132	149	160	170				
17	86	113	140	161	182	196	209				
18	100	132	164	190	215	232	248				
19	118	156	194	225	256	276	297				
20	135	180	225	261	297	322	346	364	383		
21	154	207	260	302	344	374	404	428	452		
22	174	234	295	344	392	427	462	492	521		
23	195	264	332	388	444	483	522	558	594		
24	216	293	370	433	496	539	582	625	668		
25	241	328	414	486	558	609	660	709	758		
26	266	362	459	539	619	678	737	793	849		
27	292	398	505	594	684	749	814	877	940		
28	317	434	551	650	750	820	890	961	1,032	1,096	1,169
29	346	475	604	714	824	902	980	1,061	1,142	1,218	1,294
30	376	517	658	778	898	984	1,069	1,160	1,251	1,339	1,427
31	408	562	717	850	983	1,080	1,176	1,273	1,370	1,470	1,570
32	441	608	776	922	1,068	1,176	1,283	1,386	1,488	1,600	1,712
33	474	654	835	994	1,152	1,268	1,385	1,497	1,609	1,734	1,858
34	508	700	894	1,064	1,235	1,361	1,487	1,608	1,730	1,866	2,003
35	544	754	964	1,149	1,334	1,472	1,610	1,743	1,876	2,020	2,163
36	581	308	1,035	1,234	1,434	1,583	1,732	1,878	2,023	2,173	2,323
37	618	860	1,102	1,318	1,534	1,694	1,854	2,013	2,172	2,332	2,492
38	655	912	1,170	1,402	1,635	1,805	1,973	2,148	2,322	2,491	2,660
39	698	974	1,250	1,498	1,746	1,932	2,118	2,298	2,479	2,662	2,844
40	740	1,035	1,330	1,594	1,858	2,059	2,260	2,448	2,636	2,832	3,027

FORM CLASS 78

TABLE 47.—*Gross volume of tree, Scribner log rule*

Tree diameter (inches)	VOLUME (board feet) BY NUMBER OF USABLE 16-FOOT LOGS										
	1	1½	2	2½	3	3½	4	4½	5	5½	6
10	28	36	41	48	52						
11	38	49	60	67	74						
12	47	61	75	85	95	100	106				
13	58	76	94	107	120	128	136				
14	69	92	114	130	146	156	166				
15	82	109	136	157	178	192	206				
16	95	127	159	185	211	229	247				
17	109	146	184	215	246	268	289				
18	123	166	209	244	280	306	331				
19	140	190	240	281	322	352	382				
20	157	214	270	317	364	398	432	459	486		
21	176	240	304	358	411	450	490	523	556		
22	194	266	338	398	458	504	549	588	626		
23	214	294	374	441	508	558	607	652	698		
24	234	322	409	484	558	611	665	718	770		
25	258	355	452	534	617	678	740	799	858		
26	281	388	494	585	676	745	814	880	945		
27	304	420	536	636	736	811	886	959	1,032		
28	327	452	578	686	795	877	959	1,040	1,120	1,196	1,261
29	354	491	628	746	864	953	1,042	1,132	1,222	1,306	1,389
30	382	530	678	806	933	1,028	1,124	1,224	1,325	1,421	1,517
31	411	571	731	871	1,011	1,117	1,223	1,328	1,434	1,541	1,648
32	440	612	784	936	1,089	1,206	1,322	1,432	1,543	1,661	1,779
33	469	654	838	1,001	1,164	1,289	1,414	1,534	1,654	1,783	1,912
34	498	695	892	1,066	1,239	1,375	1,507	1,636	1,766	1,906	2,046
35	530	742	954	1,141	1,328	1,473	1,618	1,757	1,896	2,044	2,192
36	563	789	1,015	1,216	1,416	1,572	1,728	1,877	2,026	2,182	2,338
37	596	836	1,075	1,290	1,506	1,670	1,835	1,998	2,160	2,324	2,488
38	629	882	1,135	1,366	1,596	1,769	1,942	2,118	2,295	2,466	2,637
39	666	935	1,204	1,449	1,694	1,881	2,068	2,251	2,434	2,616	2,799
40	703	988	1,274	1,532	1,791	1,993	2,195	2,384	2,574	2,768	2,961

Retrospective Comments Regarding MG78

1. Skipping the initial reminders could be accomplished by using flag 03 and inserting the following:

   a. FS? 03
      GTO 05 after step 07 (AVIEW)

   b. LBL 05 after step 19 ("D FOR DOYLE")

2. Reasonable printer output is obtained only with printer in NORM mode.

Program No. 41F023

Calculator: HP-41C/CV

Program Name: MYERS

Author: T.W. Beers

Date: June, 1981

Purpose: To provide a convenient calculation procedure to be used instead of the set of 14 tables of upland hardwood weights and volumes summarized by Myers, et. al. (U.S.F.S. Gen. Tech. Rep. NC-60, 1980). Provision is made for all six species cited: black, red, and white oak, hickory, white ash, and yellow poplar. In addition to the obvious advantage of the calculator over the table look-up procedure, "interpolation" between classes given in the tables is routinely handled by the programmed procedure.

A. Storage assignments

Register	Use
00	control number for the recall of the appropriate set of coefficients (a, b, and c)
01	table number (as specified in NC-60, cited above)
02	species number code: 1 = black oak (key A) 2 = red oak (key B) 3 = white oak (key C) 4 = hickory (key c, i.e., shift C) 5 = white ash (key D) 6 = yellow poplar (key E)
03	a copy of the control number stored in R_{00}
04	tree DBH, in inches or centimeters
05	tree height, in feet or meters
06	calculated tree weight or volume in pounds (LBS.), kilograms (KG.), cubic feet (CU. FT.), or cubic meters (M↑3)
07-90	the 84 _a_ coefficients listed in parts E and F
91-174	the 84 _b_ coefficients listed in parts E and F
175-258	the 84 _c_ coefficients listed in parts E and F

B. Labels

Global	Use
MYERS	program start and initialization
Local	
A	black oak control routine
B	red oak control routine
C	white oak control routine
c	hickory control routine
D	white ash control routine
E	yellow poplar control routine
F	prompts for the desired table, then tests for and displays the nature of the table; volume, green weight, or oven dry weight
00	constructs the control number from the keyed-in table number and generated species code number
01	prompts for DBH and height and calculates the desired answer
02	a guiding routine for label 00
03	to affix "LBS." to answer
04	to affix "CU.FT." to answer
05	to affix "KG." to answer
06	displays answer and provides 2-second pause
07	provides the prefix "GREEN WT."
08	provides the prefix "OVEN DRY WT."
09	provides the prefix "VOLUME"
10	displays prefix for answer and prompts for keys A through E or c (i.e., for desired species)
11	tests for the appropriate units suffix for the answer; defaults to "M↑3" if labels 03, 04, or 05 are not used

C. Flags

Type and Number	Use
User:	
12	printer double-wide flag, used in label F
21	printer enable flag, used in label 01 to skip printed documentation of DBH and height if printer is not attached

C. Flags *(Continued)*

Type and Number	Use
User:	
22	numeric data input flag, used in label F to detect a newly input table number
29	digit grouping flag, used in the initialization to suppress the decimal point (in FIX 0)
System:	
55	printer existence flag, used in label F to suppress print statement

D. Program procedure and example

I. In PRGM mode, load the program "MYERS"
(SIZE needed is 259)

II. In RUN mode

1. XEQ MYERS (suggest assignment to shift X≷Y key), and follow the prompts; USER mode is assumed in the following description, and the printer is not attached.

	Prompt	*Example* Input	Key	Output
a. start	-	-	shift X≷Y	-
	TABLE: Ø OR ?	3	R/S	GREEN WT. (pause)
	KEY: A-E OR c	-	A	*BLACK OAK* (pause)
	DBH↑HT, R/S	6↑40	R/S	273 LBS. (pause)
	DBH↑HT, R/S	20↑64	R/S	5678 LBS. (pause)
	etc., for black oak in Table 3			
b. another species in same table	-	-	B	*RED OAK* (pause)
	DBH↑HT, R/S	22↑56	R/S	6052 LBS.
	etc., for red oak in Table 3			

D. Program procedure and example *(Continued)*

	Prompt	Input	Key	Output
		Example		
c. different table	-	-	F	-
	TABLE 3 OR ?	9*	R/S	VOLUME (pause)
	KEY: A-E OR c	-	shift C	*HICKORY* (pause)
	DBH↑HT, R/S	20↑80	R/S	67.3 CU.FT. (pause)
	DBH↑HT, R/S	etc.		

*If the same table is desired (here, Table 3) R/S is pressed instead of another number

III. Notes and Cautions

1. The user is cautioned that in certain instances the calculator answer will not agree with the answer printed in the NC-60; it is presumed that this is occasioned by the fact that the coefficients presented in Table 17 of the cited publication were rounded-off to facilitate printing. The printed ones were employed in the program described here in.

2. If the printer is attached and the program is executed, paper documentation of the input and output is provided; however if the calculator display only is to be monitored, the printer should be turned off <u>and</u> unplugged to avoid program interruptions.

E 1. Program Listing and Example

```
01♦LBL "MYERS"
02 0
03 STO 01
04 CF 29
05♦LBL F
06 FIX 0
07 CF 22
08 "TABLE: "
09 ARCL 01
10 "⊢ OR ?"
11 PROMPT
12 FS?C 22
13 STO 01
14 RCL 01
15 SF 12
16 "TABLE "
17 ARCL X
18 FS? 55
19 PRA
20 CF 12
21 6
22 X>Y?
23 GTO 07
24 RCL 01
25 8
26 X>Y?
27 GTO 08
28 RCL 01
29 10
30 X>Y?
31 GTO 09
32 RCL 01
33 13
34 X>Y?
35 GTO 07
36 RCL 01
37 15
38 X>Y?
39 GTO 08
40♦LBL 09
41 "VOLUME"
42 GTO 10
43♦LBL 08
44 "OVEN DRY WT."
45 GTO 10
46♦LBL 07
47 "GREEN WT."
48♦LBL 10
49 ADV
50 AVIEW
51 PSE
52 "KEY: A-E OR c"
53 PROMPT
54♦LBL A
55 "*BLACK OAK*"
56 AVIEW
57 1
58 GTO 02
59♦LBL B
60 "*RED OAK*"
61 AVIEW
62 2
63 GTO 02
64♦LBL C
65 "*WHITE OAK*"
66 AVIEW
67 3
68 GTO 02
69♦LBL c
70 "*HICKORY*"
71 AVIEW
72 4
73 GTO 02
74♦LBL D
75 "*WHITE ASH*"
76 AVIEW
77 5
78 GTO 02
79♦LBL E
80 "*Y. POPLAR*"
81 AVIEW
82 6
83♦LBL 02
84 STO 02
85 XEQ 00
86♦LBL 01
87 RCL 03
88 STO 00
89 "DBH↑HT, R/S"
90 PROMPT
91 CLA
92 STO 05
93 X<>Y
94 STO 04
95 RCL IND 00
96 X<>Y
97 ISG 00
98 RCL IND 00
99 Y↑X
100 *
101 ISG 00
102 RCL 05
103 RCL IND 00
104 Y↑X
105 *
106 STO 06
107 FIX 0
108 FC? 21
109 GTO 11
110 "D:"
111 ACA
112 RCL 04
113 ACX
114 "  H:"
115 ACA
116 RCL 05
117 ACX
118 ADV
119♦LBL 11
120 CLA
121 7
122 RCL 01
123 X<=Y?
124 GTO 03
125 10
126 X>Y?
127 GTO 04
128 14
129 RCL 01
130 X<=Y?
131 GTO 05
132 FIX 2
133 ARCL 06
134 "⊢ M↑3"
135 GTO 06
136♦LBL 03
137 FIX 0
138 ARCL 06
139 "⊢ LBS."
140 GTO 06
141♦LBL 04
142 FIX 1
143 ARCL 06
144 "⊢ CU. FT."
145 GTO 06
146♦LBL 05
147 FIX 0
148 ARCL 06
149 "⊢ KG."
150♦LBL 06
```

E 1. *(Continued)*

```
151 AVIEW
152 PSE
153 PSE
154 GTO 01
155♦LBL 00
156 RCL 01
157 3
158 -
159 6
160 *
161 RCL 02
162 6
163 +
164 +
165 STO 00
166 168
167 +
168 1 E3
169 /
170 ST+ 00
171 .00084
172 ST+ 00
173 RCL 00
174 STO 03
175 END

MYERS:422 BYTES
```

```
TABLE 3

GREEN WT.
*BLACK OAK*
                D: 6  H: 40
273 LBS.
               D: 26  H: 72
11152 LBS.

TABLE 6

OVEN DRY WT.
*RED OAK*
                D: 6  H: 40
103 LBS.
*HICKORY*
                D: 6  H: 56
221 LBS.

TABLE 8

VOLUME
*WHITE OAK*
                D: 6  H: 40
4.0 CU. FT.
*Y. POPLAR*
                D: 6  H: 40
2.1 CU. FT.
               D: 26  H: 96
140.8 CU. FT.

TABLE 10

GREEN WT.
*WHITE ASH*
               D: 40  H: 15
1220 KG.

TABLE 16

VOLUME
*BLACK OAK*
               D: 60  H: 30
3.77 M↑3
```

E 2. Printer Output of Regression Coefficients Stored in Registers 07 through 258

R07= 0.201100
R08= 0.057000
R09= 1.098900
R10= 0.816300
R11= 0.775600
R12= 0.114150
R13= 0.095300
R14= 0.112100
R15= 3.252120
R16= 0.559400
R17= 0.022770
R18= 0.007380
R19= 0.090500
R20= 0.111000
R21= 1.727200
R22= 0.417100
R23= 0.081930
R24= 0.004080
R25= 0.047820
R26= 0.056290
R27= 2.006780
R28= 0.297170
R29= 0.104910
R30= 0.003920
R31= 0.045530
R32= 0.055840
R33= 0.840760
R34= 0.224620
R35= 0.045270
R36= 0.001960
R37= 0.001260
R38= 0.001530
R39= 0.040800
R40= 0.006530
R41= 0.002500
R42= 0.000110
R43= 0.001190
R44= 0.001520
R45= 0.020360
R46= 0.005020
R47= 0.001150
R48= 0.000060
R49= 0.035330
R50= 0.011130

R51= 0.111370
R52= 0.170260
R53= 0.079360
R54= 0.039230
R55= 0.017610
R56= 0.017870
R57= 0.149020
R58= 0.092870
R59= 0.009750
R60= 0.004680
R61= 0.014660
R62= 0.015470
R63= 0.106340
R64= 0.063780
R65= 0.015090
R66= 0.003690
R67= 0.008880
R68= 0.008980
R69= 0.091020
R70= 0.049740
R71= 0.015280
R72= 0.002460
R73= 0.007430
R74= 0.007800
R75= 0.057330
R76= 0.034830
R77= 0.008680
R78= 0.001800
R79= 0.000015
R80= 0.000015
R81= 0.000137
R82= 0.000068
R83= 0.000024
R84= 0.000005
R85= 0.000013
R86= 0.000013
R87= 0.000087
R88= 0.000049
R89= 0.000015
R90= 0.000004
R91= 2.167900
R92= 2.308900
R93= 2.221800
R94= 1.853200
R95= 2.242020
R96= 1.809920
R97= 2.237600
R98= 2.299000
R99= 2.481480
R100= 2.018400

R101= 1.920410
R102= 1.848400
R103= 2.309400
R104= 2.365600
R105= 2.398100
R106= 2.097400
R107= 2.214080
R108= 1.722330
R109= 2.233690
R110= 2.296010
R111= 2.471640
R112= 2.018330
R113= 2.239780
R114= 1.847630
R115= 2.304000
R116= 2.361400
R117= 2.358190
R118= 2.092700
R119= 2.197630
R120= 1.718520
R121= 2.214210
R122= 2.285730
R123= 2.392330
R124= 2.019780
R125= 2.216510
R126= 1.817990
R127= 2.278140
R128= 2.348390
R129= 2.357390
R130= 2.091340
R131= 2.176900
R132= 1.704520
R133= 2.167900
R134= 2.308900
R135= 2.221800
R136= 1.853200
R137= 2.242020
R138= 1.809920
R139= 2.237600
R140= 2.299000
R141= 2.481480
R142= 2.018400
R143= 1.920410
R144= 1.848400
R145= 2.309400
R146= 2.365600
R147= 2.398100
R148= 2.097400
R149= 2.214080
R150= 1.722330

E 2. *(Continued)*

R151= 2.233690
R152= 2.296010
R153= 2.471640
R154= 2.018330
R155= 2.239780
R156= 1.847630
R157= 2.304000
R158= 2.361400
R159= 2.358190
R160= 2.092700
R161= 2.197630
R162= 1.718520
R163= 2.214210
R164= 2.285730
R165= 2.392330
R166= 2.019780
R167= 2.216510
R168= 1.817990
R169= 2.278140
R170= 2.348390
R171= 2.357390
R172= 2.091340
R173= 2.176900
R174= 1.704520
R175= 0.902600
R176= 1.102000
R177= 0.481800
R178= 0.800100
R179= 0.505690
R180= 1.186350
R181= 1.999500
R182= 0.923500
R183= 0.017440
R184= 0.737600
R185= 1.458060
R186= 1.732280
R187= 0.945500
R188= 0.862500
R189= 0.200600
R190= 0.730400
R191= 0.978720
R192= 1.932250
R193= 1.000400
R194= 0.921430
R195= 0.001130
R196= 0.744440
R197= 0.800920
R198= 1.723500
R199= 0.947000
R200= 0.860980

R201= 0.255320
R202= 0.738450
R203= 0.999260
R204= 1.941680
R205= 1.003570
R206= 0.914640
R207= 0.082890
R208= 0.746400
R209= 0.841790
R210= 1.783230
R211= 0.955630
R212= 0.855270
R213= 0.256710
R214= 0.739070
R215= 1.024880
R216= 1.976320
R217= 0.902600
R218= 1.102000
R219= 0.481800
R220= 0.800100
R221= 0.505690
R222= 1.186350
R223= 0.999500
R224= 0.923500
R225= 0.017440
R226= 0.737600
R227= 1.458060
R228= 1.732280
R229= 0.945500
R230= 0.862500
R231= 0.200600
R232= 0.730400
R233= 0.978720
R234= 1.932250
R235= 1.000400
R236= 0.921430
R237= 0.001130
R238= 0.744440
R239= 0.800920
R240= 1.723500
R241= 0.947000
R242= 0.860980
R243= 0.255320
R244= 0.738450
R245= 0.999260
R246= 1.941680
R247= 1.003570
R248= 0.914640
R249= 0.082890
R250= 0.746400
R251= 0.841790
R252= 1.783230
R253= 0.955630
R254= 0.855270
R255= 0.256710
R256= 0.739070
R257= 1.024880
R258= 1.976320

F. Formulas used and background material

1. Basic computation formula

As described in the NC-60 publication, the tree volume or weight (Y) was predicted using the non-linear regression model

$$Y = aD^bH^c$$

where D is tree diameter breast high
H is tree height
and a,b,c, are the fitted coefficients

2. The 252 coefficients are stored in R_{07} to R_{258} as follows:

Table	Species	a	b	c
3	1	R_{07}	R_{91}	R_{175}
	2	R_{08}	R_{92}	R_{176}
	3	R_{09}	R_{93}	R_{177}
	4	R_{10}	R_{94}	R_{178}
	5	R_{11}	R_{95}	R_{179}
	6	R_{12}	R_{96}	R_{180}
4	1	R_{13}	R_{97}	R_{181}
.	.	.	.	.
.	.	.	.	.
.	.	.	.	.
16	6	R_{90}	R_{174}	R_{258}

3. Retrieval of the proper set of coefficients for table T and species code S is achieved by indirect addressing, making use of a control number developed (in label 00) from the following algorithm:

$$\text{control number} = X + \frac{X + 168}{1000} + .00084$$

where $X = (T - 3)\,6 + (S + 6)$

note: (1) the "3" is needed because table numbers start with 3, not 1

(2) the first "6" is needed because there are 6 species, the second "6" because register 7 is the start of the stored coefficients

The coefficients for species 3 (white oak) in Table 4 would be retrieved by:

$$X = (4 - 3)\,6 + (3 + 6) = 15$$

$$C.N. = 15 + \frac{15 + 168}{1000} + .00084$$

$$= 15.18384 \text{ (i.e., } R_{15}, R_{99}, \text{ and } R_{183})$$

F. Formulas used and background material *(Continued)*

4. Copy of Table 17 from NC-60

Table 17. *Equation coefficients*[1]—$W = aD^bH^c$

Table reference	Species	a	b	c
3	Black oak	0.20110	2.16790	0.90260
	Red oak	0.05700	2.30890	1.10200
	White oak	1.09890	2.22180	0.48180
	Hickory	0.81630	1.85320	0.80010
	White ash	0.77560	2.24202	0.50569
	Yellow poplar	0.11415	1.80992	1.18635
4	Black oak	0.09530	2.23760	1.99950
	Red oak	0.11210	2.29900	0.92350
	White oak	3.25212	2.48148	0.01744
	Hickory	0.55940	2.01840	0.73760
	White ash	0.02277	1.92041	1.45806
	Yellow poplar	0.00738	1.84840	1.73228
5	Black oak	0.09050	2.30940	0.94550
	Red oak	0.11100	2.36560	0.86250
	White oak	1.72720	2.39810	0.20060
	Hickory	0.41710	2.09740	0.73040
	White ash	0.08193	2.21408	0.97872
	Yellow poplar	0.00408	1.72233	1.93225
6	Black oak	0.04782	2.23369	1.00040
	Red oak	0.05629	2.29601	0.92143
	White oak	2.00678	2.47164	0.00113
	Hickory	0.29717	2.01833	0.74444
	White ash	0.10491	2.23978	0.80092
	Yellow poplar	0.00392	1.84763	1.72350
7	Black oak	0.04553	2.30400	0.94700
	Red oak	0.05584	2.36140	0.86098
	White oak	0.84076	2.35819	0.25532
	Hickory	0.22462	2.09270	0.73845
	White ash	0.04527	2.19763	0.99926
	Yellow poplar	0.00196	1.71852	1.94168
8	Black oak	0.00126	2.21421	1.00357
	Red oak	0.00153	2.28573	0.91464
	White oak	0.04080	2.39233	0.08289
	Hickory	0.00653	2.01978	0.74640
	White ash	0.00250	2.21651	0.84179
	Yellow poplar	0.00011	1.81799	1.78323
9	Black oak	0.00119	2.27814	0.95563
	Red oak	0.00152	2.34839	0.85527
	White oak	0.02036	2.35739	0.25671
	Hickory	0.00502	2.09134	0.73907
	White ash	0.00115	2.17690	1.02488
	Yellow poplar	0.00006	1.70452	1.97632
10	Black oak	0.03533	2.16790	0.90260
	Red oak	0.01113	2.30890	1.10200
	White oak	0.11137	2.22180	0.48180
	Hickory	0.17026	1.85320	0.80010
	White ash	0.07936	2.24202	0.50569
	Yellow poplar	0.03923	1.80992	1.18635
11	Black oak	0.01761	2.23760	0.99950
	Red oak	0.01787	2.29900	0.92350
	White oak	0.14902	2.48148	0.01744
	Hickory	0.09287	2.01840	0.73760
	White ash	0.00975	1.92041	1.45806
	Yellow poplar	0.00468	1.84840	1.73228
12	Black oak	0.01466	2.30940	0.94550
	Red oak	0.01547	2.36560	0.86250
	White oak	0.10634	2.39810	0.20060
	Hickory	0.06378	2.09740	0.73040
	White ash	0.01509	2.21408	0.97872
	Yellow poplar	0.00369	1.72233	1.93225
13	Black oak	0.00888	2.23369	1.00040
	Red oak	0.00898	2.29601	0.92143
	White oak	0.09102	2.47164	0.00113
	Hickory	0.04974	2.01833	0.74444
	White ash	0.01528	2.23978	0.80092
	Yellow poplar	0.00246	1.84763	1.72350
14	Black oak	0.00743	2.30400	0.94700
	Red oak	0.00780	2.36140	0.86098
	White oak	0.05733	2.35819	0.25532
	Hickory	0.03483	2.09270	0.73845
	White ash	0.00868	2.19763	0.99926
	Yellow poplar	0.00180	1.71852	1.94168
15	Black oak	.000015	2.21421	1.00357
	Red oak	.000015	2.28573	0.91464
	White oak	.000137	2.39233	0.08289
	Hickory	.000068	2.01978	0.74640
	White ash	.000024	2.21651	0.84179
	Yellow poplar	.000005	1.81799	1.78323
16	Black oak	.000013	2.27814	0.95563
	Red oak	.000013	2.34839	0.85527
	White oak	.000087	2.35739	0.25671
	Hickory	.000049	2.09134	0.73907
	White ash	.000015	2.17690	1.02488
	Yellow poplar	.000004	1.70452	1.97632

[1]Metrics of equation variables coincide with metrics of designated tables.

Retrospective Comments Regarding MYERS

1. Although the species names and table identifiers are generated within the program, the 252 numeric coefficients must be loaded into registers 7 through 258 prior to program execution. This can be done by loading 8 magnetic data cards as follows:

    a. key 7.258
    b. execute RDTAX, and feed in cards as prompted

2. The numeric coefficients are present on the KRON-1 tape under the name MYDATA and can be downloaded to the calculator by the following (make sure SIZE=259):

    a. MYDATA in alpha register, zero in X register
    b. XEQ SEEKR
    c. key 7.258, then XEQ READRX

3. The MYERS program and the necessary coefficients are also grouped under the file name MYERFL on the KRON-1 tape and can be downloaded in one step by:

    a. MYERFL in the alpha register
    b. XEQ READA

Program No. 41F024

Calculator: HP-41C/CV

Program Name: SCALE

Author: T.W. Beers

Date: February 1982

Purpose: The program SCALE provides an accurate alternative to the scale-stick approach to log scaling. Gross scale, net scale and amount of defect can be displayed for each log after scaling diameter, length and defect type and dimensions are input. Provision is made for summaries of gross and net scale and numbers of logs by species (9 groups possible), and by log grade (4 possible) within species. Various flags are used to enable the selection of log rule from Doyle, Scribner, Scribner Decimal C, International 1/4" and International 1/8". The defect calculation formulas are those proposed by Grosenbaugh and described in Husch, Miller and Beers (FOREST MENSURATION, 1982. John Wiley & Sons). A printed copy of the summary by species is also made available by the program, if the HP-82143A printer is attached.

A. Storage assignments

Register	Use
00	species code; 1 through 9
01	current log gross scale
02	current log defect
03	current log net scale
04	current defect proportion regarding length
05	dd, current log scaling diameter (small end, inside bark) in inches; must be integer
06	LL, current log length in feet; an integer
	(Note: registers 00 through 06 are also used in label J for indexing)
07	temporary storage in INT1/4
08	current log grade; 1 through 4
09	general temporary storage

A. Storage assignments *(continued)*

Register	Use
10	Net scale sum, species 1
11	Net scale sum, species 1, grade 1
12	Net scale sum, species 1, grade 2
13	Net scale sum, species 1, grade 3
14	Net scale sum, species 1, grade 4
15	Gross scale sum, species 1
16	Gross scale sum, species 1, grade 1
17	Gross scale sum, species 1, grade 2
18	Gross scale sum, species 1, grade 3
19	Gross scale sum, species 1, grade 4
20 . . . 29	Same as 10-19, but for species 2
. . .	
90 . . . 99	Same as 10-19, but for species 9
100	not used
101-109	not used; anticipated storage for species names or abbreviations
110	no. logs, species 1
111	no. logs, species 1, grade 1
112	no. logs, species 1, grade 2
113	no. logs, species 1, grade 3
114	no. logs, species 1, grade 4
115 . . . 119	Same as 110-114, but for species 2
. . .	

A. Storage assignments *(continued)*

Register	Use
150 . . . 154	Same as 110-114, but for species 9
155	total number of logs
156	total gross scale
157	total net scale
158	total defect

B. Labels

Global	Use
SCALE	program start and initialization
DOYLE	volume calculation by Doyle log rule
INT1/4	volume calculation by Int. 1/4" log rule
INT1/8	volume calculation by Int. 1/8" log rule
SCRIB	volume calculation by Scribner log rule
SCRIBC	volume calculation by Scribner decimal C rule
SPEC	prompts for and stores species code

Local	
A	circular defect prompt and master processing loop for defect calculation
a	elliplical defect prompt
B	square defect prompt
b	rectangular defect prompt
C	crook defect prompt
D	sweep defect prompt
E	entire end defect prompt
e	wedge shape end defect prompt
J	prints a summary table with printer attached
00	clears registers 00-158
01	accumulates total log count, in label 19
02	subtracts 1 from log diameter, in label 19
03	calculates, displays and accumulates defect and net scale, in label 19
04	length proportion of defect calculation, in label 19
05	flag test to select desired log rule; defaults to Doyle rule
06	internal use in INT1/4
07	internal use in INT1/4
08	internal use in INT1/4
09	prompt for diameter and length (dd.LL?) and grade (GRADE?)

B. Labels *(continued)*

Local	Use
10	in SCRIB, to enable Scribner decimal C calculation
11	in INT1/4 to enable Int. 1/8" calcualtion
12	in the printing routine, label J, tests for zero entries and aligns and prints each table entry
13	increments the species code and source register number, in J
14	subroutine used in alignment of table entries, in J
15	"line" printing subroutine, in J
16	"bar" printing subroutine, in J
17,18	to provide various skips when flag 00 is set
19	accumulation of gross volume, log count and prompt for defect type
20	access to skip log count by grade, in label 19

C. Flags

Type and Number	Use
User:	
00	when set, assumes non-summary use, ignoring grade and species prompts, and the initialization steps
05	when set, enables display of defect amount
09	when set, enables prompt and summary by grade
01	set, implies Doyle log rule*
02	set, implies Scribner log rule
03	set, implies Scribner Dec. C log rule
04	set, implies Int. 1/4 inch log rule
08	set, implies Int. 1/8 inch log rule
06	used in INT1/4
07	used in INT1/4
26	can be cleared to silence the tones throughout the program
29	cleared in the program to suppress decimal points and separators

*when none of the log rule flags are set, the program defaults to Doyle rule, setting flag 01 in the process.

System:

none tested

D. Size and Key assignments

SIZE: 159

Suggested key assignments:

"SCALE" on COS
"J" on TAN (this is done internally)

E. Program procedure and example

I. Load the program "SCALE" into the calculator.

II. In RUN mode, process the following data:

Log No.	Species	dd.LL Diameter (dd) and Length (LL)	Grade	Defect Type	Defect Dimensions
1	7	20.16	1	A-round	diam.=5", len.=4'
2	3	24.08	2	a-ellipse	major d =6" minor d =4", len.=5'
3	5	16.08	3	B-square	side=8", len.=6'
4	5	30.20	4	b-rectangle	side 1=6" side 2=9", len.=10'
5	1	14.16	1	C-crook	deflection=4", len.=4'

1. First, assume that summaries are to be made (flag 00 clear) by species (no flag necessary) and by log grade (flag 09 set) and that defect amount is to be displayed for each log (flag 05 set) in addition to the usual gross and net scale, Doyle rule (flag 01 set). Therefore, we need to CF 00, SF 09, SF 05, and SF 01. This can be done at the first prompt if flag 00 is initially clear. Let's assume this is the case (flag 00 clear). We'll also presume USER mode and the printer is not attached until final summary printout is desired.

	Prompt	Input	*Example* Key	Output
a. start	-	-	COS	-
	CHECK FLAGS	-	SF 09 SF 05 SF 01 R/S	-
	CLEARING	(takes 45 seconds)		

E. Program procedure and example *(continued)*

		Prompt	Input	*Example* Key	Output
b.	log 1	SPEC: 0 OR ?	7	R/S	-
		dd.LL ?	20.16	R/S	-
		GRADE ?	1	R/S	*DOYLE* G.SCALE = 256
		DEF: AaBbCDEe	A	-	-
		DEF. DIAM.?	5	R/S	-
		DEF. LENGTH ?	4	R/S	N.SCALE = 250 DEFECT = 6
c.	log 2	SPEC: 7 OR ?	3	R/S	-
		dd.LL ?	24.08*	R/S	-
		GRADE ?	2	R/S	*DOYLE* G.SCALE = 200
		DEF: AaBbCDEe	a	-	-
		MAJOR AXIS ?	6	R/S	-
		MINOR AXIS ?	4	R/S	-
		DEF. LENGTH ?	5	R/S	N. SCALE = 192 DEFECT = 8
d.	log 3	SPEC: 3 OR ?	5	R/S	-
		dd.LL ?	16.08	R/S	-
		GRADE ?	3	R/S	*DOYLE* G.SCALE = 72
		DEF: AaBbCDEe	B	-	-
		DEF. SIDE ?	8	R/S	-
		DEF. LENGTH ?	6	R/S	N.SCALE = 47 DEFECT = 25
e.	log 4	SPEC: 5 OR ?	-**	R/S	-
		dd.LL ?	30.2***	R/S	-
		GRADE ?	4	R/S	*DOYLE* G.SCALE = 845
		DEF: AaBbCDEe	b	R/S	-
		DEF. SIDE 1 ?	6	R/S	-
		SIDE 2 ?	9	R/S	-
		DEF. LENGTH ?	10	R/S	N.SCALE = 800 DEFECT = 45
f.	log 5	SPEC: 5 OR ?	1	R/S	-
		dd.LL ?	14.16	R/S	-
		GRADE ?	1	R/S	*DOYLE* G.SCALE = 100
		DEF: AaBbCDEe	C	R/S	-
		DEFLECTION ?	4	R/S	-
		DEF. LENGTH ?	4	R/S	N.SCALE = 93 DEFECT = 7
		SPEC: 1 OR ?			

*Use of the zero in "08" is required for log lengths less than 10.
**If the species code does not change (as here, for code 5) simply press R/S without keying a code.
***The trailing zero in "20" need not be keyed.

E. Program procedure and example *(continued)*

At this point, since there is no log 6 in the example, one can either recall the various totals manually (step 1g.) or automatically with the aid of an attached printer (step 1h.)

g. Manual recall of totals.

Reference to part A, Storage assignments, shows the location of the various totals by species and grade. Net and gross scale can be obtained by direct register recall, whereas number of logs and grand totals must be recalled indirectly. Both procedures are shown, using the example.

(1) Gross and net scale.

Species	Grade	Net	Gross
1	All	RCL 10: 93	RCL 15: 100
1	1	RCL 11: 93	RCL 16: 100
3	All	RCL 30: 192	RCL 35: 200
3	2	RCL 32: 192	RCL 37: 200
5	All	RCL 50: 847	RCL 55: 917
5	3	RCL 53: 47	RCL 58: 72
5	4	RCL 54: 800	RCL 59: 845
7	All	RCL 70: 250	RCL 75: 256
7	1	RCL 71: 250	RCL 76: 256

Note that "defect" is not summarized, nor are net and gross scale by log grades only; Since allowance is made for up to 9 species groups, there is insufficient memory space to accomplish this.

(2) Number of logs (for brevity this will be shown only for species 5).

Species	Grade	Reg. No.*	Input	Keys **	Output
5	All	130	130	RCL f.X	2
5	3	133	133	RCL f.X	1
5	4	134	134	RCL f.X	1

*The register number having the log count for species n can be found by the formula

$$R_n = (n + 1)5 + 100$$

The log count by grade within species follows logically from the species total.

**Note the "f" implies the shift key

E. Program procedure and example *(continued)*

(3) Grand totals

Type of total	Reg. No.	Input	Keys	Output
number of logs	155	155	RCL f.X	5
gross scale	156	156	RCL f.X	1473
net scale	157	157	RCL f.X	1382
defect	158	158	RCL f.X	91

h. Automatic recall of some totals

To obtain a print out of gross scale, net scale, and number of logs by species and grand total:

(1) Turn calculator OFF and ensure that the HP 82143A printer is OFF.

(2) Attach printer to lowest available port.

(3) Turn both calculator and printer ON and put printer in MAN mode.

(4) Press TAN key (in USER mode); this excutes label J and provides a printout as shown for the example.

Note that species are in numeric order and "zero entry species" are not printed.

i. Printed output.

```
************
      GROSS  NET   NO.
SPEC. SCALE SCALE LOGS
------------
  1    100    93    1
  3    200   192    1
  5    917   847    2
  7    256   250    1

TOTAL  1473  1382   5
************
```

F. Formulas used.

1. For the calculation of gross scale in board feet, V, of a log having a scaling diameter in inches, d, and log length in feet, L:

   a. Doyle rule: $V_D = (d-4)^2 \frac{L}{16}$

   b. Scribner: $V_S = (.79d^2-2d-4)\ \frac{L}{16}$

   c. Scribner Decimal: $V_C = V_S/10$, rounded to an integer

   d. International 1/8 inch: $V_{I1/8}$ was calculated using the algorithm developed by Beers (see FOREST MENSURATION, by Husch, Miller and Beers, 1982. John Wiley & Sons).

      The algorithm makes use of the traditional formula, $V = .22d^2-.71d$, for the top 4-foot section, followed by the diameter of each subsequent 4-foot section incremented by .5 inch, and re-use of the formula.

   e. International 1/4 inch: $V_{I1/4} = .905(V_{I1/8})$

2. For the calculation of various defects the formulas proposed by Grosenbaugh and cited in FOREST MENSURATION were used:

Defect Type	Key	Formula for Defect Proportion
round	A	$p = \frac{(\text{defect diameter} + 1)^2}{(d - 1)^2} \cdot \frac{\ell}{L}$
ellipse	a	$p = \frac{(\text{major axis} + 1)\ (\text{minor axis} + 1)}{(d - 1)^2} \cdot \frac{\ell}{L}$
square	B	$p = \frac{(\text{defect side} + 1)^2}{(d - 1)^2} \cdot \frac{\ell}{L} \cdot \frac{4}{\Pi}$
rectangle	b	$p = \frac{(\text{side 1} + 1)\ (\text{side 2} + 1)}{(d - 1)^2} \cdot \frac{\ell}{L} \cdot \frac{4}{\Pi}$
crook	C	$p = \frac{\text{deflection}}{d} \cdot \frac{\ell}{L}$



F. Formulas used *(continued)*

Defect Type	Key	Formula for Defect Proportion
sweep	D	$p = \frac{(\text{departure} - 2)}{d}$



D = departure

Defect Type	Key	Formula for Defect Proportion
entire	E	$p = \frac{\ell}{L}$
wedge	e	$p = \frac{\text{defect central angle}}{360} \cdot \frac{\ell}{L}$

In all the above formulas

ℓ = defect length in feet

L = log length in feet

d = scaling diameter, i.e. diameter inside bark at the small end of log

p = proportion of defect.

3. Defect amount and net scale:

defect = (p) (gross scale), where

p = proportion of defect

net scale = gross scale - defect

G. Program listing and example.

```
♦LBL "SCALE"
FS? 00
GTO 09
"CHECK FLAGS"
AVIEW
PROMPT
" *CLEARING*"
AVIEW
CF 29
FIX 0
.158
0
♦LBL 00
STO IND Y
TONE 3
ISG Y
GTO 00
♦LBL "SPEC"
RCL 00
TONE 7
"SPEC: "
ARCL X
"⊦ OR ?"
PROMPT
STO 00
♦LBL 09
TONE 7
TONE 7
"dd.LL ?"
PROMPT
INT
STO 05
LASTX
FRC
1 E2
*
INT
STO 06
FC? 09
GTO 05
TONE 7
"GRADE ?"
PROMPT
STO 08
♦LBL 05
FS? 01
GTO "DOYLE"
FS? 02
GTO "SCRIB"
FS? 03
GTO "SCRIBC"
FS? 04
GTO "INT1/4"
FS? 08
GTO "INT1/8"
♦LBL "DOYLE"
" *DOYLE*"
AVIEW
RCL 05
4
-
X↑2
RCL 06
16
/
*
RND
STO 01
SF 01
♦LBL 19
156
RDN
ST+ IND T
FS? 00
GTO 17
RCL 00
1
+
5
*
100
+
1
ST+ IND Y
FC? 09
GTO 20
X<>Y
RCL 08
+
X<>Y
ST+ IND Y
♦LBL 20
RCL 00
10
*
5
+
RCL 01
ST+ IND Y
FC? 09
GTO 17
X<>Y
RCL 08
+
X<>Y
ST+ IND Y
♦LBL 17
"G. SCALE= "
ARCL X
AVIEW
PSE
TONE 7
"DEF: AaBbCDEe"
PROMPT
♦LBL 01
155
RDN
ISG IND T
X<> X
FS? 00
GTO 09
GTO "SPEC"
♦LBL A
TONE 7
"DEF. DIAM. ?"
PROMPT
1
+
X↑2
♦LBL 02
RCL 05
1
-
X↑2
/
♦LBL 04
TONE 7
"DEF. LENGTH ?"
PROMPT
*
RCL 06
/
STO 04
♦LBL 03
RCL 04
RCL 01
*
RND
STO 02
158
```

G. Program listing and example *(continued)*

```
151 RDN
152 ST+ IND T
153 CHS
154 RCL 01
155 +
156 157
157 RDN
158 ST+ IND T
159 FS? 00
160 GTO 18
161 RCL 00
162 10
163 *
164 X<>Y
165 ST+ IND Y
166 FC? 09
167 GTO 18
168 X<>Y
169 RCL 08
170 +
171 X<>Y
172 ST+ IND Y
173 STO 03
174♦LBL 18
175 "N. SCALE= "
176 ARCL X
177 AVIEW
178 PSE
179 FC? 05
180 GTO 01
181 RCL 02
182 "DEFECT= "
183 ARCL X
184 AVIEW
185 GTO 01
186♦LBL a
187 TONE 7
188 "MAJOR AXIS ?"
189 PROMPT
190 1
191 +
192 TONE 7
193 "MINOR AXIS ?"
194 PROMPT
195 1
196 +
197 *
198 GTO 02
199♦LBL B
200 TONE 7
201 "DEF. SIDE ?"
202 PROMPT
203 1
204 +
205 X↑2
206 4
207 *
208 PI
209 /
210 GTO 02
211♦LBL b
212 TONE 7
213 "DEF. SIDE 1 ?"
214 PROMPT
215 1
216 +
217 TONE 7
218 "  SIDE 2 ?"
219 PROMPT
220 1
221 +
222 *
223 4
224 *
225 PI
226 /
227 GTO 02
228♦LBL C
229 TONE 7
230 "DEFLECTION ?"
231 PROMPT
232 RCL 05
233 /
234 GTO 04
235♦LBL D
236 TONE 7
237 "DEPARTURE ?"
238 PROMPT
239 2
240 -
241 RCL 05
242 /
243 STO 04
244 GTO 03
245♦LBL E
246 1
247 GTO 04
248♦LBL e
249 TONE 7
250 "DEFECT ∡ ?"
251 PROMPT
252 360
253 /
254 GTO 04
255♦LBL "SCRIB"
256 " *SCRIBNER*"
257 AVIEW
258♦LBL 10
259 RCL 05
260 X↑2
261 .79
262 *
263 RCL 05
264 2
265 *
266 -
267 4
268 -
269 RCL 06
270 16
271 /
272 *
273 FS? 03
274 RTN
275 RND
276 STO 01
277 GTO 19
278♦LBL "SCRIBC"
279 "*SCRIB. DEC.C*"
280 AVIEW
281 XEQ 10
282 10
283 /
284 RND
285 STO 01
286 GTO 19
287♦LBL "INT1/4"
288 CF 07
289 "  *INT1/4*"
290 AVIEW
291♦LBL 11
292 SF 06
293 RCL 06
294 4
295 /
296 INT
297 STO 07
298 LASTX
299 FRC
300 XEQ 07
```

G. Program listing and example *(continued)*

```
CF 06
*
STO 01
RCL 07
X=0?
GTO 08
♦LBL 06
1
ST- 07
♦LBL 07
RCL 07
2
/
RCL 05
+
STO 09
X↑2
.22
*
RCL 09
.71
*
-
FS? 06
RTN
ST+ 01
RCL 07
X>0?
GTO 06
♦LBL 08
FS?C 07
RTN
RCL 01
.905
*
RND
STO 01
GTO 19
♦LBL "INT1/8"
SF 07
"  *INT1/8*"
AVIEW
XEQ 11
RCL 01
RND
STO 01
GTO 19
♦LBL J
CF 29
FIX 0
15.09510
STO 01
1.009
STO 02
XEQ 16
"        GROSS"
ACA
"   NET   NO."
ACA
PRBUF
"SPEC.  SCALE"
ACA
"  SCALE LOGS"
ACA
PRBUF
XEQ 15
♦LBL 12
RCL IND 01
X=0?
GTO 13
1
SKPCHR
RCL 02
ACX
RCL IND 01
LOG
INT
CHS
7
+
SKPCHR
RCL IND 01
ACX
RCL 01
5
-
STO 03
RCL IND 03
LOG
INT
CHS
5
+
SKPCHR
RCL IND 03
ACX
RCL 03
100
+
RCL 02
INT
1
-
5
*
-
STO 04
RCL IND 04
LOG
INT
CHS
3
+
SKPCHR
RCL IND 04
ACX
PRBUF
♦LBL 13
ISG 01
X<> X
ISG 02
GTO 12
ADV
"TOTAL"
ACA
156
STO 05
5
STO 06
XEQ 14
157
STO 05
XEQ 14
155
STO 05
3
STO 06
XEQ 14
♦LBL 16
SF 12
"************"
PRA
CF 12
RTN
♦LBL 14
RCL IND 05
ENTER↑
LOG
INT
CHS
```

G. Program listing and example *(continued)*

```
451 RCL 06
452 +
453 SKPCHR
454 X<>Y
455 ACX
456 RTN
457♦LBL 15
458 SF 12
459 "------------"
460 PRA
461 CF 12
462 CLA
463 END
```

```
*************
      GROSS  NET   NO.
SPEC. SCALE SCALE LOGS
------------
  1     100    93    1
  3     200   192    1
  5     917   847    2
  7     256   250    1

TOTAL  1473  1382    5
*************
```

```
                 CAT 1
LBL'SCALE
LBL'SPEC
LBL'DOYLE
LBL'SCRIB
LBL'SCRIBC
LBL'INT1/4
LBL'INT1/8
END        1109 BYTES
```

Retrospective Comments Regarding SCALE

1. Because of the program length and required SIZE, if SCALE is used for summary purposes, the user is cautioned that no error correction routine is present, therefore faulty data, once processed, must be recorded and deleted "by hand."

2. For specific applications, a shorter program should be written, incorporating an error correction procedure.

Program No. 41F025

Calculator: HP-41C/CV

Program Name: D2BVVMC (Diameter To Basal Area and Vice-Versa with Metric Capability)

Author: T.W. Beers

Date: February, 1982

Purpose: To solve the formula relating tree diameter in inches or centimeters, D, and crossectional area ("basal area") in square feet or square meters, B, for either D or B, given the other as input.

A. Storage assignments-- none used.

B. Labels

Global	Use
D2BVVMC	start of program, reminder prompts
Local	
A	program start if reminder prompts are not wanted
a	to convert an answer (B or D) to the "other" units (U.S. or metric)
00	basic input prompt and its reversal
01	calculation and display of basal area
02	calculation and display of diameter
03	internal to label a
04	to enable "one-way" calculation with flag 01 set

C. Flags

Number	Use
01	if clear, intermittent calculation of D and B is assumed, therefore the prompt is always present before an input

C. Flags *(continued)*

Number	Use
	if set (externally), the user is assumed to want "one-way" calculations, therefore after the initial choice of direction (B to D or D to B) the input prompt is skipped and an input, followed by R/S leads to the displayed answer
02	if set, metric input (cm or m^2) is assumed and metric output is obtained
20	set internally to imply the direction of calculation from D to B
22	numeric data input flag set automatically when numbers are keyed in; tested to detect this condition.

D. Size and key assignments

SIZE needed: any

key assign : D2BVVMC on X≷Y
A on A (done internally)
a on shift A (done internally)

E. Program procedure and example

I. Load the program into the calculator.

II. XEQ D2BVVMC (X≷Y in USER mode) and follow the prompts; note the following guidelines:

1. With flag 01 clear, if the prompt is in the correct direction, key the number, press R/S and read the answer. To reverse the direction of the prompt and calculation, press R/S without first keying the number.

2. With flag 01 set, prompting is skipped and the current direction of the calculator prevails. If the direction is correct for you, key the number, press R/S, read answer; key the next number, R/S, read answer, etc. To change the direction to your liking, press A, observe prompt, then R/S.

3. Metric input and output prevail with flag 02 set.

4. An answer can be converted to the other units by pressing a (shift A).

E. Program procedure and example *(continued)*

III. Some examples:

1. U.S. (CF 02)

Direction	Input	Output	Using key a converts to
D to B	10	BA = 0.545 SF	= 0.051 M↑2
"	12	BA = 0.785 SF	= 0.073 M↑2
"	30	BA = 4.909 SF	= 0.456 M↑2
B to D	.5454	DIAM = 10.0 IN.	= 25.4 CM.
"	4.3	DIAM = 28.1 IN.	= 71.3 CM.

2. Metric (SF 02)

Direction	Input	Output	Converts to
D to B	25.4	BA = 0.051 M↑2	= 0.545 SF
"	100	BA = 0.785 M↑2	= 8.454 SF
B to D	.051	DIAM. = 25.5 CM.	= 10.0 IN.
B to D	.456	DIAM. = 76.2 CM.	= 30.0 IN.

F. Formulas

1. Basic formula:

U.S.

$$B(\text{in. sq. ft.}) = \frac{\Pi D^2}{576}, \ D \text{ in inches}$$

Metric

$$B(\text{in. } m^2) = \frac{\Pi D^2}{40000}, \ D \text{ in cm}$$

2. Conversion factors used:

1 in. = 2.54 cm, and its inverse

1 sq. ft. = 0.09290304 m^2 , and its inverse

G. Program listing and example.

```
01♦LBL "D2BVVMC"
02 "CHECK FLAGS:"
03 AVIEW
04 PSE
05 "SF1: ONE WAY"
06 AVIEW
07 PSE
08 "SF2: METRIC"
09 AVIEW
10 PSE
11 "*DO IT NOW*"
12 PROMPT
13♦LBL A
14 CF 22
15♦LBL 00
16 "BA TO DIAM."
17 FS? 20
18 "DIAM. TO BA"
19 TONE 7
20 TONE 7
21 PROMPT
22 FS?C 22
23 GTO 01
24 FC?C 20
25 SF 20
26 GTO 00
27♦LBL 01
28 FC? 20
29 GTO 02
30 X↑2
31 PI
32 *
33 FC? 02
34 576
35 FS? 02
36 40000
37 /
38 FIX 3
39 "BA= "
40 ARCL X
41 FC? 02
42 "⊢ SF"
43 FS? 02
44 "⊢ M↑2"
45 AVIEW
46 STOP
47 FS? 01
48 GTO 01
49 GTO 00
50♦LBL 02
```

```
51 FC? 02
52 576
53 FS? 02
54 40000
55 *
56 PI
57 /
58 SQRT
59 FIX 1
60 "DIAM.= "
61 ARCL X
62 FC? 02
63 "⊢ IN."
64 FS? 02
65 "⊢ CM."
66 AVIEW
67 STOP
68 FS? 01
69 GTO 02
70 GTO 00
71♦LBL a
72 FC? 20
73 GTO 03
74 .09290304
75 FS? 02
76 1/X
77 *
78 " = "
79 ARCL X
80 FS? 02
81 "⊢ SF"
82 FC? 02
83 "⊢ M↑2"
84♦LBL 04
85 AVIEW
86 STOP
87 FS? 01
88 GTO 01
89 GTO 00
90♦LBL 03
91 2.54
92 FS? 02
93 1/X
94 *
95 " = "
96 ARCL X
97 FC? 02
98 "⊢ CM."
99 FS? 02
100 "⊢ IN."
101 GTO 04
102 END
```

```
CHECK FLAGS:
SF1: ONE WAY
SF2: METRIC
                        SF 01
                          RUN
BA TO DIAM.
                          RUN
DIAM. TO BA
               10.000     RUN
BA= 0.545 SF
                        XEQ a
 = 0.051 M↑2
               12.000     RUN
BA= 0.785 SF
                        XEQ a
 = 0.073 M↑2
               30.000     RUN
BA= 4.909 SF
                        XEQ a
 = 0.456 M↑2
                        XEQ A
DIAM. TO BA
                          RUN
BA TO DIAM.
                .5454     RUN
DIAM.= 10.0 IN.
                        XEQ a
 = 25.4 CM.
                  4.3     RUN
DIAM.= 28.1 IN.
                        XEQ a
 = 71.3 CM.


                        SF 02
                        XEQ A
BA TO DIAM.
                          RUN
DIAM. TO BA
                 25.4     RUN
BA= 0.051 M↑2
                        XEQ a
 = 0.545 SF
              100.000     RUN
BA= 0.785 M↑2
                        XEQ a
 = 8.454 SF
                        XEQ A
DIAM. TO BA
                          RUN
BA TO DIAM.
                 .051     RUN
DIAM.= 25.5 CM.
                        XEQ a
 = 10.0 IN.
                 .456     RUN
DIAM.= 76.2 CM.
                        XEQ a
 = 30.0 IN.
```

Program No. 41F026

Calculator: HP-41C/CV

Program Name: F2CVV (Fahrenheit To Centigrade and Vice Versa)

Author: T.W. Beers

Date: February, 1982

Purpose: To convert either Fahrenheit and Centigrade degrees to the other.

A. Storage assignments

Register	Use
01	degrees Fahrenheit
02	degrees Centigrade (i.e., Celsius)
03	Kelvin units

B. Labels

Global	Use
F2CVV	start of program, reminder prompts
Local	
B	program start if reminder prompts are not wanted
b	to convert the displayed answer to Kelvin units
00	basic input prompt and its reversal
01	calculation and display of the answer

C. Flags

Number	Use
01	if clear, intermittent calculation of either degrees F or degrees C is assumed, therefore the prompt is always present before an input if set (externally), the user is assumed to want "one-way" calculations, therefore after the initial choice of direction (F to C or C to F) the input prompt is skipped, and an input, followed by R/S leads to the displayed answer

C. Flags *(continued)*

Number	Use
20	set internally to imply the direction of conversion from C to F
22	numeric data input flag; tested to detect this condition (numeric data input)

D. Size and key assignments

SIZE needed: $\geq$ 004

Key assign :				
	F2CVV	on	R↓	
	B	on	B	(done internally)
	b	on	b	(done internally)

E. Program procedure and example

I. Load the program into the calculator.

II. XEQ F2CVV (R↓ in USER mode) and follow the prompts; note the following guidelines:

1. With flag 01 clear, if the prompt is in the correct direction, key the number, press R/S, and read the answer. To reverse the direction of the prompt and conversion, press R/S without first keying the number.

2. With flag 01 set, prompting is skipped and the current conversion direction prevails. If the direction is correct for you, key the number, press R/S, and read the answer; key the next number, R/S, read the answer, etc. To change the direction to your liking, press B, observe prompt, then R/S.

3. A displayed answer (Fahrenheit or Centrigrade) can be converted to Kelvin units by pressing b (i.e., shift b) in USER mode; then, pressing R/S returns the calculations to the main stream.

III. Some examples:

Direction	Input	Output	using key b, converts to
F to C	32	DEG C = 0.0	= 273.2 KELVIN
F to C	86.9	DEG C = 30.5	= 303.7 KELVIN
C to F	100	DEG F = 212.0	= 373.2 KELVIN
C to F	-10	DEG F = 14.0	= 263.2 KELVIN

F. Formulas

1. The primary conversion is based on the algorithm: add 40 to the number, multiply by $\frac{5}{9}$ or $\frac{9}{5}$, depending on the direction of conversion, and subtract 40.

Thus:

from F to C

$$°C = \frac{5}{9}(40 + °F) - 40$$

from C to F

$$°F = \frac{9}{5}(40 + °C) - 40$$

2. Kelvin units are obtained by the relation

KELVIN units = 273.15 + °C

G. Program listing and example

```
01♦LBL "F2CVV"
02 "CHECK FLAGS:"
03 AVIEW
04 PSE
05 "SF1: ONE WAY"
06 AVIEW
07 PSE
08 "*DO IT NOW*"
09 PROMPT
10♦LBL B
11 CF 22
12 FIX 1
13♦LBL 00
14 "FAHR. TO CENT"
15 FS? 20
16 "CENT. TO FAHR"
17 TONE 7
18 TONE 7
19 PROMPT
20 FS?C 22
21 GTO 01
22 FC?C 20
23 SF 20
24 GTO 00
25♦LBL 01
26 FC? 20
27 STO 01
28 FS? 20
29 STO 02
30 40
31 +
32 1.8
33 FC? 20
34 1/X
35 *
36 40
37 -
38 FS? 20
39 STO 01
40 FC? 20
41 STO 02
42 "DEG C= "
43 FS? 20
44 "DEG F= "
45 ARCL X
46 AVIEW
47 STOP
48 FS? 01
49 GTO 01
50 GTO 00
51♦LBL b
52 RCL 02
53 273.15
54 +
55 STO 03
56 "= "
57 ARCL X
58 "⊢ KELVIN"
59 AVIEW
60 RTN
61 FS? 01
62 GTO 01
63 GTO 00
64 END
```

```
                  XEQ "F2CVV"
CHECK FLAGS:
SF1: ONE WAY
*DO IT NOW*
                        SF 01
                        XEQ B
FAHR. TO CENT
                  32.0    RUN
DEG C= 0.0
                        XEQ b
= 273.2 KELVIN
                  86.9    RUN
DEG C= 30.5
                        XEQ b
= 303.7 KELVIN
                        XEQ B
FAHR. TO CENT
                          RUN
CENT. TO FAHR
                 100.0    RUN
DEG F= 212.0
                        XEQ b
= 373.2 KELVIN
                 -10.0    RUN
DEG F= 14.0
                        XEQ b
= 263.2 KELVIN
```

Program No. 41F027

Calculator: HP41C/CV

Program Name: CONVRT

Author: T.W. Beers

Date: 1982

Purpose: A general program for commonly occurring conversions. Up to 15 local labels can be used to program user-specified conversions. Those presently programmed are inches-centimeters, feet-meters, square feet per acre-square meters per hectare, cubic feet per acre-cubic meters per hectare, ounces-milliliters (including "fifths" and "quarts"). Either direction of conversion is easily selected.

A. Storage assignments:

Register	Use
00	the six-character prompt indicating the "forward" direction of conversion
01	the reverse prompt from that in register 00
02	the six-character identifier for the answer in the "forward" direction
03	the identifier for the "reverse" direction answer
04	the conversion constant; a multiplier in the forward direction, e.g., 2.54 for inches to centimeters
05	the number of decimals to be displayed

B. Labels

Global	Use
CONVRT	program start, initial prompts

B. Labels *(continued)*

Local	Use
A	loads the registers for the conversion of inches to and from centimeters (IN→CM or CM→IN)
B	same, for feet and meters (FT→M. or M.→FT)
C	same, for square feet per acre and square meters per hectare (SFA-MH or SMH-FA)
D	same, for cubic feet per acre and cubic meters per hectare (CFA-MH or CMH-FA)
E	same, for ounces and milliliters (OZ→ML or ML→OZ)
e	converts ounces or milliliters to "fifths" (i.e., 4/5 of a quart, 1/5 of a gallon, or 25.6 ounces) and to quarts (1.25 of a fifth, or 32 ounces)
00	provides the initial prompt for direction of conversion and its automatic reversal by R/S in the absence of a numeric key stroke
01	converts milliliters (ML) to ounces in label e. Note: 1 ML = .001 liter = 1 cubic centimeter, i.e., ML = cc
11	performs the acutal conversion arithmetic, displays the answer and directs control dictated by flag 01

C. Flags

Type and Number	Use
User:	
01	when set (externally), implies the user desires consistent use of one conversion label for several successive numbers, therefore one need only key the number and press R/S after the first conversion; otherwise the inital prompt is obtained by R/S
20	set in label 00 and used throughout to indicate the direction of conversion; "clear" implies the forward direction (inches to centimeters, feet to meters, etc.) while "set" implies the reverse direction.
22	numeric data entry flag, tested in label 00 to reverse the direction of conversion in the absence of a numeric key stroke prior to R/S

C. Flags *(continued)*

Type and Number	Use
System:	none tested

D. Size and key assignments

SIZE: $\geq$006

Suggested key assignments:

"CONVRT" on shift R↓

A
B
C
D
E
e

done internally on these respective keys

F through J
and a through d

potential local labels for future conversions

E. Program procedure and example

I. Load the program into the calculator

II. XEQ CONVRT (shift R↓ in USER mode) and follow the prompts. As examples, assume we need the following conversions made and require one decimal place in the answer:

(a) using key A and with flag 01 set:

10
12
14

to centimeters

(b) using key B and with flag 01 clear:

4.5 feet to meters
1.0 meters to feet

(c) using key C:

10 sq. ft. per acre to sq. meters per hectare
1 sq. m. per hectare to sq. feet per acre
100 sq. ft. per acre to sq. m. per hectare

E. Program procedure and example *(continued)*

(d) using key D:

128 cu. ft. per acre to cu. m. per hectare
175 cu. m. per hectare to cu. ft. per acre

(e) using key E (and e)

32 oz. to ml., then to fifths and quarts
750 ml. to oz., then to fifths and quarts

Step	Prompt	Input	*Example* Key	Output
1. Initialize	-	-	SF 01*	-
	-	-	shift R↓	-
	NO. DECIMALS	1	R/S	-
	LOCAL LABEL?			
2. Answers:				
(a)		-	A	-
	IN→CM**	10	R/S	CM.= 25.4
		12	R/S	CM.= 30.5
		14	R/S	CM.= 35.6
(b)		-	CF 01	-
		-	B	-
	FT→M.	4.5	R/S	M.= 1.4
		-	R/S	-
	FT→M.	-	R/S	-
	M.→FT	1	R/S	FT.= 3.3
(c)		-	C	-
	SMH-FA	-	R/S	-
	SFA-MH	10	R/S	SM/H= 2.3
		-	R/S	-
	SFA-MH	-	R/S	-
	SMH-FA	1	R/S	SF/A= 4.4
		-	R/S	-
	SMH-FA	-	R/S	-
	SFA-MH	100	R/S	SM/H= 23.0
(d)		-	D	-
	CFA-MH	128	R/S	CM/H= 9.0
		-	R/S	-
	CFA-MH	-	R/S	-
	CMH-FA	175	R/S	CF/A= 2,501.1

E. Program procedure and example *(continued)*

			Example	
Step	Prompt	Input	Key	Output
(e)		-	E	-
	ML→OZ	-	R/S	-
	OZ→ML	32	R/S	ML.= 946.2
		-	shift E (i.e., e)	= 1.2 FIFTHS
				= 1.0 QTS.
		-	R/S	-
	OZ→ML	-	R/S	-
	ML→OZ	750	R/S	OZ.= 25.4
		-	e	= 1.0 FIFTHS
				= 0.8 QTS.

*setting flag 01 is needed only in part (a) of the example since only there do you find a "group" of conversions in the same direction

**if the prompt is not in the correct direction for your needs, press R/S without keying a number, and the direction will be reversed.

III. Other comments

1. This program as described and listed in part F, provides for only 5 conversions. The addition of others to keys F, G, H, I, J, and a, b, c, and d is easily accomplished by inserting appropriate labels, prompts, and multipliers into the program. For example, a conversion routine for miles to kilometers and vice versa could be "assigned" to key F and achieved by inserting the following program steps just prior to the END statement (step 130):

Step		
130	LBL F	
131	CF 22	
132	FIX IND 05	
133	"MI->KM"	
134	ASTO 00	
135	"KM->MI"	
136	ASTO 01	
137	"MI.= "	
138	ASTO 02	
139	"KM.= "	
140	ASTO 03	
141	1.6093	(the multiplier to convert miles to kilometers)
142	STO 04	
143	GTO 00	

E. Program procedure and example *(continued)*

2. Careful examination of the program listing, steps 39, 41, 53, 55, 95 and 97, reveals the attempt to create a right-facing arrow using the minus sign and a right parenthesis; the right parenthesis can be generated "from scratch" only by use of the technique known as synthetic programming (reference: Wickes, W.C. 1980. Synthetic Programming on the HP-41C. Larken Publications, 4517 NW Queens Ave. Corvallis, OR. 97330). A reasonable approximation to the right parenthesis is on the standard 41C keyboard as the "greater than" symbol (shift TAN in ALPHA mode).

3. The generation of the conversion routines as described above can also be achieved by use of the program CONBLD (conversion builder). This method is very inefficient timewise, but does provide an insight into the use of a program to write another program. Synthetic programming is employed in CONBLD (program number 41F028), and the PPC ROM is required.

4. The program as written assumes a printer is <u>not</u> attached; if printed copy is desired, appropriate changes should be made to provide neatly formatted output. Part F shows the example, if printer is attached and in NORM mode.

5. The use of conversions that require more than a simple multiplier or divisor, such as Fahrenheit to centigrade and vice versa, can be built into this CONVRT program but it might be just as efficient to develop individual programs. This has been done in programs F2CVV (program number 41F026) and D2BVVMC (program number 41F025).

F. Program listing and example

```
01♦LBL "CONVRT"
02 "NO. DECIMALS?"
03 PROMPT
04 STO 05
05 "LOCAL LABEL?"
06 PROMPT
07♦LBL 00
08 CLA
09 FC? 20
10 ARCL 00
11 FS? 20
12 ARCL 01
13 TONE 7
14 PROMPT
15 FS?C 22
16 GTO 11
17 FC?C 20
18 SF 20
19 GTO 00
20♦LBL 11
21 RCL 04
22 FS? 20
23 1/X
24 *
25 CLA
26 FS? 20
27 ARCL 02
28 FC? 20
29 ARCL 03
30 ARCL X
31 AVIEW
32 STOP
33 FS? 01
34 GTO 11
35 GTO 00
36♦LBL A
37 CF 22
38 FIX IND 05
39 "IN->CM"
40 ASTO 00
41 "CM->IN"
42 ASTO 01
43 "IN.= "
44 ASTO 02
45 "CM.= "
46 ASTO 03
47 2.54
48 STO 04
49 GTO 00
50♦LBL B
51 CF 22
52 FIX IND 05
53 "FT->M."
54 ASTO 00
55 "M.->FT"
56 ASTO 01
57 "FT.= "
58 ASTO 02
59 "M.= "
60 ASTO 03
61 .3048
62 STO 04
63 GTO 00
64♦LBL C
65 CF 22
66 FIX IND 05
67 "SFA-MH"
68 ASTO 00
69 "SMH-FA"
70 ASTO 01
71 "SF/A= "
72 ASTO 02
73 "SM/H= "
74 ASTO 03
75 2.296 E-1
76 STO 04
77 GTO 00
78♦LBL D
79 CF 22
80 FIX IND 05
81 "CFA-MH"
82 ASTO 00
83 "CMH-FA"
84 ASTO 01
85 "CF/A= "
86 ASTO 02
87 "CM/H= "
88 ASTO 03
89 6.997 E-2
90 STO 04
91 GTO 00
92♦LBL E
93 CF 22
94 FIX IND 05
95 "OZ->ML"
96 ASTO 00
97 "ML->OZ"
98 ASTO 01
99 "OZ.= "
100 ASTO 02
101 "ML.= "
102 ASTO 03
103 2.957 E1
104 STO 04
105 GTO 00
106♦LBL e
107 FC? 20
108 XEQ 01
109 25.6
110 /
111 " = "
112 ARCL X
113 "├ FIFTHS"
114 AVIEW
115 PSE
116 PSE
117 1.25
118 /
119 " = "
120 ARCL X
121 "├ QTS."
122 AVIEW
123 RTN
124 FS? 01
125 GTO 11
126 GTO 00
127♦LBL 01
128 29.573
129 /
130 END
```

F. Program listing and example *(continued)*

Example with printer in NORM mode

```
                   SF 01
            XEQ "CONVRT"
NO. DECIMALS?
              1.0    RUN
LOCAL LABEL?
                   XEQ A
IN->CM
             10.0    RUN
CM.= 25.4
             12.0    RUN
CM.= 30.5
             14.0    RUN
CM.= 35.6
```

```
                   CF 01
                   XEQ B
FT->M.
              4.5    RUN
M.= 1.4
                     RUN
FT->M.
                     RUN
M.->FT
              1.0    RUN
FT.= 3.3
```

```
                   XEQ D
CFA-MH
            128.0    RUN
CM/H= 9.0
                     RUN
CFA-MH
                     RUN
CMH-FA
            175.0    RUN
CF/A= 2501.1
```

```
                   XEQ C
SMH-FA
                     RUN
SFA-MH
             10.0    RUN
SM/H= 2.3
                     RUN
SFA-MH
                     RUN
SMH-FA
              1.0    RUN
SF/A= 4.4
                     RUN
SMH-FA
                     RUN
SFA-MH
            100.0    RUN
SM/H= 23.0
```

```
                   XEQ E
ML->OZ
                     RUN
OZ->ML
             32.0    RUN
ML.= 946.2
                   XEQ e
 = 1.2 FIFTHS
 = 1.0 QTS.
                     RUN
OZ->ML
                     RUN
ML->OZ
            750.0    RUN
OZ.= 25.4
                   XEQ e
 = 1.0 FIFTHS
 = 0.8 QTS.
```

Retrospective Comments Regarding CONVRT

1. Since many local alpha labels are used, one must be careful to not have other programs assigned to keys meant for use in CONVRT; otherwise the assignment takes precedence over the local label and calculator execution transfers to the assigned program.

2. Inspired by the need to convert liters to gallons on a recent auto trip, and the desire to calculate miles per gallon easily, a program called CVERT was prepared incorporating these needs with the conversions also present in CONVRT. The program CVERT is on the KRON-1 tape and operates similar to CONVRT with the following local label functions:

   A through E and e — same as in CONVRT

   F -- the miles-kilometers conversion routine suggested in the CONVRT write-up

   G -- gallons-liters conversion

   H -- calculation of miles per gallon; assumes label G was just used to convert liters to gallons, therefore "gallons used" is in X register when H is pressed.

Program No. 41F028

Calculator: HP-41C/CV

Program Name: CONBLD (for Conversion Builder)

Authors: Thomas W. and Ted W. Beers

Date: 1982

Purpose: CONBLD was written to demonstrate that program statements and indeed entire programs can be composed by another program. Thus, CONBLD can be used to insert user-designed subroutines in the CONVRT program (No. 41F027) as an alternative to keying the instructions directly into the program. "Synthetic programming" (see Wickes, W.C. 1980. Synthetic Programming on the HP-41C. Larken Publications, 4517 NW Queens Ave. Corvallis, OR. 97330) is used and the PPC ROM is required (Personal Programming Center, 2545 West Camden Place, Santa Ana, CA 92704).

A. Storage registers

Register	Use
00	the six-character prompt indicating the "forward" direction of the conversion
01	the reverse of the prompt stored in register 00
02	the six-character identifier (units) for the answer in the "forward" direction
03	the identifier for the answer in the reverse direction
04	the conversion constant; a multiplier in the forward direction, e.g., 3.785 for U.S. gallons to liters.
05	the local label (A through J or a through e)
06-11	used in the synthetic programming process.
13	decrementing used in label 15 to isolate numerals in the conversion factor
17	decrementing used in label 07 to determine text length

A. Storage registers *(continued)*

Register	Use
18	index for indirect recall of text line by labels 07, 08, 09
20	incrementing used in label 10 to select alpha text to be assembled
21	first part of the conversion factor (alpha form)
22	second part of the conversion factor (alpha form)

B. Labels

Global	Use
CONBLD	program start, prompts for input and assembles first three program lines in the conversion subroutine
Local	
00	determines the number of digits in the conversion factor
07	determines the length of and generates the text length
08	generates a text line
09	isolates, decodes and loads the text characters
10	assembles the next eight program lines
11	checks for decimal point and if factor is negative
12	assembles the conversion factor and the last two program lines and terminates the byte loading process.
13	locates and decodes each digit of the conversion factor
14	generates the exponent byte for the factor
15	sets up the decoding and assembly of the conversion factor

C. Flags

Type and Number	Use
User:	
08,09	used in the byte-loading subroutine in PPC ROM
21	cleared, to enable normal VIEW and AVIEW; set, if printer is attached (Flag 55 set) to enable printing if printer is ON.

C. Flags *(continued)*

Type and Number	Use
System:	
55	tested to determine if printer is attached

D. Size and key assignments

SIZE: $\geq$ 23

E. Program procedure and example

I. Load the program "CONBLD" into the calculator

II. Assume that we want to insert a subroutine, label F, in the program CONVRT to convert from pounds to kilograms and vice-versa. The multiplicative conversion factor in the "forward" direction is 0.4536 (pounds x 0.4536 = kilograms), and the six-character prompts and identifiers we choose are:

"LB->KG"
"KG->LB"
"LB·= "
"KG·= "

The procedure is as follows:

1. Go to the place in the CONVRT program that we want the subroutine inserted. A logical place is just before the END statement. (GTO CONVRT, then BST twice.)

2. In PRGM mode, insert the following steps:

```
LBL++
 +  )
 +  |
 +  |
 .  }   at least 62 plusses
 .  |
 .  |
 +  )
XEQ LB    (this will appear as XROMᵀLB)
```

3. In RUN mode, XEQ CONBLD and follow the prompts

E. Program procedure and example *(continued)*

	Prompt	Example Input	Key	Output
a. local label	LOCAL LBL?	F	R/S	-
b. first prompt	1ST PROMPT?	LB->KG	R/S	-
c. second prompt	2ND PROMPT?	KG->LB	R/S	-
d. first units	1ST UNITS?	LB.= _	R/S	-
e. second units	2ND UNITS?	KG.= _ space	R/S	-
f. conversion factor	FACTOR?	.4536	R/S	blank, then *WORKING*

g. after approximately 3 minutes, during which time tone 87 sounds as each instruction is loaded into program memory, proceed to the next step.

	Prompt	Input	Key	Output
h. program clean-up	SST, DEL oox	-	SST	x.xxx
		PRGM mode	XEQ DEL oox	-
	-	-	SST to XROM⊢LB and delete this and any ⊢'s that remain in program.	

i. Program CONVRT has now been augmented by the addition of Label F which will convert pounds to kilograms and vice-versa.

III. Other comments

1. In the program listing tone 87 prints as TONE 7 in steps 36, 41, 46, 53, 59, 115, and 120. In order to achieve tone 87, one can XEQ 1K after first keying in: 159↑87↑XX, where XX is the key location code where tone 87 is to be assigned.

2. In the generation of the conversion factor, SCI 3 (4 significant digits) is the most accurate possible due to PPC ROM program NC limitation; if greater precision is desired the factor can be replaced manually.

F. Program listing and example

```
01♦LBL "CONBLD"
02 AON
03 "LOCAL LBL ?"
04 PROMPT
05 ASTO 05
06 "1ST PROMPT ?"
07 PROMPT
08 ASTO 00
09 "2ND PROMPT ?"
10 PROMPT
11 ASTO 01
12 "1ST UNITS ?"
13 PROMPT
14 ASTO 02
15 "2ND UNITS ?"
16 PROMPT
17 ASTO 03
18 AOFF
19 "FACTOR ?"
20 PROMPT
21 STO 04
22 XROM "L-"
23 "  *WORKING*"
24 CF 21
25 AVIEW
26 FS? 55
27 SF 21
28 207
29 XROM "-B"
30 CLA
31 ARCL 05
32 XROM "CD"
33 37
34 +
35 XROM "-B"
36 TONE 7
37 169
38 XROM "-B"
39 22
40 XROM "-B"
41 TONE 7
42 156
43 XROM "-B"
44 133
45 XROM "-B"
46 TONE 7
47 CLA
48 .003
49 STO 20
50♦LBL 10
51 RCL 20
52 XEQ 08
53 TONE 7
54 154
55 XROM "-B"
56 RCL 20
57 INT
58 XROM "-B"
59 TONE 7
60 ISG 20
61 GTO 10
62♦LBL 15
63 CLA
64 SCI 3
65 ARCL 04
66 ASTO 21
67 ASHF
68 ASTO 22
69 11
70 STO 13
71♦LBL 00
72 DSE 13
73 CLA
74 ARCL 21
75 ARCL 22
76 RCL 13
77 XROM "NC"
78 XROM "CD"
79 X=0?
80 GTO 00
81♦LBL 13
82 RCL 13
83 CLA
84 ARCL 21
85 ARCL 22
86 XROM "NC"
87 XROM "CD"
88 48
89 X>Y?
90 GTO 11
91 RDN
92 69
93 X=Y?
94 GTO 14
95 RDN
96 32
97 -
98 GTO 12
99♦LBL 11
100 45
101 RCL Z
102 X≠Y?
103 GTO 11
104 28
105 GTO 12
106♦LBL 11
107 26
108 GTO 12
109♦LBL 14
110 27
111♦LBL 12
112 XROM "-B"
113 DSE 13
114 GTO 13
115 TONE 7
116 52
117 XROM "-B"
118 177
119 XROM "-B"
120 TONE 7
121 CF 09
122 XROM "-B"
123♦LBL 08
124 STO 18
125 7
126 STO 17
127♦LBL 07
128 DSE 17
129 CLA
130 ARCL IND 18
131 RCL 17
132 XROM "NC"
133 XROM "CD"
134 X=0?
135 GTO 07
136 RCL 17
137 240
138 +
139 XROM "-B"
140♦LBL 09
141 CLA
142 ARCL IND 18
143 RCL 17
144 XROM "NC"
145 XROM "CD"
146 XROM "-B"
147 DSE 17
148 GTO 09
149 END
```

F. Program listing and example *(continued)*

```
          GTO "CONVRT"

130♦LBL "++"
131 +
132 +
133 +
134 +
135 +
136 +
137 +
138 +
139 +
140 +
141 +
142 +
143 +
144 +
145 +
146 +
147 +
148 +
149 +
150 +
151 +
152 +
153 +
154 +
155 +
156 +
157 +
158 +
159 +
160 +
161 +
162 +
163 +
164 +
165 +
166 +
167 +
168 +
169 +
170 +
171 +
172 +
173 +
174 +
175 +
176 +
177 +
178 +
179 +
180 +
181 +
182 +
183 +
184 +
185 +
186 +
187 +
188 +
189 +
190 +
191 +
192 +
193 XROM "LB"
            XEQ "CONBLD"
LOCAL LBL ?
F                   RUN
1ST PROMPT ?
LB->KG              RUN
2ND PROMPT ?
KG->LB              RUN
1ST UNITS ?
LB.=                RUN
2ND UNITS ?
KG.=                RUN
FACTOR ?
            .4536   RUN

SST, DEL 006
                DEL 006
```

The completed Label F prepared by CONBLD

```
130♦LBL F
131 CF 22
132 FIX IND 05
133 "LB->KG"
134 ASTO 00
135 "KG->LB"
136 ASTO 01
137 "LB.= "
138 ASTO 02
139 "KG.= "
140 ASTO 03
141 4.536 E-1
142 STO 04
143 GTO 00
144 END
```

Program No. 41F029

Calculator: HP-41C/CV

Program Name: BPROB (Binomial Probability)

Author: T.W. Beers

Date: February, 1982

Purpose: To calculate individual and cumulative probabilities of obtaining k successes in n trials assuming a specified probability of success in one trial, p.

A. Storage assignments

Register	Use
00	p, the probability of success, 1 trial
01	n, the number of trials
02	k, the number of successes; changes throughout the program if the automatic incrementing feature is enabled (flag 01 set)
03	cumulative probabilities, i.e., $\sum P_k$; P_k = probability of obtaining k success

B. Labels

Global	Use
BPROB	program start; initial prompts and storage of input
Local	
A	prompt and storage of first k
01	calculation and display of individual probability, incrementing k if flag 01 set, and accumulation of the probabilities
02	display of the cumulative probability
03	printing of table heading
04	printing of table "line"
05	printing of table "bar"

C. Flags

Type and Number	Use
User:	
01	if set, the initial k is incremented by 1 and display proceeds automatically
12	to print double-wide characters
29	to supress decimal point in FIX 0
System:	
55	to detect printer attached

D. Size and key assignments

SIZE needed: $\geq$004

Key assign: BPROB on X≷Y
A on A (done internally)

E. Program procedure and example

I. Load the program into the calculator.

II. XEQ BPROB (X≷Y in USER mode) and follow the prompts. For the following steps, consider the case where a die is to be rolled 4 times (n = 4) and we want to know the probability of obtaining 2 ones, 3 ones, and 4 ones (k = 2, k = 3, k = 4) and the probability of obtaining at least 2 ones [P(2) + P(3) + P(4)]. Note that the probability of obtaining a "one" in one trial (a roll) is p = 1/6 = .1667.

1. Assume flag 01 clear, FIX 3

Prompt	Input	*Example* Key	Output
-	-	X≷Y	-
KEY P, R/S	.1667	R/S	-
KEY N, R/S	4	R/S	-
1ST K?, R/S	2	R/S	P(2) = 0.116
	-	R/S	P(3) = 0.015
	-	R/S	P(4) = 0.001
	-	R/S	CUM P = 0.132

For the same N and p, press A and the prompt is 1st K? R/S

For new N and/or p, XEQ BPROB or press X≷Y in USER mode.

E. Program procedure and example *(continued)*

2. If flag 01 is set, R/S need not be pressed between answers.

F. Formulas used.

1. Basic binomial distribution formula

$$P(k) = \binom{n}{k} p^{k} (1-p)^{(n-k)}$$

where:

$P(k)$ = the probability of obtaining k "successes" in n trials

p = the probability of a "success" in 1 trial

n = number of trials

$\binom{n}{k}$ = the combination of n items taken k at a time

$$= \frac{n!}{k!\,(n-k)!}$$

G. General comments regarding the program.

The PPC ROM was used to calculate $\binom{n}{k}$, and to generate the parentheses in the display. If the ROM is not available, one should change the program as follows:

replace steps 39 with "P<"

replace step 41 with "├>= "

replace steps 20-24 with

```
XEQ FACT
RCL 02
XEQ FACT
   ÷
RCL 01
RCL 02
   -
XEQ FACT
   ÷
```

H. Program listing and example.


```
01♦LBL "BPROB"
02 FIX 3
03 CF 29
04 "KEY P, R/S"
05 PROMPT
06 STO 00
07 "KEY N, R/S"
08 PROMPT
09 STO 01
10 FS? 55
11 XEQ 03
12♦LBL A
13 0
14 STO 03
15 "1ST K?, R/S"
16 PROMPT
17 STO 02
18♦LBL 01
19 RCL 01
20 X<>Y
21 X≠0?
22 XROM "CM"
23 X=0?
24 1
25 RCL 00
26 RCL 02
27 Y↑X
28 *
29 1
30 RCL 00
31 -
32 RCL 01
33 RCL 02
34 -
35 Y↑X
36 *
37 ST+ 03
38 FIX 0
39 "P("
40 ARCL 02
41 "⊢)= "
42 FIX 3
43 ARCL X
44 AVIEW
45 FS? 01
46 PSE
47 FC? 01
48 STOP
49 RCL 02
50 RCL 01
51 X=Y?
52 GTO 02
53 1
54 ST+ 02
55 RCL 02
56 GTO 01
57♦LBL 02
58 ADV
59 "CUM. P= "
60 ARCL 03
61 AVIEW
62 FS? 55
63 XEQ 05
64 RTN
65♦LBL 03
66 XEQ 05
67 CLA
68 "BINOMIAL "
69 ACA
70 "DISTRIBUTION"
71 ACA
72 PRBUF
73 XEQ 04
74 "PROB. OF "
75 ACA
76 "SUCCESS= "
77 ACA
78 RCL 00
79 ACX
80 ADV
81 FIX 0
82 "NO. TRIALS= "
83 ACA
84 RCL 01
85 ACX
86 ADV
87 XEQ 04
88 FIX 3
89 RTN
90♦LBL 04
91 SF 12
92 "------------"
93 PRA
94 CF 12
95 RTN
96♦LBL 05
97 SF 12
98 "************"
99 PRA
100 ADV
101 CF 12
102 .END.
```

```
*************

BINOMIAL DISTRIBUTION
------------
PROB. OF SUCCESS=  0.167
        NO. TRIALS=  4
------------
P(0)= 0.482
P(1)= 0.386
P(2)= 0.116
P(3)= 0.015
P(4)= 0.001

CUM. P= 1.000
*************


P(2)= 0.116
P(3)= 0.015
P(4)= 0.001

CUM. P= 0.132
*************


*************

BINOMIAL DISTRIBUTION
------------
PROB. OF SUCCESS=  0.167
        NO. TRIALS=  4
------------
P(2)= 0.116
P(3)= 0.015
P(4)= 0.001

CUM. P= 0.132
*************
```

Retrospective Comments Regarding BPROB

1. The printed output from BPROB can be cleaned up slightly by the following:

    a. delete step 100 (ADV)

    b. insert 5 advances (ADV) after step 63 (XEQ 05)

2. The above changes were made in the BPROB program which is on the KRON-1 tape.

Program No. 41F030

Calculator: HP41C/CV

Program Name: SCHNUR

Author: T.W. Beers

Date: 1982

Purpose: To calculate tree height or site index, given the other, after age has been specified. Based on Wiant's (Jour. Forestry, 1975, Vol. 73 page 429) prediction equation for G. Luther Schnur's (1937) classic site index curves for upland oaks.

A. Storage assignments:

Register	Use
00	not used
01-06	constants in the prediction equation
07	average tree age
08	site index
09	average tree height

B. Labels

Global	Use
SCHNUR	program start, generates constants
Local	
A	prompts for and stores age
B	converts height or site index in feet to meters
00	provides the prompt for height or site index
01	directs solution for site index (flag 20 set) or height (flag 20 clear)
02	views the input prompt, stops for data input, and provides the "toggle" to change the input prompt in the absence of numeric data key depression

B. Labels *(continued)*

Local	Use
03	solves the prediction equation for height in feet
04	solves the prediction equation for site index in feet

C. Flags

Type and Number	Use
User:	
20	set in label 02 to indicate height is keyed-in and solution is for site index; clear indicates the reverse
22	numeric data input flag, tested in label 02 to provide the alternative input prompt in the absence of a numeric key stroke prior to R/S
29	cleared, to eliminate decimal in FIX 0
System:	none tested

D. Size and key assignments

SIZE $\geq$ 010

Suggested key assignments:

"SCHNUR" on TAN

A on Σ+ (done internally)
B on 1/x (done internally)

E. Program procedure and example

I. Load the program into the calculator

II. XEQ SCHNUR (TAN in USER mode) and follow the prompts. Consider the following examples:

(a) Age: 60, SI= 40 , HT= ?
SI= 50 , HT= ?
SI= 60 , HT= ? , HT in meters= ?

(b) Age: 90, HT= 70 , SI= ?
HT= 100, SI= ? , SI in meters= ?

E. Program procedure and example *(continued)*

			Example	
Step	Prompt	Input	Key	Output
1. Initialize	-	-	TAN	*SCHNUR*
2. Example a	AGE ?	60	R/S	-
	KEY SI, R/S	40	R/S	HT.= 45 FT.
	-	50	R/S	HT.= 55 FT.
	-	60	R/S	HT.= 66 FT.
	-	-	B	= 20 METERS
3. Example b	-	-	A	-
	AGE ?	90	R/S	-
	KEY SI, R/S	-	R/S	-
	KEY HT., R/S	70	R/S	S.I.= 54 (FT.)
	-	100	R/S	S.I.= 78 (FT.)
	-	-	B	= 24 METERS

4. For more data in the same age class, press R/S and observe the prompt, if it is not the right one, press R/S again.

5. For a new age class, press A to obtain the AGE? prompt.

F. Program listing and example

```
01♦LBL "SCHNUR"
02 "  *SCHNUR*"
03 AVIEW
04 FIX 0
05 CF 29
06 CF 22
07 62.7
08 STO 01
09 8.37
10 STO 02
11 81.63249
12 STO 03
13 .00786
14 STO 04
15 4.09382
16 STO 05
17 4.40767
18 STO 06
19♦LBL A
20 "AGE ?"
21 PROMPT
22 CF 22
23 STO 07
24♦LBL 00
25 FS? 20
26 "KEY HT.,R/S"
27 FC? 20
28 "KEY SI, R/S"
29♦LBL 02
30 TONE 7
31 PROMPT
32 FS?C 22
33 GTO 01
34 FC?C 20
35 SF 20
36 GTO 00
37♦LBL 01
38 FS? 20
39 GTO 04
40♦LBL 03
41 STO 08
42 RCL 07
43 .29
44 Y↑X
45 RCL 05
46 *
47 RCL 06
48 -
49 RCL 08
50 RCL 01
51 -
52 *
53 RCL 02
54 /
55 RCL 03
56 100
57 RCL 07
58 -
59 X↑2
60 RCL 04
61 *
62 -
63 +
64 STO 09
65 "HT.= "
66 ARCL X
67 "⊢ FT."
68 GTO 02
69♦LBL B
70 .3048
71 *
72 " = "
73 ARCL X
74 "⊢ METERS"
75 PROMPT
76 GTO 00
77♦LBL 04
78 STO 09
79 100
80 RCL 07
81 -
82 X↑2
83 RCL 04
84 *
85 CHS
86 RCL 03
87 +
88 CHS
89 RCL 09
90 +
91 RCL 07
92 .29
93 Y↑X
94 RCL 05
95 *
96 RCL 06
97 -
98 /
99 RCL 02
100 *
101 RCL 01
102 +
103 STO 08
104 "S.I.= "
105 ARCL X
106 "⊢ (FT.)"
107 GTO 02
108 END
```

Examples (a) and (b) with printer in NORM mode

```
                XEQ "SCHNUR"
  *SCHNUR*
AGE ?
                    60    RUN
KEY SI, R/S
                    40    RUN
HT.= 45 FT.
                    50    RUN
HT.= 55 FT.
                    60    RUN
HT.= 66 FT.
                         XEQ B
 = 20 METERS


                         XEQ A
AGE ?
                    90    RUN
KEY SI, R/S
                          RUN
KEY HT.,R/S
                    70    RUN
S.I.= 54 (FT.)
                   100    RUN
S.I.= 78 (FT.)
                         XEQ B
 = 24 METERS
```

G. Formulas used

The formula used to fit Schnur's site index curves (in USDA Technical Bull. No. 560, 1937) was that derived by H.V. Wiant, Jr. (Jour. Forestry 73: 429):

$$SI = 62.7 + 8.37\left(\frac{H - [81.63249 - .00786(100 - A)^2]}{4.09382A^{.29} - 4.40767}\right)$$

where:

SI= site index
H= average tree total height
A= average tree age

The above formula was also solved for H and used in the part of the program which prompts the user for site index.

Schnur's site index curves for upland oaks formulated

Harry V. Wiant, Jr.

THE AUTHOR—Harry V. Wiant, Jr., is professor, Division of Forestry, West Virginia University, Morgantown. (Published with the approval of the Director of the West Virginia University Agricultural Experiment Station as Scientific Paper No. 1352.)

ABSTRACT—An equation for Schnur's site index curves for oaks is presented, facilitating computerized computations. The average absolute difference between formula and table values was 0.5 foot and the maximum difference was 1.4 feet.

Researchers (*2,3*) have formulated site index curves previously published for some forest species to eliminate the need for reading and interpolating from sets of curves and to facilitate computer processing. Complicated growth functions requiring sophisticated computer programs were used in developing these formulas, and maximum errors—the maximum difference between observed and predicted heights over the range of data points analyzed—have ranged between 0.6 and 4.6 feet. Schnur's (*4*) site index curves for upland oaks have been widely used for many years in the Appalachians, and there is a need for a formulated approach.

Procedure

Schnur presents a formula for ascertaining site index (base age, 50 years) as follows:

$$\text{Site index} \quad 62.7 + 8.37\left(\frac{H - Ha}{Sa}\right)$$

where H = average total height (ft.) of dominant and codominant oaks, with an age of "a" (yrs.).

Ha = average height of oaks from Schnur's data for trees of age "a."

Sa = standard deviation of height about the average at age "a."

The "matchacurve system" (*1*) was used to develop predictions of Ha and Sa using data presented by Schnur.

Results

The prediction equation for Ha, with r = 0.999, was:

$$Ha = 81.63249 - 0.00786(100 - A)^2$$

Sa was related to age, with r = 0.998, as follows:

$$Sa = 4.09382A^{.29} - 4.40767$$

Substituting these relations into Schnur's equation, we have:

$$\text{Site index} = 62.7 + 8.37\left(\frac{H - [81.63249-0.00786(100-A)^2]}{4.09382A^{.29} - 4.40767}\right)$$

Prediction of the 70 points Schnur used to plot his site index curves showed an average absolute difference of 0.5 foot and a maximum error of 1.4 feet. Considering the inaccuracies inherent in reading graphs and in height measurements as usually made in the field, the error in the formulated approach is negligible. ■

Literature Cited

1. JENSEN, C. E., and J. W. HOMEYER. 1971. Matchacurve-2 for algebraic transforms to describe curves of the class X^n. USDA Forest Serv. Res. Paper INT-106.
2. LUNDGREN, A. L., and W. A. DOLID. 1970. Biological growth functions describe published site index curves for Lake States timber species. USDA Forest Serv. Res. Paper NC-36.
3. PAYANDEH, B. 1974. Formulated site index curves for major timber species in Ontario. Forest Sci. 20:143-144.
4. SCHNUR, G. L. 1937. Yield, stand, and volume tables for even-aged upland oak forests. USDA Tech. Bul. 560.

Retrospective Comments Regarding SCHNUR

1. A frequently occurring problem is encountered in this program — that is, program stoppage when an AVIEW is executed and the printer is OFF! If one wants to have the flexibility of running the program with the printer attached and either ON or OFF, the AVIEW-encountered stoppage (if printer OFF) can be avoided by the following program changes:

   a. insert CF 21 just before the AVIEW (step 03 in SCHNUR)

   b. insert FS? 55
      SF 21 just after the AVIEW

   Thus, clearing flag 21 disables printing and the calculator acts as if the printer is not attached and therefore the AVIEW does not cause a stoppage with the printer OFF; setting flag 21 in the presence of flag 55 set (55 is set automatically when printer is connected), restores the condition required to print, if the printer is then turned on.

   The effect on a running program when a VIEW or AVIEW is encountered, is summarized below:

		Flag 55, printer existence flag	
		SET (printer attached)	CLEAR (printer disconnected)
Flag 21, print enable flag	SET:	printer ON -- prints, no stoppage printer OFF — stoppage, no print	stoppage, no print
	CLEAR:	no print, no stoppage	no print, no stoppage

2. In the above example, the printing of the AVIEW "message" was not necessary; if one wants to ensure that an AVIEW stoppage will not occur and that the AVIEW message will be printed, a routine developed by Roger Hill for the PPC ROM module is useful. Instead of AVIEW, one would use XEQ VA in the program, assuming the PPC ROM is inserted, or that the following subroutine is present in the calculator:

```
LBL VA
SF 25
PRA
SF 25
FS?C 21
CF 25
AVIEW
FC?C 25
SF 21
RTN
```

Calculator: HP-41C/CV

Program Name: TVOLSA (Tree Volume and Surface Area)

Author: T.W. Beers

Date: 1982

Purpose: To calculate tree volume in cubic feet and lateral surface area in square feet, assuming the following:

1. Input is [DBH ↑ upper diameter ↑ merchantable length] for the main stem, and [lower diameter ↑ upper diameter ↑ length] for additional sections. All diameters are consistently outside (or inside) bark in inches and merchantable length is the distance in feet from stump height (here assumed to be .5 foot) to the upper diameter.

2. The section between breast height and stump height is a 4-foot cylinder.

3. The section above breast height is a frustum of a cone and volume is determined by Newton's formula.

A. Storage assignments

Register	Use
0	accumulated tree volume
1	R_ℓ, radius at breast height (or lower diameter point) in feet
2	R_u, radius at upper diameter point in feet
3	L', section length in feet
4	R_m, mid-section radius in feet = $\frac{R_\ell + R_u}{2}$
5	accumulated surface area
6	individual section volume
7	individual section surface area
8	sum of volumes for all trees keyed in
9	sum of surface areas for all trees keyed in

B. Labels

Global	Use
TVOLSA	program start, clearing of certain registers

B. Labels *(continued)*

Local	Use
A	displays the volume and surface area (with flag 01 set) for the tree and adds these values to the grand totals
a	displays the grand total volume and surface area (with flag 01 set)
B	provides a repeat of LBL A output, without affecting the grand totals
00	initial prompt involving DBH, upper end diameter, d, and first section length, L
02	calculation of main stem volume from stump height of .5 foot to main stem upper diameter
03	calculation of main stem surface area above breast height
04	primary routine for the calculation and display of both section volume and surface area; also provides the prompt for the sections above the one that includes DBH, D↑d↑L', D = lower diameter, d = upper diameter, L' = section length
05	subroutine to display section surface area
06	subroutine to calculate surface area below breast height
07	to provide program re-entry for the second and subsequent trees in a data set

C. Flags

Type and Number	Use
User:	
00	set internally to signal calculation for a section not including the main stem below breast height
01	when set (externally) causes surface area to be displayed in addition to volume
22	tested to provide branch to label A in the absence of data input for an "upper section"
System:	
	none tested

D. Size and key assignments

SIZE: $\geq$010

Suggested key assignments:

"TVOLSA"	on	shift TAN
A	on	A (done internally)
a	on	shift A (done internally)
B	on	B (done internally)

E. Program procedure and example

I. Load the program into the calculator

II. XEQ TVOLSA (shift TAN in USER mode) and follow the prompts. As an example, consider the following trees, for which we want to calculate cubic-foot volume and surface area (set flag 01) in square feet.



Tree 1:	Tree 2:
main stem:	main stem:
DBH = 20"	DBH = 16"
d = 12"	d = 13"
L = 20'	L = 8'
upper sections:	upper sections:
D = 10"	D = 13"
d = 6"	d = 11"
L' = 8'	L' = 14'
D = 8"	
d = 4"	
L' = 10'	

E. Program procedure and example *(continued)*

Step	Prompt	Input	*Example* Key	Output
1. Initialize	-	-	SF 01	-
	-	-	shift TAN	-
2. Data input	DBH↑d↑L, R/S	20↑12↑20	R/S	VOL. = 31.5 CF S.A. = 88.0 SF
	D↑d↑L', R/S	10↑6↑8	R/S	VOL. = 2.9 CF S.A. = 16.8 SF
	D↑d↑L', R/S	8↑4↑10	R/S	VOL. = 2.0 CF S.A. = 15.7 SF
3. Tree 1 total	D↑d↑L', R/S (press B for another look)	-	A	ΣVOL = 36.4 CF ΣSA = 120.4 SF
4. Tree 2 input	-	-	R/S	-
	DBH↑d↑L, R/S	16↑13↑8	R/S	VOL. = 10.2 CF S.A. = 31.9 SF
	D↑d↑L', R/S	13↑11↑14	R/S	VOL. = 11.0 CF S.A. = 44.0 SF
5. Tree 2 total	D↑d↑L', R/S (press B for another look)	-	A	ΣVOL = 21.2 CF ΣSA = 75.9 SF
6. Grand totals	-	-	a	TOT V = 57.6 TOT SA = 196.3

F. Formulas

1. Volume-- all radii and lengths are assumed in feet

a. main stem from stump to breast height:

$$V = \pi R^2 L, \text{ where } L = 4$$

$$\text{and } R = \frac{DBH}{2}$$

b. main stem above breast height; Newton's formula is used:

$$V = \frac{L'}{6} \pi (R_\ell^2 + 4R_m^2 + R_u^2)$$

where L' = merchantable length - 4
R_ℓ = radius at breast height
R_u = radius at merchantable top
R_m = radius at mid-section
$= (R_\ell + R_u)/2$

F. Formulas *(continued)*

c. upper sections; Newton's formula is again used, but:

L' = section length

R_ℓ, R_u, and R_m = lower, upper, and mid-section radii

2. Surface area-- all radii and lengths are assumed in feet

a. main stem from stump to breast height:

$$S = 2\Pi RL$$

$$= 8\Pi R, \quad \text{where} \quad L = 4$$

$$\text{and} \quad R = \frac{DBH}{2}$$

b. main stem above breast height; upper and lower circumferences are "unrolled" to form a trapezoid, leading to:

$$S = (2\Pi R_\ell + 2\Pi R_u)\frac{L'}{2}$$

$$= \Pi(R_\ell + R_u)L'$$

c. upper sections are handled also as trapezoids, using the same altered definitions as described for volume determination.

G. Program listing and example

```
01♦LBL "TVOLSA"
02 0
03 STO 08
04 STO 09
05♦LBL 07
06 CF 00
07 ΣREG 00
08 CLΣ
09♦LBL 00
10 TONE 9
11 "DBH↑d↑L, R/S"
12 PROMPT
13 4
14 -
15♦LBL 02
16 STO 03
17 RDN
18 24
19 /
20 STO 02
21 RDN
22 24
23 /
24 STO 01
25 FS? 00
26 GTO 04
27 X↑2
28 PI
29 *
30 4
31 *
32 STO 00
33♦LBL 04
34 RCL 01
35 RCL 02
36 +
37 2
38 /
39 STO 04
40 X↑2
41 4
42 *
43 RCL 01
44 X↑2
45 +
46 RCL 02
47 X↑2
48 +
49 PI
50 *
51 6
52 /
53 RCL 03
54 *
55 STO 06
56 ST+ 00
57 XEQ 03
58 FC? 00
59 XEQ 06
60 FIX 1
61 "VOL.= "
62 FS? 00
63 ARCL 06
64 FC? 00
65 ARCL 00
66 "⊦ CF"
67 AVIEW
68 PSE
69 FS? 01
70 XEQ 05
71 CF 22
72 TONE 7
73 TONE 7
74 "D↑d↑L', R/S"
75 PROMPT
76 FC? 22
77 GTO A
78 SF 00
79 GTO 02
80♦LBL 03
81 RCL 01
82 RCL 02
83 +
84 PI
85 *
86 RCL 03
87 *
88 STO 07
89 ST+ 05
90 RTN
91♦LBL 05
92 "S.A.= "
93 FS? 00
94 ARCL 07
95 FC? 00
96 ARCL 05
97 "⊦ SF"
98 AVIEW
99 PSE
100 RTN
101♦LBL 06
102 RCL 01
103 PI
104 *
105 8
106 *
107 ST+ 05
108 ST+ 07
109 RTN
110♦LBL A
111 RCL 00
112 ST+ 08
113 RCL 05
114 ST+ 09
115♦LBL B
116 "ΣVOL= "
117 ARCL 00
118 "⊦ CF"
119 AVIEW
120 FC? 01
121 STOP
122 FC? 01
123 GTO 07
124 FS? 01
125 PSE
126 "ΣSA= "
127 ARCL 05
128 "⊦ SF"
129 AVIEW
130 STOP
131 GTO 07
132♦LBL a
133 "TOT V= "
134 ARCL 08
135 AVIEW
136 FC? 01
137 STOP
138 FS? 01
139 PSE
140 "TOT SA= "
141 ARCL 09
142 AVIEW
143 STOP
144 GTO 07
145 END
```

G. Program listing and example *(continued)*

Printer in NORM. mode

with flag 01 set

```
                SF 01
           XEQ "TVOLSA"
DBH↑d↑L, R/S
           20.0 ENTER↑
           12.0 ENTER↑
           20.0    RUN
VOL.= 31.5 CF
S.A.= 88.0 SF
D↑d↑L', R/S
           10.0 ENTER↑
            6.0 ENTER↑
            8.0    RUN
VOL.= 2.9 CF
S.A.= 16.8 SF
D↑d↑L', R/S
            8.0 ENTER↑
            4.0 ENTER↑
           10.0    RUN
VOL.= 2.0 CF
S.A.= 15.7 SF
D↑d↑L', R/S
                 XEQ A
ΣVOL= 36.4 CF
ΣSA= 120.4 SF
                   RUN
DBH↑d↑L, R/S
           16.0 ENTER↑
           13.0 ENTER↑
            8.0    RUN
VOL.= 10.2 CF
S.A.= 31.9 SF
D↑d↑L', R/S
           13.0 ENTER↑
           11.0 ENTER↑
           14.0    RUN
VOL.= 11.0 CF
S.A.= 44.0 SF
D↑d↑L', R/S
                 XEQ A
ΣVOL= 21.2 CF
ΣSA= 75.9 SF
                 XEQ a
TOT V= 57.6
TOT SA= 196.3
```

with flag 01 clear

```
           XEQ "TVOLSA"
DBH↑d↑L, R/S
             20 ENTER↑
             12 ENTER↑
             20    RUN
VOL.= 31.5 CF
D↑d↑L', R/S
           10.0 ENTER↑
            6.0 ENTER↑
            8.0    RUN
VOL.= 2.9 CF
D↑d↑L', R/S
            8.0 ENTER↑
            4.0 ENTER↑
           10.0    RUN
VOL.= 2.0 CF
D↑d↑L', R/S
                 XEQ A
ΣVOL= 36.4 CF
                   RUN
DBH↑d↑L, R/S
           16.0 ENTER↑
           13.0 ENTER↑
            8.0    RUN
VOL.= 10.2 CF
D↑d↑L', R/S
           13.0 ENTER↑
           11.0 ENTER↑
           14.0    RUN
VOL.= 11.0 CF
D↑d↑L', R/S
                 XEQ A
ΣVOL= 21.2 CF
                 XEQ a
TOT V= 57.6
```

Program No. 41F032

Calculator: HP-41CV

Program Name: QFACTOR

Author: T.W. Beers

Date: March, 1982

Purpose: To provide a means of fitting data of the form Y = number of trees and X = diameter class to the negative exponential model, $Y = ke^{-aX}$, using the linear least squares approximation $\log Y = b_0 + b_1 X$. Estimates of a, k, b_0, and b_1 are displayed, as is the commonly used index, q, which reflects the ratio of trees in successive diameter classes. Provision is made to obtain predicted Y (i.e., $\hat{Y}$) for any X and to obtain q for class widths other than that assumed initially.

A. Storage assignments

Register	Use
00	D_i, current diameter, i.e., X_i
01	$\overline{X}$, arithmetic mean of X
02	Σx^2, $\Sigma(X_i-\overline{X})^2$
03	$\overline{Y}'$, arithmetic mean Y'; Y' = log Y
04	$\Sigma y'^2$, $\Sigma(Y_i'-\overline{Y}')^2$
05	$\Sigma xy'$, $\Sigma(X_i-\overline{X})(Y_i'-\overline{Y}')$
06	b_0, the Y' intercept (a base 10 logarithm)
07	b_1, the slope (change in Y' per unit change in X)
08	r, the simple correlation coefficient between Y' and X
09	a, the "rate-of-change coefficient"
10	k, the "density coefficient"
11	ΣX, sum of X
12	ΣX^2, sum of squared Xs
13	$\Sigma Y'$, sum of log Y

A. Storage assignments *(continued)*

Register	Use
14	$\Sigma Y'^2$, sum of squared log Ys
15	$\Sigma XY'$, sum of products of X and Y'
16	n, number of sets of data
17	h, basic class width for which q is desired
18	q_h, q for class width h
19	$q_{h'}$, q for class width h'
20	h', keyed-in alternative class width

B. Labels

Global	Use
QFACTOR	program start, register clearing and identification of variables
LRS	simple linear regression calculation subroutine.
Local	
A	basic calculations, automatic display of a, k, and q(h); after R/S, prompt for h' and calculation and display of q(h')
B	prompts and calculation of $\hat{Y}$, predicted number of trees
00	basic summation loop in QFACTOR
01	input error correction loop, activated by a negative value for Y
02	calculation and display loop in B
03	in A, to enable return to the basic q(h) by R/S after q(h') calculation

C. Flags -- flag 29 cleared to suppress decimal in FIX 0

D. Size and key assignments

SIZE needed: $\geq$021

Key assign:			
QFACTOR	on	shift	X≷Y
A	on	A	(done internally)
B	on	B	(done internally)

E. Program procedure and example

I. Load the program into the calculator

II. XEQ QFACTOR (shift X≷Y in USER mode) and follow the prompts and reminders. As an example, consider the following data for which we want to calculate a "q" factor assuming a class width of 2 inches, and we also want q based on 1 inch classes.

Selected two-inch DBH class	Per-acre basis: Observed no. trees	Per-acre basis: Predicted no. trees
X	Y	$\hat{Y}$
10	13.79	?
20	3.97	?
30	1.12	?

Step	Prompt	Input	Example Key	Output
1. Data input	-	-	Shift X≷Y	-
	CLASS WIDTH?	2	R/S	X=D. CLASS Y=NO. TREES
	X↑Y, R/S	10↑13.79	R/S	X DONE: 10
	X↑Y, R/S	20↑3.97	R/S	X DONE: 20
	X↑Y, R/S	30↑1.12	R/S	X DONE: 30
	X↑Y, R/S			
2. Calculation	X↑Y, R/S	-	A	b(0)=1.68620 b(1)=-0.05452 *WORKING* a=0.126 K=48.552 Q(2)=1.29
3. Alternative q	-	-	R/S	-
	FOR Q(H) KEY "H", R/S	1	R/S	Q(1)=1.13
4. Predicted Ys	-	-	B	-
	FOR NO. TREES, KEY D, R/S	10	R/S	Y(10)=13.84
		20	R/S	Y(20)=3.94
		30	R/S	Y(30)=1.12

5. Repeat -- for a repeat of step 2 output and as a lead-in to step 3, press A again

-- for more predicted Ys, press B

E. Program procedure and example *(continued)*

6. Error correction -- if in the input of data, a faulty value has been processed, at the X↑Y, R/S prompt one can delete the faulty data by again keying the faulty pair and changing the sign (CHS key) of the Y (actually in the X register at this point) and pressing R/S. The correct pair can now be keyed and processed as indicated in step 1.

F. Formulas

1. Linear regression.

The usual simple linear regression formulas using the least squares approach were employed to obtain b_0 and b_1 in

$Y' = b_0 + b_1 X$, where

$Y' = \log Y = \log$ (no. of trees)
X = diameter class

Note that base 10 logarithms were used.

2. Non-linear form.

The values of the coefficients a and k in the basic form of the fitted equation,

$Y = ke^{-aX}$, where

Y = no. of trees
X = diameter class
e = base of natural logarithms

were calculated by

$$a = \frac{-b_1}{\log e} = \frac{-b_1}{0.43429}$$

$$k = \text{antilog } (b_0) = 10^{b_0}$$

Note that if natural logs had been used the formulas would have been different; see FOREST MENSURATION (Husch, Miller, and Beers, 1982, John Wiley & Sons.)

3. q values

a. for the basic class width, h,

$$q = e^{ha}$$

F. Formulas *(continued)*

b. for the alternative class width, h',

$$q_{h'} = 10^{-h'b_1}$$

c. another convenient formula relating q_h and $q_{h'}$ is

$$q_h = (q_{h'})^{h/h'}, \quad \text{or the reverse}$$

$$q_{h'} = (q_h)^{h'/h}$$

4. Predicted number of trees

Predicted values for Y were obtained by solving the basic model for Y given the class midpoint, X:

$$\hat{Y}_i = k'e^{-aX_i}$$

where

$\hat{Y}_i$ = value on the fitted curve

X_i = i<u>th</u> diameter class midpoint

k' = the density coefficient appropriate, as used here, for per acre data and <u>two-inch</u> classes

a = the rate-of-change coefficient as defined earlier; independent of area and class width

Note that if the class width is not 2, and the number of trees in a class of width h is desired, the right side of the equation must be multiplied by h/2. Refer to FOREST MENSURATION for more details.

G. Program listing and example


```
01♦LBL "QFACTOR"
02 CF 29
03 FIX 0
04 ΣREG 11
05 CLΣ
06 "CLASS WIDTH?"
07 PROMPT
08 STO 17
09 "X= D. CLASS"
10 AVIEW
11 SIN
12 SIN
13 "Y= NO. TREES"
14 AVIEW
15 SIN
16 SIN
17♦LBL 00
18 " X↑Y, R/S"
19 PROMPT
20 X<0?
21 GTO 01
22 LOG
23 X<>Y
24 STO 00
25 Σ+
26 RCL 00
27 "X DONE: "
28 ARCL X
29 AVIEW
30 PSE
31 GTO 00
32♦LBL A
33 FIX 5
34 XEQ "LRS"
35 PSE
36 FIX 3
37 "*WORKING*"
38 AVIEW
39 RCL 07
40 CHS
41 1
42 E↑X
43 LOG
44 /
45 STO 09
46 RCL 06
47 10↑X
48 STO 10
49 RCL 09
50 RCL 17
51 *
52 E↑X
53 STO 18
54 "a= "
55 ARCL 09
56 AVIEW
57 PSE
58 "K= "
59 ARCL 10
60 AVIEW
61 PSE
62♦LBL 03
63 FIX 0
64 "Q("
65 ARCL 17
66 "⊢)= "
67 FIX 2
68 ARCL 18
69 AVIEW
70 STOP
71 "FOR Q(H)"
72 AVIEW
73 PSE
74 "KEY "H", R/S"
75 AVIEW
76 STOP
77 STO 20
78 RCL 07
79 CHS
80 *
81 10↑X
82 STO 19
83 FIX 0
84 "Q("
85 ARCL 20
86 "⊢)= "
87 FIX 2
88 ARCL 19
89 AVIEW
90 STOP
91 GTO 03
92♦LBL 01
93 ABS
94 LOG
95 X<>Y
96 Σ-
97 GTO 00
98♦LBL B
99 "FOR NO. TREES,"
100 AVIEW
101 SIN
102 SIN
103 "KEY D, R/S"
104 PROMPT
105♦LBL 02
106 STO 00
107 RCL 09
108 *
109 CHS
110 E↑X
111 RCL 10
112 *
113 FIX 0
114 "Y("
115 ARCL 00
116 "⊢)= "
117 FIX 2
118 ARCL X
119 AVIEW
120 STOP
121 GTO 02
122♦LBL "LRS"
123 MEAN
124 STO 01
125 X↑2
126 STO 02
127 X<>Y
128 STO 03
129 X↑2
130 STO 04
131 RCL 01
132 RCL 03
133 *
134 STO 05
135 RCL 16
136 CHS
137 ST* 02
138 ST* 04
139 ST* 05
140 RCL 12
141 ST+ 02
142 RCL 14
143 ST+ 04
144 RCL 15
145 ST+ 05
146 RCL 05
147 RCL 02
148 /
149 STO 07
150 RCL 01
```

G. Program listing and example *(continued)*


```
151 *
152 CHS
153 RCL 03
154 +
155 STO 06
156 "b(0)= "
157 ARCL X
158 AVIEW
159 PSE
160 "b(1)= "
161 ARCL 07
162 AVIEW
163 RTN
164 END
```

```
                 XEQ "QFACTOR"
CLASS WIDTH?
                      2    RUN
X= D. CLASS
Y= NO. TREES
 X↑Y, R/S
                 10 ENTER↑
              13.79    RUN
X DONE: 10
 X↑Y, R/S
                 20 ENTER↑
               3.97    RUN
X DONE: 20
 X↑Y, R/S
                 30 ENTER↑
               1.12    RUN
X DONE: 30
 X↑Y, R/S
                      XEQ A
b(0)= 1.68620
b(1)= -0.05452
*WORKING*
a= 0.126
K= 48.552
Q(2)= 1.29
                        RUN
FOR Q(H)
KEY "H", R/S
               1.00    RUN
Q(1)= 1.13
                      XEQ B
FOR NO. TREES,
KEY D, R/S
              10.00    RUN
Y(10)= 13.84
              20.00    RUN
Y(20)= 3.94
              30.00    RUN
Y(30)= 1.12
```

Retrospective Comments Regarding QFACTOR

1. The QFACTOR program provides a useful means of fitting the negative exponential model to data commonly encountered in forest management, and for calculating the q-value. However, careful reading of FOREST MENSURATION (Husch, et al. 1982) will show that while the a coefficient is independent of land area and diameter class width, and q depends on diameter class width, the k coefficient depends upon both the land area on which the numbers of trees are based, and diameter class width.

2. As a result of comment 1, future versions of QFACTOR should prompt for area (per-acre would be A=1) as well as class width. Thus, one could designate for display and printing:

   k, and q, without subscripts, to indicate per-acre data by one-inch classes,

   q(h) to indicate q appropriate for h-inch classes,

   and k'(A,h) to imply the k coefficient appropriate for number of trees on an area of A acres, by h-inch diameter classes.

   In programming, then, one would fit the regression of number of trees per area, A, by diameter classes of width, h, and calculate

$$a = \frac{-b_1}{\log e}$$

$$k'_{A,h} = 10^{b_o}$$

and

$$q_h = e^{ha}$$

   Then, the basic per acre, one-inch class coefficients (unsubscripted a, k, and q) are calculated from:

$$a = a \quad \text{(i.e., no change)}$$

$$q = (q_h)^{1/h}$$

and

$$k = \frac{k'_{A,h}}{hA}$$

3. One should also be careful in estimating numbers of trees (i.e., predicted Ys). That is, for per-acre estimates and one-inch classes, number of trees in the i^{th} class, n_i, is

$$n_i = ke^{-aX_i}$$

But, for class width, h, on an area, A, number of trees in the i^{th} class, N_i, is

$$N_i = hAn_i = hAke^{-aX_i}$$

4. The foregoing comments and the pattern of previous "Retrospective Comments" should lead the reader to expect that a revised program to accomplish the refinements can be found on the KRON-1 tape. However, such major changes deserve a new write-up and, alas!, the author finally recognizes that it is time to bring this nearly three-year project to a close. Writing the QFAC1 program must wait for another day.

```
           +
           rr
          rrrr
         rrrrrr
        rrrrrrrr
       rrrrrrrrrr
      rrrrrrrrrrrr
     rrrrrrrrrrrrrr
    rrrrrrrrrrrrrrrr
   rrrrrrrrrrrrrrrrrr
  rrrrrrrrrrrrrrrrrrrr
 rrrrrrrrrrrrrrrrrrrrrr
          jjjj
          jjjj
         nnnnnn
```

```
           +
           11
          1111
         111111
        11111111
       1111111111
      111111111111
     11111111111111
    1111111111111111
   111111111111111111
  11111111111111111111
 1111111111111111111111
          ssss
          ssss
         hhhhhh
```

```
           +
           PP
          PPPP
         PPPPPP
        PPPPPPPP
       PPPPPPPPPP
      PPPPPPPPPPPP
     PPPPPPPPPPPPPP
    PPPPPPPPPPPPPPPP
   PPPPPPPPPPPPPPPPPP
  PPPPPPPPPPPPPPPPPPPP
 PPPPPPPPPPPPPPPPPPPPPP
          PPPP
          PPPP
         CCCCCC
```

```
           +
           ff
          ffff
         ffffff
        ffffffff
       ffffffffff
      ffffffffffff
     ffffffffffffff
    ffffffffffffffff
   ffffffffffffffffff
  ffffffffffffffffffff
 ffffffffffffffffffffff
          CCCC
          CCCC
         BBBBBB
```

```
           +
           tt
          tttt
         tttttt
        tttttttt
       tttttttttt
      tttttttttttt
     tttttttttttttt
    tttttttttttttttt
   tttttttttttttttttt
  tttttttttttttttttttt
 tttttttttttttttttttttt
          WWWW
          WWWW
         BBBBBB
```

LITERATURE CITED

Beers, T. W. 1969. Slope corrections in horizontal point sampling. Jour. Forestry 67:188-92.

Dearing, J. 1981. Calculator tips and routines. Corvallis Software, Inc. Box 1412, Corvallis, Oregon 97330.

Grosenbaugh, L. R. 1952. Shortcuts for cruisers and scalers. U.S.F.S. Southern For. Exp. Sta. Occ. Paper No. 126.

Husch, B., Miller, C. I., and Beers, T. W. 1982. Forest mensuration. 3rd ed. John Wiley & Sons, New York.

Laidly, P. R. 1979. Metric site index curves for aspen, birch and conifer in the Lake States. U.S.D.A. Forest Service Gen. Tech. Rep. NC-54.

Mesavage, C., and Girard, J. W. 1946. Tables for estimating board foot volume of timber. U.S. Forest Service, Washington, D.C.

Myers, C., Polak, D. J., Raisanen, D., Schlesinger, R. C., and Stortz, L. 1980. Weight and volume equations and tables for six upland hardwoods in southern Illinois. U.S.D.A. Forest Service Gen. Tech. Rep. NC-60.

PPC. 1981. PPC ROM user's manual. Personal Programming Center, 2545 W. Camden Place, Santa Ana, CA 92704.

Schnur, G. L. 1937. Yield, stand, and volume tables for even-aged upland oak forests. U.S. Dept. Agric. Tech. Bull. No. 560.

Scott, C. T. 1979. Northeastern forest survey board-foot volume equations. U.S.D.A. Forest Service Res. Note NE-271.

Shepperd, W. D. 1980. Hand-held calculator programs for the field forester. U.S.D.A. Forest Service Gen. Tech. Rep. RM-76.

Wiant, H. V., Jr. 1975. Schnur's site index curves for upland oaks formulated. Jour. Forestry 73:429.

Wiant, H. V., Jr., and Castaneda, F. 1977. Mesavage and Girard's volume tables formulated. Resource Inventory Notes March 1977. U.S.D.I. Bureau of Land Mgt. Denver, CO.

Wickes, W. C. 1980. Synthetic programming on the HP-41C. Larkin Publications, 4517 NW Queens Ave., Corvallis, OR 97330.

SF 00

CLEARING

FLAG SET ?
00 AUTO PRNT
01 FOR FREQ.
02 FOR WTD.
SF XX, R/S
KEY Y↑X, R/S

50.00↑ 1890.00
35.00↑ 1900.00
25.00↑ 1920.00
11.98↑ 1931.00
15.00↑ 1934.00
13.00↑ 1935.00
6.98↑ 1940.00
10.00↑ 1941.00
5.99↑ 1944.00
8.98↑ 1948.00
6.98↑ 1950.00
4.99↑ 1952.00
5.98↑ 1955.00
4.98↑ 1960.00

Y-INT.= 1233.81
SLOPE= -0.63
B.F.O.= 0.01

S(YX)= 3.80
S(a)= 100.62
S(b)= 0.05
S(BFO)= 1.85E-3

R= -0.96
R↑2= 0.92

T(R)= -12.12
T(a)= 12.26
T(b)= -12.12
T(BFO=1)=-535.81

****
bbb.dddII: 1900.06020

STUDENTS T= 2.18

X: 1900.00
Y-HAT= 37.13
MCI= 32.51 TO 41.74
ICI= 27.65 TO 46.60

X: 1920.00
Y-HAT= 24.53
MCI= 21.69 TO 27.37
ICI= 15.78 TO 33.28

CF 04
0.00 STO 27
XEQ "RCHEK1"
NO. ROLLS?
100.00
SEED=0.000000
GENERATING

DONE: HIT R/S
RUN
ROLL CT.= 100
N0= 12
N1= 8
N2= 9
N3= 9
N4= 17
N5= 15
N6= 8
N7= 7
N8= 7
N9= 8

CHI SQ.= 11.00
MEAN= 4.21
BIAS= -6.44%

****

SPEC.	GROSS SCALE	NET SCALE	NO. LOGS
1	100	93	1
3	200	192	1
5	917	847	2
7	256	250	1
TOTAL	1473	1382	5

****

TABLE 3

GREEN WT.
BLACK OAK
D: 8 H: 38
487 LBS.
D: 24 H: 70
9140 LBS.

TABLE 6

OVEN DRY WT.
RED OAK
D: 10 H: 48
394 LBS.
WHITE OAK
D: 10 H: 54
597 LBS.
HICKORY
D: 6 H: 56
221 LBS.

TABLE 8

VOLUME
WHITE ASH
D: 16 H: 44
28.2 CU. FT.
Y. POPLAR
D: 6 H: 40
2.1 CU. FT.
D: 26 H: 96
140.8 CU. FT.

TABLE 10

GREEN WT.
Y. POPLAR
D: 40 H: 34
2042 KG.

TABLE 16

VOLUME
RED OAK
D: 60 H: 30
3.57 M↑3

```
          XEQ "LOGVOL"
BF CF OR M?
BF
KEY DU ↑ L
           14.0 ENTER↑
           10.0    RUN
KEY A TO D OR E
                  XEQ E
DOYLE= 62.5
SCRIB.= 76.8
INT. 1/4= 80.2
INT. 1/8= 88.6
```

```
TALLY BY
LUMBER GRADE

 GRADE 1= 450
 GRADE 2= 240
 GRADE 3= 80
 GRADE 4= 210
 GRADE 5= 100
 GRADE 6= 150
                  XEQ c
SPECIES ?
              1    RUN
WALNUT:
 GRADE 1= 150
 GRADE 2= 40
 GRADE 3= 80
 GRADE 4= 60
 GRADE 5= 100
 GRADE 6= 150
```

```
          XEQ "LOGVOL"
BF CF OR M?
CF
KEY a,b,c, OR e
                  XEQ e
DB↑ DM↑ DU↑,L
          12.40 ENTER↑
          10.00 ENTER↑
           9.10 ENTER↑
          14.00    RUN
HUB.= 7.64 CF
SMAL.= 9.03 CF
NEWT.= 8.10 CF
```

```
*************
SPECIES     VOLUME
------------
HEMLOK       218.6
CEDARS       343.7
SUMAPL       834.4
BIRCHS       212.7
ABEECH       198.1
BASSWD       615.1
HICKRY       909.1
OTHERS       251.8
 TOTAL      3583.5
*************
```

```
                  XEQ a
HUBER VOLUME
KEY DM ↑ L
          25.40 ENTER↑
           4.27    RUN
HUB.= 0.22 M↑3
                    RUN
   = 7.64 CF
                  XEQ e
DB↑ DM↑ DU↑,L
          31.50 ENTER↑
          25.40 ENTER↑
          23.10 ENTER↑
           4.27    RUN
HUB.= 0.22 M↑3
SMAL.= 0.26 M↑3
NEWT.= 0.23 M↑3
```

```
SPECIES:  4
       D: 20.0  H: 32.0
VOLUME= 218.6
```